SHARP

New Life Now

パソコンテレビ
パーソナルコンピュータ

形名 CZ-803C
CZ-804C

BASIC MANUAL

目　　次

ページ

SHARP HuBASICの特長……1

第1章　SHARP HuBASICの概要

1.1　SHARP HuBASICの起動……4
1.2　動作モード……5
　1.2.1　ダイレクトモード……5
　1.2.2　プログラムモード……5
1.3　コマンドとステートメント……6
1.4　定数……7
　1.4.1　数値定数……7
　1.4.2　文字定数……8
1.5　変数……8
1.6　予約語……9
1.7　配列変数……9
1.8　関数……10
1.9　式……10
1.10　演算子……11
　1.10.1　算術演算子……11
　1.10.2　関係演算子……11
　1.10.3　文字列の比較……11
　1.10.4　論理演算子……12
1.11　区切り記号……13
1.12　スクリーンエディタ……14
1.13　コントロールコード……17
1.14　テレビのダイレクトコントロール……18

第2章　コマンド・ステートメント
　コマンド・ステートメント・関数説明の形式
　コマンド・ステートメント・関数の 書式 の見方

2.1　コマンド……20
　2.1.1　CLEAR/LIMIT……20
　2.1.2　NEWON……20
　2.1.3　NEW……21
　2.1.4　AUTO……22
　2.1.5　LIST/LLIST……23
　2.1.6　RUN……24
　2.1.7　CONT……25
　2.1.8　EDIT……26
　2.1.9　DELETE……27
　2.1.10　RENUM……28
　2.1.11　SEARCH……29
　2.1.12　TRON/TROFF……30
　2.1.13　DEVICE……31
　2.1.14　FILES/LFILES……32
　2.1.15　LOAD/LOADM……33

2.1.16 SAVE/SAVEM ····· 34
2.1.17 LOAD?/VERIFY ····· 35
2.1.18 CHAIN ····· 36
2.1.19 MERGE ····· 37
2.1.20 KILL ····· 37
2.2 一般ステートメント ····· 38
2.2.1 LET ····· 38
2.2.2 DEFINT/DEFSNG/DEFDBL/DEFSTR ····· 39
2.2.3 PRINT/LPRINT ····· 40
2.2.4 WRITE ····· 41
2.2.5 INPUT ····· 41
2.2.6 LINPUT/LINE INPUT ····· 42
2.2.7 CLEAR/CLR ····· 42
2.2.8 OPTION BASE ····· 43
2.2.9 DIM ····· 43
2.2.10 LABEL ····· 44
2.2.11 GOTO/GO TO ····· 44
2.2.12 GOSUB/GO SUB ····· 45
2.2.13 RETURN ····· 45
2.2.14 IF～THEN…ELSE ····· 46
2.2.15 FOR～TO…STEP ····· 48
2.2.16 NEXT ····· 49
2.2.17 REPEAT ····· 49
2.2.18 UNTIL ····· 50
2.2.19 WHILE ····· 51
2.2.20 WEND ····· 51
2.2.21 ON～ ····· 53
2.2.22 STOP ····· 55
2.2.23 END ····· 55
2.2.24 SWAP ····· 56
2.2.25 REM ····· 56
2.2.26 READ ····· 57
2.2.27 DATA ····· 58
2.2.28 RESTORE ····· 59
2.2.29 DEF FN ····· 60
2.2.30 DEF USR ····· 61
2.2.31 CALL ····· 61
2.2.32 POKE ····· 62
2.2.33 OUT ····· 63
2.2.34 RANDOMIZE ····· 63
2.3 ファイル処理ステートメント ····· 64
2.3.1 MAXFILES ····· 64
2.3.2 OPEN ····· 64
2.3.3 CLOSE ····· 65
2.3.4(1) PRINT＃ ····· 65
2.3.4(2) PRINT＃0 ····· 65
2.3.5 WRITE＃ ····· 66
2.3.6 INPUT＃ ····· 66
2.3.7 LINPUT＃/LINE INPUT＃ ····· 67
2.3.8 DEVI$ ····· 67
2.3.9 DEVO$ ····· 68

2.3.10 INIT……68
2.4 エラー処理ステートメント……69
2.4.1 ON ERROR GOTO ……69
2.4.2 RESUME ……70
2.4.3 ERROR……70
2.5 画面制御ステートメント……71
2.5.1 WIDTH ……71
2.5.2 SCREEN/GRAPH……72
2.5.3 CONSOLE ……73
2.5.4 LOCATE/CURSOR ……74
2.5.5 COLOR……75
2.5.6 CREV……75
2.5.7 CFLASH ……76
2.5.8 CGEN……77
2.5.9 CSIZE ……77
2.5.10 DEF CHR$……78
2.5.11 WINDOW ……80
2.5.12 CLS……83
2.5.13 PALET……84
2.5.14 PRW……86
2.5.15 CANVAS ……87
2.5.16 LAYER……88
2.5.17 GET@……92
2.5.18 PUT@……93
2.5.19 POKE@……94
2.5.20 OPTION SCREEN……94
2.6 グラフィックステートメント……95
2.6.1 PSET……95
2.6.2 PRESET ……96
2.6.3 LINE ……97
2.6.4 CIRCLE……100
2.6.5 POLY……101
2.6.6 PAINT ……102
2.6.7 POSITION……103
2.6.8 PATTERN……104
2.7 画面制御・グラフィックステートメントの使用可能な画面……106
2.8 特殊ステートメント……107
2.8.1 KEY/DEF KEY……107
2.8.2 KEYLIST/KLIST……107
2.8.3 KEY0……108
2.8.4 REPEAT ON/REPEAT OFF ……109
2.8.5 CLICK ON/CLICK OFF……109
2.8.6 ON KEY GOSUB……110
2.8.7 KEY ON/KEY OFF/KEY STOP……111
2.8.8 MID$……111
2.8.9 MEM$……112
2.8.10 MON……112
2.8.11 HCOPY……113
2.8.12 BOOT……113
2.8.13 PAUSE……114

2.8.14 WAIT……114
2.9 タイマー制御ステートメント……115
2.9.1 TIME$ ……115
2.9.2 DAY$……115
2.9.3 DATE$ ……116
2.9.4 TIME ……116
2.9.5 ASK……117
2.10 テレビ制御ステートメント……118
2.10.1 TVPW……118
2.10.2 CRT……119
2.10.3 CHANNEL……119
2.10.4 VOL……120
2.10.5 SCROLL ……120
2.11 カセット制御ステートメント……121
2.11.1 EJECT ……121
2.11.2 CSTOP……121
2.11.3 FAST……121
2.11.4 REW ……122
2.11.5 APSS……122
2.11.6 CMT……123
2.12 サウンド制御ステートメント……124
2.12.1 BEEP……124
2.12.2 MUSIC ……124
2.12.3 TEMPO……126
2.12.4 PLAY……126
2.12.5 SOUND……127
2.13 書式指定(PRINT USING)……131
2.13.1 数値型……131
2.13.2 文字型……133
2.13.3 その他の型……133
2.13.4 文字定数の出力……133

第3章 関数・システム変数

3.1 数値関数……136
3.1.1 SIN……136
3.1.2 COS……136
3.1.3 TAN……137
3.1.4 ATN……137
3.1.5 ABS……138
3.1.6 SGN……138
3.1.7 INT……139
3.1.8 FIX……139
3.1.9 FRAC……140
3.1.10 CINT/CSNG/CDBL……141
3.1.11 SQR……142
3.1.12 EXP……142
3.1.13 LOG……143
3.1.14 PAI……143
3.1.15 RAD……144

3.1.16 FAC……145
3.1.17 SUM……146
3.1.18 RND……146
3.2 文字関数……147
3.2.1 ASC……147
3.2.2 CHR$……147
3.2.3 VAL……148
3.2.4 STR$……148
3.2.5 HEX$……149
3.2.6 OCT$……149
3.2.7 BIN$……150
3.2.8 LEFT$……151
3.2.9 RIGHT$……151
3.2.10 MID$……152
3.2.11 STRING$……152
3.2.12 SPACE$……153
3.2.13 INSTR……153
3.2.14 LEN……153
3.2.15 HEXCHR$……154
3.2.16 MIRROR$……154
3.2.17 MKI$/MKS$/MKD$……155
3.2.18 CVI/CVS/CVD……156
3.3 特殊関数……157
3.3.1 INP……157
3.3.2 PEEK……158
3.3.3 PEEK@……159
3.3.4 POS……159
3.3.5 LPOS……160
3.3.6 VARPTR……160
3.3.7 FRE/SIZE……161
3.3.8 POINT……161
3.3.9 STICK……162
3.3.10 STRIG……162
3.3.11 CMT……163
3.3.12 EOF……164
3.3.13 KANJI$……164
3.3.14 FN……165
3.3.15 USR……165
3.4 入出力用文字関数……166
3.4.1 MEM$……166
3.4.2 SCRN$……166
3.4.3 CHARACTER$……166
3.4.4 CGPAT$……167
3.4.5 INKEY$……169
3.4.6 INPUT$……170
3.5 タイマー変数……171
3.5.1 TIME$……171
3.5.2 DAY$……171
3.5.3 DATE$……172
3.5.4 TIME……172

3.6 システム変数…… 173
3.6.1 CSRLIN…… 173
3.6.2 STRPTR …… 173
3.6.3 DTL…… 174
3.6.4 ERL…… 174
3.6.5 ERR…… 175
第4章 ファイル操作
4.1 ファイルとは…… 178
4.2 デバイスとは…… 178
4.3 ファイルディスクリプタとは…… 179
4.4 ファイルの使い方…… 180
4.4.1 プログラムファイルの使い方…… 180
4.4.2 データファイルの使い方…… 180
4.4.2.1 データファイルの作り方…… 181
4.4.2.2 データファイルの読み出し方…… 184
4.4.3 ファイル関連命令…… 185
4.5 ファイルの種類…… 187
4.6 シーケンシャルファイルの利用例…… 189
付録
A.1 テキスト画面とその属性ポートへのアクセス方法…… 192
A.2 コマンド・ステートメント・関数の省略形一覧表…… 197
A.3 機械語とのリンク方法…… 200
A.3.1 モニターのコマンド…… 200
A.3.2 USR命令の使い方…… 203
A.4 コード表…… 205
A.4.1 キャラクターコード表(ASCIIコード表) …… 205
A.4.2 JIS第1水準漢字一覧表…… 206
A.4.3 非漢字一覧表…… 210
A.4.4 図形文字用符号コード一覧表…… 211
A.5 エラーメッセージ一覧表…… 212
A.6 ファイルディスクリプタ表…… 214
A.7 メモリーマップ…… 215
A.8 SHARP HuBASICプログラマのためのワンポイントアドバイス…… 216
A.9 サンプルプログラム集…… 217
A.9.1 キャラクタ定義1 …… 217
A.9.2 〃 2 …… 217
A.9.3 チェック模様と斜線…… 218
A.9.4 線と図…… 218
A.9.5 ペイントデモプログラム…… 219
A.9.6 ペイントタイルパターンデモプログラム…… 219
A.9.7 星…… 220
A.9.8 三原色…… 220
A.9.9 カエルの歌…… 221
A.9.10 TEL.サウンド …… 221
A.9.11 SOUND命令による効果音…… 222
A.9.12 サイコロ1 …… 224
A.9.13 〃 2 …… 224
A.9.14 〃 3 …… 225
A.9.15 漢字を任意の位置に表示…… 225
BASICテープのコピー作成方法 …… 226
A.10 類似命令一覧表…… 227

コマンド・ステートメント・関数

アルファベット順索引

A
ABS ····· 138
APSS ····· 122
ASC ····· 147
ASK ····· 117
ATN ····· 137
AUTO ····· 22
B
BEEP ····· 124
BIN$ ····· 150
BOOT ····· 113
C
CALL ····· 61
CANVAS ····· 87
CDBL ····· 141
CFLASH ····· 76
CGEN ····· 77
CGPAT$ ····· 167
CHAIN ····· 36
CHANNEL ····· 119
CHARACTER$ ····· 166
CHR$ ····· 147
CINT ····· 141
CIRCLE ····· 100
CLEAR ····· 20, 42
CLICK ON ····· 109
CLICK OFF ····· 109
CLOSE ····· 65
CLR ····· 42
CLS ····· 83
CMT ····· 123, 163
COLOR ····· 75
CONSOLE ····· 73
CONT ····· 25
COS ····· 136
CREV ····· 75
CRT ····· 119
CSIZE ····· 77
CSNG ····· 141
CSRLIN ····· 173
CSTOP ····· 121
CURSOR ····· 74
CVD ····· 156
CVI ····· 156
CVS ····· 156
D
DATA ····· 58
DATE$ ····· 116, 172
DAY$ ····· 115, 171
D
DEF CHR$ ····· 78
DEFDBL ····· 39
DEF FN ····· 60
DEFINT ····· 39
DEF KEY ····· 107
DEFSNG ····· 39
DEFSTR ····· 39
DEF USR ····· 61
DELETE ····· 27
DEVICE ····· 31
DEVI$ ····· 67
DEVO$ ····· 68
DIM ····· 43
DTL ····· 174
E
EDIT ····· 26
EJECT ····· 121
END ····· 55
EOF ····· 164
ERL ····· 174
ERR ····· 175
ERROR ····· 70
EXP ····· 142
F
FAC ····· 145
FAST ····· 121
FILES ····· 32
FIX ····· 139
FN ····· 165
FOR～TO…STEP… ····· 48
FRAC ····· 140
FRE ····· 161
G
GET@ ····· 92
GOSUB ····· 45
GOTO ····· 44
GRAPH ····· 72
H
HCOPY ····· 113
HEXCHR$ ····· 154
HEX$ ····· 149
I
IF～THEN…ELSE ····· 46
INIT ····· 68
INKEY$ ····· 169
INP ····· 157
INPUT ····· 41
INPUT# ····· 66
INPUT$ ····· 170
INSTR ····· 153
INT ····· 139
K
KANJI$ ····· 164
KEY ····· 107
KEY 0 ····· 108
KEY LIST ····· 107
KEY OFF ····· 111
KEY ON ····· 111
KEY STOP ····· 111
KILL ····· 37
KLIST ····· 107
L
LABEL ····· 44
LAYER ····· 88
LEFT$ ····· 151
LEN ····· 153
LET ····· 38
LFILES ····· 32
LIMIT ····· 20
LINE ····· 97
LINE INPUT ····· 42
LINE INPUT# ····· 67
LINPUT ····· 42
LINPUT# ····· 67
LIST ····· 23
LLIST ····· 23
LOAD ····· 33
LOADM ····· 33
LOAD? ····· 35
LOG ····· 143
LOCATE ····· 74
LPOS ····· 160
LPRINT ····· 40
M
MAXFILES ····· 64
MEM$ ····· 112, 166
MERGE ····· 37
MID$ ····· 111, 152
MIRROR$ ····· 154
MKD$ ····· 155
MKI$ ····· 155
MKS$ ····· 155
MON ····· 112
MUSIC ····· 124
N
NEW ····· 21
NEW ON ····· 20
NEXT ····· 49
O
OCT$ ····· 149

O
ON～ 53
ON ERROR GOTO 69
ON KEY GOSUB 110
OPEN 64
OPTION BASE 43
OPTION SCREEN 94
OUT 63
P
PAI 143
PAINT 102
PALET 84
PATTERN 104
PAUSE 114
PEEK 158
PEEK@ 159
PLAY 126
POINT 161
POKE 62
POKE@ 94
POLY 101
POS 159
POSITION 103
PRESET 96
PRINT 40
PRINT USING 131
PRINT# 65
PRW 86
PSET 95
PUT@ 93
R
RAD 144
RANDOMIZE 63
READ 57
REM 56
RENUM 28
REPEAT 49
REPEAT OFF 109
REPEAT ON 109
RESTORE 59
RESUME 70
RETURN 45
REW 122
RIGHT$ 151
RND 146
RUN 24
S
SAVE 34
SAVEM 34
SCREEN 72
SCRN$ 166
SCROLL 120
SEARCH 29
SGN 138
SIN 136
SIZE 161
SOUND 127
SPACE$ 153
SPC 40
SQR 142
STICK 162
STOP 55
STRIG 162
STR$ 148
STRING$ 152
STRPTR 173
SUM 146
SWAP 56
T
TAB 40
TAN 137
TEMPO 126
TIME 116, 172
TIME$ 115, 171
TROFF 30
TRON 30
TVPW 118
U
UNTIL 50
USR 165
V
VAL 148
VARPTR 160
VERIFY 35
VOL 120
W
WAIT 114
WEND 51
WHILE 51
WIDTH 71
WINDOW 80
WRITE 41
WRITE# 66

ご　注　意

このマニュアルは、本機の同梱BASICに基づいて作成されています。

(1) 本機では、システムソフトウェアを全てファイル形態のソフトウェアパック（カセットテープ）によってサポートされます。
各システムソフトウェアおよび本書の内容は、改良のため予告なく変更することがありますので、ファイルバージョンナンバーには、特にご注意されるよう、お願いいたします。

(2) 本機のシステムソフトウェアならびに本書の内容を個人で使用する場合を除き、当社に無断で複製することは禁止します。

(3) 本機は非常に複雑な機能および組合せを有する製品であり、出荷に際してマニュアルを含め充分なチェックをして万全を期しておりますが、万一ご使用中ご不審な点、お気付きのことがありましたら、もよりのお客様ご相談窓口までご連絡ください。
なお、運用した結果、生じる影響については、責任を負いかねますので、あらかじめご了承ください。

SHARP HuBASICの特長

SHARP HuBASICは、以下の特長をもっています。

1．本機のもつハードウェア機能を十分に活用できます。BASIC言語で操作可能なものとしては、以下のものがあげられます。

- プログラマブル・ファンクションキー
- 専用カセット
- タイマー・カレンダーつきクロック
- セントロニクス・インターフェース機器
- テレビジョンコントロール（専用ディスプレイテレビ使用時のみ）
- グラフィック・プライオリティ・ロジック
- ユーザー定義キャラクタジェネレータ
- プログラマブル・サウンドジェネレータ
- テレビ画面とコンピュータ画面の合成（専用ディスプレイテレビ使用時のみ）

2．48Kバイトのグラフィック専用RAMにより各種のグラフィック処理に対応できます。

（1）640 ×200 ドット高分解能グラフィック
　　フルカラー8色1画面、または単色（8色中1色選択）3画面

（2）320 ×200 ドット複数画面
　　フルカラー8色2画面、または単色（8色中1色選択）6画面

3．WINDOW命令によって論理座標を設定することにより、簡単なプログラムで、多彩な表示を実現できます。

4．テキスト用VRAMとグラフィック用VRAMを別々にもっているため、テキスト画面とグラフィック画面の合成を容易に行なえます。

5．パレットの概念によって、色の変更を瞬時に行なうことができます。

6．特殊なハードウェアとそれらをサポートする強力なソフトウェアにより、グラフィック画面同士や、グラフィックとテキスト画面との間で優先順位をもたせて重ね合わせることができます(プライオリティ機能)。この機能により、簡単なプログラムで複雑な動きの画面をコントロールすることができます。

7．通常のキャラクタ用ROM（英数字、カナ、セミグラフィックパターンなど）のほかに、ユーザーが自由に定義できるキャラクタ・ジェネレータRAMをもっており、256種のパターンを定義することができます。これらのパターンはドット単位で色を指定することができます。

8．当漢字ROMには、JIS第1水準漢字(2,965字)をはじめとしてひらがな・カタカナ、英・数字、記号などを含め3,707文字のキャラクタジェネレータとしての字体が収納されており、コンピュータを操作することにより、ディスプレイ画面にこれ等の文字を表示することができます。なお、当漢字ROMに収納されている漢字の字体構成は、16ドット×16ドットで構成されています。
（CZ-804Cのみ、漢字ROMを標準装備しています。）

9．4種類のキャラクタ表示サイズをもっており、混在して用いることができ、組合わせて使用することにより、見やすい画面を構成することができます。

10．プログラマブル・サウンドジェネレータ回路を持っており、各種の効果音を容易に作り出すことができます。

11．本機専用のディスプレイテレビに接続することにより、テレビ画面とコンピュータ画面を合成して扱うことができます。

12．本機専用のディスプレイテレビのチャンネル、電源、音量などをコントロールする命令をもっているため、BASICプログラム中でこれらのコントロールができます。

13. タイマー機能をもっており、会話形式で時刻設定が可能です。

14. プログラム中のジャンプ先として、ラベルを扱えるため、わかりやすいプログラムを作ることができます。

15. IF THEN ELSE, WHILE WEND, REPEAT UNTILなど構造化プログラミングに対処したステートメントを備えています。

16. 専用カセットに対応した命令をいくつか用意しているので、プログラム中で自由にカセットのコントロールが行なえます。

17. 単精度実数の有効桁が8桁になっているため、高精度の演算を少ない容量で実現できます。

第1章 SHARP HuBASICの概要

1.1 SHARP HuBASICの起動

本機の背面にあるメイン電源スイッチおよび前面の電源スイッチをONにすると、接続のディスプレイテレビの画面に、

```
IPL is under preparing
```
……（1）

と一瞬表示されます。続いて、

```
Make ready any device
Push(F,R,C or T) key
    F:FIoppy
    R:ROM
    C:CMT
    T:Timer
```
……（2）

と表示されて、カセットのフタが開きますので、BASICカセットファイルをセットすると、イニシャルローディングを行なうIPL(initial program loader)[注1]が自動的にスタートして、画面に

```
IPL is looking for a program from CMT
```
……（3）

と表示され、テープが回り始めます。BASICの実際のローディングが始まると、

```
IPL is loading BASIC CZ8CB01
```
……（4）

と表示されます。数分後、イニシャルローディングが終了すると、画面に

```
SHARP-HuBASIC CZ-8CB01 VX.X
Copyright (C) 1982 by SHARP/Hudson

××××× Bytes free
Ok
■
```
……（5）

とメッセージを表示し、カセットを自動的に巻き戻します。

Okの下でカーソルが点滅しているのは、システムコントロールがBASICのコマンドレベル[注2]にあり、コマンドの入力待ちの状態にあることを示しています。

もし、電源投入前にあらかじめBASICカセットファイルをカセットレコーダーにセットしておくと、画面上のメッセージが(1)、(3)、(4)と連続的に変わり、IPLが自動的にスタートして、(5)のメッセージを表示してイニシャルローディングを終了します。

カセットファイルをセットするときは、テープが先頭まで巻き戻してあるかどうか必ず確かめるようにしてください。

イニシャルローディング終了後は、次の命令を実行した状態になっています。

```
WIDTH40 : COLOR 7 , 0 : CGEN : CFLASH : CREV : CSIZE : PALET : PRW : WINDOW : CONSOLE
: SCREEN 0 , 0 , 0 : DEVICE "CAS": : PLAY" V12 : V 0 :V 0".
```

注1）IPL……BASIC等のシステムソフトウェアを外部ファイル（カセット、ディスクなど）から内部メインメモリーに格納（ローディング）するプログラム。
BASIC等のロード後は、コンピュータの管理はBASIC等のシステムソフトウェアに移され、IPLは自動的に切りはなされます。

注2）BASICのコマンドレベル……BASICのコントロール下にあり、BASICの任意の命令を実行したり、プログラムを作成したりすることができます。

1.2 動作モード

SHARP HuBASICを起動すると、本機はディスプレイにOkを表示し、入力待ちになっています。この状態のときは、キー入力（キーボードから入力）によるすべてのBASICコマンドを受けつけます。

BASICのコマンドの実行には、ダイレクトモードとプログラムモードがあります。

1.2.1 ダイレクトモード

行番号（文番号）をつけずにステートメントを入力し、  キーを押すとすぐ実行しますが、これをダイレクトモードにおける実行といいます。

1.2.2 プログラムモード

行番号（1～65534までの整数）をつけて入力すると、それはBASICプログラムとして行番号順にメモリーに蓄えられます。

一度メモリに蓄積されたプログラムは、RUN,GOTO文、およびGOSUB文によって実行させることができます。これを、上記のダイレクトモードにおける実行に対し、プログラムモードにおける実行といいます。

 キーはBASIC側へデータを与えるためのキーで各行の入力後に必ず押して下さい。

○BASICの一般的記述形式

行番号 ステートメント〔：ステートメント〔：ステートメント〕…〕　注)〔 〕内は省略可能

ⓐ　　ⓑ

ⓐ行番号は、1～65534までの整数で、プログラムの順番を示し、GOTO文、GOSUB文などのジャンプ先の番号として使います。プログラムとしてメモリーに蓄積するには、この行番号をつけて、一行入力したら  キーを押す必要があります。

行番号として0または65535を使用しても、メモリーに蓄積されません。

65536以上の数を使用した場合エラーになります。

ⓑステートメントは、後に記述するSHARP HuBASICがもつ各種のコマンド、一般ステートメント、その他のステート、関数、および代入式のことです。

このステートメントはコロン（：）で区切ることにより、つなげて書くことができます。これをマルチステートメントと呼び、一行には、行番号も含めて、255文字まで入力可能です。

（例）



1.3 コマンドとステートメント

コマンドおよびステートメントは、処理機能を表すキーワードと、処理内容を指示するオペランドから構成されます。

(例1)

```
AUTO 500
```

キーワード　オペランド

《意味》　プログラムの行番号を500から自動的に発生せよ。(AUTO文参照)

(例2)



《意味》　変数Ｉを0から100まで増分2で増やせ。(FOR文参照)

(例3)



オペランドは数値定数、文字定数、変数、関数、演算子、および区切り記号から構成されます。

1.4 定数

SHARP HuBASICで扱う定数は、数値定数と文字定数の2種類です。

1.4.1 数値定数

数値定数には整数定数と実数定数とがあります。これら数値定数は、正及び負の数、および0を含んでいます。負の数には－符号をつける必要がありますが正の数には＋符号を省略してもかまいません。

（1）整数定数

●10進定数は、0～9までの数字を用い、範囲が－32767～32767までの整数です。他の数と区別する為、数字の最後に％をつけて表します。

32768以上の数に％をつけると、オーバーフローになり入力できません。

（例） 32767％
－23％
－32767％

●16進定数は、最初に&H(Hexadecimal)を伴った4桁までの16進数です。16進数は、0～9までの数字と、A～Fまでの英文字を使って表します。範囲は&H0～&HFFFFまでです。(－符号をつけることも可能)

（例） &HEFD8
&H3F
&H400

●8進定数は、最初に&O(Octal)を伴った6桁までの8進数です。8進数は、0～7までの数字を使って表します。範囲は&O0～&O177777までです。(－符号をつけることも可能)

（例） &O301
&O177777
&O2350

●2進定数は、最初に&B(Binary)を伴った16桁までの2進数です。2進数は、0と1の2つの数字を使って表します。範囲は&B0～&B1111111111111111までです。(－符号をつけることも可能)

（例） &B1010
&B1111111111111111
&B10000

（2）実数定数

実数定数は、有効桁数によって単精度定数と倍精度定数にわけられます。

●単精度定数

単精度は、有効数字8桁まである実数定数で、数字の最後に！をつけて表します。8桁を越える値になると、自動的に指数形式に変わります。その場合、仮数部が8桁、指数部が－38～38の範囲内になります。単精度であることを示すため、指数部はE±nとして表示されます。

（例） 124.6
358692
－2.85E＋08（＝-2.85×10^{8}）

●倍精度定数

倍精度は、有効数字16桁まである実数定数で、数字の最後に＃をつけて表します。16桁を越える値になると、自動的に指数形式に変わります。その場合、仮数部が16桁、指数部が－38～38の範囲内になります。倍精度であることを示すために、指数部はD±nとして表示されます。

（例） 1192＃
1.234567890123457D＋16（＝$1.234567890123457\times10^{16}$）
76895.4345265

1.4.2 文字定数

文字定数は最大255文字で、ダブルクォーテーションマーク（"）で囲まれた英数字、カナ、セミグラフィック、記号などの文字列です。ただし"は文字定数にはできませんが、必要な場合にはCHR＄関数（第2章参照）を用いてください。
文字数が0の文字定数を特にヌルストリングとよび ""で表します。

（例） "SHARP HuBASIC"
"1982/09/15"
"ワタシ␣ノ␣ナマエ␣ハ␣X1␣デス"

1.5 変数

変数は、英字で始まる英数字列で表わし長さは240文字以内ならば任意です。しかし、コマンドやステートメントに使われている予約語で始まる変数名は変数名として使えません。

（例） RUNX →変数にならない。
XRUN →変数になる。

変数の型を示すために、変数のうしろに英記号が1文字伴います。
変数の型を指定する英記号
％……整数型 ！……単精度型 ＃……倍精度型 ＄……文字型

ⅰ）整数型には％をつけます。格納できる数値の範囲は、−32767～32767までの整数です。

（例） X％
A％
A2％

ⅱ）単精度型には！をつけますが、通常は省略してかまいません。
BASIC起動直後の変数のデフォルト値は、単精度です。（変数の後に変数の型を指定する英記号を省略すると単精度となります。）
格納できる数値の範囲は、有効数字8桁、指数部−38～38のE±n表現です。

（例） N
A！
D1！

（注）デフォルト値……BASICコマンド、ステートメントにおいて、パラメータを指定なしで使用したとき、パラメータに自動的にセットされる値。

ⅲ）倍精度型には＃をつけます。格納できる数値の範囲は、有効数字16桁、指数部−38～38のD±n表現です。

（例） B＃
PA＃
M3＃

ⅳ）文字型には＄をつけます。格納できるのは、255文字までの文字列です。

（例） A＄
XN＄
X1＄

（注1）A、A％、A＃、A＄はそれぞれ区別されます。BASIC起動時はA！とAは同じです。
（DEFINT、DEFSNG、DEFDBL、DEFSTR参照）
（注2）変数名として小文字も使用可能ですが、大文字との区別は行なっていません。

1.6 予約語

予約語（キーワード）はBASICインタープリタが管理している語で、ユーザが変数名として使うことはできません。すべてのコマンド、ステートメント、関数、およびシステム変数が予約語になっています。
その他、下に示す語も予約語になっています。これらの予約語で始まる変数名は使えません。

AND	ATTR$	BASE	DSKF	ELSE	ERASE
EQV	FIELD	FPOS	IMP	KBUF	LOC
LOF	LSET	MOD	NAME	NOT	OR
POP	PUSH	RSET	SET	SPC	STEP
SUB	TAB	THEN	TO	TRACE	USING
VDIM	XOR				

1.7 配列変数

配列は、1つの変数名で参照できる同じ型の値の集まりです。プログラムで配列を使うときは、配列の変数名に続けて（ ）の中に配列の次元と同じ個数の添字を指定します。配列の次元は255次元まで（添字は最大255個まで）許されます。
配列の大きさはDIM文で宣言しますが、宣言しなくとも配列を使うことができます。ただし宣言なしのときは添字の上限は10です。

（例）

$A(x_1, x_2, \cdots, x_n)$　　nは最大255
　　xiの最大32767
↑
変数名

1.8 関数

SHARP HuBASICには、下のような関数が用意されています。

数値関数
文字関数
特殊関数
入出力用文字関数
タイマー関数

関数とは、ユーザの与えた任意の値に対し、1対1または多対1の対応で、BASIC本体が1つの値を返してくる機能のことです。実際にプログラムの文のなかでは、

関数名(引数、引数、……)

という形で変数と同じように使い、引数に数値、数式および文字列を与えると、演算処理され、その結果が関数の値になります。

(例)

```
SIN(3.14)
CHR$(76,79,86,69,76,89,33)
VAL("&H"+"EFC0")
RIGHT$("ABCDEF",3)
```

この関数はBASIC本体のなかにすでに組み込まれているので組み込み関数と呼びます。

また、BASICは上述の組み込み関数のほかに、ユーザ自身が定義できるユーザ定義関数を備えていますが、詳細はDEF FN文の項で述べます。

数値関数のほとんどは引数に倍精度の数値を与えると、関数の値を倍精度で返してきます。πは円周率を与えるシステム変数です。(πは GRAPH + A キーで入力されます。)

1.9 式

式とは定数、変数、関数を演算子(後述)で結んだものや、単独の定数、変数、関数をいいます。
論理式とは、式のうちで特に関係演算子などを用いて真(式の値が-1)か、偽(式の値が-1以外)かに着目するものです。

(例)

```
235
XA
 "SHARP"
 "&H"+A$
COS(X-Y)+.5
(X-A)*(X-B)
(A+B)/2
```

1.10 演算子

演算子には算術演算子、関係演算子、および論理演算子の3種があります。

1.10.1 算術演算子

算術演算子を下に示します。

(算術演算子)	(演算内容)	(優先順位)	(例)
^	累乗演算	1	X^3, X^Y
*	乗　算	2	A*3, A*B
/	除　算	2	A/12, A/B(ただし、B≠0)
−	マイナス符号	3	−X
＋	加　算	4	A＋B
−	減　算	4	B−B, A−B
¥	整数除算	5	10¥3 (結果は整数の割算の商で、10¥3＝3となります。)
MOD	剰余演算	5	32.8MOD7.25 (結果は整数の割算の余りで、33÷7＝4余り5になります。)

(変数の後にMODを書く場合には、変数とMODの間にスペースをあけて下さい。)

優先順位が等しい場合は、左から先に演算されます。

演算を( )で囲むと、かっこ中の演算が先に実行されます。

(例)

$(A+B)/2\cdots\frac{A+B}{2}$　　$A+B/2\cdots\cdots A+\frac{B}{2}$

1.10.2 関係演算子

関係演算子はＩＦ文などの条件式において、2つの数値や文字列を比較するのに用います。結果は真のとき－1、偽のとき0で得られます。

下に関係演算子を示します。(IF～THEN…ELSE文参照)

(関係演算子)	(関係演算)	(例)
＝	両辺が等しい。	X＝Y
＜＞，＞＜	両辺が等しくない。	X＜＞Y，X＞＜Y
＜	左辺が右辺より小さい。	X＜Y
＞	左辺が右辺より大きい。	X＞Y
＜＝，＝＜	左辺が右辺より小さいか等しい。	X＜＝Y，X＝＜Y
＞＝，＝＞	左辺が右辺より大きいか等しい。	X＞＝Y，X＝＞Y

1.10.3 文字列の比較

数値の比較と同様に、文字列も関係演算子を用いて比較することができます。

比較する文字列の短い方の長さだけ、最初から1文字づつ比較して行き、1カ所でも違っているときには、そのキャラクタコードの大きい方の文字列が大きいと判断され、最後まで同じだった場合には、長さの長い方が大きいと判断され、長さも同じの場合のみ等しいと判断されます。キャラクタコードについては後ろのコード表を参照してください。

ヌルストリング("")はどの文字列よりも小さいと判断されます。

(例)

```
""      <  "A"
"AAA"   >  "AA"
"AB  "  >  "AB"
"ABC"   =  "ABC"
```

1.10.4 論理演算子

論理演算子はいくつかの条件を判断するときに使い、論理演算をして、ビットごとに0か1を与えます。また、関係演算を2つ以上つなぐことができます。下に論理演算子を示します。

《論理演算子》《論理演算》　　《真理値表》

NOT　否定（not）

X	NOT X
1	0
0	1

AND　論理積（and）

X	Y	X AND Y
1	1	1
1	0	0
0	1	0
0	0	0

OR　論理和（inclusive or）

X	Y	X OR Y
1	1	1
1	0	1
0	1	1
0	0	0

XOR　排他的論理和（exclusive or）

X	Y	X XOR Y
1	1	0
1	0	1
0	1	1
0	0	0

IMP　包含（implication）

X	Y	X IMP Y
1	1	1
1	0	0
0	1	1
0	0	1

EQV　同値（equivalence）

X	Y	X EQV Y
1	1	1
1	0	0
0	1	0
0	0	1

注）変数と論理演算子をつづけて書く場合は、間にスペースをあけてください。

(例)

●NOT 13　　13＝（0000 0000 0000 1101）$_2$
∴NOT13＝（1111 1111 1111 0010）$_2$
＝－14

●15AND 5　　15＝（0000 0000 0000 1111）$_2$
5＝（0000 0000 0000 0101）$_2$
∴15AND 5＝（0000 0000 0000 0101）$_2$
＝5

●50 OR 44　　50＝（0000 0000 0011 0010）$_2$
44＝（0000 0000 0010 1100）$_2$
∴ 50 OR 44＝（0000 0000 0011 1110）$_2$
＝62

●42XOR36　　42＝（0000 0000 0010 1010）$_2$
36＝（0000 0000 0010 0100）$_2$
∴42XOR36＝（0000 0000 0000 1110）$_2$
＝14

●204 IMP136　　204＝（0000 0000 1100 1100）$_2$
136＝（0000 0000 1000 1000）$_2$
∴204 IMP 136＝（1111 1111 1011 1011）$_2$
＝-69

●235 EQV440　　235＝（0000 0000 1110 1011）$_2$
440＝（0000 0001 1011 1000）$_2$
∴235 EQV440＝（1111 1110 1010 1100）$_2$
＝－340

```
IF -10<X AND X<10 THEN 2000 ELSE 1000
```

(意味) Xの値が－10＜X＜10だったら2000行へ、そうでなかったら1000行へジャンプせよ。

```
IF X<-10 OR X>10 THEN 2000 ELSE 1000
```

(意味) Xの値がX＜－10またはX＞10だったら2000行へ、そうでなかったら1000行へジャンプせよ。

論理演算はプログラムではIF～THEN…ELSE…文の中で使うことが多い。

1.11 区切り記号

,（カンマ）　PRINT, INPUT, DATAなどオペランドが並ぶ場合の区切りとして使用します。
（例） INPUT A, B, C

;（セミコロン）PRINTなどの区切りとして使用します。
（例） PRINT "A＝" ; A

:（コロン）　マルチステートメントの区切りとして使用します。
（例） A＝A＋1 : PRINT A

1.12 スクリーンエディタ

BASICプログラムを作成するときに、編集作業はつきものですが、これは、画面上のカーソル(注)を使って行います。

新しいプログラムを入力するときは、先に述べたように行番号、ステートメントの順にキーボードから1文字ずつ入力して行きますが、このとき、カーソルは文字の入力に伴って1文字分ずつ画面の右方向へ移動し、画面右隅まで行くと、次の行の左隅に移動します。行番号を含めて1行につき255文字まで入力可能で、プログラムの1行は画面表示上の1行とは異なります。つまり一つの行番号が含むステートメントの集合のことです。

プログラムを1行入力したとき画面に表示されているステートメントの文字列は、この時点では、まだBASICのメインメモリーの中に取り込まれず、テキスト画面に置かれているにすぎません。メモリーに登録するには [↵] キーを押す必要があります。以上が最も基本的なプログラムの入力方法ですが、実際には、いくつかの行からなるプログラムを入力している途中で、前のステートメントを修正したり、行番号のつけかえを行ったり、部分的な削除を行ったりするのが普通です。このような場合、カーソルによるスクリーンエディタを使って効率よくプログラムの編集を行う必要があります。

プログラムの編集はカーソルの操作方法を知ることから始まりますので、まずカーソルの移動方法を説明します。

〈カーソルの移動〉

[⇨] このキーはカーソルを右へ移動するためのもので、押し続けるとカーソルを最下行まで移動します。最下行で押し続けるとテキスト画面が1行ずつ上にスクロールします。

[⇦] このキーはカーソルを左へ移動するためのもので、押し続けるとカーソルを最上行まで移動します。

[⇧] このキーはカーソルを上へ移動するためのものです。最上行まで移動したら止まります。

[⇩] このキーはカーソルを下へ移動するためのもので、カーソルが最下行まで行ったとき、さらに押し続けるとテキスト画面が上にスクロールし始めます。

通常、カーソルキーに限らずキーボードのキーを押し続けると、同じ文字ないしは同じ運動が継続して入力されますが、REPEAT OFFと入力して [↵] キーを押すと、この機能がなくなり、1文字しか受けつけなくなります。これはREPEAT ONと入力して [↵] キーと押すことによって解除することができます。

〈カーソルのホーム位置への移動〉

[CLR HOME] キーを押すとカーソルは画面の左上隅（これをホーム位置といいます）へ瞬時に移動します。

また、[SHIFT] を押しながら [CLR HOME] を押すと、テキスト画面がすべてクリアされカーソルはホーム位置に移動します。

〈文字の挿入〉

文字の挿入を行う場合は、[SHIFT] を押したまま [INST DEL] キーを必要な回数入力して、カーソル位置の右側に空白（スペース）を入れます。

《例》

```
10X＝100
20PRINT X
```

PRINTとXの間にTAB(5);と入れたい場合は、Xのところにカーソルを持って行き、[SHIFT] を押したまま7回 [INST DEL] キーを打ってください。

```
10 X＝100
20 PRINT          X
```

上図のようにPRINTとXの間が空くので、PRINTの後ろにTAB(5);と入力して [↵] キーを押すと、TAB(5);の挿入が完了します。

```
10 X＝100
20 PRINT TAB(5);X
```

注）カーソルとは、画面上で点滅している白い正方形のことで、キー入力の位置を示します。

〈文字の削除〉

文字の削除を行う場合は、 INST DEL を必要な回数入力して、カーソル位置の左側を削除して行きます。

《例》

```
10  X=100
20  PRINT  TAB(5);X
```

TAB(5);を削除する場合は、Xのところにカーソルを持って行き、 INST DEL を7回打ってください。(INST DEL を押し続けて7回打つ分、入力することもできます。)

```
10  X=100
20  PRINT X
```

 を押すと、TAB(5);の削除が完了します。

〈行全体の削除〉

1つの行全体を削除する場合は、その行の行番号のみ入力して、  を押します。

《例》

```
10  X=100
20  PRINT X
10
```

上の図のように、カーソルをテキストに重ならない位置に持って行って、  を押すと、10行目は削除されます。試しに、LIST  と押してプログラムリストを出すと、10行目が完全に削除されたことがわかります。

```
10  X=100
20  PRINT X
10
Ok
LIST
20  PRINT X   ←  10行目はなくなったのでリストされない。
Ok
```

〈1行の挿入〉

1つの行を挿入する場合は、挿入したい行の行番号と、ステートメントを入力して、  を押します。

《例》

```
LIST
20  PRINT X
Ok
```

10行目にDIM A(20)と入力したい場合は、カーソルをテキストに重ならない位置に持って行って10 DIM A(20)
 と押します。

```
LIST
20  PRINT X
Ok
10  DIM A(20)
Ok
```

試しに、LIST [↵] と押して、プログラムのリストを出すと、10行目が挿入されたリストが出ます。

```
10 DIM A(20)
Ok
LIST
10 DIM A(20)
20 PRINTX
Ok
```

〈その他のスクリーンエディット機能〉

CTRL + F (CTRL キーを押しながら F キーを押す)
カーソルが1ワード[注1]分ずつ右へ移動します。

CTRL + B
カーソルが1ワード分ずつ左へ移動します。

CTRL + T　水平TAB(タビュレーション) の設定をします。

CTRL + I　CTRL + T で設定した位置にカーソルを飛ばします。(H TAB キーを押したのと同じ)

CTRL + Y　CTRL + T で設定した位置で、このキーを押すと、その位置だけ水平TABが解除されます。

CTRL + W　このキーを押した行と次の行とをつなぎます。

CTRL + J　カーソルの位置から右側の文字列をラインフィード[注2]して、次の行へ移します。

CTRL + N　カーソルのある行から上のテキストを上方向に1行スクロール[注3]します。

CTRL + O　カーソルのある行から下のテキストを下方向に1行スクロールします。

CTRL + Z　カーソルのある行から下のテキストをすべてクリアします。

CTRL + M　[↵] に同じ

BASICプログラムのテキストは、プログラムの実行やエラーの発生、ブレイクなどによって壊れることはありません。ただし、行番号のついた行で [↵] を押すと、そのままメモリーに1行登録されてしまいます。

プログラムの作成過程は、プログラムを実行してエラーを発見し、スクリーンエディタによって必要な修正を行うという一連の作業から成り立つといってよいでしょう。

なお、コマンドを入力するときは SHIFT + CLR HOME キーを押して、画面をクリアしてから行うようにすると誤りが少なくてすみます。

注1) ワード……ワードとは、空白、英記号を除く、連続した文字列のこと。(ただし、コロン(:)はワードとみなされますが、πはワードとはみなされません。)

注2) ラインフィード……行おくりのことです。

注3) スクロール……カーソルのある行を区切りとして、その行から上又は下の画面を垂直方向に1行移動します。

1.13 コントロールコード

SHARP HuBASICは、通常のオペレーションのほか、さらに扱いやすいように、数多くのキー機能を用意しています。以下、キー機能を表で示します。(なお、下の表でCTRL＋Aは CTRL キーを押しながら A キーを押すことを意味します)

CTRL ＋	出力コード	処　理　内　容	参　考
@	00		
A　または　a	01	INSTモード[注1]の設定・解除をする。	
B　　b	02	カーソルを1ワード分左へ戻す。	
C　　c	03	実行を停止する。	SHIFT＋BREAK
D　　d	04	スクリーンモードなどを初期状態にする。	INIT
E　　e	05	カーソルから右を行の終わりまで消す。	
F　　f	06	カーソルを1ワード分右へ進める。	
G　　g	07	ベルを鳴らす。	
H　　h	08	文字を抹消する。	DEL
I　　i	09	水平TAB(スペース出力) の実行。	HTAB
J　　j	0A	ラインフィードをする。	
K　　k	0B	ホームポジションへカーソルを移動する。	HOME
L　　l	0C	画面を消去する。	CLR
M　　m	0D	キャリッジリターンをする。	↵
N　　n	0E	カーソル行から上を上方向へスクロールする。	
O　　o	0F	〃　下を下方向へ　〃	
P　　p	10		
Q　　q	11	一時停止を解除する。	
R　　r	12	空白を挿入する。	INS
S　　s	13	一時停止をする。[注2]	BREAK
T　　t	14	水平TABを設定する。	
U　　u	15		
V　　v	16		
W　　w	17	次の行と結合する。	
X　　x	18		
Y　　y	19	水平TABを抹消する。	
Z　　z	1A	カーソル以下の画面をすべてクリアする。	
[	1B		
¥	1C	カーソルを右へ移動	→
]	1D	カーソルを左へ移動	←
^	1E	カーソルを上へ移動	↑
-	1F	カーソルを下へ移動	↓
0		画面の背景色を黒にする。	COLOR , 0
1		〃　青　〃	〃　1
2		〃　赤　〃	〃　2
3		〃　マゼンタ　〃	〃　3
4		〃　緑　〃	〃　4
5		〃　シアン　〃	〃　5
6		〃　黄色　〃	〃　6
7		〃　白　〃	〃　7

注1）INSTモード……文字列を入力するたびに、カーソルより右にある文字列を1文字分ずつ水平移動させて入力した文字を間に挿入させるモード。

注2）プログラム実行中は、キーを押している間のみ停止し、キーを離すと再び実行を開始します。
(完全に一時停止をするのは、リスト出力時などのBASIC管轄下にある場合です。)

(テンキーの) 0		文字グラフィックの色を黒にする。	COLOR 0
1		〃 青にする。	〃 1
2		〃 赤にする。	〃 2
3		〃 マゼンタにする。	〃 3
4		〃 緑にする。	〃 4
5		〃 シアンにする。	〃 5
6		〃 黄色にする。	〃 6
7		〃 白にする。	〃 7
/		キャラクタジェネレータのROM/RAMを切り換える。	CGEN
*		文字を点滅モードとノーマルモードとに切り換える。	CFLASH
—		文字を反転モード 〃	CREV

1.14 テレビのダイレクトコントロール

専用ディスプレイテレビに対してのキー機能を下の表で示します。(SHIFT＋0は SHIFT キーを押しながら 0 キーを押すことを意味します。)

SHIFT＋	出力コード	処理内容	参考
(テンキーの) 0		音声ミュート(注)のON/OFFを切り換える。	
1		チャンネル1の設定をする。	CHANNEL 1
2		チャンネル2 〃	〃 2
3		チャンネル3 〃	〃 3
4		チャンネル4 〃	〃 4
5		チャンネル5 〃	〃 5
6		チャンネル6 〃	〃 6
7		チャンネル7 〃	〃 7
8		チャンネル8 〃	〃 8
9		チャンネル9 〃	〃 9
/		チャンネル10 〃	〃 10
*		チャンネル11 〃	〃 11
—		チャンネル12 〃	〃 12
＋		テレビ画面とコンピュータ画面を重ねる。	CRT 2
=		テレビ画面のみの表示。	CRT 0
.		コンピュータ画面のみの表示。	CRT 1
,		音量のノーマル設定。	
↑		音量を上げる。	
↓		音量を下げる。	
←		チャンネルの切り換え(12→11→…→2→1→12→…)	
→		〃 (1→2→…→11→12→1→…)	

注) 音声ミュート……音声出力を(一時的に)カットすること。

第2章 コマンド・ステートメント

● コマンド・ステートメント・関数説明の形式

機　能	各コマンド・ステートメント・関数の働きを簡潔に記述します。
書　式	各コマンド・ステートメント・関数の書き方の一般形を示します。
省略形	各コマンド・ステートメント・関数の省略形を示します。 (省力形を入力しても、フル入力したのとまったく同じです。)
解　説	各コマンド・ステートメント・関数の使用方法、働き、注意事項を詳しく述べています。
サンプルプログラム	使い方の実例を示し、特に説明を要するものには説明をつけています。

● コマンド・ステートメント・関数の 書式 の見方

コマンドやステートメントの 書式 の中に書かれている記号・文字には、次のような意味があります。

1. [ ]　　[ ]の中は任意に省略できます。
2. { }　　{ }の中の縦に並べて書いてあるものは、任意に選択することを示しています。
3. ……　　……は連続することを示します。
4. 空白　　1つの空白を置くことを示します。
5. " "　　" "の中は文字列データであることを示します。
6. 語句　　語句については書式の下で説明しています。(変数、式 etc.)
7. ⇐ ⇒ ⇒ ⇐ ⇒　　関連命令を示します。

2.1 コマンド

2.1.1 CLEAR・LIMIT

機　能　BASICが使用するメモリーの上限を指定します。(下限はBASICの管理エリアによってきまります。)

書　式

```
CLEAR〔アドレス〕
```

```
LIMIT アドレス
```

アドレス…使用するメモリーの上限のアドレス+1。& HFF00以下。

省略形　CLE.
LIM.

解　説　この2つのコマンドはいずれも、メインメモリー内でBASICが使用する領域の上限のアドレス+1を指定するためのもので、変数、配列変数のクリアは行わず、FOR～NEXT、WHILE～WEND、REPEAT～UNTIL、GOSUB文などが格納される領域(スタック)のみクリアします。(付録「メモリーマップ」参照)

CLEARでアドレスを省略すると、全ての変数と配列変数について、数値型変数は0に、文字型変数は　""(ヌルストリング)にクリアします。これは、CLR(2.5.12CLR文参照)と全く同じ機能です。

この2つのコマンドは通常、プログラムの頭に置かれ、サブルーチンの中では使用できませんので注意してください。

サンプルプログラム

```
CLEAR &HC000  ← BASICが使用する上限アドレスを&HBFFFとする。
CLEAR         ← 全ての変数と配列変数についてクリアします。
LIMIT &HDF50  ← BASICが使用する上限アドレスを&HDF4Fとする。
```

2.1.2 NEWON

機　能　メインメモリー内のテキストの開始アドレスを設定し、テキスト、および変数をすべて消します。

書　式

```
NEWON　〔アドレス〕
```

アドレス…メインメモリー内のテキストの先頭アドレス。

省略形　なし。

解　説　指定したアドレスを、メインメモリー内のテキスト(BASICプログラム)の開始アドレスとして、テキストおよび変数をすべてクリアします。アドレスを省略すると、BASIC起動時のテキストアドレスの設定となります。

サンプルプログラム

```
NEWON &HA000  ← テキストの開始アドレスを&HA000としテキストおよび変数をクリアする。
NEWON         ← BASIC起動時のテキストアドレスの設定に開始アドレスを変え、テキ
OK              ストおよび変数をクリアする。
```

2.1.3 NEW

機　能　メモリー内のプログラムおよび変数をすべて消します。

書　式

```
NEW
```

省略形　なし。

解　説　現在メモリー内にあるテキスト（BASICプログラム）および変数をすべて消します。新しいプログラムを入れる場合には、このコマンドを実行してください。

サンプルプログラム

```
LIST
10 A=0
20 PRINT "A=";A
30 END
Ok
NEW
Ok
LIST
Ok
```

サンプルプログラムは、行番号10から30までがメインメモリー内にある場合、これをリストしその後、NEWコマンドによってメモリーをクリアしたのち、再度LISTしてメインメモリー内のプログラムが消されたことを確認しています。

(注意)

★NEW ONとCLEAR(or LIMIT)コマンドによってBASICプログラムのエリアを指定します。
これらのコマンドは、機械語のプログラムやデータ等の領域を確保するための命令です。
BASICプログラム中で、機械語プログラムとリンクして使用する場合、あらかじめNEW ONとCLEAR(LIMIT)コマンドによってBASICの領域を限定した後、その領域外に機械語プログラム等を格納してください。
限定されていないと機械語プログラムがくずされてしまう場合があります。

★NEWコマンドでは、NEW ONとCLEAR(or LIMIT)で指定した領域がクリアされます。
領域外はクリアされません。

2.1.4 AUTO

機　能　行番号を自動的に発生させます。

書　式

AUTO{m / .} [,n]

m…行番号。1～65534。
n…増　分。1～65534。

省略形　A.

解　説　行番号を自動的に発生させ、プログラムの入力を行いやすいようにするためのコマンドです。mで指定される行番号から発生を開始し、増分nずつ増やしながら行番号を発生させます。このとき、行番号mを省略すると行番号10から発生を開始します。
すでに行番号つきのステートメントが入力されている場合にはそのリストが表示されますが、そのまま [↵] を押せばリストは変更されません。
また、・(ピリオド)には、BASIC内部の持っている行管理ポインタの値が入っているので、数値と同様に扱うことができます。
なお、AUTOの中止には [SHIFT] + [BREAK] キーを押すか [CTRL] + [C] キーを押してください。

サンプルプログラム

```
AUTO
Ok
10
20
30                ← [SHIFT] + [BREAK] キーを押す。
Ok
AUTO 100,5
Ok
100
105
110
115
120               ← [SHIFT] + [BREAK] キーを押す。
Ok
AUTO 2350
Ok
2350
2360
2370
2380              ← [SHIFT] + [BREAK] キーを押す。
Ok
```

サンプルプログラムは、AUTOコマンドによりAUTO、AUTO100、5およびAUTO2350としたとき、自動的に発生する行番号を示しています。
それぞれ10から10おき、100から5おき、2350から10おきで行番号が発生しています。

2.1.5 LIST・LLIST

機　能　プログラムのリストを表示します。

書　式

LIST 〔"ファイルディスクリプタ：ファイル名"〔,〕〕[[{m|.}][－[{n|.}]]]

LLIST[[{m|.}][－[{n|.}]]]

m……開始行番号。

n……終了行番号。

ファイルディスクリプタ……SCR：,CRT：,LPT：,CAS：,MEM：,EMM0：～EMM9：。

省略形　L.

LL.

解　説　開始行mより終了行nまでのプログラムリストについて、LISTは画面に、LLISTはラインプリンターに出力します。行番号を指定することによってプログラムのどこの部分でも表示させることができます。開始行mを省略すると最初から、終了行nを省略すると最後まで出力します。CTRL＋S キーを押すか又はBREAKキーを押すと表示は一旦ストップし、スペースキー等のキーを押すと表示を再開します。

.（ピリオド）には行管理ポインタの値が入っています。

SHIFT＋BREAK やERRORで実行が停止した時、最後に実行中であった行番号が行管理ポインタに設定されていますので、LIST.（L..）によって停止した行番号の内容を表示し、訂正や変更を行なうことができます。

サンプルプログラム

```
10 INPUT A,B,X
20 Y=A*X+B
30 PRINT A,B,X,Y
40 END
```

Ⓐ サンプルプログラムは最初にⒶで示されるプログラムを入力し、

```
LIST
10 INPUT A,B,X
20 Y=A*X+B
30 PRINT A,B,X,Y
40 END
Ok
```

← LISTのみ入力した場合、メインメモリー上にあるプログラムを全て表示する。

```
LIST 30
30 PRINT A,B,X,Y
Ok
```

← LIST30と入力した場合、行番号30のプログラムのみ表示する。

```
LIST 20-
20 Y=A*X+B
30 PRINT A,B,X,Y
40 END
Ok
```

← LIST20-と入力した場合、行番号20から以後のプログラムを表示する。

```
LIST -20
10 INPUT A,B,X
20 Y=A*X+B
Ok
```

← LIST-20と入力した場合、行番号20までのプログラムを表示する。

```
LIST 20-30
20 Y=A*X+B
30 PRINT A,B,X,Y
Ok
```

← LIST20-30と入力した場合、行番号20から30までのプログラムを表示する。

2.1.6 RUN

機　能

プログラムの実行を始めます。

書　式

RUN [{ n / "ファイルディスクリプタ:ファイル名" }]

n…行番号。

ファイルディスクリプタ……CAS：, MEM：, EMM0：～EMM 9 ：。

省略形

R.

解　説

行番号 n からプログラムの実行を開始します。ただし、変数をすべてクリアしてから実行を開始しますので、変数をクリアせずに実行したい場合は GOTO（2.2.11 GOTO 文参照）を使ってください。行番号 n を省略すると、プログラムの最初から実行します。また、ファイルディスクリプタでファイルを直接指定すると、ロードした後、そのプログラムを即実行します。

RUN "ファイルディスクリプタ：ファイル名 " の実行は、LOAD " ファイルディスクリプタ：ファイル名 " 実行後にRUN を実行させたのと同じです。

サンプルプログラム

```
LIST
10 INPUT A,B,X
20 Y=A*X+B
30 PRINT A,B,X,Y
40 END
Ok
RUN
? 2,5,4
 2            5            4            13
Ok
```

サンプルプログラムはLISTをとって示したごとく、行番号10から40までのプログラムをRUNコマンドにより実行した例です。

RUNすると？を表示しカーソルが表示され入力待ちとなりますので、3つの数字を ,（カンマ）で区切って入力してください。

ここでは、2と5と4を入力した例です。

2.1.7 CONT

機　能　実行を中断したプログラムを再開します。

書　式

```
CONT
```

省略形　C.

解　説　STOP（2.2.22 STOP 文参照）や SHIFT ＋ BREAK でプログラムを止めた場合、CONT を実行すると、止まった次のステートメントから実行を再開します。
エラー発生時、CLEAR（2.1.1 CLEAR 文参照）実行直後、プログラムの書き換えなど特定条件下では CONT が実行できずエラーが出ます。
実行再開ができるときは Ok. が表示され、できないときは Ok が表示されます。

サンプルプログラム

```
LIST
 10 A=0
 20 PRINT A
 30 A=A+1
 40 GOTO 20
Ok.
RUN
  0
  1
  2
  3              ← SHIFT ＋ BREAK キーを押す。
Break in 20
Ok.
PRINT A          ← 変数A の内容を出力させる
  3
Ok.
A=10             ← 変数A の内容を変える
Ok.
CONT
  11
  12
  13
  14
  15
  16
  17
```

サンプルプログラムは、LISTに示したプログラムを実行した場合で、行番号40から20へもどる無限ループで数字を0から順に表示しつづけます。
これを SHIFT ＋ BREAK を同時にキー入力することで中断し、この時の変数A の内容を直接実行により確認した後、変数A の値を変えてCONTコマンドによりプログラムを再実行させた例です。

2.1.8 EDIT

機　能　指定した行番号をリストし、カーソルを行の先頭にセットします。

書　式

```
EDIT {n}
     {.}
```

n …行番号。
省略は・をつけるのと同じ。

省略形　E.

解　説　行番号nをリストし、nの右側にカーソルを移動させて、入力待ちになります。
エラーの発生した行の修正を能率よく行うために、このコマンドを使います。エラーが発生するとBASIC内部の・(ピリオド)にその行番号を入れるので、EDIT.(またはE..)を実行すると、エラーの発生行を直接リストして修正することができます。LIST.(またはL..)とは異なり、カーソルは行番号のうしろにきています。

サンプルプログラム

```
10 A=0
20 PLINT A
30 A=A+1
40 GOTO 20
RUN
Syntax error in 20 ←エラー発生
OK
EDIT.
20 PLINT A                ←[P]の所にカーソルが出る

20 PRINT A                ←[L]を[R]に変えて [↵] を押す。
LIST
 10 A=0
20 PRINT A
30 A=A+1
40 GOTO 20
OK
```

サンプルプログラムは、2.1.7のCONTコマンドの説明に使用したプログラムを入力する際、誤って行番号20のPRINTコマンドをPRINTと入力した場合の修正にEDITコマンドを使用した例です。

2.1.9 DELETE

機　能　指定した範囲のプログラムを削除します。

書　式

DELETE [m/.] [,/_] [n/.]

m…開始行番号。
n…終了行番号。

省略形　D.

解　説　開始行番号mから終了行番号nまでのプログラムを削除するためのコマンドです。プログラムの一部の行の削除は、該当する行の番号のみを入れることでできますが、まとまった行の集まりをプログラム中から削除したいときにこのコマンドを使います。なお、開始行mを省略すると、プログラムの最初から、終了行nを省略すると、プログラムの最後まで削除します。mとnの両方を省略すると、プログラム全部を削除します。(・(ピリオド)には、行管理ポインタの値が入っています。)

サンプルプログラム

```
LIST
10 INPUT A,B,X
20 Y=A*X+B
30 PRINT A,B,X,Y
40 END
OK
DELETE 20-30
OK
LIST
10 INPUT A,B,X
40 END
OK
```

サンプルプログラムは、2.1.6のRUNコマンドの説明に使用したプログラムにDELETE 20-30として、プログラムの行番号20および30が削除されたことを確認した例です。

2.1.10 RENUM

機　能　プログラムの行番号を指定通り付け換えます。

書　式

```
RENUM〔〔l〕,〔m〕,〔n〕〕
```

l …新行番号。省略すると10。
m …旧行番号。省略するとプログラムの最初の行。（l ≧ m）
n …増分。省略すると10。

省略形　REN.

解　説　旧行番号mで指定した行以降の行番号を、新行番号1で始まる行番号に付け換えます。新しい行番号は増分nの一定間隔で増えます。GOTO（2.2.11 GOTO文参照）、GOSUB（2.2.12 GOSUB文参照）などの飛び先の行番号も同時に変化します。
新行番号は旧行番号より大きくなければなりません。ただし、旧行番号を省略した場合は小さくてもかまいません。プログラム中で未定義の行番号は65535にセットされます。
旧行番号mを省略する場合、新行番号と増分の間に ,（カンマ）が2つなければいけません。

サンプルプログラム

Ⓐ

```
5 REM G.C.M & L.C.M
10 INPUT "A=";AI
15 IF AI<1 GOTO 130
20 INPUT "B=";BI
25 IF BI<1 GOTO 130
26 IF A<B THEN SWAP AI,BI
27 A=AI:B=BI
30 M=A MOD B
40 IF M=0 GOTO 100
50 A=B
60 B=M
70 GOTO 30
100 PRINT"L.C.M ";AI*(BI/B)
110 PRINT"G.C.M ";B
120 GOTO 10
130 END
OK
RENUM
```

サンプルプログラムは、最大公約数および最小公倍数を求めるプログラムの作成中にRENUMコマンドを実行した例です。
サンプルⒶのプログラムがRENUMを実行した後、サンプルⒷのごとくとなります。

（注）最大公約数と最小公倍数のプログラムは、2.1.12のTRON・TROFFを参照ください。

Ⓑ

```
10 REM G.C.M & L.C.M
20 INPUT "A=";AI
30 IF AI<1 GOTO 160
40 INPUT "B=";BI
50 IF BI<1 GOTO 160
60 IF A<B THEN SWAP AI,BI
70 A=AI:B=BI
80 M=A MOD B
90 IF M=0 GOTO 130
100 A=B
110 B=M
120 GOTO 80
130 PRINT"L.C.M ";AI*(BI/B)
140 PRINT"G.C.M ";B
150 GOTO 20
160 END
```

2.1.11 SEARCH

機　能　指定文字列を含んでいる行のリストを出します。

書　式

```
SEARCH"x"
```

x …文字列。

省略形　SE.

解　説　プログラム中より指定文字列を探し出し、該当する行をすべてリストします。
文字列中に "（ダブルクォーテーションマーク）を含むことはできません。

サンプルプログラム

```
LIST
10 REM SAMPLE
20 INPUT A,B,X
30 Y=A*X+B
40 PRINT A;B;X;Y
50 END
Ok
SEARCH"A"
10 REM SAMPLE
20 INPUT A,B,X
30 Y=A*X+B
40 PRINT A;B;X;Y
Ok
SEARCH" A"
20 INPUT A,B,X
40 PRINT A;B;X;Y
Ok
```

SEARCH "A" を実行すると、A を含む10～40行目のリストを出力します。
SEARCH "　A" を実行すると、スペースも 1 文字とみるので"　A" を含む20と40行目のリストを出力します。

2.1.12 TRON・TROFF

機　能　プログラムの実行過程の追跡と解除を行います。

書　式

```
TRON
```

```
TROFF
```

省略形

T.

TROF.

解　説　TRONを入力して、Ok. と表示された後、RUNを実行すると、それ以降に実行した行の番号を[　]（かっこ）で囲んで、実行順に画面に表示します。

TROFFを実行すると、TRONの機能を解除します。

サンプルプログラム

```
LIST
10 REM G.C.M. & L.C.M.
20 INPUT "A=";A
30 IF A<1 THEN 20
40 INPUT "B=";B
50 IF B<1 THEN 40
60 IF A<B THEN SWAP A,B
70 X=A:Y=B
80 M=X MOD Y
90 IF M=0 THEN 130
100 X=Y:Y=M
120 GOTO 80
130 PRINT"G.C.M.";Y
140 PRINT"L.C.M.";A*B/Y
150 GOTO 20
Ok
TRON
Ok.
RUN
[10][20]A=? 9
[30][40]B=? 12
[50][60][70][80][90][100][120][80][90][1
30]G.C.M. 3
[140]L.C.M. 36
[150][20]A=?←  SHIFT  ＋  BREAK  キーを押す。
Break in 20
Ok.
TROFF
Ok.
RUN
A=? 9
B=? 12
G.C.M. 3
L.C.M. 36
A=?          ←  SHIFT  ＋  BREAK  キーを押す。
Break in 20
Ok.
```

最大公約数と最小公倍数のプログラムをTRONコマンド入力後、実行させると実行した行番号を〔　〕で囲んで表示します。9と12の最大公約数と最小公倍数をもとめたところで中断し、TROFFコマンドでトレースを中止し、再度RUNコマンドで実行した例です。
トレース時との相異を確認してください。

2.1.13 DEVICE

機　能

デフォルトのファイルディスクリプタを決めます。

書　式

DEVICE "ファイルディスクリプタ:"

省略形

DEV.

解　説

FILES, LEILES, LOAD, LOADM, SAVE, SAVEM, LOAD?, VERIFY, CHAIN, MERGE, OPEN, INIT, DEVI$, DEVO$, KILL, LISTおよびRUNのそれぞれのコマンドにおいて、それぞれのファイルディスクリプタを省略すると、DEVICEで指定されたファイルディスクリプタとして実行されます。ただし、ROM、カセットバージョンBASIC起動時は CAS: にディスクリプタが指定されています。
一般に、ある結果を画面に表示したり、プリンタなどに出力したりする様な時、入出力デバイスにより別々にプログラミングしてやらねばなりませんが、ファイルディスクリプタを用いればDEVICE命令により、入出力デバイスを簡単に変更 でき、プログラムも短かくて済むといった利点があります。
ただし、コマンドによって使用できないファイルディスクリプタがあります。
(詳細は第4章を参照ください。)

サンプルプログラム

```
LIST
10 DEVICE"LPT:"
20 OPEN"O",#1,"A"
30 PRINT#1,"SHARP HuBASIC"
40 PRINT#1,STRING$(13,"-")
50 CLOSE
60 END
OK
RUN
OK
```

ファイルディスクリプタをLPT:に指定し、アウトプットモード"0"、ファイル番号#1、ファイル名Aのファイルを開けます。そしてSHARP HuBASICとライン13個をファイルAに格納します。
LPT:ですから、プリンタに出力されます。

```
SHARP HuBASIC
-------------
```
} プリンタへの出力

2.1.14 FILES・LFILES

機　能

ファイルの一覧表を表示します。

書　式

```
{FILES }[ "ファイルディスクリプタ:"]
{LFILES}
```

ファイルディスクリプタ…CAS:, MEM:, EMM0:～EMM9:。

省略形

FIL.
LF.

解　説

ファイルディスクリプタで指定したデバイス内にあるファイルの名前(ファイル名)の一覧表が、FILESのときには画面に表示され、 LFILESのときにはラインプリンターにプリントされます。ファイルディスクリプタの指定がないときは、DEVICE（2.1.13 DEVICE文参照）で指定されたデバイスのファイル名の一覧表が表示されます。

サンプルプログラム

```
LFILES"MEM:"
Bas      "MEM0:PROGRAM 1    .    "          '80/01/01 SUN 08:20
Bas      "MEM0:PROGRAM 2    .    "          '80/01/01 SUN 08:20
Bas      "MEM0:PROGRAM 3    .    "          '80/01/01 SUN 08:20
Bas      "MEM0:PROGRAM 4    .    "          '80/01/01 SUN 08:20
Bas      "MEM0:PROGRAM 5    .    "          '80/01/01 SUN 08:20
Bas      "MEM0:PROGRAM 6    .    "          '80/01/01 SUN 08:20
Bas      "MEM0:SAMPLE #1    .    "          '80/01/01 SUN 08:20
Bas      "MEM0:SAMPLE #2    .    "          '80/01/01 SUN 08:20
Ok.
```

上の例は、グラフィックRAMをグラフィック表示用としてではなく外部記憶用として使用した場合の実行例で、グラフィックメモリー内のファイルの名前(ファイル名)の一覧表をプリントアウトさせたものです。

注）ファイル名の前についているBasはファイルの形式を表わすもので、
Bas……BASICプログラム
Asc……アスキー形式で書きこまれたファイル
Bin ……バイナリー形式で書きこまれたファイル
の3種類があります。

2.1.15 LOAD・LOADM

機　能

ファイルに記録されているプログラムを、メインメモリーに入れます。

書　式

```
LOAD〔"〔ファイルディスクリプタ:〕ファイル名"〕
```

```
LOADM〔"〔ファイルディスクリプタ:〕ファイル名"〕〔,ad〕
                                              〔,〔ad〕,R〕
```

ad…ロード開始アドレス

ファイルディスクリプタ…CAS:, MEM:, EMM0:～EMM9:。

省略形

LO.

LO.M

解　説

BASICプログラムをメインメモリーに入れるときにはLOADを、機械語プログラムをメインメモリーに入れるときにはLOADMを実行します。

LOADにおいて、ファイル名を指定すると、そのファイルを見つけ出すまで、ほかのファイルを読み飛ばし、指定したファイルの所に来るとロードを始めます。(ファイルからメインメモリーにプログラムを移すことをロードといいます。) また、ファイル名を省略すると、最初に見つけたファイルをロードします。

LOADMにおいて、adは機械語プログラムのロードを開始するアドレスで、これを省略するとSAVEM (2.1.16 SAVEM文参照) で指定したセーブ開始アドレスからロードが始まります。書式の最後に, Rをつけると、ロード終了後プログラムを直ちに実行します。



サンプルプログラム

```
LOAD
LOAD "TEST"
LOAD "CAS0:TEST"

LOADM
LOADM "TEST"

LOADM "TEST",,R
```

特にDEVICEコマンドでファイルディスクリプタを指定していない場合、BASICの初期設定により、カセットよりプログラムをロードします。この時、最初に見つけたファイルがロードされます。

ファイル名の指定があるのでTESTというファイル名のプログラムをロードします。

ファイルディスクリプタとしてカセットを指定し、ファイル名はTESTというファイルを指定しています。

最初に見つけたプログラムが機械語プログラムである場合、ロードします。

ファイル名が"TEST"という機械語プログラムをロードします。

Rオプションを付けた例でTESTというファイル名の機械語プログラムをロードし、終了後、直ちにプログラムを実行します。

,(カンマ)は2つとも省略してはいけません。

2.1.16 SAVE・SAVEM

機　能　メインメモリー内のプログラムをファイルに記録します。

書　式

SAVE〔"〔ファイルディスクリプタ:〕ファイル名"〕〔,A〕

SAVEM〔"〔ファイルディスクリプタ:〕ファイル名"〕,ad_1, ad_2〔,ad_3〕

ad_1…セーブの開始アドレス。
ad_2…セーブの終了アドレス。
ad_3…実行開始アドレス。
ファイルディスクリプタ…CAS:, MEM:, EMM0:～EMM9:。

省略形　SA.
SA.M

解　説　SAVEを実行すると、メインメモリー内のBASICプログラムを外部ファイルに記録します。(このことをセーブするといいます。) 通常のSAVEは、バイナリィ形式[注1]にプログラムを圧縮して行いますが、書式の最後に，Aをつけると、アスキー形式[注2]で記録するので、MERGE（2.1.19 MERGE文参照）の実行が可能になります。アスキー形式のセーブは、リストそのままの形で行うのでバイナリー形式よりファイルのスペースを必要としますが、外部ファイルのプログラムをMERGEするためには必ずアスキー形式でセーブしておかなければなりません。
SAVEMを実行すると、メインメモリー内の機械語プログラムを、ad_1で指定されたアドレスからad_2で指定されたアドレスまで、外部ファイルに記録します。ad_3を省略するとad_1がad_3として設定されます。



（ファイル名中には ,(カンマ) "(ダブルクォーテーション) :(コロン); (セミコロン)を使用しないでください。）

サンプルプログラム

```
SAVE                                  ← 特にDEVICEコマンドにより、ファイルディスクリプタを指定していない場合、ファイル名なしでプログラムをセーブする。
SAVE"TEST"                            ← TESTというファイル名をつけてプログラムをセーブする。
SAVE"CAS0:TEST"                       ← カセットテープにTESTというファイル名をつけてプログラムをセーブする。
SAVE"CAS0:TEST",A                     ← カセットテープにアスキー形式でTESTというファイル名をつけてセーブする。
SAVEM"CAS0:TEST",&H8000,&H80FF,&H8000 ← メインメモリー内アドレス&H8000から&H80FFにある機械語プログラムをTESTというファイル名でカセットテープにセーブする。この場合、実行アドレスとセーブ開始アドレスが同一であるので最後の&H8000は省略可能です。
SAVEM"TEST",&H7800,&H9EFF,&H9010      ← メインメモリー内アドレス&H7800から&H9EFFにある機械語プログラムをTESTというファイル名で実行開始アドレスを&H9010としてセーブする。
```

注1）バイナリー形式…BASICプログラムが、メモリー中に格納されているままの状態(中間コード)によりファイルにセーブする。

注2）アスキー形式…セーブする内容を各文字毎にキャラクターコード(アスキーコード)に変更して、ファイルにセーブする。

2.1.17 LOAD？・VERIFY

機　能　セーブしたプログラムが、ファイルに正しく記録されているかどうか調べます。

書　式

```
LOAD？〔"〔ファイルディスクリプタ:〕ファイル名"〕
```

```
VERIFY 〔"〔ファイルディスクリプタ:〕ファイル名"〕
```

ファイルディスクリプタ…CAS:。

省略形　VE.（LOAD？の省略形はありません）

解　説　LOAD？とVERIFYはまったく同じ意味のコマンドで、メインメモリー内のプログラムと、ファイル名で指定した外部ファイル内のプログラムとの照合を行います。このコマンドの実行によって、セーブしたプログラムがファイルに正しく記録されているかどうか調べることができます。また、ファイル名を省略すると、最初に見つけたファイルとメモリーとの照合を始めます。アスキー形式のファイルをVERIFY、LOAD？した場合は、ファイルの最初から最後まで空読みします。これにより、アスキー形式のファイル内の不良ブロックを見つけ出すことができます。



サンプルプログラム

```
LOAD?
LOAD?"CAS0:TEST"

VERIFY
VERIFY"TEST"
```

LOAD？　メインメモリー内にあるプログラムとセーブされている内容の照合を行ないます。

LOAD？"CAS0:TEST"　カセットテープにTESTというファイル名でセーブされている内容とメインメモリー内にあるプログラムとの照合を行ないます。

VERIFY　LOAD？と同じ意味です。

VERIFY"TEST"　TESTというファイル名でセーブされている内容とメインメモリー内にあるプログラムとの照合を行ないます。

2.1.18 CHAIN

機　能　プログラムをロードし実行します。

書　式

```
CHAIN "〔ファイルディスクリプタ:〕ファイル名"
```

ファイルディスクリプタ…CAS:，MEM:，EMM0:～EMM9:。

省略形　CH.

解　説　メインメモリー内にあるプログラムの変数を保護し、ファイルディスクリプタで指定されたファイルからプログラムをロードして実行します。大きなプログラムを複数のプログラムに分割し、CHAIN命令で次々に実行させることにより、メインメモリーにはいりきらないプログラムも実行可能にすることができます。

サンプルプログラム

```
CHAIN"CAS:test"
```
← メインメモリー内にあるプログラムの変数の値をそのまま保護し、カセットより test というファイル名のプログラムをロードし実行します。

```
CHAIN"test"
```
← 特にDEVICEコマンドによりファイルディスクリプタを指定していない場合、上記サンプルプログラムと同一内容を実行します。

サンプル(1)

```
LIST
5 REM ﾃﾞｰﾀ ﾘｰﾄﾞ
10 CLS
20 DIM A(50)
30 FOR I=0 TO 50
40 A(I)=PEEK(I)
50 NEXT
OK
```

サンプル(2)

```
LIST
5 REM ﾃﾞｰﾀ ｼｮﾘ
10 CLS
20 FOR I=0 TO 50
30 PRINT HEX$(A(I)),
40 NEXT
OK
```

まず、あらかじめサンプル(2)を打ちこんで、カセットなどにセーブしておきます。

SAVE "CAS：SAMPLE2"↵

その上で、サンプル(1)を打ちこんで実行します。

NEW↵
OK
(サンプル(1)を打ちこむ)
OK
RUN↵
OK

すると、配列変数Aの中に、メインメモリーの先頭から50個のデータが読みこまれます。
次に、CHAINコマンドを実行します。

CHAIN "CAS：SAMPLE2"↵

これによって、サンプル(2)がロードされ実行されます。CHAINコマンドは変数を保護しますから、サンプル(1)で読みこんだ配列変数Aのデータは残っています。サンプル(2)では、それを表示させています。
このように、CHAIN命令によって、複数のプログラムで同一データを処理することができますので大きなプログラムの実行に有効です。

2.1.19 MERGE

機　能　メインメモリー内のプログラムとファイル内のプログラムとを併合します。

書　式

```
MERGE〔"〔ファイルディスクリプタ:〕ファイル名"〕
```

ファイルディスクリプタ…CAS:, MEM:, EMM0:～EMM9:。

省略形　M.

解　説　ファイル名で指定したファイルからプログラムを読み込み、メインメモリー内のプログラムと組み合わせて1つのプログラムにします。このとき、双方の行番号が重複した場合は、ファイルから読み込んだ方を有効にします。ファイル内のプログラムは、アスキー形式でセーブされていなければなりません。



サンプルプログラム

```
MERGE
MERGE"TEST"
MERGE"CAS0:TEST"
```

←アスキー形式でファイルされたプログラムを、メインメモリー内のプログラムと合わせて一本のプログラムにします。

←アスキー形式でTESTというファイル名をつけてセーブされているプログラムと、メインメモリー内のプログラムを1つにします。

←カセットテープにアスキー形式でTESTというファイル名をつけてセーブされているプログラムと、メインメモリー内のプログラムを1つにします。

2.1.20 KILL

機　能　ファイルに記録されたプログラムを抹消します。

書　式

```
KILL "[ファイルディスクリプタ:]ファイル名"
```

ファイルディスクリプタ… MEM:, EMM0:～EMM9:

省略形　KI.

解　説　ファイル名で指定した外部ファイル内のプログラムを消します。ファイルディスクリプタは省略することができますがカセットに対しては実行されません。

サンプルプログラム

```
KILL "TEST"
```

←TESTというファイル名のファイルを抹消します。

2.2 一般ステートメント

2.2.1 LET

機　能　式の値を変数に代入します。

書　式

〔LET〕 変数＝ { 式 / 定数 / 変数 }

省略形　LETは完全に省略できます。

解　説　式の演算結果を変数に代入します。左辺が数値変数のとき右辺は数値、左辺が文字列変数のとき右辺は文字列というように、両辺のタイプは一致させなければなりません。なお、LETは完全省略ステートメントなので、なくてもかまいません。

サンプルプログラム

```
10 LET A%=14
20 LET B=10.532
30 LET C!=12.234
40 LET D=12*B-C/2
50 LET E#=1.2345678901234#
60 LET F$="ABCDEFG"
```

```
10 A%=14
20 B=10.532
30 C!=12.234
40 D=12*B-C/2
50 E#=1.2345678901234#
60 F$="ABCDEFG"
```

上の2つのサンプルは全く同じ意味です。

2.2.2 DEFINT・DEFSNG・DEFDBL・DEFSTR

機　能　変数の型を宣言します。

書　式

```
{DEFINT  }
{DEFSNG  } { 文字1〔,文字2,…〕 }
{DEFDBL  } { 文字1〔- 文字2〕  }
{DEFSTR  }
```

文字はアルファベット1文字。

省略形

DEFI.
DEFS.
DEFD.
DEFST.

解　説　この4つのDEFは、変数の型を宣言し、型の識別記号をつけなくても、宣言した型の変数として使うためのステートメントです。たとえば、Aは単精度型変数ですが、DEFSTR Aと宣言すると、文字型になり、A$とわざわざ$をつけなくても文字列の代入ができます。
この4つの型宣言は、変数の頭1文字によって行いますので、この頭文字で始まる変数は、型記号（%, !, #, $）がついていなければ、すべて宣言した型になります。
DEFINTは変数を整数型として、
DEFSNGは変数を単精度型として、
DEFDBLは変数を倍精度型として、
DEFSTRは変数を文字型として、
それぞれ定義します。
また、変数と変数の間を－（ハイフン）でつなぐと、その間に含まれるアルファベットはすべて、宣言した型になります。なお、まったく宣言しなければすべて単精度型変数になります。

サンプルプログラム

```
10 DEFINT C-H
20 DEFSTR A
30 DEFSNG I,J,K
40 DEFDBL B,L-Z
```

上のサンプルではC～Hを整数型、Aを文字型、I, J, Kを単精度型、BとL～Zを倍精度型の変数として使えるよう定義しています。

2.2.3 PRINT・LPRINT

機　能　画面に情報を出力します。

書　式

PRINT[a_1] $\left(\left\{\begin{matrix};\\,\end{matrix}\right\} a_2\right) \left(\left\{\begin{matrix};\\,\end{matrix}\right\} a_3\right)$ ……

LPRINT[a_1] $\left(\left\{\begin{matrix};\\,\end{matrix}\right\} a_2\right) \left(\left\{\begin{matrix};\\,\end{matrix}\right\} a_3\right)$ ……

a_1, a_2, a_3……式、変数、定数。

省略形　？またはP.
LP.

解　説　PRINTは,（カンマ）;（セミコロン）で区切られた式の値、文字列、変数、定数の値を画面に表示します。PRINTのみのときは1行改行し、式、変数、定数の間や最後に;（セミコロン）をつければ、改行せずに続けて表示することができます。また,（カンマ）をつければ10文字間隔の表示となります。LPRINTは指定した式の値、文字列、変数、定数の値をラインプリンターにプリントします。改行についてはPRINTと同様です。
なお、？（クエスチョンマーク）をPRINTと同じステートメントとして使うことができます。

サンプルプログラム

```
LIST
10 REM SAMPLE
20 A=15
30 B=-30
40 C$="HELLO!"
50 D$="Hi!"
60 PRINT TAB(5);A;C$
70 PRINT B;TAB(5);D$
80 PRINT B;SPC(5);D$
90 PRINT "ABCDEFGHIJK";22,"SHARP X-1";-82
100 PRINT A,B,C$,D$
OK
RUN
       15 HELLO!
-30  Hi!
-30       Hi!
ABCDEFGHIJK 22        SHARP X-1-82
 15          -30      HELLO!     Hi!
OK
```

TAB（n）は画面の左端からn文字分飛ぶという意味で、SPC（n）は位置に関係なくn文字の空白を書くという意味です。PRINTの書式(USINGの使い方)については書式指定の章(2.13 PRINT USING)で詳述します。

2.2.4 WRITE

機　能　画面にデータを表示します。

書　式

WRITE〔a_1〕$\left\{\begin{matrix},\\;\end{matrix}\right\}$ a_2 $\left\{\begin{matrix},\\;\end{matrix}\right\}$ a_3 ……

a_1, a_2, a_3……式、変数、定数。

省略形　WR.

解　説　画面に式の値を ,(カンマ)で区切り、詰めて表示します。文字型のときは"で囲んで表示します。
PRINTとは違い、データ間は自動的に ,(カンマ)で区切られます。またTAB(n)の使用はできません。

サンプルプログラム

```
WRITE 1,-2,"123"
1,-2,"123"
Ok
WRITE 1;-2;"123"
1,-2,"123"
Ok
```

2.2.5 INPUT

機　能　キーボードからデータを入力します。

書　式

INPUT〔"文字列"〕$\left\{\begin{matrix};\\,\end{matrix}\right\}$ 変数1〔, 変数2,……〕

文字列……画面に表示する文字列。
変　数……キーボードから入力する変数。

省略形　I.

解　説　キーボードから入力した数値や文字を変数に入れます。"(ダブルクォーテーションマーク)で文字列を囲むことによって、ステートメントの実行時に、その文字列を画面に表示し、キーボードからの入力を待ちます。このとき、文字列と変数が ;(セミコロン)で区切られていると文字列の後ろに?(クエスチョンマーク)がつき ,(カンマ)で区切られていると?は表示しません。キーボードからデータを入力して、  キーを押すとそのデータを変数に入れます。
変数の型とキーボードから入力する変数の型は一致していなければなりません。複数個の変数を「,」で区切って指定しているときには、その指定順に、対応するデータを「,」で区切って一度に入力します。
何も入力しないで  キーを押すと変数の値を保存して次のステートメントに進みます。

サンプルプログラム

```
10 INPUT "A=";A
20 INPUT "B,C,D=";B,C,D
30 PRINT "A=";A
40 PRINT "B,C,D=";B;C;D
50 END
RUN
A=? 5                          ←5を入力
B,C,D=? 5,15,3.14              ←5と15と3.14を入力
A= 5
B,C,D= 5   15   3.14
Ok
```

2.2.6 LINPUT•LINE INPUT

機　能　キーボードから文字を入力します。ブランク（空白）やカンマ（,）も入力できます。

書　式

```
LINPUT〔"文字列"{;}〕文字式
      {,}
```

```
LINE INPUT〔"文字列"{;}〕文字式
           {,}
```

文字列……画面に表示する文字列。
文字式……キーボードから入力する文字変数。

省略形　LIN.
LINEI.

解　説　キーボードから入力した文字列を文字式に入れます。このとき，（カンマ）や空白（スペース）も1文字として入力できます。INPUT（2.2.5 INPUT文参照）と同様に　"（ダブルクォーテーションマーク）で囲った文字列を画面に表示できますが、LINPUTでは、これも同時に入力してしまいますので、プログラム中で取り除く処理が必要です。なお，と；（セミコロン）のどちらを使っても同じ結果となります。入力できる最大文字数は255文字までです。

サンプルプログラム

```
LINE INPUT"STR=";A$
STR=123,456,""""""",,,,
OK
PRINT A$
STR=123,456,""""""",,,,
OK
```

2.2.7 CLEAR•CLR

機　能　変数および配列をすべてクリアします。

書　式

```
CLEAR
```

```
CLR
```

省略形　CLE.（CLRにはありません）

解　説　すべての変数および配列のうち数値型のものを0に、文字型のものをヌルストリング（何も入ってない状態）にします。（CLEARについては、2.1.1を参照してください。）

サンプルプログラム

```
A=1000:B$="123"
OK
?A,B$;"*"
 1000       123*
OK
CLEAR
OK
?A,B$;"*"
 0          *
OK
```

2.2.8 OPTION BASE

機　能　配列の添字の下限を宣言します。

書　式

```
OPTION BASE n
```

n……0, 1

省略形　OP. B.

解　説　配列の添字は通常下限値が0になっていますが、このステートメントによって、1にすることができます。配列のディメンジョンが切られる前に一度だけ宣言でき、再宣言できません。
CLR, CLEAR, RUN, NEW [ON] によって、この宣言は解除されます。

サンプルプログラム

```
NEW
Ok
OPTION BASE 1
Ok
DIM A(10)
Ok
OPTION BASE 0                ←― 一度OPTION BASE 1 を実行しているので、再宣言するとエラーとなります。
Duplicate Definition
Ok
```

2.2.9 DIM

機　能　配列変数を定義します。

書　式

```
DIM 配列名1 (m1〔, m2, ……〕)〔, 配列名2 (n1〔, n2, …〕)〕……
```

配列名……配列型変数名。

m_1, m_2, n_1, n_2 …… : 添字の上限。

省略形　DI.

解　説　配列変数の名前とその添字の上限を設定します。1つのDIMで複数の配列変数を定義でき、配列ごとに、メモリー範囲内で255個の添字（255次元）の指定ができます。(一行の入力文字数の制限から実際には255次元までは指定できません。)DIMの実行後、数値型配列には0が入り、文字型配列はヌルストリングになります。配列添字の下限は、通常は0ですがOPTION BASE 1を宣言すると、1に変更することができます。

サンプルプログラム

```
10 OPTION BASE 0
20 DIM A(10),B(5,6)
```

上の例では、具体的には

```
A(0)        B(0,0), B(1,0),……,B(5,0)
A(1)        B(0,1),……          ⋮
 ⋮            ⋮      ……       ⋮
A(10)       B(0,6),……,B(5,6)
```

というような2つの配列が定義され、すべての配列要素に0が入ります。

```
10 OPTION BASE 1
20 DIM A(10),B(5,6)
```

OPTION BASE 1のときは、

```
A( 1)       B( 1, 1), B(2, 1),……, B(5,1)
A( 2)       B(1,2) ……          ⋮
 ⋮            ⋮      ……       ⋮
A(10)       B(1,6), ……, B(5,6)
```

というような2つの配列が定義され、すべての配列要素には0が入ります。

2.2.10 LABEL

機　能　プログラム中にラベルをつけます。

書　式

```
LABEL "ラベル名"
```

ラベル名……255文字までの文字列。

省略形　LA.

解　説　プログラム中にラベルをつけます。このラベルは、GOTO, GOSUBなどのジャンプ先として参照される目印で、これによってプログラムにドキュメント性をもたせることができます。ラベル名の制限は特にありません。プログラム中に同じラベルがある場合は最初に出てきたラベルが優先されます。

サンプルプログラム

```
10 A=0
20 LABEL"インサツ":PRINT A;
30 A=A+1
40 GOTO "インサツ"
```

```
10 A=0
20 PRINT A;
30 A=A+1
40 GOTO 20
```

上のサンプルはどちらも同じ意味です。
止めるときは、SHIFT ＋ BREAK キーを押してください。

2.2.11 GOTO・GO TO

機　能　指定した行番号かラベルへ無条件にジャンプします。

書　式

```
{GOTO  }{ 行番号    }
{GO TO }{"ラベル名"}
```

行番号……ジャンプ先の行番号。
ラベル名……LABELで定義された文字列。

省略形　G.

解　説　GOTOとGO TOは同じ機能を持ち、指定した行番号かラベルへ、無条件にジャンプするもので、プログラムの流れを変えるときに使用します。ラベルはLABEL（2.2.10 LABEL文参照）によってエントリーされた文字列です。

サンプルプログラム

```
10 A=0
20 LABEL"インサツ":PRINT A;
30 A=A+1
40 GOTO "インサツ"
```

2.2.12 GOSUB・GO SUB

機　能　BASICプログラム内のサブルーチンを呼び出します。

書　式

```
{GOSUB }{ 行番号   }
{GO SUB}{"ラベル名"}
```

行番号……サブルーチンの開始行番号。
ラベル名……LABELで定義された文字列。

省略形　GOS.

解　説　メモリー中にあるBASICのサブルーチンを呼び出し、指定された行番号またはラベル名からRETURN（2.2.13 RETURN文参照）までを実行させます。呼び出されるサブルーチンの最後には必ずRETURNがなければなりません。

サンプルプログラム

```
10 INPUT"A,B=";A,B
20 GOSUB 100
30 PRINT "C=";C
40 END
100 C=(A-B)/2
110 RETURN
RUN
A,B=? 3,12
C=-4.5
Ok
```

サンプルプログラムは、行番号20で行番号100からはじまるサブルーチンを呼びだしています。
行番号40のENDがないと行番号30を実行後、サブルーチン内へ実行がうつり、行番号110でエラーが発生します。

2.2.13 RETURN

機　能　GOSUBで呼ばれたサブルーチンの最後につけて戻りに使用します。

書　式

```
RETURN{ 行番号   }
      {"ラベル名"}
```

行番号……サブルーチンからの戻り先行番号。
ラベル名……LABELで定義された文字列。

省略形　RE.

解　説　RETURNはGOSUBで呼ばれたサブルーチンから戻り先へ戻るときに使用します。行番号、ラベル名をつけると、その行へ戻りますが、省略するとGOSUB文の後ろに戻ります。

サンプルプログラム　GOSUB文参照。

```
RETURN                  ←― GO SUB文の後ろにもどります。
RETURN 100              ←― 行番号100へもどります。
RETURN "MAIN PROG"      ←― ラベル〝MAIN PROG〟の行へもどります。
```

2.2.14 IF～THEN……ELSE

機　能

論理式を判断して次に進みます。

書　式

IF 論理式 THEN { 行番号 / "ラベル名" / 文 } [ELSE { 行番号 / "ラベル名" / 文 }]

IF 論理式 [THEN] GOTO { 行番号 / "ラベル名" } [ELSE { 行番号 / "ラベル名" / 文 }]

行番号……ジャンプ先の行番号。
ラベル名……LABEL（2.2.10 LABEL文参照）で定義された文字列。
文……任意のステートメント。

省略形

IF～TH.……EL.……

解　説

論理式が正しいかどうか判断して、指定の行番号やステートメントに制御を移して実行を続けます。
ELSE以降がある場合、論理式が正しければ、THENの後ろで指定された行番号・ラベルの行、または文の実行に移り、まちがっていれば、ELSEの後ろで指定された行番号・ラベルの行、または文の実行に移ります。
ELSE以降がない場合、論理式が正しければ、THENの後ろで指定された行番号・ラベルの行、または文の実行に移り、まちがっていれば、次の行へ移ります。
THENとELSEの後ろの文として、IF文を使ってもかまいません。ただし、この場合は最後のIF文のELSEおよび文のみ省略することができます。なお、THENおよびELSEの前は1文字分スペースをあけてください。次に、このステートメントの流れ図を示します。

〈ELSE以降がある場合〉



〈ELSE以降がない場合〉



サンプル
プログラム

```
LIST
10 INPUT A
20 IF A=0 THEN PRINT"A=0" ELSE PRINT"A<>0"
OK
RUN
? 1
A<>0              ← 変数Aに1を入れるとA＜＞0がプリントされます。
OK
RUN
? 0
A=0               ← 変数Aに0を入れるとA＝0がプリントされます。
OK

LIST
10 INPUT A,B
20 IF A=0 THEN IF B=0 THEN PRINT"A=0,B=0" ELSE PRINT"A=0,B<>0"
ELSE IF B=0 THEN PRINT"A<>0,B=0" ELSE PRINT"A<>0,B<>0"
30 GOTO 10
OK
RUN
? 0,0
A=0,B=0
? 0,1
A=0,B<>0
? 1,0
A<>0,B=0
? 1,1
A<>0,B<>0
?                 ← SHIFT + BREAK
Break in 10         キーを押す。
Ok.
```

行番号20の内容は、次のフローチャートの様になります。



2.2.15 FOR～TO…STEP…

機　能　FORとNEXTとの間を繰返し実行します。

書　式

```
FOR 変数名＝初期値 TO 終了値〔STEP 増分〕
```

変数名……単精度型変数、整数型変数。
初期値、終了値、増分……定数、変数、式。負の数でもよい。

省略形　F.

解　説　FORはNEXTとともに用いられ、変数の値が終了値からはみ出すまでFORからNEXTへの実行を続け、はみ出すとNEXTの次へ実行を移します。変数は初期値の値から始まって、この文が実行されるたびに増分だけ加えられて行き、終了値をはみ出すと繰返しが終わります。なお、STEP増分を省略すると増分は1になります。
対応するNEXTがない場合はFOR without NEXTエラーが出ます。変数の初期値が終了条件をすでに満たしている場合は、そのままNEXTの次へ実行を移します。TOとSTEPの前には1文字分スペースをあけてください。

サンプルプログラム

```
LIST
10 FOR I=1 TO 4
20 FOR J=1 TO I
30 FOR K=J TO I
40 PRINT"*";
50 NEXT K
60 PRINT
70 NEXT J
80 PRINT
90 NEXT I
OK
RUN
*

**
*

***
**
*

****
***
**
*

OK
```

最初、行番号10で変数Iは1、行番号20でJも1、これにより行番号30でKも1となります。
行番号40を実行し、＊を1つ印字し、行番号50でKが終了値Iになっているので、行番号60を実行し改行されます。
次に行番号70でJの値が終了値I（ここでは1）になっているので、行番号80を実行し改行されます。
次に行番号90でIの値が終了値4になっていないので、実行が行番号10へもどりIの値を2として行番号20へ進みます。行番号20ではI＝1の時、このFOR NEXTループは終了しているのであらたにJ＝1となります。行番号30でも同様にJ＝1、I＝2よりK＝1となり、40行により＊を1つ印字し、50行でKが終了時2になっていないので30行にもどり、Kに増加分（ここではSTEP省略により1）1が加えられK＝2となり、40行が実行され、＊が＊のあとに1つ追加印字され、50行でKが終了値になっていることより60行が実行され改行され、70行でJが終了値2になっていないので実行が20行にうつり、Jが1増加し、2となって30行でK＝2となり、40行で＊が1つ印字され、50行でKが終了値2によっているので60行により改行され、70行でJが終了値2になっているので80行が実行され、再度改行し、90行でIが終了値4になっていないので10行へ実行がうつります。
以下、Iが4となるまで実行くりかえし、左の様な結果となります。

注）NEXTの後ろの変数を省略するほうが実行速度が上がりますが、ここではわかりやすくするため、変数をつけています。

2.2.16 NEXT

機　能　FORの終端を示します。

書　式

```
NEXT〔変数名〔,変数名,……〕〕
```

省略形　N.

解　説　NEXTの変数名は、FORの変数名と1対1に対応していなければなりません。 FORとNEXTの中にさらに別のFORとNEXTを置くことができますが、これを入れ子の構造といい、入れ子は何重でもかまいません。

```
FOR I=～
.....
FOR J=～
.....
FOR K=～
.....
NEXT
NEXT
NEXT
```

```
FOR I=～
.....
FOR J=～
.....
FOR K=～
.....
NEXT K,J,I
```

どちらも同じ構造です。

変数を省略するほうがループ速度が速くなりますが、NEXTはFOR文の数と同じ数だけ必要です。

サンプルプログラム　FOR文参照。

```
NEXT
NEXT I
NEXT C,B,A
```

2.2.17 REPEAT

機　能　REPEAT～UNTILの初めを示します。

書　式

```
REPEAT
```

省略形　REP.

解　説　REPEATは、UNTILと対にして使用し、ループの初めを示すのみです。
対応するUNTILがない場合はエラーになります。

サンプルプログラム　UNTIL文参照。

2.2.18 UNTIL

機　能

REPEATとUNTILとの間を繰返し実行します。

書　式

```
REPEAT
    ⋮
UNTIL 論理式
```

省略形

U.

解　説

このステートメントはREPEATループの端末で、論理式がまちがっていればREPEATへ戻り、正しければ次の行へ通り抜けます。
なお、REPEAT～UNTILの間にGOTO文を入れてループの外にジャンプさせることはできません。

サンプルプログラム

初期値を入力すると、ループを回るたびに1ずつ減じて行き、1より小さくなると次のPRINT文に移ります。

```
 10 INPUT"ショキチ=";A
20 REPEAT
30 PRINT A
40 A=A-1
50 UNTIL A<1
60 PRINT "オワリ"
RUN
ショキチ=? 5
 5
 4
 3
 2
 1
オワリ
OK
RUN
ショキチ=? -5
 -5
オワリ
OK
```

2.2.19 WHILE

機　能　WHILEとWENDとの間を繰返し実行します。

書　式

```
WHILE 論理式
    ⋮
WEND
```

省略形　W.

解　説　WHILEはWENDとともに用いられ、論理式が正しければWHILEとWENDとの間の実行を続け、まちがっているとき WEND の次の文へジャンプします。注意するのは、論理式が最初で判断されている点で、場合によっては、WHILEとWENDの中を一度も通らないで次にジャンプすることもあります。ここに、REPEAT～UNTILとの相違点があります。WHILEとWENDが正しく対応していない場合エラーになります。なお、WHILE～WENDの間にGOTO文を入れてループの外にジャンプさせることはできません。

サンプルプログラム　INPUT 文で入力した初期値が1 以上であれば、出力して1 減じますが、 1, より小さくなると 60行目のPRINT文にジャンプします。

```
10 INPUT"ショキチ=";A
20 WHILE A>=1
30 PRINT A
40 A=A-1
50 WEND
60 PRINT "オワリ"
RUN
ショキチ=? 5
 5
 4
 3
 2
 1
オワリ
OK
RUN
ショキチ=? -5
オワリ
OK
```

2.2.20 WEND

機　能　WHILEの終端を示します。

書　式

```
WEND
```

省略形　WE.

解　説　このステートメントはWHILEの終端として使用されるのみです。

サンプルプログラム　2.2.19 WHILE文参照。

〈ループ文のフローチャート〉

(1) FOR NEXT ループ（FOR文参照）



FOR i=m TO n STEP s
変数 iが mからnまでsずつ増えながら、NEXTとの間でループします。
mの値がnをはみ出すときにループが終了します。
ループに入る前から、mの値がnをはみ出している時は1回もループを通りません。

(2)REPEAT UNTIL ループ（UNTIL文参照）



REPEATとUNTILとの間でループします。論理式は終端の UNTIL に記述し、真の値をとるまでループします。よって、少なくとも1回はループを通ることになります。

(3) WHILE WEND ループ（WHILE文参照）



WHILEとWENDとの間でループします。論理式は入口の WHILE に記述し、真の値をとっている間ループします。よって、1回もループを通らない場合もあり得ます。

2.2.21 ON～

機　能

式の値によって、いくつかの行へジャンプします。

書　式

ON 式	GOTO	行番号1〔, 行番号2, 行番号3,……〕
	GOSUB	〃
	RETURN	〃
	RESTORE	〃
	RESUME	〃

行番号のかわりにラベル名も使えます。式の後ろには必ず空白を1つ以上置いてください。

省略形

O.

解　説

式の値（数値）がiのとき、i番目に指定した行番号の行にジャンプして、それ以降の行を実行します。式の値が1ならば1番目の行番号、2ならば2番目、3ならば3番目、……というように対応しており、式の値に対応する行番号がなければON文の次を実行します。なお式の値が整数でなければ、小数第1位で四捨五入された値になります。

〈行番号が4つあるとき〉



サンプル
プログラム

```
LIST
5 REM SAMPLE PROGRAM
10 PRINT "1:ﾀｼｻﾞﾝ"
20 PRINT "2:ﾋｷｻﾞﾝ"
30 PRINT "3:ｶｹｻﾞﾝ"
40 PRINT "4:ﾜﾘｻﾞﾝ"
50 INPUT "1-4?",A:IF A<1 GOTO 50
60 IF A>4 GOTO 50
70 ON A GOSUB 100,200,300,400
80 PRINT "ｺﾀｴ";C
90 GOTO 10
100 GOSUB 500
110 C=A+B
120 RETURN
200 GOSUB 500
210 C=A-B
220 RETURN
300 GOSUB 500
310 C=A*B
320 RETURN
400 GOSUB 500
410 IF B=0 THEN PRINT"ｺﾀｴﾊ ｱﾘﾏｾﾝ!":RETUR
N 10
420 C=A/B
430 RETURN
500 INPUT "ｶｽﾞｦ2ﾂｲﾚﾃ",A,B
510 RETURN
Ok
```

サンプルプログラムは、行番号50で入力したAの値により70行にて計算内容を選んで処理しています。

2.2.22 STOP

機　能　プログラムの実行をストップします。

書　式

```
STOP
```

省略形　S.

解　説　プログラム中にSTOPがあると、「Break in 行番号」を画面に表示し、止まります（行番号はSTOPのあった行を示しています)。プログラムの停止後にCONT（2.1.7 CONT文参照）を実行すると、止まった次の文から実行を再開します。

サンプルプログラム

```
LIST
10 PRINT"No 10"
20 STOP
30 PRINT"No 30"
OK
RUN
No 10
Break in 20
OK.
CONT
No 30
OK
```

2.2.23 END

機　能　プログラムの実行の終了を宣言します。

書　式

```
END
```

省略形　EN.

解　説　このステートメントを実行するとプログラムは終了し、すべてのファイルを閉じて、画面上にはOkが表示されて入力待ちとなります。

サンプルプログラム

```
LIST
10 PRINT"No 10"
30 PRINT"No 30"
OK
RUN
No 10
No 30
OK
20 END
LIST
10 PRINT"No 10"
20 END
30 PRINT"No 30"
OK
RUN
No 10
OK
```

サンプルプログラムは、行番号10と30の間に、行番号20としてEND文を入れた時の実行内容の相異を示しています。

2.2.24 SWAP

機　能　2つの変数の値を交換します。

書　式

```
SWAP x1 , x2
```

x_1,x_2…値を交換する変数。

省略形　SW.

解　説　x_1の値をx_2に、x_2の値をx_1に入れて、値の交換を行います。変数は数値、文字型いずれでもかまいませんが、2つの変数の型は一致していなければなりません。

サンプルプログラム

```
 10 REM SAMPLE PROGRAM
 20 A=10
 30 B=-9
 40 C$="CC"
 50 D$="DD"
 60 E$="EE"
 70 PRINT A;B
 80 PRINT C$;D$;E$
 90 SWAP A,B
100 SWAP C$,D$:SWAP D$,E$
110 PRINT A;B
120 PRINT C$;D$;E$
130 END
Ok
RUN
 10 -9
CCDDEE
-9  10
DDEECC
Ok
```

サンプルプログラムは、数値変数A、Bと文字変数C$、D$、E$に定数と文字列をあらかじめ定義しておき、AとBの交換、C$とD$の交換およびD$とE$の交換を行なった結果を表示させています。

実行結果は、

交換前　A＝10，B＝－9，C$＝"CC"，D$＝"DD"，E$＝"EE"

交換後　A＝－9，B＝10，C$＝"DD"，D$＝"EE"，E$＝"CC"

となっていることを示しています。

2.2.25 REM

機　能　プログラム文にコメントを入れます。

書　式

```
REM〔コメント〕
```

コメント……任意の文字列

省略形　'（アポストロフィ）

解　説　プログラムにコメントを挿入するためのステートメントで、プログラムの実行にはかかわりありません。REMは '（アポストロフィ）で代用できます。

サンプルプログラム

```
 10 REM Sample program
 15 REM
 20 REM SHARP/Sharp/ｼｬｰﾌﾟ
 30 REM
 40 REM  「REM」ﾊ ﾌﾟﾛｸﾞﾗﾑ ﾆ ｺﾒﾝﾄ ｦ ｲﾚﾙﾀﾒ ﾉ
 50 REM  ｽﾃｰﾄﾒﾝﾄ ﾃﾞｽ。
 60 '
 70 '       「'」=「REM」
 80 '
```

2.2.26 READ

機　能

DATA文で用意されたデータを変数に入れます。

書　式

READ x_1〔, x_2, x_3,……〕

x_1, x_2, x_3……DATA文のデータを入れる変数

省略形

REA.

解　説

DATA文で用意されたデータを読み込んで、変数に割り当てます。READ文はいつもDATA文と対にして使い、DATA文の定数データは、READ文の変数と1対1に対応し、かつ双方同じ型でなければなりません。DATA文は行番号の若い順にデータの並びの先頭から読み込まれ、DATA文の定数データの数がREAD文の変数の数より多いときには、次のREAD文により引き続き読み込まれます。
データの数が不足しているときは、Out of data エラーになります。RESTORE（2.2.28 RESTORE文参照）を使うと、DATA文を読み直したり、DATA文の読み込み順を変えることができます。

サンプルプログラム

90行の数値データを20行で読み込み、Xに入れて30行のPRINT文で出力します。続いて、100行の文字列データを60行で読み込み、A$, B$, C$に入れて70, 80行のPRINT文で出力します。

```
LIST
10 FOR I=1 TO 7
20 READ X
30 PRINT X;
40 NEXT
50 PRINT
60 READ A$,B$,C$
70 PRINT A$;" ";B$
80 PRINT A$;" ";C$
90 DATA 8,0,1.2,-4,&H1A,&B101,12
100 DATA Good,morning,evening
Ok
RUN
 8  0  1.2 -4  26  5  12
Good morning
Good evening
Ok
```

2.2.27 DATA

機　能

READで読むデータを定義します。

書　式

DATA 定数1〔, 定数2,……〕

定数……数値定数、文字列のデータ。

省略形

DA.

解　説

READ文で読み込むデータを用意するステートメントで、何かを実行するというステートメントではないので、プログラム中のどこにでも、いくつでも置くことができます。1つのDATA文によって用意できるデータは、最大255文字分の数値定数および文字列定数です。文字列をデータとして用意する場合、"(ダブルクォーテーション) で囲まなくても使用できますが,(カンマ) を区切りに使うので、カンマをデータにする場合にはダブルクォーテーションで囲まなければなりません。また、:(コロン) をデータにする場合にもダブルクォーテーションで囲む必要があります。

なお、現在、読みこみ中のDATA文の行番号を知りたい時は、システム変数DTL (3.6.3 DTL文参照) を使います。

サンプルプログラム

```
1000 DATA 23,43,55,65,12,54,34,54,99
1010 DATA 0,33,67,93,10,20,55,77,53,45

1500 DATA Suzuki,Tanaka,Abe,Yamada,Kubo
1510 DATA Nomura,Chiba,Uchida,Kimura

2000 DATA "A,A","A,B","A,C","B,A","B,B"
2010 DATA "B,C","C,A","C,B","C,C"
```

1000と1010行は数値データ、
1500と1510行は文字列データ、
2000と2010行は文字列データにカンマが入った場合
をそれぞれ示しています。
なお、数値データと文字列データを合わせて、定数データと呼ぶことがあります。

2.2.28 RESTORE

機　能

READ文で読み始めるDATA文を指定します。

書　式

RESTORE { 行番号 / "ラベル名" }

行番号……読み始めるDATA文の行番号。
"ラベル名" ……LABELで指定した文字列。

省略形

RES.

解　説

このステートメント実行後のREAD文は、行番号、"ラベル名" で指定したDATA文のデータから読み込みを開始し、RESTOREのみのときは、プログラム中最初に現れるDATA文から読み込みを開始します。

サンプルプログラム

20行で1020行からデータを読み込むよう指定し、80行で "データ" というラベル名のところからデータを読み込むよう指定しています。

```
LIST
10 REM
20 RESTORE 1020
30 FOR I=1 TO 5
40 READ A
50 PRINT A;
60 NEXT
70 PRINT
80 RESTORE "ﾃﾞｰﾀ"
90 FOR I=1 TO 5
100 READ A
110 PRINT A;
120 NEXT
1000 '
1010 LABEL "ﾃﾞｰﾀ":DATA 23,43,55,65,12
1020 DATA 12,56,34,78,56
OK
RUN
 12  56  34  78  56
 23  43  55  65  12
OK
```

2.2.29 DEF FN

機　能

ユーザーが関数を定義するのに使います。

書　式

DEF FN関数名 (x_1〔,x_2……〕)＝式

関数名…FNに続けて書きます。数値型、文字型のどちらでも可能です。

x_1, x_2…パラメータ。

省略形

なし。

解　説

ユーザーの作った関数を自由に定義することができ、プログラム中で使うことができます。関数名と定義式の型が一致していれば、文字列関数も定義できます。

サンプルプログラム

3つの関数

$$FNA\ (X, Y, Z) = \frac{X+Y+Z}{3}$$

$$FNB\ (X, Y, Z) = \sqrt[3]{XYZ}$$

$$FNC\ (X, Y, Z) = \frac{XYZ}{3}$$

を定義して、X, Y, Zに任意の値を代入して関数の値を計算してみましょう。

```
LIST
10 DEF FNA(X,Y,Z)=(X+Y+Z)/3
20 DEF FNB(X,Y,Z)=(X*Y*Z)^(1/3)
30 DEF FNC(X,Y,Z)=(X*Y*Z)/3
40 INPUT"D=";D:INPUT"E=";E:INPUT"F=";F
50 PRINT "FNA=";FNA(D,E,F),"FNB=";FNB(D,E,F),"FNC=";FNC(D,E,
60 PRINT
70 PRINTX,Y,Z
OK
RUN
D=? 1
E=? 2
F=? 3
FNA= 2      FNB= 1.8171206        FNC= 2

 0            0            0
OK
```

X, Y, Zの各変数は
影響を受けません。

2.2.30 DEF USR

機　能　ユーザーが機械語サブルーチンを定義するのに使います。

書　式

```
DEF USR[番号]＝アドレス
```

番号……機械語サブルーチンの番号。0～9の整数で、省略すると0。
アドレス…機械語サブルーチンの先頭のアドレス。

省略形　なし。

解　説　ユーザーが機械語で作った機械語サブーチンを呼び出すUSR関数に番号をつけその実行開始の先頭アドレスを設定します。USR関数は、0～9まで登録テーブルをBASIC内部に持っていて、最大10まで関数を定義することができます。いくつかのUSR関数の識別にこの0～9の番号を使いますが、同じ番号のUSR関数を何度でも設定しなおすことができます。

機械語サブルーチンの使い方については付録「A.3.2 USR命令の使い方」を参照してください。

サンプルプログラム

```
LIST
10 CLEAR &HFE00
20 POKE &HFE00,&HC9
30 DEF USR1=&HFE00
40 PRINT USR1(10)
OK
RUN
 10
OK
```

サンプルプログラムは、機械語のエリアとして&HFE00以上とし、&HFE00のアドレスに機械語でRETURNを示す&HC9を入れ、ユーザーの機械語ルーチンUSR1として&HFE00からを指定し、40行でこのUSR1を呼びだし何もせずBASICへもどし、引数の値を表示させています。

2.2.31 CALL

機　能　機械語サブルーチンを直接呼び出します。

書　式

```
CALL アドレス
```

アドレス…サブルーチンの実行開始番地

省略形　CA.

解　説　アドレスを実行開始番地とする機械語サブルーチンを呼び出します。
アドレスには16進定数（&HXXXXの形）、10進整定数（—32768～65536）などの整数定数、整数変数、式が指定されます。CALL命令はUSR命令と異なり、引数はもてません。

機械語からBASICへ戻るには機械語命令のRET（&HC9）を実行させなければなりません。
機械語サブルーチンは、CLEARまたはLIMIT文で指定したアドレス以降に置いてください。



サンプルプログラム

```
LIST
10 CLEAR &HFE00
20 POKE &HFE00,&HC9
30 CALL &HFE00
OK
RUN
OK
```

2.2.32 POKE

機　能　メインメモリー内の指定アドレスにデータを書き込みます。(⇐⇒PEEK)

書　式

```
POKE アドレス, データ[,データ,データ,…]
```

アドレス…データの書き込みアドレス。0～&HFFFF。
データ……0～255までの整数

省略形　PO.

解　説　メインメモリー内の指定したアドレスに1バイト（8ビット）のデータを書き込みます。
データを,（カンマ）で区切って続けて書くと、連続したアドレスに書き込むことができます。
POKEの使用に際しては、BASICプログラムなどメインメモリー内のシステムを破壊してしまう恐れがあるので十分注意してください。

メインメモリーからデータを読みこむ時は、PEEK関数（3.3.2 PEEK参照）を使います。
逆に、V-RAM（ビデオRAM）に対してデータの書きこみ、読み出しを行なう時には、それぞれPOKE@, PEEK@（2.5.19, 3.3.3各命令参照）を使います。

サンプルプログラム

```
10 CLEAR &HBFFF:DT=0
20 FOR AD=&HC000 TO &HC01F
30 POKE AD,DT                ｲﾒﾓﾘｰ ﾍﾉ ｶｷｺﾐ。
40 DT=DT+1
50 NEXT
60 FOR AD=&HC000 TO &HC01F
70 DT=PEEK(AD)               ｲﾒﾓﾘｰ ｶﾗﾉ ﾖﾐｺﾐ。
80 PRINT " ";HEX$(DT);
90 NEXT
```

このプログラムを実行するとメモリー内の&HC000～&HC01Fの各番地に0～31の数値データを書き込み、実際に書き込まれたかどうか調べるため読み込んで、16進で表示します。

2.2.33 OUT

機　能　I/Oポートに1バイトのデータを送ります。(⇒2.5.19 POKE@)

書　式

```
OUT ポートアドレス, 出力データ
```

ポートアドレス…出力するポートのアドレス 0～&HFFFF
出力データ……0～255の整数

省略形　OU.

解　説　出力データをポートアドレスで指定したポートに出力します。

テキスト画面とその属性ポートへのアクセスの仕方については、巻末の「テキスト画面とその属性ポートへのアクセス方法」を参照してください。
I/Oポートから、データを読みこむ場合は、INP関数（3.3.1 INP参照）を使います。

サンプルプログラム　画面の左上隅（&H3000番地）から256文字表示し、そのおのおのに点滅、反転、色指定の各属性をつけるプログラムを下に示します。

```
10 SCREEN0,0,0:WIDTH 40
20 FOR I=0 TO 255
30  OUT &H3000+I,I
35  OUT &H2000+I,(I MOD 32)
40 NEXT
50 LOCATE 0,23
```

2.2.34 RANDOMIZE

機　能　RND関数で発生させる乱数の系列を設定します。

書　式

```
RANDOMIZE〔x〕
```

x…乱数系列を発生するための初期値。－32767～32767。

省略形　RA.

解　説　RND関数はプログラムを実行するたびに同じ系列の乱数を発生しますが、このステートメントを与えることによって、乱数の系列を変更します。

式は-32767～32767までの数値になるように指定しますが、省略した場合は、BASICがランダムに乱数系列を設定します。

サンプルプログラム

```
LIST
10 PRINT RND(1)
20 PRINT RND
30 RANDOMIZE 10
40 PRINT RND
50 RANDOMIZE 10
60 PRINT RND
OK
RUN
 .81971014
 .405249
 .50813067
 .50813067
OK
```

（注）この値は乱数ですので サンプル通りでるとは限りません。

2.3 ファイル処理ステートメント

2.3.1 MAXFILES

機　能　同時にOPENできるファイル番号の最大値を指定します。

書　式

```
MAXFILES n
```

n…ファイル番号の最大値 0 ～15。

省略形　MA.

解　説　同時にファイル番号の最大値nまでのファイルをOPENできるように設定します。このとき、変数エリアはクリアされます。

BASIC起動時にファイル番号の最大値は 1 に指定されます。

サンプルプログラム

```
MAXFILES 3
Ok
```

2.3.2 OPEN

機　能　ファイルを開きプログラムで使用できるようにします。

書　式

```
OPEN "{O|I}",〔#〕ファイル番号,〔"〔ファイルディスクリプタ:〕ファイル名"〕
```

ファイル番号…1 からMAXFILESで設定した数まで。

省略形　OPE.

解　説　ファイルディスクリプタで指定したファイル名をつけて、ファイル番号で指定した番号で設定し、Oのときはアウトプットモード、Iのときはインプットモードにします。
以後このファイルは、ファイル番号を使ってアクセスします。
O…シーケンシャルファイルのアウトプットオープン（ファイルにデータを書き込む）
I…　〃　インプットオープン　（ファイルからデータを読み込む）

サンプルプログラム

```
10 REM SAMPLE PROGRAM
20 OPTION SCREEN 2              'ｸﾞﾗﾌｨｯｸﾒﾓﾘﾖ ﾌｧｲﾙﾆ ﾂｶｴﾙﾖｳﾆ ｽﾙ
30 INIT"MEM:"                   'ｼｮｷｶ ｽﾙ
40 OPEN "O",#1,"MEM:TEST"       'ｶｷｺﾐﾌｧｲﾙﾖ ﾋﾗｸ
50 FOR I=0 TO 255
60 PRINT #1,I;                  'ﾌｧｲﾙﾆ ｶｷｺﾑ
70 NEXT
80 CLOSE                        'ﾌｧｲﾙﾖ ﾄｼﾞﾙ
90 OPEN "I",#1,"MEM:TEST"       'ﾖﾐｺﾐﾌｧｲﾙﾖ ﾋﾗｸ
100 IF EOF(1)=-1 THEN 140       'ﾃﾞｰﾀｶﾞｵﾜｯﾀｶ ﾄﾞｳｶ ｼﾗﾍﾞﾙ
110 INPUT#1,A                   'ﾌｧｲﾙｶﾗ ﾖﾐｺﾑ
120 PRINTA;
130 GOTO 100
140 CLOSE                       'ﾌｧｲﾙﾖ ﾄｼﾞﾙ
150 END
```

2.3.3 CLOSE

機　能　ファイルを閉じます。

書　式

```
CLOSE〔[#]ファイル番号　,[#]ファイル番号　,…〕
```

ファイル番号…OPENで指定したファイル番号。

省略形　CLO.

解　説　OPENによって開かれたファイルを閉じます。CLOSEで閉じたファイルはOPENで再指定しないかぎり、使用できません。

ファイル番号を省略するとすべてのファイルが一斉に閉じられます。

なお、NEW、RUN、LOADのときもすべてのファイルは閉じられます。

サンプルプログラム　2.3.2 OPEN文参照。

```
CLOSE          ← 全てのファイルを閉じます。
CLOSE #1,#3    ← 1と3のファイルのみ閉じます。
```

2.3.4(1) PRINT#

機　能　シーケンシャルファイルにデータを記録します。

書　式

```
PRINT#ファイル番号[,x1,x2,…]
```

ファイル番号…OPENで指定したファイル番号

x_1, x_2,……出力する式または文字列

省略形　?# またはP.#

解　説　x_1, x_2,…で指定した式または文字列をファイル番号で指定したファイルへ出力して記録します。このステートメントを実行するためには、OPEN（2.3.2 OPEN文参照）ですでにファイルを開いておかねばなりません。なお、一連の処理が終了したら必ずCLOSE(2.3.3 CLOSE文参照)してください。

出力する最初の式または文字列の前に，(カンマ) を忘れないように注意してください。

サンプルプログラム　2.3.2 OPEN文参照。

```
PRINT#1,A,B,C  ← 数値変数AとBとCの値をファイルへ出力します。
```

2.3.4(2) PRINT#0

機　能　CSIZEで指定した文字サイズで表示します。およびコントロールキャラクタを表示します。

書　式

```
PRINT#0　〔,x1,x2,…〕
```

x_1, x_2……表示する式または文字列。

省略形　?#0またはP.#0

解　説　PRINT#0は、次の2つの機能があります。

- CSIZEで指定した文字サイズで表示します。
- PRINT命令の機能（式の値、文字列、変数・定数の値を表示する）に加えて、コントロールコードをコントロールキャラクタに変えて表示します。

2.3.5 WRITE＃

機　能　データをシーケンシャルファイルに記録します。

書　式

```
WRITE＃ファイル番号[,x1,x2,…]
```

ファイル番号…OPENで指定したファイル番号
x_1, x_2,……出力する式または文字列

省略形　WR.＃

解　説　ファイル番号で指定したシーケンシャルファイルに、x_1, x_2,…で指定した式または文字列を記録します。WRITE＃はPRINT＃と異なり、データの区切りにカンマ（,）を入れ、文字列データのときはダブルクォーテーション（"）をつけて、詰めて出力するので、ファイルの使用領域の節約になります。

サンプルプログラム

```
WRITE#1,A,B,C
WRITE#0,X;"X"
```

← 数値変数AとBとCの値をシーケンシャルファイルへ記録します。
← 数値変数Xの値とダブルクォーテーションでかこまれた文字Xとを間に，(カンマ)を入れて表示します。

2.3.6 INPUT＃

機　能　シーケンシャルファイルからデータを変数に読み込みます。

書　式

```
INPUT＃ファイル番号,x1[,x2,…]
```

ファイル番号…OPENで指定したファイル番号
x_1, x_2,……入力したデータを入れる変数

省略形　I.＃

解　説　ファイル番号で指定したファイルからシーケンシャル（記録されている順）に読み込んで変数x_1，x_2，…へ入れます。このステートメントを実行するためにはOPEN（2.3.2 OPEN文参照）で、すでにファイルを開いておく必要があります。なお、一連の処理が終了したら必ずCLOSE（2.3.3 CLOSE文参照）してください。

サンプルプログラム

```
INPUT#1,A,B,C
```

← シーケンシャルファイルよりA，B，Cの順に入力します。

2.3.7 LINPUT＃・LINE INPUT＃

機　能　シーケンシャルファイルから1行分（255文字まで）のデータを読み込みます。

書　式

```
LINPUT＃ファイル番号，文字変数
LINE INPUT＃ファイル番号，文字変数
```

ファイル番号…OPENで指定したファイル番号
文字変数……入力したデータを入れる文字型変数

省略形　LI.＃　LINEI.＃

解　説　ファイル番号で指定したシーケンシャルファイルから255文字までのデータを読み込んで文字変数に入れます。

サンプルプログラム

```
LINPUT#1,A$          ← シーケンシャルファイルのファイル番号1のファイルよりA$へ
                       255文字読み込みます。
LINE INPUT#1,A$      ← シーケンシャルファイルのファイル番号1のファイルよりA$へ
                       255文字読み込みます。
```

2.3.8 DEVI$

機　能　指定された入力デバイスから1レコードのデータを読み込みます。

書　式

```
DEVI$ "ファイルディスクリプタ：" ,レコード番号，文字変数，文字変数
```

ファイルディスクリプタ…CAS：　カセットテープ
（入力デバイス名）　MEM：　グラフィックメモリー
EMM0：外部メモリー0
〜　〜
EMM9：外部メモリー9
レコード番号……各入力デバイスの先頭からのレコード番号
1レコード＝256バイト

省略形　なし

解　説　ファイルディスクリプタで指定した入力デバイス内のレコード番号で指定したレコードから、256バイト分の文字列データを読み込み128バイトずつ2つの文字変数に入れます。

サンプルプログラム

```
DEVI$"EMM0:",1,A$,B$     ← 外部メモリーよりA$、B$にレコード番号1のレコ
                           ードから256バイト文字列データを読み込みます。
```

（上の例では、オプションの320KB外部メモリー(外部メモリー)が必要です。）

2.3.9 DEVO$

機　能　1レコード分の文字列データを指定された出力デバイスに記録します。

書　式

```
DEVO$ "ファイルディスクリプタ:", レコード番号, 文字式, 文字式
```

ファイルディスクリプタ…CAS:　カセットテープ
MEM:　グラフィックメモリー
EMM0:　外部メモリー0
〜　〜
EMM9:　外部メモリー9
レコード番号……データを記録するレコードの番号。

省略形　DEVO.

解　説　ファイルディスクリプタで指定した出力デバイス内の、レコード番号で指定したレコードに、文字式の文字列データを256文字分記録します。このとき1つの文字式には128バイトの文字が入っている必要があります。

サンプルプログラム

```
DEVO$"EMM0:",1,A$,B$
```

← 外部メモリーのレコード番号1のレコードにA$、B$の文字合計256文字出力します。

(上の例では、オプションの320KB外部メモリー(外部メモリー)が必要です。)

2.3.10 INIT

機　能　外部デバイスの初期化を行います。

書　式

```
INIT〔"ファイルディスクリプタ:"〕
```

ファイルディスクリプタ…外部デバイスを指定する（第4章 4.3 ファイルディスクリプタ参照)。

省略形　なし。

解　説　指定した外部デバイスの初期化（イニシャライズ）を行います。

ファイルディスクリプタ	処理内容
CRT:, SCR:	COLOR7,0:CGEN:CFLASH:CREV:CSIZE :PALET:PRW:WINDOW:CONSOLE:SCREEN 0,0,0
KEY:, LPT:	Bad file modeエラーが出ます。
CAS:	カセットの巻き戻しと先頭からの消去をします。
EMM0:〜EMM9:, MEM:	フォーマットの管理テーブルを初期化します。(外部メモリー内のプログラム及びファイルは使用不可能となります。)

"ファイルディスクリプタ" を省略すると、スクリーン（CRT:,SCR:）の初期化が行われます。

ダイレクトモードでINITを実行すると、「Are you sure?(yorn)」と表示されますので、実行するときはYキーを押し、実行を中止するときはそれ以外のキーを押してください。

プログラム中でINITを実行しても「Are you sure?(yorn)」と聞いて来ないで、直接実行してしまいますので注意してください。

サンプルプログラム

```
INIT"CRT:"
Are you sure ? (y or n)Y
Ok
```

2.4 エラー処理ステートメント

2.4.1 ON ERROR GOTO

機　能　エラーが発生したとき指定行番号からのエラー処理ルーチンに制御を移します。

書　式

```
ON ERROR GOTO { 行番号
              { "ラベル名"
```

行 番 号 …エラーが発生したとき実行させるエラー処理ルーチンの開始行番号。
"ラベル名"…　〃　ラベル名。

省 略 形　O. ERR. G.

解　説　このステートメントを前もって実行していると、エラーが発生したときに、行番号で指定した行からのエラー処理ルーチン（ユーザーが設定）へ制御を移し、処理ルーチンの出口のRESUME（2.4.2 RESUME文参照）までの実行を行います。

これによって、ほとんどのエラーに対し、BASICレベルで対策可能になります。
エラー処理ルーチンの中でエラーが起きた場合は、エラーを出力してコマンド待ちにもどります。

またエラー処理ルーチンを実行中、プログラムの最後に達した場合は、NO RESUMEエラーが発生します。END文で終った場合は、エラーが発生しないで、コマンド待ちにもどります。
普通このステートメントは、プログラムの先頭で実行しておきます。

なお、エラー処理ルーチン内でON ERROR GOTO 0 とすると通常のエラーが発生し、プログラムの実行がとまります。

サンプルプログラム

```
LIST
10 ON ERROR GOTO 1000
20 INPUT"A,B=";A,B
30 PRINT A/B
40 GOTO 20
1000 IF ERR<>11 THEN ON ERROR GOTO 0
1010 PRINT"ｹｲｻﾝ ﾃﾞｷﾏｾﾝ !!!";CHR$(7)
1020 RESUME NEXT
OK
RUN
A,B=10,2
 5
A,B=10,0
ｹｲｻﾝ ﾃﾞｷﾏｾﾝ !!!
A,B=10,3
 3.3333333
A,B=5E40,1          ← Aに5×10^40を入力したので、オーバーフローエラー(エラーコード6)になって、行番号1000でON ERROR GOTO 0が実行されました。そのため、通常のエラーが出力されています。
Overflow in 20
OK
```

サンプルプログラムは、A÷Bの計算をさせています。
B＝0の時、ケイサンデキマセン!!と表示します。

2.4.2 RESUME

機　能

エラールーチン終了後、プログラムの実行を再開します。

書　式

```
RESUME { { 行番号 } }
       { {"ラベル名"} }
       { { NEXT   } }
```

行番号、"ラベル名"…エラー処理ルーチン終了後復帰する行番号・ラベル

省略形

RESU.

解　説

ON ERROR GOTOで飛び込んだエラー処理ルーチンから戻る場合は、必ずRESUMEによらなければなりません。

RESUMEのみならば、エラーの発生した命令から、

RESUME NEXTならば、エラーの発生した次の命令から、

RESUME 行番号
　〃　"ラベル名"　}ならば、その行番号またはラベルから、

実行を再開します。

もし、RESUMEではなく、GOTOなどで、もとのプログラムに実行を移した場合は、2回目のエラーが発生したときに、エラー処理ができなくなります。

サンプルプログラム

```
RESUME "モトノ ルーチン"
```

2.4.3 ERROR

機　能

エラーを故意に発生させます。

書　式

```
ERROR x
```

x…BASICで定義されているエラーコード

省略形

ERR.

解　説

エラーコードで指定したエラーが発生し、ERR関数（3.6.5 ERR参照）にはエラーコードが、ERL変数（3.6.4 ERL参照）にはそのステートメントの行番号が入ります。これによって、エラー処理ルーチンが正常に作動するか確めることが可能です。エラー処理ルーチンのデバッグが終了したら、プログラム中より取りのぞいて下さい。

サンプルプログラム

```
ERROR 3
RETURN without GOSUB
Ok
ERROR 15
String too long
Ok
```

サンプルプログラムは、直接実行でERROR 3 およびERROR15を入力した時の結果です。

2.5 画面制御ステートメント

2.5.1 WIDTH

機　能　画面サイズを設定します。

書　式

```
WIDTH  x
```

x……一行当たりのカラム数
（電源投入時は40カラム25ラインに設定されます。）

省略形　WI.

解　説　画面のサイズを設定するときに使います。
一行当たりのカラム数を40に指定すると、画面のサイズは40カラム25ラインに設定され、80に指定すると、80カラム25ラインに設定されます。
このステートメントを実行すると、WINDOW（2.5.11 WINDOW文参照）とCONSOLE（2.5.3 CONSOLE文参照）を解除し、CLS4（2.5.12 CLS文参照）と同じ処理をします。
なお、一行当たりのカラム数に0～39を指定すると、40のときと同じ画面サイズに、41～255を指定すると、80のときと同じサイズに設定されます。

サンプルプログラム

```
WIDTH 40
WIDTH 80
```



2.5.2 SCREEN・GRAPH

機　能　グラフィック画面の使用モードを設定します。

書　式

```
SCREEN〔〔出力ページ〕〔,〔入力ページ〕〔,グラフィックモード〕〕〕
GRAPH〔〔出力ページ〕〔,〔入力ページ〕〔,グラフィックモード〕〕〕
```

出力ページ……0，1（0のときページ0、1のときページ1を指定します。）
入力ページ……0，1（　〃　）
グラフィックモード…0，1，2，3

省略形　SC.
GR.

解　説　SCREENとGRAPHは同じ命令で、出力・入力のページ指定とグラフィックモードを指定します。
出力・入力のページとはWIDTH40の時のみ有効で、2ページのうち、どちらを表示してどちらに書き込むか指定するものです。入力ページというのは、画面制御命令やPRINT、INPUT命令等が実行されるページであり、出力ページとは、実際画面に出されているページのことをいいます。
また、グラフィックモードとはG1、G2、G3の3枚のグラフィック画面の使用モードで、この命令実行以後のグラフィック命令は次に示す画面に対してのみ有効となります。
グラフィックモード0…指定のパレット番号でドットをセット・リセットできます。
〃　1…グラフィック1のみドットをセット・リセットできます。
〃　2…グラフィック2のみドットをセット・リセットできます。
〃　3…グラフィック3のみドットをセット・リセットできます。
この命令はPALETおよびPRWも制御しており、すべてのパラメーターを省略した場合は、グラフィックが見えなくなり(LIST時も同じ)、パラメーターを指定した時には、省略した場合やLIST等で見えなくなる前のPALET、PRWにもどします。カーソルが見えなくなってこまった時は、CTRL ＋ D キーを押してください。

サンプルプログラム

```
SCREEN 0,0,0
SCREEN 0,1,0
SCREEN 0,1,1
```

← 入力を0ページ、出力も0ページ、グラフィックモードは指定パレットとする。
← 入力を0ページ、出力を1ページ、グラフィックモードは指定パレットとする。
← 入力を0ページ、出力を1ページ、グラフィックモードはG1のみとする。

2.5.3 CONSOLE

機　能　画面内での文字の表示エリアを設定します。

書　式

```
CONSOLE  Ys, Yℓ[, Xs, Xℓ]
```

Y_s……垂直方向の表示開始ライン　　0～24
Y_ℓ……垂直方向の表示ライン数　　1～25(正確には1～25-Y_s)
X_s……水平方向の表示開始カラム
　　WIDTH 40のとき　　0～39
　　WIDTH 80のとき　　0～79
X_ℓ……水平方向の表示カラム数
　　WIDTH 40のとき　　1～40(正確には1～40-X_s)
　　WIDTH 80のとき　　1～80(正確には1～80-X_s)

省略形　CONS.

解　説　テキスト画面に対して、文字の表示エリアを設定します。
このステートメントの実行後は、指定した長方形のエリア内だけに文字を表示することができ、テキスト画面のクリアやスクロールもこのエリア内で行われます。設定後カーソルは(X_s、Y_s)の位置へ移動します。
CONSOLEの後を省略すると、最大文字数表示する画面に戻ります。

サンプルプログラム

```
LIST
10 CLS
20 CONSOLE 0,10,0,40
30 FOR X=1 TO 500
40  PRINT X;
50 NEXT
60 CONSOLE 0,25,0,20
70 FOR X=1 TO 500
80  PRINT X;
90 NEXT
Ok
```

20行目 ← テキスト画面の表示エリアを始点座標(0、0)、終点座標(39、9)の長方形に設定する。
30～50行目 ← 水平方向に連続して1から500までの数を表示。
60行目 ← テキスト画面の表示エリアを始点座標(0、0)、終点座標(19、24)の長方形に設定する。





このプログラムを実行すればCONSOLEの設定の違いが分かります。

CTRL ＋ D キーを押せば全エリア表示にもどります。

注）●CONSOLE命令は垂直方向表示開始ライン、垂直方向表示ライン数、水平方向表示開始カラム、水平方向表示カラム数を設定する命令であり、表示エリアの始めと終りの座標を設定するのではないことに注意ください。
●グラフィック画面の表示エリア設定は、WINDOW命令によって行ないます。
(2.5.11 WINDOW参照)

2.5.4 LOCATE・CURSOR

機　能　カーソルを指定位置へ移動します。

書　式

```
LOCATE  x, y
```

```
CURSOR  x, y
```

x, y……CONSOLE命令で指定された表示エリア内の値。
　x……水平方向の表示カラム　　WIDTH 40のとき　0～39
　　　　　　　　　　　　　　WIDTH 80のとき　0～79
　y……垂直方向の表示ライン　　WIDTHに関係なく　0～24

省略形　LOC.
CU.

解　説　LOCATEとCURSORは同じ命令で、テキスト画面のカーソル位置を指定します。範囲はCONSOLEで指定された表示エリア内で、それを外れた場合はエラーになります。

サンプルプログラム

```
LIST
10 FOR I=0 TO 39
20 FOR J=0 TO 24
30   LOCATE I,J:PRINT"*";
40 NEXT
50 NEXT
OK
```

サンプルプログラムは、画面左上端から右下端まで、垂直方向に文字＊を連続表示させるプログラムです。

WIDTH 40



WIDTH 80



2.5.5 COLOR

機　能　画面の色を設定します。

書　式

```
COLOR〔表示色〕〔,背景色〕
```

表示色…文字の表示色を下のような数字で指定します。

0　黒
1　青
2　赤
3　マゼンタ
4　緑
5　シアン
6　黄
7　白

これをカラーコードといい、省略すると表示色は変化しません。

背景色…表示色と同様にカラーコードで指定し、省略すると背景色は変化しません。

省略形　COL.

解　説　文字の色と背景の色を指定する命令です。

電源投入時は、自動的にCOLOR 7 , 0 にセットされています。グラフィック画面において、パレットコード（2.5.13 PALET文参照）を省略すると、COLOR文の表示色のコードの値が、パレットコードの値として使用されます。

なお、表示色と背景色はCTRLキーと数字のキーを同時に押しても設定することができます。(「コントロールコード」参照)。

サンプルプログラム

```
COLOR 2

COLOR 6

COLOR 3,0

COLOR 4,0
```

2.5.6 CREV

機　能　文字の表示モードの切り換えをします。

書　式

```
CREV〔文字表示モード〕
```

文字表示モード…0 , 1

省略及び0のとき　ノーマルモード(標準モード)

1のとき　リバースモード(反転モード)

省略形　CR.

解　説　文字の表示モードを反転モードにしたり標準モードにしたりするためのステートメントです。

このステートメントの実行以降、文字は指定された表示モードで画面に表示されます。

ノーマルモード　A , B , C , … , 1 , 2 , 3 , … , _（ブランク）

反転モード　Ⓐ , Ⓑ , Ⓒ , … , ① , ② , ③ , … , ▒

サンプルプログラム

CFLASH参照。

2.5.7 CFLASH

機　能

文字の表示モードの切り換えをします。

書　式

CFLASH〔文字表示モード〕

文字表示モード…0，1
省略及び0のとき　ノーマルモード　　(標準モード)
　　1のとき　フラッシングモード(点滅モード)

省略形

CF.

解　説

文字の表示モードをフラッシングモードにしたり、ノーマルモードにしたりするためのステートメントです。

このステートメントの実行以降、文字は指定された表示モードで画面に表示されます。ノーマルモードはCREV（2.5.6 CREV文参照）と同様です。フラッシングモードとは、CREVのノーマルモードと反転モードとを交互に繰返すモードです。このとき、切り変わりの周期は、画面上のカーソルの点滅の周期と同じになります。

サンプルプログラム

```
CFLASH 1
OK                    ←┐
CFLASH 0              ←┘── 画面上では点滅しています。
OK
```

```
10 REM SAMPLE PROGRAM
20 INIT:WIDTH 40
30 LOCATE 5,5
40 CREV 0:CFLASH 0
50 PRINT"NORMAL           CHARACTER";
60 LOCATE 5,7
70 CREV 1:CFLASH 0
80 PRINT"REVERSE          CHARACTER";
90 LOCATE 5,9
100 CREV 0:CFLASH 1
110 PRINT"FLASH            CHARACTER";
120 LOCATE 5,11
130 CREV 1:CFLASH 1
140 PRINT"REVERSE & FLASH CHARACTER";
150 CREV:CFLASH
160 END
```

サンプルプログラムは、ノーマル、リバース、フラッシュ、リバースフラッシュで文字を表示します。

2.5.8 CGEN

機　能　文字の表示モードの切り換えをします。

書　式

```
CGEN〔文字表示モード〕
```

文字表示モード…0，1
省略及び0のとき　ノーマルモード　　(ROMCGモード)
　　1のとき　ユーザー定義キャラクタモード(RAMCGモード)

省略形　CG.

解　説　文字の表示モードをROMCG(ROMキャラクタジェネレータ)モードにしたり、RAMCG(RAMキャラクタジェネレータ)モードにしたりするためのステートメントです。
RAMCGモードにすると、それ以後表示される文字はすべて、ユーザー定義のキャラクタとなります。
BASIC起動時は何も定義されていないので、DEFCHR$ 命令で定義してから使用してください。

サンプルプログラム

```
CGEN 1  ← ユーザー定義キャラクタモードとなります。(もとにもどすには、CTRL
          +D を押してください。)
CGEN 0  ← ROMキャラクタジェネレータモードとなります。
```

2.5.9 CSIZE

機　能　文字の大きさを変えます。

書　式

```
CSIZE〔n〕
```

n……0，1，2，3
省略及び0　ノーマル文字
　　1　垂直2倍文字
　　2　水平2倍文字
　　3　垂直2倍・水平2倍文字

省略形　CS.

解　説　PRINT＃0 命令で表示する文字の大きさを指定します。ノーマル以外を指定した場合、PRINT＃0 命令を実行する時、次の注意が必要です。
垂直2倍文字…その二行中に、ノーマル及び水平2倍文字があってはなりません。また他の文字とは垂直座標の行間隔を偶数行あけて(0を含む)表示させてください。
水平2倍文字…必ず偶数のX座標から出力すること。一行中にノーマル文字があってもかまいません。
垂直2倍・水平2倍文字…その二行中にノーマル及び水平2倍文字があってはなりませんが、垂直2倍文字はあってもかまいません。必ず偶数のX座標から出力してください。また、他の文字とは垂直座標の行間隔を偶数行あけて(0を含む)表示させるようにしてください。
LOCATE及びCURSORによる座標指定は、水平2倍　垂直2倍　水平2倍・垂直2倍　のようになります。(斜線部)

サンプルプログラム

```
100 A$="ABCDEFG":CLS
110 CSIZE 0
120 LOCATE 0,0:PRINT#0 A$   } ← ノーマル文字
130 CSIZE 1
140 LOCATE 0,1:RRINT#0 A$   } ← 垂直2倍文字
150 CSIZE 2
160 LOCATE 0,3:PRINT#0 A$   } ← 水平2倍文字
170 CSIZE 3
180 LOCATE 0,4:PRINT#0 A$   } ← 垂直、水平2倍文字
190 CSIZE 0                 ←―― ノーマル文字にもどす。
200 END
```

注) CSIZE 2 及びCSIZE 3 の使用中PRINT TAB(n)等のTAB関数による表示位置の移動が、従来の動作と異なってきますのでご注意ください。

2.5.10 DEF CHR$

機　能

ユーザー定義のキャラクタジェネレータにキャラクタパターンを定義します。

書　式

DEF␣CHR$(n)＝文字式

n ……0～255 キャラクタコード。

文字式…24文字のパターンデータ。

省略形

DEFCHR.

解　説

ユーザーが定義できるキャラクタは、通常のキャラクタ(文字)と同様に横方向に8ドット、縦方向に8ドットの長さをもつタイル型のドットパターンです。このキャラクタを定義するためには、8×3＝24文字の文字列(B.R.G.に対応)が必要で、1文字で横8ドット、縦1ドットのパターンを指定し、8文字で縦横8ドットの長さのキャラクタを指定します。すなわち、1文字に対応するキャラクタコードのビットパターンが、横8ドットのドットパターンを表します。

文字列の最初の8文字は青、次の8文字は赤、最後の8文字は緑のキャラクタパターンを示します。

ユーザーの定義できるキャラクタの数は256個までで、キャラクタコードnに対応しています。表示するには、前もってCGEN 1 (2.5.8 CGEN文参照)を実行し、PRINT文を使います。

サンプルプログラム

キャラクタとして斜線を定義して、8つのカラーで表示してみましょう。

HEXCHR$ ("01 02 04 08 10 20 40 80") のパターンは下の様になります。



HEXCHR＄（"0000000000000000"）は、全部ドットが消えている状態で、下の様になります。



……（2）

（1）のパターンをA＄、（2）のパターンをN＄として、これらを組み合せてCHR＄(10)～CHR＄(17)の8色の斜線パターンを定義します。

たとえば、CHR＄(15)はA＄＋N＄＋A＄として



10～17はコントロールコードなので、PRINT＃0を使って表示します。

```
10 SCREEN 0,0,0:CGEN 1
20 A$=HEXCHR$("0102040810204080")
30 N$=HEXCHR$("0000000000000000")
40 '
50 '                  ｱｵ ｱｶ ﾐﾄﾞﾘ
60 '                  |  |  |
69 DEF CHR$(10)=N$+N$+N$
70 DEF CHR$(11)=A$+N$+N$
80 DEF CHR$(12)=N$+A$+N$
90 DEF CHR$(13)=A$+A$+N$
95 DEF CHR$(14)=N$+N$+A$
96 DEF CHR$(15)=A$+N$+A$
97 DEF CHR$(16)=N$+A$+A$
98 DEF CHR$(17)=A$+A$+A$
110 '
130 FOR I=10 TO 17
140 PRINT#0,CHR$(I);
150 NEXT
200 CGEN 0:END
```

2.5.11 WINDOW

機　能　グラフィック画面の表示エリアを設定します。

書　式

```
WINDOW (Xs,Ys)－(Xe,Ye) [,(X1,Y1)－(X2,Y2)]
```

X_s, Y_s……始点座標。

X_s…水平座標　WIDTH40のとき0～319
　　WIDTH80のとき0～639

Y_s…垂直座標　0～199

X_e, Y_e……終点座標。

X_e…X_sに同じ

Y_e…Y_sに同じ

X_1, Y_1……始点の論理座標。-1.7014118E＋38～1.7014118E＋38。

X_2, Y_2……終点の論理座標。-1.7014118E＋38～1.7014118E＋38。

省略形　WIN.

用語の説明

〈グラフィック画面の座標系〉

グラフィック画面の分解能は、WIDTH40のとき320×200、WIDTH80のとき640×200で、各ドットの位置は下の図のような座標を使って表します。

WIDTH40のとき



WIDTH80のとき



〈絶対座標系と論理座標系〉

グラフィック画面の座標系には、絶対座標系と論理座標系の2種類があり、グラフィック画面に対してのみ意味をもちます。

○絶対座標系

上の〈グラフィック画面の座標系〉で説明した座標に一致するもので、画面上の1ドットが座標単位の1に対応します。WIDTH40とWIDTH80とでは水平方向(X軸方向)の大きさが、0～319、0～639と異なります。

○論理座標系

WINDOW命令によってユーザーが指定した座標系で、画面の垂直方向・水平方向とも-1.7014118E＋38～1.7014118E＋38までの単精度の範囲で指定できます。後に詳述するPSET、PRESET、LINE、CIRCLE、POLY、PAINTのグラフィックステートメント、およびPOINT関数はこの論理座標系に対して機能します。

下の図は、この座標系と絶対座標系の概念図です。



★広大な平面を想像してください。

解　説

ユーザーはWINDOW命令を使うことによって、広大な論理座標平面の中から任意の領域を指定して、画面(絶対座標平面)上の任意の領域にそれを表示させることができます。たとえば、論理座標系の(x_1, y_1)から(x_2, y_2)を対角線とする長方形の領域を、画面上の(x_s, y_s)から(x_e, y_e)を対角線とする長方形の領域に表示させるには、

WINDOW $(x_s, y_s)-(x_e, y_e), (x_1, y_1)-(x_2, y_2)$と指定します。

下にその概念図を示します。



このように、論理座標系(単精度)の範囲内ならば、マクロ・ミクロのどんな領域でも、画面に表示することができます。

論理座標の指定を省略すると、絶対座標系のサイズと同じ領域が自動的に指定されます。



サンプルプログラム

0°～720°までのサインカーブを描いてみます。論理座標で水平方向に0～720、垂直方向に-1～1の枠を設定し、絶対座標の(50, 20)と(400, 120)を対角線とする長方形の領域に、その後(400,50)と(600, 90)を対角線とする長方形の領域にサインカーブを描きます。




```
10 SCREEN0,0,0:WIDTH 80
20 WINDOW(50,20)-(400,120),(0,-1)-(720,1)
30 S=2:GOSUB 100
40 WINDOW(400,50)-(600,90),(0,-1)-(720,1)
50 S=4:GOSUB 100
60 LOCATE 0,20:END
100 FOR X=0 TO 720 STEP S
110 Y=SIN(RAD(X))
120 PSET(X,Y)
130 NEXT
140 RETURN
```

2.5.12 CLS

機　能　画面をクリアします。

書　式

```
CLSn
CLS
```

n……0，1，2，3，4

省略形　CL.

解　説　画面表示を消して、背景色のみの画面にします。

nの値によって、グラフィック1、グラフィック2、グラフィック3のそれぞれをアクセスします。

nの値

0　グラフィック1、グラフィック2、グラフィック3を同時にクリアします。

1　グラフィック1をクリアします。

2　グラフィック2をクリアします。

3　グラフィック3をクリアします。

4　グラフィック1、グラフィック2、グラフィック3とテキスト(文字)を同時にクリアします。

省略　テキスト画面をクリアします。CONSOLE（2.5.3 CONSOLE文参照）で画面が指定されているときには、指定された範囲のテキスト画面をクリアします。

グラフィック画面をクリアする場合、WINDOW（2.5.11 WINDOW文参照）で画面が指定されているときには、指定された絶対座標（2.5.11 WINDOW文参照）の範囲の画面をクリアします。

サンプルプログラム

```
CLS
Ok.          ← 画面左上隅に表示
CLS 0
Ok.
CLS 4
Ok.                〃
```

2.5.13 PALET

機　能　パレットのカラーを設定します。

書　式

PALET〔パレットコード，カラーコード〕

パレットコード…0～7
カラーコード……0～7（COLOR文参照）

省略形　PAL.

解　説　パレットコードで指定されるパレットに、カラーコードで指定されるカラーを設定します。
SHARP HuBASICは、0～7のパレットコードで表される8つのパレットをもっているので、ユーザーはこのパレットに任意のカラーコードを設定して、カラーをダイナミックに使用することができます。
下の図は、パレットの概念を表すもので、パレットコードとカラーコードの間はゴムひもでつながっていて、パレットコード側は固定され、カラーコード側にはフックがついています。



BASIC起動時には、上の図のように、パレットコード0がカラーコード0（黒）、パレットコード1がカラーコード1（青）……というようにフックがかかっていて、パレットコード＝カラーコードとして使用することができます。
また、PALET1，7：PALET3，2：PALET4，2：PALET5，7：PALET6，7と設定すると、下のようにフックがかかり、パレットコード0が黒、パレットコード2、3、4が赤、パレットコード1、5、6、7が白の3色使用できることになります。



フックの入れかえは瞬時に行えますので、プログラム中でパレット1が白のとき、画面には白い線が描けますが、PALET1, 4となった瞬間、白い線を緑色に切り替えるというような融通性に富んだプログラムを作ることが可能です。パラメータを省略したときは、BASIC起動時のパレット状態にもどります。

サンプルプログラム

```
10 SCREEN0,0,0:WIDTH 40
30 LINE(0,0)-(300,150),PSET,1
40 FOR C=1 TO 7
50 PALET 1,C
60 FOR I=0 TO 200:NEXT I
70 NEXT C:GOTO 40
```

画面を斜めに走る直線の色が、青、赤、マゼンタ、緑、シアン、黄色、白と変化します。
止めるときは CTRL + C キーを押すか、 SHIFT + BREAK キーを押してください。

2.5.14 PRW

機　能

テキスト画面のグラフィック画面に対する優先順位を設定します。

書　式

```
PRW〔n〕
```

n…0～255の整数(省略は0)

省略形

なし

解　説

各色について、テキスト画面とグラフィック画面のどちらを優先して表示するかを設定するステートメントです。nの値0～255(2進で&B00000000～&B11111111)の各ビットが、左の桁からパレットコードの7、6、5、4、3、2、1、0に対応していて、2進でビット1のところがグラフィックのパレット優先、ビット0のところがテキスト画面優先となります。

たとえば、nが100のとき2進表現では01100100なので、パレットコードが6、5、2のものの表示がテキスト画面より優先されます。パレットコードの指定がないうちは、パレットコードはカラーコードに一致するので、テキスト画面のキャラクタ(文字)が黄色、シアン、赤の陰に隠れることになります。

ビット	7	6	5	4	3	2	1	0
	白	黄	シアン	緑	マゼンタ	赤	青	黒

0のとき　テキスト優先

1のとき　グラフィック優先

サンプルプログラム

```
10 SCREEN0,0,0:WIDTH 40
15 PRINT"AAAAAAAAAAAAAAAAAAAAAAAAAAAAAAAAAAAAAAAA"
20 FOR I=0 TO 7
30 LINE(I*40,0)-(I*40+39,199),PSET,I,BF
40 NEXT
45 FOR T=0 TO 5000:NEXT
50 PRW &B11010101
60 FOR T=0 TO 5000:NEXT
70 PRW &B10101010
75 FOR T=0 TO 5000:NEXT
80 CLS4:END
```

このプログラムを実行すると、はじめキャラクタAが1行と、カラータイルが表示され、しばらく経つとキャラクタAが黄色、緑、赤、黒の陰に隠れて見えなくなり、続いて、シアン、マゼンタ、青の陰に隠れて見えなくなります。

画面右端の白いタイルの部分はキャラクタAの保護色となっているため、Aは見えません。

2.5.15 CANVAS

機　能

マルチ画面モード時の各グラフィック画面の色を指定します。

書　式

```
CANVAS  c1, c2, c3
```

c_1…グラフィック1の画面の"カラーコード"(COLOR文参照)
c_2…グラフィック2の画面の"カラーコード"(　〃　)
c_3…グラフィック3の画面の"カラーコード"(　〃　)

省略形

CAN.

解　説

マルチ画面モード時のグラフィック画面1、2、3のそれぞれの色を指定します。グラフィック画面がかさなっている部分の表示色はLAYERで指定された優先順位により、画面1、2、3いずれかの色となります。

PALET命令やPRW命令と同時に使わないで下さい。

サンプルプログラム

```
CANVAS 1,2,3    ← グラフィック1の画面を青、2の画面を赤、3の画面をマゼンタ
CANVAS 7,2,4    ← とする。
                  グラフィック1の画面を白、2の画面を赤、3の画面を緑とする。
```

2.5.16 LAYER

機　能

テキスト画面とグラフィック画面の優先順位を指定します。

書　式

```
LAYER  t,g,g,g
```

t,g,g,g …1、2、3、4の4つの数が1対1に対応。

t　テキスト画面の優先順位番号。

g_1　グラフィック1の画面の優先順位番号。

g_2　グラフィック2の画面の優先順位番号。

g_3　グラフィック3の画面の優先順位番号。

省略形

LAY.

解　説

テキスト画面(文字)とグラフィック画面3つの表示の優先順位を指定します。

t、g_1、g_2、g_3にはそれぞれ1番から4番の順位が入り、画面には順位の高い画面が優先的に表示されます。

PALETやPRWと同時には使わないでください。

サンプルプログラム

```
CLS0                                          ←グラフィック画面をクリアする。
OK
CANVAS 1,2,4                                  ←画面1の色を青、画面2の色を赤、画面3の色
OK                                              を緑と決めます。
LAYER 1,2,3,4                                 ←画面の優先順位を最優先のテキストから順にグ
OK                                              ラフィック画面1, 2, 3の順に設定します。
SCREEN 0,0,1                                  ←グラフィック画面1をアクセスするモードに設
OK                                              定します。
LINE(0,60)-(200,160),PSET,1,BF                ←これで長方形を書きます。CANVAS 1 (青)のた
OK                                              めに青い長方形が書かれます。
SCREEN 0,0,2                                  ←グラフィック画面2をアクセスするモードに設
OK                                              定します。
LINE(80,40)-(220,120),PSET,1,BF               ←長方形を書きます。CANVASの2のために赤い
OK                                              長方形が書かれます。
SCREEN 0,0,3                                  ←グラフィック画面3をアクセスするモードに設
OK                                              定します。
LINE(180,0)-(319,199),PSET,1,BF               ←長方形を書きます。CANVASの4のために緑の
OK                                              長方形が書かれます。
LAYER 1,3,2,4                                 ←画面の優先順位を変更します。そのため今まで
OK                                              青の長方形にかくれていた赤い長方形が前に出
LAYER 1,4,3,2                                   てきます。
OK                                            ←画面の優先順位を変更します。テキスト、緑の
                                                長方形、赤の長方形、青の長方形の順に長方形
                                                がかさなります。
```

■PALET. PRW. CANVAS. LAYERについて

PALET命令が8色の各色を任意の色に設定する命令であるのに比べ、CANVAS命令は3枚のグラフィック画面について、そこに表示される色を指定する命令です。したがって、CANVAS命令では、8色中3色しか使用することができません。

また、PRW命令がTEXT画面8色の各色についてTEXT画面とグラフィック画面の優先順位を設定する命令であるのに比べ、LAYER命令は、TEXT画面と3枚のグラフィック画面の計4枚の画面の優先順位を設定する命令です。

このように、PALETとPRW、CANVASとLAYERをそれぞれペアにして考えるとわかり易いと思います。

もう少し詳しく説明しますと、グラフィック用メモリー（48Kバイト）を3分割して、16KバイトずつをそれぞれB(青)、R(赤)、G(緑)に対応させています。16Kバイトのメモリー容量は、画面表示モードによって、40カラムモード(WIDTH40)では320×200ドットを2画面、80カラムモード(WIDTH80)では640×200ドットを1画面表示するのに必要なのです。(SCREEN命令は、40カラムモード時の2画面を制御しているのです。) また、B.R.G.がそれぞれ独立なメモリー領域をもっているため、この3色の重ね合せにより、8色(黒含む)を各ドット(1画素)毎に指定できます。

たとえば、あるドットについて、それに対応するB(青)とR(赤)が1でG(緑)が0ならば、そのドットの色はマゼンタとなります。このようにB.R.G. 3種類のグラフィックメモリーを1枚のグラフィック画面として使用することにより、任意の8色を表示させることができます。PALETとPRW命令は、グラフィックメモリーをこのように使用した時の命令です。逆に3種類のグラフィックメモリーを各々独立な1画面と考えれば、それぞれ1色ずつしか対応できないかわりに、3枚のグラフィック画面として使うことができます(マルチ画面)。この状態でCANVAS命令により各画面毎に8色中から1色を選択できます。(CANVAS命令前のPALET命令は無効となります。) また、この状態で各グラフィック画面とテキスト画面の優先順位を決めるのがLAYER命令です。

このように、PALET,PRWとCANVAS,LAYERを同時に使用することはできません。

(これらの関係は、SCREEN命令においてグラフィックモード0とグラフィックモード1～3のそれぞれに対応しています。)

PALET ↵ によってパレット状態を初期化すると、8色フルカラー表示できる初期状態にもどりますが、PRWはクリアされません。PRW命令をもクリアしたい時は、INIT ↵ を実行してください。

ここで注意しなければならないことは、PALET及びINIT命令でパレット状態を初期化すると、CANVAS及びLAYER命令はクリアされたように見えますが、再び、たとえばCANVAS命令を使うと初期化前に設定したLAYER命令も有効になっています。つまり、これらは本当にクリアされたのではなく、表面上クリアされたように見えるだけなのです。

ですから、次のように考えてください。

PALET ↵ ……PALET0,0：PALET1,1：PALET2,2：PALET3,3：PALET4,4：
PALET5,5：PALET6,6：PALET7,7
(PRWは変化なし)

INIT ↵ ……上記プラスPRW 0

CANVAS,LAYERをクリアしてBASIC起動時の状態にもどしたい時には次の命令を実行してください。

```
CANVAS  1, 2, 4：LAYER  1, 2, 3, 4
INIT
```

CANVAS，LAYER命令のサンプルプログラム説明

Sample 1

```
LIST
 10 INIT:CLS4
 20 FOR I=0 TO 7
 30 LINE (I*40,0)-(I*40+39,199),PSET,I,BF
 40 NEXT
 50 PAUSE 10
 60 LAYER 1,2,3,4:CANVAS 1,2,4
 70 PAUSE 10
 80 PALET
 90 END
Ok
```

図1（Sample 1）



解　説

サンプルプログラム1を実行しますと、最初に①の画面が現われます。これは、8色表示可能なグラフィックモード0の状態です。

次に、行番号60のLAYER，CANVAS命令を実行すると、8色中3色しか表示できなくなります。

たとえば、G1、G2，G3のカラーコードを、それぞれ青、赤、緑に設定し、優先順位を、TEXT，G1，G2，G3の状態にしてやると、まず、優先順位の高い青の含まれている色が青に変わります。マゼンタ(青＋赤)、シアン(青＋緑)、白(青＋赤＋緑)がこれに相当します。次に、青に次ぐ優先順位の赤を含む色が赤に変わります。黄(赤＋緑)がそれです。黒色というのは、青も赤も緑も含まない色、すなわちバックそのものですので変化しません。(図1、画面②)

次に、行番号80を実行しますと、パレットが初期化されますので、CANVAS，LAYER命令の影響がなくなり、8色表示可能な画面にもどります。

行番号60の優先順位とカラーコードを変えることにより、②の画面は様々な色に変化しますので、適当に変えて確認してみて下さい。

Sample 2

Sample 1のプログラムを次のように変更してみます。

```
60    LAYER 1,4,3,2：PAUSE10：CANVAS 3,5,6
```

図2（Sample 2）



解　説

行番号60のLAYER命令が実行されますと、以前にCANVAS命令が実行された時は、その色に従って優先順に表示しますが、このSampleのようにCANVASによってグラフィック画面の色が指定されていない時は、CANVAS 1，2，4（すなわちG1が青，G2が赤，G3が緑）が実行されたものとしてLAYERを実行します。Sampleでは、優先順位がG3，G2，G1の順になっていますので、緑を含む、シアン、黄、白が緑に変わり、マゼンタ(青＋赤)は赤になります。(図2、画面②)

同様のことは逆の場合にも起こり、LAYER命令が実行されていない時にCANVASを実行すると、LAYER 1，2，3，4が実行されたものと解釈してCANVAS命令を行ないます。

次に、行番号60のCANVAS 3，5，6を実行しますが、これは、G1をマゼンタ、G2をシアン、G3を黄に指定する命令なので、青の部分がマゼンタに、赤の部分がシアンに、緑の部分が黄に変化します。(図2、画面③)

もし、行番号60のLAYERとCANVAS命令の順序が以下のように逆になっていたら、画面は図3のように変化します。

```
60    CANVAS 3,5,6：PAUSE10：LAYER 1,4,3,2
```

(注意)

ただし、実行前に次の命令をダイレクトに実行して下さい。

CANVAS 1 , 2 , 4 ：LAYER 1 , 2 , 3 , 4 ：INIT ↵ というのは、前述の命令

60　LAYER 1 , 4 , 3 , 2：PAUSE10：CANVAS 3 , 5 , 6

を実行した後では、行番号60のLAYER命令とCANVAS命令の順序を逆に実行しても、図3の通りには変化しません。それは、最初CANVAS 3 , 5 , 6 を実行する時には、前に実行したLAYER命令LAYER 1 , 4 , 3 , 2 が有効ですので、(前に説明したようにPALET命令によってクリアされたわけではない)。

LAYER 1 , 4 , 3 , 2：CANVAS 3 , 5 , 6

を実行したことと同じになり、図2③の画面になってしまいます。

よって、比較のためには、前もって同じLAYER状態にもどしてやる必要があるわけです。

図3 (Sample 3)



2.5.17 GET@

機　能

グラフィック画面ドットデータを読み込みます。

書　式

GET@ (x_1, y_1)-(x_2, y_2), 配列名 [,パレットコード]

x_1, y_1 ……ドットデータ読み込み開始座標。
x_2, y_2 ……　〃　終了座標。
} (絶対座標)

配列名……数値型配列変数。(予めDIM文で用意する。)

省略形

なし。

解　説

グラフィック画面上のドットデータを配列名で指定される配列変数に読み込みます。読み込む画面の領域は、開始座標(x_1, y_1)と終了座標(x_2, y_2)を対角とする長方形の枠内です。必要なカラー情報はパレットコードで指定しますが、この指定を省略するとテキストのデータを指定することになります。この場合座標は、テキストの座標とみなされます。

読み込むのに必要な配列のバイト数は次の式で計算できます。

テキストの場合　(ABS(x_1 - x_2)+1)*(ABS(y_1 - y_2)+1)*2

WIDTH40　$0 \leqq x_1, x_2 \leqq 39$
　　$0 \leqq y_1, y_2 \leqq 24$
WIDTH80　$0 \leqq x_1, x_2 \leqq 79$
　　$0 \leqq y_1, y_2 \leqq 24$

グラフィックの場合　INT(((ABS(x_1 - x_2)+1)*(ABS(y_1 - y_2)+1)+7)/8)*m

mの値はパレットコードにより異なる……　0の場合…m＝0
　　1,2,4　〃　…m＝1
　　3,5,6　〃　…m＝2
　　7　〃　…m＝3

WIDTH40　$0 \leqq x_1, x_2 \leqq 319$
　　$0 \leqq y_1, y_2 \leqq 199$
WIDTH80　$0 \leqq x_1, x_2 \leqq 639$
　　$0 \leqq y_1, y_2 \leqq 199$

必要な配列の要素数は、必要なバイト数を、次の数で割ったものとなります。

整数型配列の場合……2
単精度　〃　……5
倍精度　〃　……8

必ず、予め必要バイト数を計算して、DIM文によって配列変数を用意してから実行するようにしてください。

グラフィック画面の場合、グラフィックRAM上のデータを必要に応じてB(青)、R(赤)、G(緑)の順に読みとっていって、配列の型に応じたバイト数のメモリー中につめこんでいきます。

mの値がパレットコードによって異なるのは、グラフィックメモリーがそれぞれB(青)、R(赤)、G(緑)の3枚で構成されているので、パレットコードが1、2、4(青、赤、緑)のいずれかの時、1枚の画面データを読みこんでやればよいのに対し、パレットコードが3、5、6(マゼンタ、シアン、黄)の場合、3枚中2枚のグラフィックメモリーのデータを読み込む為、2倍の記憶場所が必要となります。

また、パレットコード7(白)の場合、3枚とも読み込む為、3倍の記憶場所が必要になるわけです。

テキスト画面の場合、テキスト画面とその属性（付録「A.1テキスト画面とその属性ポートへのアクセス方法」参照）の各座標のデータを配列に格納します。

サンプルプログラム

2.5.18 PUT@参照。

2.5.18 PUT@

機　能　グラフィックドットデータをグラフィック画面に表示します。

書　式

PUT@'$(x_1, y_1)-(x_2, y_2)$, 配列名 {, PSET, パレットコード / , PRESET, パレットコード / , XOR, パレットコード / , OR, パレットコード / , AND, パレットコード / , NOT, パレットコード}

x_1, y_1 ……ドット情報の書き込み開始座標。
x_2, y_2 ……　〃　　終了座標。
（絶対座標）

配列名……数値型配列変数。

（モード）PSET ……ビット1の点(dot)をつけ、ビット0の点を消す。
PRESET…ビット1の点を消す。
XOR ……ビット1の点を反転する。
OR……ビット1の点をつける。
AND ……ビット0の点を消す。
NOT ……ビット1の点を消して、ビット0の点をつける。

省略形　なし。

解　説　GET@（2.5.17 GET@文参照）で配列変数に入れられたグラフィック画面のドットデータを、画面の指定した領域にグラフィックとして表示します。表示領域は開始座標（x_1, y_1）と終了座標（x_2, y_2）を対角とする長方形の枠内です。

モードについては、指定した領域内に既にあるドットのカラーデータについて、該当する配列変数中のドット情報と、カラービットごとの論理演算を行い、その結果を画面にセットします。テキスト情報の入った配列変数のときには、モード以降の設定はいりません。〔　〕内を指定すると、グラフィックの座標とみなされます。

GET@、PUT@の数値型配列変数名は（　）を省略することができ、その場合、配列名(0)と同じになります。配列の添字を変えることによって、同一配列に2つ以上のパターンを定義することができます。

サンプルプログラム

```
10 INIT:CLS4
20 DIM A%(24)
30 POLY(5,5),5,7,144,0,720
40 POLY(5,5),5,4,144,90,810
50 GET@(0,0)-(10,10),A%,7
60 FOR I=0 TO 190 STEP 10
70 PUT@(I,I)-(I+10,I+10),A%,PSET,7
80 NEXT
```

重なった星形をななめに19個描きます。20個目は、(190,190)－(200,200)の範囲ですのでY座標がオーバーし表示されません。

このサンプルプログラムでは、行番号50でグラフィック座標（0，0）、(10，10)を対角とする正方形のわく内のデータ(すなわち、星形のグラフィックデータ)を整数配列A%内に読み込ませています。配列の大きさは次のように計算されます。

INT（(11×11＋7)／8）＊3／2＝24

2.5.19 POKE@

機　能　VRAM内の指定されたアドレスに1バイトのデータを書き込みます。(⇒OUT)

書　式

```
POKE@　アドレス，データ〔，データ，データ…〕
```

アドレス…VRAM内の指定アドレス。＆H2000～&HFFFF
データ……メモリーに書き込む1バイトのデータ。0～255の整数

省略形　PO.@

解　説　VRAM内の指定したアドレスに、1バイトのデータを書き込みます。データを，(カンマ)で区切って続けて書くと、連続したアドレスに書き込むことができます。(付録「A.1テキスト画面とその属性ポートへのアクセス方法」参照)

サンプルプログラム　画面のカーソル位置（＆H3000番地）から256個の文字を表示し、その属性として、黒、青、赤、マゼンタ、緑、シアン、黄、白の色を指定するプログラムを下に示します。

```
10 INIT:WIDTH 40
15 FOR I=0 TO 255
20 POKE@ &H3000+I,I
25 POKE@ &H2000+I,(I MOD 8)
30 NEXT
40 LOCATE 0,23
```

注）MODは剰余演算子で結果は整数の割算の余りになります。
(1.10.1 算術演算子参照)

2.5.20 OPTION SCREEN

機　能　グラフィック用メモリーの使用目的の設定をします。

書　式

```
OPTION SCREEN n
```

n…1，2

省略形　OP. SC

解　説　グラフィック用メモリーの使用目的の設定に使います。

nの値

1．グラフィック用メモリーをグラフィック表示用に使います。
2．グラフィック用メモリーを外部記憶用に使います。

OPTION SCREEN 2を行った後は、必ずINIT "MEM:" を実行してください。

サンプルプログラム

```
OPTION SCREEN 1
Ok
OPTION SCREEN 2
Ok
INIT"MEM:"
Are you sure ? (y or n)Y
Ok
```

2.6 グラフィックステートメント

2.6.1 PSET

機　能　グラフィック画面に点（ドット）を表示します。（⇐⇒2.6.2 PRESET）

書　式

```
PSET(x,y[,パレットコード])
```

x, y…… ドットの表示座標（論理座標）
　　x……水平座標
　　y……垂直座標
パレットコード…ドットのパレット（色を指定します）
　　省略すると現在のパレットコード（2.5.5 COLOR文参照）が使われます。

省略形　PS.

解　説　グラフィック画面の論理座標（x,y）の点を指定のパレット番号でセットします。マルチ画面モードの時は、指定の画面のみパレットコード0ならリセット、それ以外ならセットします。
WINDOW外の点を指定してもエラーにはならず表示もされません。

サンプルプログラム　ドットの水平・垂直座標とカラーをランダムに指定して、画面に表示します。

```
10 ' PSET
20 INIT:WIDTH 40
30 X=INT(RND*320)
40 Y=INT(RND*200)
50 C=INT(RND*7)+1
60 PSET(X,Y,C)
70 GOTO 30
```

止めるときは、SHIFT ＋ BREAK キーを押すか
　　CTRL ＋ C キーを押してください。

2.6.2 PRESET

機　能　グラフィック画面の点（ドット）をリセットします。（⇔2.6.1 PSET）

書　式

```
PRESET(x,y[,パレットコード])
```

x,y…… ドットの表示座標（論理座標）
x …水平座標
y …垂直座標
パレットコード…ドットの消去パレット

省略形　PRE.

解　説　グラフィック画面の論理座標（x,y）の点を指定のパレットコードでリセットします。

グラフィック画面上の各ドットについて、表示されていれば（1ビットセットされていれば）消し、されていなければ何もしません。

マルチ画面モードでは0で何もされずにそれ以外では、リセットされます。

サンプルプログラム　白い画面の各ドットをランダムにリセットします。

```
10 INIT:WIDTH 40
20 LINE(0,0)-(319,199),PSET,7,BF
30 X=INT(RND*319)
40 Y=INT(RND*200)
50 C=CINT(RND*7+.5)
60 PRESET(X,Y,C)
70 GOTO 30
```

止めるときは、SHIFT ＋ BREAK キーを押すか

CTRL ＋ C キーを押してください。

2.6.3 LINE

機　能　画面上に直線や長方形を描きます。

書　式

LINE[(x1,y1)]-(x2,y2) { ,PSET / ,PRESET / ,XOR } { [,パレットコード] / ,[パレットコード],ラインパターン / [,パレットコード],B[,ラインパターン] }

省略形　なし。

解　説　論理座標 (x_1, y_1) - (x_2, y_2) 間に直線を描いたり2点を対角とする長方形を描いたりします。(x_1, y_1) を省略するとその前の命令の終点 (x_2, y_2) から新しい終点へと描きます。

モードは、セットする、リセットする、反転するの3つのモードがありそれぞれ、PSET, PRESET, XORと指定します。パレットコードは省略された場合、COLOR文の表示色と同じパレットコードが使われます。

モードの次か、パレットコードの次にBを指定すると長方形を描きます。Bがない場合は直線を描きます。

ラインパターンは16ビットの数値を指定しますが省略の場合は&HFFFF（実線）で描きます。

点線を描く時は&HCCCC、一点鎖線は&HFFCCを指定するとよいでしょう。

サンプルプログラム　画面上にランダムに直線を引きます。

```
10 ' LINE SAMPLE
20 INIT:WIDTH 40
30 X0=INT(RND*320)
40 Y0=INT(RND*200)
50 X1=INT(RND*320)
60 Y1=INT(RND*200)
70 C=INT(RND*7)+1
80 LINE(X0,Y0)-(X1,Y1),PSET,C
90 X0=X1:Y0=Y1
100 GOTO 50
```

止めるときは、SHIFT ＋ BREAK キーを押すか CTRL ＋ C キーを押してください。

点線を描く時の指定方法

```
80 LINE(X0,Y0)-(X1,Y1),PSET,C,&HCCCC
```

一点鎖線を描く時の指定方法

```
80 LINE(X0,Y0)-(X1,Y1),PSET,C,&HFFCC
```

LINE

機　能　画面上に長方形を描き，その内部をぬりつぶします。

書　式

LINE[(x1,y1)]-(x2,y2) { ,PSET / ,PRESET / ,XOR } { ,BF / ,[パレットコード],BF / ,BF,タイルパターン }

省略形　なし。

解　説　論理座標 (x_1, y_1) - (x_2, y_2) の範囲内を指定のモードでぬりつぶします。ラインやボックスと同じパラメーターを持ちモードの次かパレットコードの次に、BFと指定する事によりボックスフルを実行します。

BFの次にタイルパターンを表わす文字式を指定する事によりそのパターンでのぬりつぶしが可能となります。(タイルパターンについては、2.6.6 PAINT参照)

サンプルプログラム

```
10 INIT:CLS0
20 LINE(0,0)-(100,100),PSET,3,BF
30 LINE(100,100)-(200,199),PSET,BF,HEXCHR$("AAAAAA555555")
                                          青 赤 緑
```



画面にマゼンタ色の4角形と白と黒の市松模様の4角形を描きます。

LINE

機　能　指定の論理座標間を実線でつなぎます。

書　式

LINE[(x1,y1)]-(x2,y2)[-(x3,y3)…-(xn,yn)]

省略形　なし。

解　説　指定の論理座標間を次のラインでつなぎます。

モード……PSET。
パレット番号…COLOR文の表示色と同じ。
ラインパターン…&HFFFFの実線。
消したい時はCOLOR 0 を実行後この命令を実行します。
始点を省略するとその前の命令の終点から
指定した終点まで線を引きます。

サンプルプログラム

```
INIT:CLS0:LINE(100,0)-(0,100)-(100,200)-(200,100)-(100,0)
```



左の様な4角形を描きます。

LINE

機　能　テキスト画面に線や長方形を描きます。

書　式

LINE[(x₁,y₁)]-(x₂,y₂),文字式[,B / ,BF]

x₁,x₂ ……テキスト上のX座標（WIDTH40のとき0～39 / WIDTH80のとき0～79）

y₁,y₂ ……テキスト上のY座標（0～24）

省略形　なし。

解　説

テキスト画面に対して、ライン(B,BF省略時)、ボックス(B指定)、ボックスフル(BF指定)の各命令を実行します。ラインを指定すると、テキスト座標(x_1,y_1)、(x_2,y_2)を、文字式の先頭キャラクタで結びます。ボックスを指定すると、両座標を対角とする四角形を先頭キャラクタで描きます。ボックスフルを指定すると、四角形内部を先頭キャラクタでぬりつぶします。(x_1,y_1)を省略しますと、その前の命令の終点(x_2,y_2)を、始点(x_1,y_1)とみなして四角形を描きます。

また、CONSOLE命令によって指定した範囲外にも描くことができます。ただ、WINDOW命令で絶対座標と論理座標を別の座標系としている時は、注意が必要です。LINEの座標は論理座標とみなされますので、上記の絶対座標系とは異なった四角形を描きます。

また、テキスト画面の属性（アトリビュート）には、LINE命令が実行された時に設定されている色、反転、点滅、CG(キャラクタジェネレータ)、サイズのデータが、描かれるキャラクタと同時に書きこまれますので、表示文字は、それらの設定通りに変化したものになります。(詳しくは、付録「A.1テキスト画面とその属性ポートへのアクセス方法」を参照して下さい。)

サンプルプログラム

```
LIST
10 WIDTH40:INIT
20 COLOR 2
30 LINE(0,0)-(39,24),"*A",BF
40 PAUSE 10
50 COLOR 6:CREV 1
60 LINE(1,1)-(38,23),"!",B
70 PAUSE 10
80 CFLASH 1
90 LINE(39,0)-(0,24),"0"
100 INIT
110 END
OK
```

- 10行：表示状態を初期化し、画面をクリアします。
- 20行：表示色を赤に指定します。
- 30行：全画面が赤色の〝*〟キャラクタでぬりつぶされます。
- 40行：約1秒間実行を停止します。
- 50行：表示色を黄色に指定し、以後の文字を反転させます。
- 60行：画面ワクより1行内側に白地に青(黄の反転)の〝!〟で長方形を描きます。
- 70行：以後の表示文字を点滅状態に設定します。
- 80行：画面右上すみから左下すみへ、点滅キャラクタ〝0〟でラインを描きます。
- 90行：表示状態を初期化します。

2.6.4 CIRCLE

機　能　円・弧を描きます。

書　式

CIRCLE(x,y),r[,[パレットコード],[扁平率],[初期角]〔,終了角〕]

x,y…… 中心の座標（論理座標）
　　x 水平座標
　　y 垂直座標
r ……半径
パレットコード…円弧の色（COLOR文参照）省略すると現在の色
扁平率……（扁平率）＝（垂直方向の半径）÷（水平方向の半径）省略すると1
初期角…… 省略すると0(度)
終了角…… 省略すると360(度)

省略形　CI.

解　説

画面にドットで円や弧を描くためのステートメントです。

中心（x,y)、半径rの円が画面に描かれます。扁平率を設定すると、x方向とy方向異なる半径の円を描くことができます。初期角と終了角を指定すると、その内角の弧が表示されます。ここで描かれた円は論理座標での円であり組合わせのディスプレイにより真円に見えない場合があります。
WIDTH80の時のx方向の半径は2rとなります。(扁平率もこの関係に従います。)

サンプルプログラム

下図の様な円をいろいろな色で描きます。

```
10 INIT:WIDTH 40
20 FOR X=20 TO 319 STEP 40
30 FOR Y=20 TO 199 STEP 40
40 C=INT(RND*7)+1
50 CIRCLE(X,Y),18,C,1,0,360
60 NEXT
70 NEXT
```



2.6.5 POLY

機　能　多角形を描きます。

書　式

POLY(x,y),r[,[パレットコード],[ステップ角],[初期角][,終了角]]

x,y…… 中心の座標（論理座標）
　　x 水平座標
　　y 垂直座標
r ……中心から頂点までの距離（ドット数）
パレットコード…多角形の辺の色（COLOR文参照）。省略すると現在の色
ステップ角…… n角形のとき$\frac{360}{n}$(度)。省略すると1（度）
初期角……省略すると0（度）
終了角……　〃　360(度)

省略形　POL.

解　説　(x,y) を多角形の中心とし、中心から各頂点までの距離をrとした多角形を描きます。頂点間の (x,y) を多角形の中心とし、中心から各頂点までの距離をrとした多角形を描きます。頂点間の角度をステップ角と呼び120で三角形、90で四角形となります。この角度が0の時は、中心から初期角の方向へ長さrの直線を引きます。

サンプルプログラム　星を描きます。

```
POLY(100,100),90,5,144,90,810
```

半径90パレットコード5の色で144度きざみ90度から810度までです。



2.6.6 PAINT

機　能

画面の特定のエリアを指定した色で塗りつぶします。

書　式

PAINT(x,y)[,パレットコード / ,タイルパターン[,境界のパレットコード,…]]

省略形

PAI.

解　説

論理座標（x,y）の点を塗り始めの点として、境界となるパレットコードで指定された図形内を、指定のパレットコードあるいはタイルパターンで塗りつぶします。

省略の場合、前のグラフィック命令のパレットコードが使われます。

タイルパターンは、横方向に8ドット、縦方向に任意の長さをもつタイル模様です。縦方向nドットのタイルを指定するためにはn×3文字の文字列が必要で、3文字で横8ドット縦1ドットの最小パターンを指定し、左から1文字ずつ青、赤、緑のドットパターンを示します。下に8×1ドットのタイルパターンを示します。

（例）

HEXCHR$（"AACC33"）

AA＝（10101010）$_2$

CC＝（11001100）$_2$

33＝（00110011）$_2$

1はセット

0はリセット



次の例は8×8ドットのタイルパターンです。

（例）

HEXCHR$（"①110011 ②220022 ③4400044 ④8800088"）

+HEXCHR$（"⑤110011 ⑥220022 ⑦4400044 ⑧8800088"）

11＝（00010001）$_2$

22＝（00100010）$_2$

44＝（01000100）$_2$

88＝（10001000）$_2$

00＝（00000000）$_2$



サンプルプログラム

```
CLS0
OK
PAINT(0,0),2
OK
```

画面を消し左上スミ（0,0）から2の色でぬりつぶします。
境界が無いので全画面ぬりつぶします。

```
CIRCLE(123,85),67,4
OK
PAINT(123,85),2,4
OK
```

境界線となる閉鎖空間（たとえば円など）を作ってそのぬりつぶす始めの点は、円の中心にするとまちがいありません。

ぬりつぶす色の次に境界線の色です。円は4の色で描いてあるので4に指定するとよいでしょう。

タイルパターンに重ねてタイルパターンでぬりつぶす場合、途中で止まってしまったり、Out of memoryエラーが出る場合がありますので、事前に単色でぬりつぶしてから、タイルパターンでPAINTする様にして下さい。

```
PAINT(123,85),HEXCHR$("1100112200224400448800B8"),4
OK
```

2.6.7 POSITION

機　能　PATTERN文によるドットパターンの表示位置を指定します。

書　式

```
POSITION x,y
```

x,y……ドットパターンの表示位置の座標
　x…ドットパターンの表示位置の水平座標
　　WIDTH40のとき　0～319
　　WIDTH80のとき　0～639
　y…ドットパターンの表示位置の垂直座標
　　0～199

省略形　POS.

解　説　このステートメントはPATTERN文によって設定されるドットパターンの開始座標パターンの左上隅の座標）を指定し、必ずPATTERN文の前に置かれます。

サンプルプログラム　2.6.8 PATTERN文参照。

```
POSITION 10,20
```

2.6.8 PATTERN

機　能　グラフィック画面の特定のエリアに任意のグラフィックパターンを描きます。

書　式

```
PATTERN n, 文字式[,文字式…]
```

n ……ドットパターンの垂直方向の段数 -255～255
　　n＜0のとき　下方向に重なる段数
　　n＞0のとき　上方向に重なる段数
　　n＝0のとき　Illegal function call を起こします。
文字式……ドットパターン（ビットパターン）を表す文字式。

省略形　PAT.

解　説　POSITIONで指定された座標を開始点として、ストリングで表わされたビットパターンを、画面にn段分（垂直方向にnドット分）表示します。

ストリングをいくつかに分けて指定すると、画面にはドットパターンが連続して表示されます。

POSITIONが1度しか定義されていないで、その後PATTERN文が続く場合は表示位置を右か下へずらして、グラフィックパターンを表示し続けます。

描くべきグラフィックパターンは1バイト（8ビット）単位のドットパターンから構成されていて、ドットパターンは各セット・リセットの組み合わせにより256通りあります。ドットパターンを表すデータには文字列を使います。

たとえば■□■□■□■□(■はセット、□はリセット)を表示するには、$(10101010)_2=(AA)_{16}=(170)_{10}$ を文字に変換したCHR$(&HAA)かCHR$(170)、またはカタカナの小文字のエ "ェ" を使い、□■□■□■□■を表示するには、$(01010101)_2=(55)_{16}=(85)_{10}$ を文字に変換したCHR$(&H55) かCHR$(85)、あるいは "U" を使います。

ドットパターンの積み重ねはnの値の符号によって決められ、nが負のとき、上から下へ、nが正のとき下から上へ積み重ねられます。文字列データが積み重ねの段数より多いときは、下方向、上方向にn段積み重ねたあと、すぐ右隣り（x軸正方向）で同じ方向に積み重ね、すべての文字列データを表示しつくすまで同じ操作を繰り返します。

たとえば、A$＝ "アイウエオカキクケコサシス"
すなわち、HEXCHR$（ "B1B2B3B4B5B6B7B8B9BABBBC" ）
または、　CHR$（&HB1）＋CHR$（&HB2）＋……＋CHR$（&HBC）
をPATTERN文を使って画面に表示すると、下の図のようなドットパターンとなります。

```
PATTERN -5,A$
```



サンプル
プログラム

左の例のサンプルプログラムを示します。

```
10 INIT:CLS4
20 A$="ｱｲｳｴｵｶｷｸｹｺｻｼｽ"
30 POSITION 0,20
40 PATTERN -5,A$
```

下にPATTERN文によって漢字を表示させる1例を上げます。実行させると、画面左上に「漢」という字が表示され、続いて右隣りに同じ「漢」が表示されます。

```
10 INIT:CLS4
20 POSITION 0,20
30 PATTERN -16,KANJI$(&H7F1)
40 PATTERN -16,KANJI$(&H7F1)
```

(上の例で、CZ-803Cではオプションの漢字ROMが必要です。)

KANJI$についてはKANJI$関数を参照してください。

注) 専用漢字ROMのついていない場合は16×16ドットの白い四角形が2つ続いて表示されます。

2.7 画面制御ステートメント／グラフィックステートメントの使用可能な画面

●CONSOLE
（文字の表示エリアの設定）
●LOCATE,CURSOR
（カーソルの位置の指定）
●CREV
（文字の反転表示）
●CFLASH
（文字のフラッシング表示）
●CGEN
（ユーザ文字モードの設定）
●CSIZE
（文字のサイズの指定）
●DEFCHR$
（キャラジェネに文字パターンを定義）

●PALET
（パレットカラーの指定）
●OPTION SCREEN
（グラフィック用メモリーの用途設定）

〈グラフィック画面〉



〈テキスト画面〉



●WIDTH
（画面サイズの設定）
●SCREEN,GRAPH
（I/Oページ、グラフィックモードの設定）
●COLOR
（カラー設定）
●CLS
（画面のクリア）
●GET@
（ドットデータの読み込み）
●PUT@
（ドットデータの表示）
●POKE@
（VRAM内へドットデータを書き込む）
●LINE
（直線、長方形の表示）

●WINDOW
（グラフィック表示エリアの設定）
●PSET
（ドットの表示）
●PRESET
（ドットのリセット）
●CIRCLE
（円、弧の表示）
●POLY
（多角形の表示）
●PAINT
（指定エリアの塗りつぶし）
●POSITION
（グラフィックパターンの表示位置の指定）
●PATTERN
（グラフィックパターンの表示）

●PRW
（Text/グラフィックの優先順位の設定）
● LAYER
（Text/G1/G2/G3の優先順位の設定）



それぞれ任意8色中の1色

〈マルチ画面〉

●CANVAS
（グラフィックのカラー指定）

2.8 特殊ステートメント

2.8.1 KEY・DEF KEY

機能 ファンクションキーに対して文字列を定義します。

書式

```
KEY ファンクションキー番号，文字列
```

```
DEF KEY ファンクションキー番号，文字列
```

ファンクションキー番号……1～10（ファンクションキー左端から1,2,3,4,5,SHIFTキーを押しながら押すと6,7,8,9,10と対応しています。）

文字列……最大15文字まで定義することができ、15文字を越えるとはみ出した右側の文字が無視されます。

省略形

K.

DEFK.

解説 ファンクション番号で指定したキーに文字列を対応させます。このステートメントによって文字列を定義すると、それ以降、そのキーを押すと定義した文字列をキーボードから打ち込むのと同じになります。

ファンクションキーはBASICロード時には次のように定義されています。

1. "AUTO"+CHR$(13)
2. "?TIME$"+CHR$(13)
3. "KEY"
4. "LIST"+CHR$(26,13)
5. "RUN "+CHR$(13)
6. "LOAD "+CHR$(13)
7. "WIDTH "
8. "CHR$("
9. "PALET "
10. "CONT"+CHR$(13)

注) CHR$(13)は  キーを押したことと同じです。

2.8.2 KEYLIST・KLIST

機能 ファンクションキーの定義状態を表示します。

書式

```
KEY LIST
```

```
KLIST
```

省略形

K.L.

KL.

解説 このステートメントを実行すると、画面にキー番号と文字列の一覧表を表示します。

サンプルプログラム

```
KEYLIST

KEY 1,"AUTO"+CHR$(13)
KEY 2,"?TIME$"+CHR$(13)
KEY 3,"KEY"
KEY 4,"LIST"+CHR$(26,13)
KEY 5,"RUN "+CHR$(13)
KEY 6,"LOAD "+CHR$(13)
KEY 7,"WIDTH "
KEY 8,"CHR$("
KEY 9,"PALET "
KEY10,"CONT"+CHR$(13)
OK
```

2.8.3 KEY0

機　能　キーボードから入力のシミュレートを行う。

書　式

KEY0 ,"文字列"

省略形　K. 0

解　説　これは、キーバッファ(注)に文字列を格納する命令です。すなわち、ユーザーがキーボードのキーを押さなくても、押したと同じ状態になります。入力できる文字列は、最大63文字で、64文字目からは無視します。プログラム実行中は、KEY 0 命令による入力文字はキーバッファにためられるだけで、INKEY＄関数などの入力命令を実行した時に、はじめて受けつけられます。

注) キーバッファとは、キー入力文字を一時ためておく場所です。コンピュータは、入力待ち状態になると、キーバッファの内容をたえず見張っていて、バッファに入力された命令を取り込み、内部で処理します。

サンプルプログラム

```
KEY0,"FORI=0TO10:?I;:N."+CHR$(13)
OK
FORI=0TO10:?I;:N.
 0  1  2  3  4  5  6  7  8  9  10
OK
```

この行を入力すると、
自動的にこの行が入力されて、
実行されます。

```
LIST
10 FOR I=1 TO 5000:NEXT
20 KEY0,"":A$=INKEY$(1)
30 IF A$="Y" THEN END
OK.
```

← このループが終了するまでは何も実行しません。
← キーバッファをクリアして、キーボードから1文字入力します。
← 入力した文字で判断する分枝命令です。

このサンプルプログラムは、KEY 0 命令を誤入力防止用に使用した例です。たとえば、行番号10実行中に誤まってキーを押しても、行番号20のINKEY＄関数を実行する前にKEY 0 命令によってキーバッファがクリアされます。(ヌルストリング" "がストアされる)ので、INKEY＄関数は、正しくキーボードから入力される文字を待ちます。

2.8.4 REPEAT ON・REPEAT OFF

機　能　キー入力のリピート機能の設定、解除を行います。

書　式

```
REPEAT {ON }
       {OFF}
```

省略形　REP.O.
REP.OF.

解　説　通常、キーボードのキーを押し続けると、同じ文字が継続して入力されますが、この機能を止めるのがREPEAT OFFです。
REPEAT OFFにすると、キーボードからのキー入力が1文字で止まり、それ以上受けつけなくなります。REPEAT ONすると元の状態に戻ります。

サンプルプログラム

```
REPEAT ON
OK
REPEAT OFF
OK
```

2.8.5 CLICK ON・CLICK OFF

機　能　キー入力時のクリック音の設定、解除を行います。

書　式

```
CLICK {ON }
      {OFF}
```

省略形　CLI.O.
CLI.OF.

解　説　キーボードのキーを押すと「クリッ」という音（クリック音）を出しますが、CLICK OFFにするとこの音を出しません。CLICK ONにすると、再びクリック音が出るようになります。

サンプルプログラム

```
CLICK ON
OK
CLICK OFF
OK
```

2.8.6 ON KEY GOSUB

機　能

ファンクションキーからの入力値によっていくつかの行へジャンプします。

書　式

```
ON KEY GOSUB　行番号1〔,行番号2,行番号3,…〕
```

省略形

O. K. GOS.

解　説

i番目のファンクションキーが押されたとき、i番目に指定した行番号の行をサブルーチンとして実行します。ファンクションキー1ならば1番目の行番号，2ならば2番目，3ならば3番目，…というように対応しており、ファンクションキーの値に対応する行番号がなければ次の文へ実行を移します。
ファンクションキーは左から1，2，……，5、シフトを押しながらのときは6，7，……，10の値をとります。
ON KEY GOSUBを実行すると本来のファンクションキーの機能である文字列定義の機能が使用できなくなります。(CTRL＋Dキーを押すとON KEY GOSUBは解除されます。)

ON式GOSUB命令とは本質的に異なっています。ON式GOSUB命令は、そのステートメントが実行される時、式の値を判断してジャンプしますが、ON KEY GOSUB命令は、このステートメントを実行してもすぐにはジャンプせずに次の行へうつります。しかし、ファンクションキーが押されるとON KEY GOSUB命令で指定された行へジャンプします。そしてRETURN命令によってジャンプ前の実行が再開されます。

サンプルプログラム

```
10 ON KEY GOSUB 100,200
20 PRINT"*";:GOTO 20
100 BEEP:RETURN
200 CLS:RETURN
```

このプログラムを実行しますと画面上に星印を書きますが、ファンクションキー1を押すとBeep音が鳴り、ファンクションキー2を押すと画面がクリアされて実行が継続されます。
SHIFT＋BREAKキーを押すとプログラムは止まりますが、その後、ファンクションキーを押しますと出力されませんので CTRL ＋ Dキーを押してください。

2.8.7 KEY ON・KEY OFF・KEY STOP

機　能　ON KEY GOSUB命令を制御します。

書　式

```
KEY〔n〕 { ON  }
        { OFF }
        { STOP}
```

n…1～10の整数。

省略形
K.O.
K.OF.
K.S.

解　説
n番目のファンクションキーについて
KEY n ONの場合、押されたときON KEY GOSUBを実行します。
KEY n OFFの場合、押されたとき本来のファンクションキーが押されたものとみなします。
KEY n STOPの場合、押されたとき、ON KEY GOSUBは実行せず、KEY n ONが実行されるまで、保留しています。
nを省略すると、1～10すべてのファンクションキーについて上記の機能をもちます。

サンプルプログラム

```
KEY ON
OK
KEY OFF
OK
KEY STOP
OK
```

2.8.8 MID$

機　能　文字式の一部の置き換えをします。

書　式

```
MID$(文字式1 ,m ,n)=文字式2
```

文字式1……置き換えされる文字式、結果になる。
m……文字式1の置き換え開始位置。
n……文字式1の置き換える文字式の長さ。
文字式2……置き換わる文字式。

省略形　MI.

解　説
文字式1のm番目の文字からn個の文字を、文字式2の最初からn個の文字で置き換えます。
mの値が文字式1の長さを越えた場合は何もしません。nの値が文字式2の長さより大きくなると、nは文字式2の長さになります。(この命令は、3．2．10　MID$とは異なります。)

サンプルプログラム

```
LIST
10 A$="ABCDEFGHIJKLMN"
20 MID$(A$,2,3)="1234"
30 PRINTA$
OK
RUN
A123EFGHIJKLMN
OK
```

2.8.9 MEM$

機　能　指定されたメモリーへ、文字式を転送します。

書　式

MEM$(アドレス，n)＝文字式

アドレス……メインメモリー内の先頭アドレス(0～&HFFFF)。
n……転送される文字数(0～255)。
文字式……転送しようとする文字式。

省略形　MEM.

解　説　指定したアドレスから始まるメインメモリー内の領域にn個の文字を転送します。

サンプルプログラム

```
MEM$(&HFF00,5)="12345"
OK
?MEM$(&HFF00,5)
12345
OK
```

←メインメモリー内の&HFF00からのアドレスに12345の5個の文字を転送します。

←メインメモリー内の&HFF00から5文字分とりだし表示します。

(3.4.1 MEM$参照)

2.8.10 MON

機　能　制御をBASIC内蔵の機械語モニターに移します。

書　式

MON

省略形　MO.

解　説　この命令は、BASIC内蔵の機械語モニターを起動します。

この機械語モニターについては「A.3.1モニターのコマンド」を参照してください。

サンプルプログラム

```
MON
*D 0000 003F
:0000=C3 FA 00 C3 7C 01 50 28 /ﾃ･.ﾃ|.P(
:0008=C3 D4 03 C3 84 04 18 05 /ﾃﾔ.ﾃ■.ﾘ.
:0010=C3 D4 03 C3 BD 04 00 18 /ﾃﾔ.ﾃｽ...
:0018=C3 D4 03 C3 9D 02 00 27 /ﾃﾔ.ﾃﾝ..'
:0020=C3 D4 03 C3 07 0E 07 20 /ﾃﾔ.ﾃ...
:0028=C3 D4 03 C3 AA 0A 0D B7 /ﾃﾔ.ﾃｪ..ｷ
:0030=C3 D4 03 C3 CA 0A 00 FF /ﾃﾔ.ﾃﾊ..ﾃ
:0038=C3 D4 03 C3 75 0B C3 79 /ﾃﾔ.ﾃu.ﾃy
*R
OK
```

2.8.11 HCOPY

機　能　画面をラインプリンターに出力します。

書　式

```
HCOPY〔n〕
```

n…0～4の整数

省略形　H.

解　説　この命令は画面をラインプリンターにコピーするものであり、nの値は0～4です。

nの値

0　グラフィック1，2，3全画面を合成したものをコピーします。
1　グラフィック1の画面をコピーします。
2　グラフィック2の画面をコピーします。
3　グラフィック3の画面をコピーします。
4　テキスト画面と、グラフィック1，2，3全画面を合成したものをコピーします。
省略　テキスト画面をコピーします。

（注意）
テキスト画面の中で、CFLASHは無視され倍文字キャラクターは正しくコピーできません。また、ユーザーキャラクタジェネレーターは青・赤・緑の各パターンの合成されたものとなります。

サンプルプログラム

```
HCOPY
OK
HCOPY 0
OK
```

コピーされた文字のサイズは下のようになります。

普通の活字のサイズ
（LPRINTやLLISTによる活字サイズ）

ABC

コピー文字のサイズ
WIDTH40のとき

ABC

WIDTH80のとき

ABC

2.8.12 BOOT

機　能　IPLを起動します。

書　式

```
BOOT
```

省略形　BO.

解　説　このコマンドを実行したら、その前の状態には戻れません。
あくまでもIPLの管理下になります。
BOOTとキーボードから打ち込むと「Are you sure ? (y or n)」と表示され、キー入力を持っているので、実行したければ[Y]キーを押し、中止したければそれ以外のキーを押してください。[Y]キーを押すと、電源投入直後の状態になり、プログラムがすべて消去されます。
プログラム中でBOOTを実行しても「Are you sure? (y or n)」と聞いて来ないで直接実行してしまいますので注意してください。

サンプルプログラム

```
BOOT
Are you sure ? (y or n)N
OK
```

2.8.13 PAUSE

機　能　プログラムの実行を一時休止します。

書　式

```
PAUSE n
```

n……0～255

省略形　PA.

解　説　このステートメントを実行すると、
設定時間分プログラムの実行を休止して、次に進みます。休む時間は1単位につき約0.1秒です。

サンプルプログラム

```
TIME=0:PAUSE 100:?TIME
 10
OK
```

10秒間プログラムの実行を休止します。

2.8.14 WAIT

機　能　ポートからの条件待機をします。

書　式

```
WAIT ポート番号、 x1 〔, x2 〕
```

ポート番号……&H0000～&HFFFF

x1 , x2……数式。0～255

省略形　WA.

解　説　(ポート番号で指定した)入力ポートから読み込んだデータと x2 とのエクスクルーシブオア(注)(exclusive or「1.10.4 論理演算子」参照) をとり、その結果と x1 とのANDをとります。そしてもし結果が0のときは、もう1度入力ポートからのデータを読み込んで、前述と同じ操作を繰り返します。結果が0でないときはWAITの次の文へ実行を移します。
このようにして、WAIT文は入力ポートが、ある状態になるまで、プログラムの実行を停止して待ちます。
また、X2を省略するとX2には0がセットされます。

サンプルプログラム

```
WAIT &HC0, I, J
```

注) エクスクルーシブオァ(排他的論理和〔XOR〕)……2つの2進数の各ビットを比較し、ビット内容が同じ時0を、ビット内容が異なる時1を与えます。
(例) A＝&B10110010, B＝&B11011001について
A XOR B＝&B01101011

2.9 タイマー制御ステートメント

2.9.1 TIME$

機　能　時間の設定をします。

書　式

```
TIME$＝"hh：mm：ss"
```

h h……時間 00～23
m m……分　00～59
s s……秒　00～59

省略形　なし。

解　説　ユーザーが時間を設定するのに使うステートメントです。

サンプルプログラム

```
TIME$="12:34:56"
Ok
?TIME$
12:34:57
Ok
```

2.9.2 DAY$

機　能　曜日の設定をします。

書　式

```
DAY$＝"XXX"
```

XXX……曜日

省略形　DAY.

解　説　ユーザーが曜日を設定するときに使うステートメントです。

SUN　日曜日
MON　月　〃
TUE　火　〃
WED　水　〃
THU　木　〃
FRI　金　〃
SAT　土　〃

サンプルプログラム

```
DAY$="SUN"
Ok
?DAY$
SUN
Ok
```

2.9.3 DATE$

機　能　年月日の設定をします。

書　式

```
DATE$="yy/mm/dd"
```

y y……年　00～99
m m……月　01～12
d d……日　01～31

省略形　DATE.

解　説　ユーザーが年月日を設定するときに使うステートメントです。初期状態で年は？？がセットされますので設定をしなおす必要があります。

サンプルプログラム

```
DATE$="82/12/29"
Ok
?DATE$
82/12/29
Ok
```

2.9.4 TIME

機　能　秒数の設定をします。

書　式

```
TIME=n
```

n…秒数　0～86399

省略形　TIM.

解　説　ユーザーが秒数を設定するのに使います。

サンプルプログラム

```
LIST
10 PRINT TIME$
20 TIME=0
30 FOR I%=0 TO 10000:NEXT
40 PRINT TIME
50 PRINT TIME$
Ok
RUN
14:16:08
 3
14:16:11
Ok
```

2.9.5 ASK

機　能　会話型のタイマー設定ルーチンを呼び出します。

書　式

```
ASK
```

省略形　AS.

解　説　会話タイプのタイマーの設定ルーチンを呼び出します。これによって、7つのタイマーのセットができます。ただし、WIDTH40のSCREEN 0 , 0 に設定されている必要があり、それ以外のときASKを実行するとBad screen modeエラーが出ます。(本機専用のディスプレイテレビにのみ有効な機能です。)

サンプルプログラム　ASK [↵] と入力すると、画面に下の図が表示されますので画面メーセージに従ってタイマーを設定してください。

年は設定できますが変化しません。初期状態では××年が設定されます。(ウルウ年の判別はできませんので、ウルウ年がある場合は再設定必要です。)

```
TV Timer control

      82/11/10 WED 10:34:54

TIMER1    XX/XX XXX XX:XX    OFF
TIMER2    XX/XX XXX XX:XX    OFF
TIMER3    XX/XX XXX XX:XX    OFF
TIMER4    XX/XX XXX XX:XX    OFF
TIMER5    XX/XX XXX XX:XX    OFF
TIMER6    XX/XX XXX XX:XX    OFF
TIMER7    XX/XX XXX XX:XX    OFF

          Month  01 - 12 or XX

[ESC]=Exit, [CLR]=Reset, [CR]=Set
```

2.10 テレビ制御ステートメント

本機専用のディスプレイテレビにのみ有効な機能です。

2.10.1 TVPW

機　能　テレビの電源のon/offを行います。

書　式

```
TVPW {ON }
     {OFF}
```

省略形　TV.O.
TV.OF.

解　説　専用ディスプレイテレビの電源のスイッチを入れたり切ったりする命令です。ＯＮのときはテレビのモードなので、コンピュータのディスプレイとして使用するときはCRT(CRT文参照)でモードを設定する必要があります。

サンプルプログラム

```
TVPW ON:CRT 1
OK
```

(注意)　連続してテレビの電源スイッチを入れたり切ったりすることは、テレビ故障の原因となりますので、絶対にやめてください。

2.10.2 CRT

機　能　画面表示の設定をします。

書　式

```
CRT n
```

n……0 , 1 , 2 , 3

省略形　なし。

解　説　このステートメントの実行によって画面表示の切り換えを行うことができます。

n の値

0　テレビ放送を表示します。
1　コンピュータ画面を表示します。
2　テレビ放送のコントラストを下げて、コンピュータ画面を重ねて表示します。
3　テレビ放送とコンピュータ画面を同時に重ねて表示します。

サンプルプログラム

```
CRT 2
Ok
```

2.10.3 CHANNEL

機　能　テレビのチャンネルを変えます。

書　式

```
CHANNEL n
```

n…テレビのチャンネル。1～12。

省略形　CHAN.

解　説　テレビ放送のチャンネルを指定します。(このチャンネルは放送局のチャンネルナンバーとは関係ありません。専用ディスプレイテレビの取扱説明書を参照してください。)

サンプルプログラム

```
CHANNEL 12
Ok
CHANNEL 4
Ok
```

2.10.4 VOL

機　能　テレビの音量を変えます。

書　式

```
VOL〔n〕
```

n……—62～63までの整数。

省略形　VO.

解　説　VOLは、相対的に音量を変化させる命令です。
引数を省略すると、標準の音量、
n＜0のとき音量はダウン。
n＞0のとき音量はアップ。
n＝0　のとき、ミュートのon/offが反転します。

サンプルプログラム

```
VOL  1
OK
VOL-1
OK
VOL  1
OK
VOL  0
OK
```

2.10.5 SCROLL

機　能　画面のスクロールを行う。

書　式

```
SCROLL n
```

n……—3～3までの整数

省略形　SCRO.

解　説　テレビ放送とコンピュータ画面が重なった状態のとき（CRT2、CRT3のとき）、コンピュータ画面の上下方向のスクロールを行います。

nの値が正の数のときは上方向に、負の数のときは下方向にスクロールし、その速さは上下ともに3段階に設定されていて、nの絶対値が大きいほど速くスクロールします。nの値が0のとき、または省略すると、スクロールはストップします。SHIFT＋BREAKキーを押してもスクロールはストップします。

サンプルプログラム

```
SCROLL -1
OK
SCROLL 1
Ok
SCROLL 0
OK
```

2.11 カセット制御ステートメント

2.11.1 EJECT

機　能　カセットのドアを開けます。

書　式

```
EJECT
```

省略形　EJ.

解　説　カセットのドアを開けて、画面は入力待ちの状態になります。
カセットがFAST (2.11.3 FAST文参照), REW(2.11.4 REW文参照)の状態にあるときは、CSTOP (2.11.2 CSTOP文参照) が行われてから、EJECTが実行されます。

サンプルプログラム

```
EJECT
```

2.11.2 CSTOP

機　能　カセットテープの走行をストップします。

書　式

```
CSTOP
```

省略形　CST.

解　説　テープ走行をストップさせます。FAST (2.11.3 FAST文参照), REW(2.11.4 REW文参照)などの命令により、カセットが作動しているときそれを止める命令です。ただし、LOAD (2.1.15 LOAD文参照) やSAVE (2.1.16 SAVE文参照) をしているとき、あるいはAPSS(2.11.5 APSS文参照)の途中では実行しません。

サンプルプログラム

```
CSTOP
```

2.11.3 FAST

機　能　カセットテープ走行の早送りをします。

書　式

```
FAST
```

省略形　FA.

解　説　カセットテープ走行を早送りの状態にします。カセットテープが入っていない場合は何もしません。
(注) プログラムモードでは、早送りの状態にしてすぐ次のステートメントを実行します。

サンプルプログラム

```
FAST
```

2.11.4 REW

機　能　カセットテープの巻き戻しをします。

書　式

```
REW
```

省略形　なし。

解　説　カセットテープを巻き戻しの状態にします。カセットテープが入っていない場合は何もしません。
(注) プログラムモードでは、巻き戻しの状態にしてすぐ次のステートメントを実行します。

サンプルプログラム

```
REW
```

2.11.5 APSS

機　能　カセットテープのプログラムの頭出しをします。

書　式

```
APSS n
```

n …… −50～50の整数

省略形　AP.

解　説　カセットテープのプログラムの頭出しをする命令で、nが0のときは何もせず、正の数のときFAST方向n番目の頭出しを、負の数のときREW方向の頭出しをします。

ダイレクト実行のAPSS中は、残り1ブロックになると、画面上でカーソルが点滅を始めますが、キー入力は受けつけられません。ただし、押されたキーは7文字まで記憶されるので、APSSが終わりしだい処理されます。SHIFT＋BREAKキーか、カセットキーを押すことにより途中で実行を止めることができます。

サンプルプログラム

```
APSS -1
```

2.11.6 CMT

機　能

カセットの状態の設定を行います。

書　式

```
CMT＝n
```

n ……0～6,10

省略形

CM.

解　説

nの値によってカセットの状態を設定します。(3.3.11 CMT参照)

nの値	カセットの状態
0	EJECT
1	STOP
2	READ
3	FF
4	REW
5	APSS+1
6	APSS−1
10	WRITE

2～10はカセットが入っていないと何もしません。
また、10はプログラムのツメが折ってあると何もしません。
カセットテープの消去はCMT＝10として下さい。

サンプルプログラム

```
LIST
10 CMT=4                          ←― カセットをREW状態にします。
20 IF CMT<>1 THEN 20              ←― カセットがSTOPするまで待ちます。
30 FOR I=1 TO 5                   ┐
40 CMT=5                          ├― 5個先のプログラムの頭出しをします。
50 IF CMT<>1 THEN 50              ┘
60 NEXT
70 END
Ok
```

サンプルプログラムは、カセットテープの先頭から5個目のプログラムの頭出しを自動的に行なうプログラムです。

(注) サンプルプログラムにおいて、20行目のTHENの後ろで20行目の頭へもどしておりますが、これを10行目にもどしますと、カセットのテープエンドで巻き戻しとオートストップをくり返すことがあります。これは、20行目で使用されているCMT関数の値をコンピュータがテープエンドを検知して修正するまでに10行目のREW命令の実行がされてしまい、CMT関数の値がREWの状態に設定されてしまうためです。
この様にREW命令やFAST, READ, FF, WRITE及びCMT＝4, CMT＝2, CMT＝3, CMT＝10といった命令は、それだけをくり返すようなループ中では使用しないでください。

2.12 サウンド制御ステートメント

2.12.1 BEEP

機　能　「ポッ」という音を出します。

書　式

```
BEEP 〔n〕
```

n……0 , 1

省略形　BE.

解　説　このステートメントは「ポッ」という音を出すためのものです。
パラメータが1のとき、チャンネルCから音を出します。
パラメータが0のとき、チャンネルA、B、Cすべての音を止めます。
パラメータを省略すると、BEEP1とBEEP0が続いて実行されたのと同様の処理が行われ、1回だけ音を出して止まります。チャンネルA、B、Cの音については、2 .12 . 5 SOUNDを参照してください。

サンプルプログラム

```
BEEP 1
Ok
BEEP 0
Ok
BEEP
Ok
```

2.12.2 MUSIC

機　能　音楽の演奏をします。

書　式

```
MUSIC 文字式
```

文字式……楽譜を表わすデータ

省略形　MU.

解　説　文字式で指定された楽譜に従って演奏します。文字式は、音程オクターブ、音の長さ、音量、和音を指定するもので、以下に説明する文字を並べたものです。
①音程…音程は下記のようにC～Bの文字を使って表されます。

音　程	指定方法
ド	C
ド＃(レb)	＃C
レ	D
レ＃(ミb)	＃D
ミ	E
ファ	F
ファ＃(ソb)	＃F
ソ	G
ソ＃(ラb)	＃G
ラ	A
ラ＃(シb)	＃A
シ	B

②和音…2重音や3和音を演奏するには、上声部と下声部の間にチャンネルセパレータ：(コロン)をはさみます。 "チャンネルA：チャンネルB：チャンネルC"

③長さ…音程の後に0～9の整数をつけて、音の長さを指定します。

音の長さ	対応する整数
1/8 (32分音符)	0
1/4 (16分音符)	1
3/8 (付点16分音符)	2
1/2 (8分音符)	3
3/4 (付点8分音符)	4
1 (4分音符)	5
1 1/2 (付点4分音符)	6
2 (2分音符)	7
3 (付点2分音符)	8
4 (全音符)	9

注) 音の長さは4分音符（整数5）を1としたときの相対的な値です。
整数の指定がない場合は前の音と同じ長さを意味します。

④休符…R0～R9で、休みの長さを指定します。

⑤オクターブ…O1～O8でオクターブの指定をします。
また、音符の前に＋をつけて1オクターブ上の音程を、－をつけて1オクターブ下の音程を表します。

⑥音量…V0～V15で音量の指定をします。(2.12.5 SOUNDを参照してください。)

サンプルプログラム

オクターブ4のドからオクターブ5のドまでドレミファソラシドを出力します。

```
MUSIC"V1204CDEFGAB05C"
OK
```

オクターブ4でドミソの和音を順に重ねていきます。

```
MUSIC"V1304CDEFG:V1204EFGAB:V1204GAB05CD"
OK
```

2.12.3 TEMPO

機　能　音楽演奏のテンポを指定します。

書　式

```
TEMPO n
```

n……30～7500

省略形　TE.

解　説　MUSIC文によって演奏される曲のテンポを指定します。指定された数値は1分間あたりの4分音符の拍数を示します。

サンプルプログラム

```
TEMPO 120
```

```
TEMPO 1600
```

2.12.4 PLAY

機　能　音楽を演奏します。

書　式

```
PLAY { n    }
     { 文字式 }
```

n……30～7500

文字式……(2.12.2 MUSIC文参照)

省略形　PL.

解　説　MUSIC文とTEMPO文の両方の機能を持ち、PLAYの後ろが数式の場合TEMPO文と同様の、文字式の場合MUSIC文と同様の機能を持ちます。

サンプルプログラム

```
PLAY"V1204CDEFGAB05C"
PLAY"V1304CDEFG:V1204EFGAB:V1204GAB05CD"
```

(MUSIC命令のサンプルと同じ動作をします。)

2.12.5 SOUND

機　能　PSG(プログラマブル・サウンドジェネレータ)を直接コントロールします。

書　式

SOUND　PSGのレジスタ番号、データ

レジスタ番号… 0～13
データ…… 0～255(もしくは2進で&B 00000000～&B11111111)

省略形　SO.

解　説　SOUND命令により、PSG内の14個のレジスタへデータを書き込んで、ゲームの効果音など様々な音を作り出す事が出来ます。
本機の内蔵しているPSGは、3つのトーンジェネレータ、ノイズジェネレータ、エンベロープジェネレータ、ミキサー、3つのボリュームコントローラーから構成されています。
そのブロック図を下に示します。



(注) 図中のR0～R13はレジスタ番号です。

そして各ブロックのコントロールは、レジスタ番号0～13で行ないます。
各レジスタの機能表を以下に示します。

レジスタ番号	レジスタ機能	ビット構成 7 6 5 4 3 2 1 0	設定データ値(15進)	
0	チャンネルA周波数	下位8ビットT_L(微調)	0(高)～255(低)	
1		上位4ビット T_H(粗調)	0(高)～15(低)	
2	チャンネルB周波数	下位8ビットT_L(微調)	0(高)～255(低)	
3		上位4ビット T_H(粗調)	0(高)～15(低)	
4	チャンネルC周波数	下位8ビットT_L(微調)	0(高)～255(低)	
5		上位4ビット T_H(粗調)	0(高)～15(低)	
6	ノイズ周波数	5ビットN	0(高)～31(低)	
7	チャンネル選択	IN/OUT IOB IOA / ノイズ C B A / トーン C B A	0～63 (各ビット共、0を設定すると選択される)	
8	チャンネルA音量	M 4ビット	0(小)～16(大)	各チャンネルとも M=0のとき 4ビット(0～15)の値で音量設定 M=1のとき(データ値16) 音量はエンベロープに依存し 4ビットデータは無効になる。
9	チャンネルB音量	M 4ビット	0(小)～16(大)	
10	チャンネルC音量	M 4ビット	0(小)～16(大)	
11	エンベロープ周期	下位8ビットE_L(微調)	0(小)～255(大)	
12		上位8ビットE_H(粗調)	0(小)～255(大)	
13	エンベロープ形状	4ビット	0～15	

(注) レジスタ番号7のビット7,6は、SOUND命令では使用しません。

レジスタ番号0～5では、A，B，C各チャンネルの音程の周波数を決定します。これはPSGの入力クロックである2MHzを16で割った周波数を、各チャンネルの12ビットデータ（Aチャンネルの場合には、レジスタ番号1の上位4ビットとレジスタ番号0の下位8ビット）で分周して発生させます。
求めたい周波数をf_T〔Hz〕とすると、

① $f_T=\dfrac{2\times10^6}{16\times T〔Hz〕}$

② $T=256\times T_H+T_L$

T ：12ビットデータによる分周値(10進)
T_H：上位4ビットデータによる分周値(10進)
T_L：下位8ビットデータによる分周値(10進)

例えば、1kHzの音を発生するには、

①式より $1000=\dfrac{2\times10^3}{16\times T}$

ですから $T=\dfrac{2000}{16\times1}=125$

これと②式より

$T_H=\left(\dfrac{125}{256}\right)$ の整数部＝ 0 ＝&B0000

$T_L=125-256\times T_H \quad =125=$&B01111101

となります。Aチャンネルに1kHzを設定するには、

```
SOUND 0,125:SOUND 1,0
```

とします。
又、100Hzの場合には、

$T=\dfrac{2\times10^6}{16\times100}=1250$

よって $T_H=\left(\dfrac{1250}{256}\right)$ の整数部＝ 4＝&B0100

$T_L=1250-256\times4 \quad =226=$&B11100010

となります。Bチャンネルに100Hzを設定するには、

```
SOUND 2,226:SOUND 3,4
```

とします。
レジスタ番号6では、ノイズの平均周波数を決定します。
ノイズ周波数の式は、求める周波数をf_N〔Hz〕とすると、

③ $f_N=\dfrac{2\times10^6}{16\times N〔Hz〕}$

N：レジスタ番号6の5ビットデータによる分周値(10進)

例えば、5kHzのノイズを発生するには、

③式より $N=\dfrac{2\times10^6}{16\times5000}=25=$&B11001

となります。
レジスタ番号7では、A，B，C各チャンネル毎にトーン／ノイズ／(トーン＋ノイズ)出力のイネーブル選択を行ないます。例えば、

```
SOUND 7,44   または、  SOUND 7,&B101100
```

と設定すると、Aチャンネルからはトーンだけ、Bチャンネルからはトーン＋ノイズ、Cチャンネルからはどちらも出力されません。
レジスタ番号8～10は、各チャンネルの音量を決定します。ただし、Mビット(5ビット目)の値により、4ビットデータによる0～15までの一定した音量が設定できるモード(M＝0)と、自動的に音量が0～15まで変化するエンベロープモード(M＝1)とを選択できます。
レジスタ番号11，12では、エンベロープの周波数を決定します。これはPSGの入力クロックを256で割った周波数を、16ビットデータで分周して発生させます。エンベロープ周波数をf_E〔Hz〕とすると、

④ $f_E = \dfrac{2 \times 10^6}{256 \times E}$〔Hz〕

⑤ $E = 256 \times E_H + E_L$

E ：16ビットデータによる分周値(10進)
E_H：レジスタ番号12の上位8ビットデータによる分周値(10進)
E_L：レジスタ番号11の下位8ビットデータによる分周値(10進)

例えば、$f_E = 0.5$〔Hz〕(エンベロープ周期$1/f_E = 2$〔秒〕)にするには、

④式から

$$E = \frac{2 \times 10^6}{256 \times 0.5} = 15625$$

したがって

$$E_H = \left(\frac{15625}{256} \right) \text{の整数部} = 61 = \&B\ 00111101$$

$$E_L = 15625 - 256 \times 61 = 9 = \&B\ 00001001$$

そして、エンベロープの形状をレジスタ番号13で決定します。データに対応したパターンを下図に示します。

R13の設定データ	エンベロープパターン
0～3	
4～7	
8	
9	
10	
11	
12	
13	
14	
15	

$\longleftrightarrow$ $1/f_E$＝エンベロープ周期

サンプルプログラム

次は、実際にSOUND命令で音を出してみます。下のプログラムを実行してください。

```
LIST
10 SOUND 0,125:SOUND 1,0                    'A ch Freq. Set
20 SOUND 6,25                               'Noise Freq. Set
30 SOUND 11,9:SOUND 12,61                   'Env. Freq. Set
40 SOUND 13,8                               'Env. Pat. Set
50 SOUND 8,15                               'A ch Vol. Set
60 SOUND 7,&B111110                         'A ch Tone Select
OK
```

Aチャンネルから1kHzの音が出ます。音を止めるには CTRL ＋ D を押して下さい。次に60行目を次のように再入力してください。

```
60 SOUND 7,&B110110                         'A ch T+N Select
```

Aチャンネルからは、音とノイズの両方が出力されます。次に50行目を次のように再入力してください。

```
50 SOUND 8,16                               'A ch Env. mode Set
```

Aチャンネルの音量にエンベロープがかかり、おもしろい音になります。
又、SOUND命令でエンベロープを指定することにより、PLAY（MUSIC）文による同一音程の演奏を音符毎に区切ることができます。

```
PLAY120:PLAY"V1504C5CDDEED"
```

上のPLAY文を、次のようにかえてみてください。ただし、文字式での音量指定（V）は省きます。

```
LIST
10 SOUND 11,125:SOUND 12,20:SOUND 13,0:SOUND 8,16
20 PLAY120:PLAY"O4C5CDDEED"
Ok
```

（注）チャンネルCを使用するときは、あらかじめCLICK OFF（2.8.5 CLICK OFF参照）してください。
というのは、キーのクリック音はサウンドジェネレータのCチャンネルを利用しているため、Cチャンネルを利用して和音等を出力した場合、命令どおりの音がでません。

2.13 書式指定（PRINT USING）

PRINT USINGによる書式指定は、次の書式指定子によってコントロールします。書式指定子には、数値型と文字型の2種類あります。

2.13.1 数値型

（1）#

表示したい最大の桁数を指定し、指定した桁数より数値の桁数が小さいときには右づめで表示されます。

```
LIST
10 A=2345
20 PRINT USING"#######";A                    （6桁の指定）
OK
RUN
   2345
OK
```

（2）

固定小数点の表示において、小数点の位置をそろえる指定で、小数点以下の桁指定も行えます。

```
10 A=23.456
20 PRINT USING"###.##";A
OK                                           （小数点以下第2位）
RUN
 23.46
OK
```

（3），

数値を3桁ごとに区切って、そこに,（カンマ）を入れて右づめで表示します。

```
10 A=23456789!
20 PRINT USING"###,###,###";A
OK                                           （3桁ごとにカンマをつける）
RUN
 23,456,789
OK
```

（4）＋と－

＋と－は、数値を表示するときに＋と－の符号をどこに表示するか指定します。

＋は#の前と後に、－は#の後のみに指定できます。

＋を指定した場合、正数の＋記号を出力します。

```
10 A=235:B=-440
20 PRINT USING"####+";A;B
30 PRINT USING"####-";A;B
40 PRINT USING"+####";A;B
OK
RUN
 235+ 440-
 235  440-
 +235 -440
OK
```

(5) **

桁指定をし、その範囲内でスペースがある場合、すべて＊で埋めるという指定をします。

＊自身も桁数に含まれます。

```
 10 A=235
 20 PRINT USING"**######";A
 OK
 RUN
 ****235
 OK
```

(6) ¥¥

数字の前に¥（円マーク）をつけます。¥自身も桁数に含まれます。

```
 10 A=235
 20 PRINT USING"¥¥######";A
 OK
 RUN
    ¥235
 OK
```

(7) **¥

この指定は、＊＊と¥¥を組み合せたもので、数字の前に¥をつけ、その範囲内でスペースがある場合、＊（アスタリスク）で埋めます。

```
 10 A=235
 20 PRINT USING"**¥####";A
 OK
 RUN
 ***¥235
 OK
```

(8) ^^^^

桁数指定の#の後ろに置くと、E表現、D表現のいずれも表示できるようになります。

```
 10 A=235:B#=.00345#
 20 PRINT USING"##.####^^^^";A;B#
 OK
 RUN
  2.3500E+02 3.4500D-03
 OK
```

(9) 1つのUSINGで複数の変数を記述すると、その変数すべてに対し、USING指定したとみなします。

```
 10 PRINT USING"####.###";101;2;10;-2;5.5;7.4
 OK
 RUN
  101.000    2.000   10.000   -2.000    5.500    7.400
 OK
```

2.13.2 文字型

（1）！

！（感嘆符）を指定すると、文字型変数にある文字列の最初の1文字のみ表示します。

```
LIST
 10 A$="123":B$="456"
20 PRINT USING"!";A$;B$
OK
RUN
 14
OK
```

（2）&　　&

指定文字列を& と& のスペースの個数＋2個分表示します。文字は、左づめで、スペースの個数＋2個以下のときは残りの右の部分をスペースで埋めます。

```
 10 A$="123":B$="ABCDEF"
20 PRINT USING"&   &";A$;B$
OK
RUN
 123 ABCD
OK
```

2.13.3 その他の型

＿（アンダーバー）

この後ろに伴った書式指定子1個を、単なる1文字とみなして表示します。

```
 10 PRINT USING"_####";10
OK
RUN
# 10
OK
```

2.13.4 文字定数の出力

USINGの書式指定子のほかの文字がある場合は、それをそのまま表示します。

```
10 A=235:B=9.5
20 PRINT USING"DATA  ###";A
30 PRINT USING"##.## Sec";B
OK
RUN
DATA  235
 9.50 Sec
OK
```

2.13.2 文字型

2.13.3 文字の操作

2.13.4 文字変数の出力

第3章 関数・システム変数

3.1 数値関数

3.1.1 SIN

機　能　数式のサインを与えます。

書　式

```
SIN（x）
```

x…サインを求める数式。単位はラジアンです。

省略形　なし

解　説　数式のサイン、sin Xが、この関数の値になります。

サンプルプログラム　πの値を3.14、PAI(1)、RAD(180)としたときのそれぞれについて、sin πの値を求めてみましょう。

```
LIST
10 PA=3.14
20 PB=PAI(1)
30 PC=RAD(180)
40 PRINT SIN(PA)
50 PRINT SIN(PB)
60 PRINT SIN(PC)
Ok
RUN
 1.5926531E-03
 0
 9.3132257E-10
Ok
```

1.5926531E－03は1.5926531×10^{-3}

9.3132257E－10は9.3132257×10^{-10}

という意味で、どちらもほとんど0に近い値といえます。

PAI関数、RAD関数については、その関数の項を参照してください。

3.1.2 COS

機　能　数式のコサインを与えます。

書　式

```
COS（x）
```

x…コサインを求める数式。単位はラジアンです。

省略形　なし

解　説　数式のコサイン、cos Xが、この関数の値になります。

サンプルプログラム

```
LIST
10 P=PAI(1)
20 PRINT COS(0)
30 PRINT COS(P)
RUN
 1
-1
Ok
```

PAIについては、PAI関数参照。

3.1.3 TAN

機　能　数式のタンジェントを与えます。

書　式

```
TAN（x）
```

x…タンジェントを求める数式。単位はラジアンです。

省略形　なし

数式のタンジェント、tan X が、この関数の値になります。

サンプルプログラム

```
LIST
10 A=.5
20 PRINT TAN(.235)
30 PRINT TAN(.235*A)
OK
RUN
 .2394237
 .11804375
OK
```

3.1.4 ATN

機　能　数式のアークタンジェントを与えます。

書　式

```
ATN（x）
```

x…アークタンジェントを求める数式。単位はラジアンです。

省略形　AT.

解　説　数式のアークタンジェント、arctan X が、この関数の値になります。

サンプルプログラム

```
LIST
10 A=1
20 X=ATN(A)
30 Y=X*4
40 PRINT X
50 PRINT Y
OK
RUN
 .78539816
 3.1415927
OK
```

3.1.5 ABS

機　能　数式の絶対値を与えます。

書　式

```
ABS（x）
```

x…絶対値を求める数式。

省略形　AB.

解　説　数式 x の絶対値| x |が、この関数の値になります。

サンプルプログラム

```
LIST
10 PRINT ABS(23)
20 PRINT ABS(-23)
30 PRINT ABS(10-33)
OK
RUN
 23
 23
 23
OK
```



3.1.6 SGN

機　能　数式の符号を与えます。

書　式

```
SGN（x）
```

x…数式。

省略形　SG.

解　説　数式 x によって、次の値がこの関数の値になります。

x ＞0 のとき　　1
x ＝0 のとき　　0
x ＜0 のとき　－1

サンプルプログラム

```
LIST
10 PRINT SGN(93)
20 PRINT SGN(0)
30 PRINT SGN(-93)
OK
RUN
 1
 0
-1
OK
```



注）○印はその点を含みません。
●印はその点を含みます。

3.1.7 INT

機　能　数式を超えない最大の整数を与えます。

書　式

```
INT（x）
```

x…数式。

省略形　なし

解　説　数式を超えない最大の整数[x]が、この関数の値になります。

サンプルプログラム

```
LIST
10 PRINT INT(9.3)
20 PRINT INT(-9.3)
30 PRINT INT(-.5)
OK
RUN
 9
-10
-1
OK
```



3.1.8 FIX

機　能　数式の整数部を与えます。

書　式

```
FIX（x）
```

x…数式。

省略形　なし

解　説　数式の整数部のみを取り出して、小数点以下を切り捨てた値が、この関数の値になります。数値の四捨五入にはCINT（3.1.10 CINT文参照）を使います。

サンプルプログラム

```
LIST
10 A=FIX(2.23)
20 B=FIX(-2.23)
30 C=FIX(A*B)
40 PRINT A,B,C
OK
RUN
 2           -2          -4
OK
```



3.1.9 FRAC

機　能　数式の小数部を与えます。

書　式

```
FRAC（x）
```

x…数式。

省略形　FR.

解　説　数式の小数部のみを取り出して、整数部を取り去った値が、この関数の値になります。

サンプルプログラム

```
LIST
10 A=23.565
20 B=FRAC(A)
30 C=FIX(A)
40 PRINT"A=";A
50 PRINT"B=";B
60 PRINT"C=";C
70 PRINT"B+C=";B+C
Ok
RUN
A= 23.565
B= .565
C= 23
B+C= 23.565
Ok
```

Aの値を23.565として、
Aの小数部をFRAC関数で求めて、Bに代入し、
Aの整数部をFIX関数で求めて、Cに代入します。
B＋Cを計算すると、もとのAの値になります。
注）整数部の桁数が大きい数のFRACは、誤差が大きくなります。

3.1.10 CINT・CSNG・CDBL

機　能

数式の型変換を行います。

書　式

CINT（x）

CSNG（x）

CDBL（x）

x…型変換を行う数式。

省略形

CIN.
CSN.
CD.

解　説

CINTはxを整数型に、（小数点第1位で四捨五入）
CSNGはxを単精度型に、(有効数字9桁目にて四捨五入)
CDBLはxを倍精度型に、
それぞれ変換した値になります。

サンプルプログラム

```
LIST
10 X%=CINT(23.54)
20 Y!=CSNG(1.234567899#)
30 Z#=CDBL(23.54)
40 PRINT X%;Y!;Z#
OK
RUN
 24   1.2345679   23.539999999910593
OK
```

変数や定数の最後に％をつけると整数型
　　！をつけると単精度型
　　＃をつけると倍精度型
の数値をそれぞれ表します。
上のサンプルプログラムでは、
23.54を整数の24に、
1.234567899を単精度8桁の1.2345679に、
23.54を倍精度16桁の23.53999999910593に、
それぞれ変換しています。

注）CDBLは単に数値の型を倍精度にするのみで、値そのものの精度は変わりません。値も倍精度にするには、VALとSTR$を用いて以下の様にして下さい。

A＃＝VAL(STR$(A))

3.1.11 SQR

機　能　数式の平方根を与えます。

書　式

```
SQR（x）
```

x…平方根を求める数式。0以上の値です。

省略形　SQ.

解　説　数式の平方根$\sqrt{x}$が、この関数の値になります。

サンプルプログラム　0から5までの平方根をSQR関数を使って求めてみましょう。

```
LIST
10 FOR I=0 TO 5
20 PRINT I;SQR(I)
30 NEXT
OK
RUN
 0  0
 1  1
 2  1.4142136
 3  1.7320508
 4  2
 5  2.236068
OK
```

3.1.12 EXP

機　能　数式の指数関数を与えます。

書　式

```
EXP（x）
```

x…指数関数を求める数式。x≦88.02969

省略形　EX.

解　説　自然対数の底eのx乗e^xが、この関数の値になります。

サンプルプログラム　e15とe0.000002345を求めてみます。

```
LIST
10 PRINT EXP(15)
20 PRINT EXP(.000002345#)
OK
RUN
 3269017.3
 1.00000234500275
OK
```

3.1.13 LOG

機　能　数式の自然対数を与えます。

書　式

```
LOG（x）
```

x…数式。x＞0

省略形　なし

解　説　数式の自然対数ln x(＝log eX)が、この関数の値になります。log B Aはlog(A)/log(B)で求めることができます。ただし、B＞0，B≠1。

サンプルプログラム

公式

$\ln A + \ln B = \ln(AB)$

$\ln A - \ln B = \ln\left(\frac{A}{B}\right)$

を実際に確かめてみましょう。

```
LIST
10 A=2
20 B=5
30 PRINT LOG(A),LOG(B)
40 PRINT LOG(A)+LOG(B),LOG(A*B)
50 PRINT LOG(A)-LOG(B),LOG(A/B)
Ok
RUN
 .69314718            1.6094379
 2.3025851            2.3025851
-.91629073           -.91629073
Ok
```

3.1.14 PAI

機　能　円周率πのx倍を与えます。

書　式

```
PAI（x）
```

x…数式。

省略形　なし

解　説　円周率πのx倍が、この関数の値になります。

球の表面積を求めるプログラムを作ってみましょう。

サンプルプログラム

球の表面積Sは、　$S = 4\pi r^2$

で求められ、下のサンプルでは、2通りの式を使って計算しています。

```
LIST
10 'キュウ ノ ヒョウメンセキ
20 INPUT"ハンケイ=";R
30 VA=4*π*R*R
40 VB=PAI(4)*R^2
50 PRINT"メンセキ=";VA;VB
Ok
RUN
ハンケイ=? 10
メンセキ= 1256.6371   1256.6371
Ok
```

注）PAI(1)の代わりに、システム変数のπ(パイ)を使うことができます。

πは単精度で3.1415927ですが、π#として倍精度型にすると、3.141592653589793として使えます。

(πは GRAPH ＋ A キーを押すと表示されます。)

3.1.15 RAD

機　能

度単位の数式をラジアン単位に変換します。

書　式

```
RAD（x）
```

x…ラジアン単位に変換する数式。単位は度です。

省略形

なし

解　説

度単位の数式を$\frac{\pi}{180}$倍して、ラジアン単位に変換します。三角関数の値を求めるときに使うと便利です。

サンプルプログラム

0°，30°，45°，60°，90°のおのおのについてsinとcosの値を求めたプログラムを下に示します。

```
LIST
10 A=RAD(0)
20 B=RAD(30)
30 C=RAD(45)
40 D=RAD(60)
50 E=RAD(90)
60 PRINT USING"##.####";SIN(A);COS(A)
70 PRINT USING"##.####";SIN(B);COS(B)
80 PRINT USING"##.####";SIN(C);COS(C)
90 PRINT USING"##.####";SIN(D);COS(D)
100 PRINT USING"##.####";SIN(E);COS(E)
OK
RUN
 0.0000 1.0000
 0.5000 0.8660
 0.7071 0.7071
 0.8660 0.5000
 1.0000 0.0000
OK
```

3.1.16 FAC

機　能

数式の階乗を与えます。

書　式

```
FAC（n）
```

n…階乗を求める数式。0～33の整数。

省略形

なし

解　説

数式の階乗x！が、この関数の値になります。

サンプルプログラム

0から10までの各整数について、階乗計算をしてみましょう。

```
LIST
10 FOR I=0 TO 10
20 PRINT I;FAC(I)
30 NEXT
OK
RUN
 0   1
 1   1
 2   2
 3   6
 4   24
 5   120
 6   720
 7   5040
 8   40320
 9   362880
 10   3628800
OK
```

3.1.17 SUM

機　能　数式 n の$\sum_{i=1}^{n} i$ を与えます。

書　式

```
SUM（n）
```

n …数式。0 以上の整数値です。

省略形　なし

解　説　$\sum_{i=1}^{n} i = 1+2+\cdots+(n-1)+n=\frac{1}{2}n(n+1)$ が、この関数の値になります。

サンプルプログラム

```
LIST
10 FOR I=0 TO 10
20 PRINT I;SUM(I)
30 NEXT
OK
RUN
 0  0
 1  1
 2  3
 3  6
 4  10
 5  15
 6  21
 7  28
 8  36
 9  45
 10  55
OK
```

3.1.18 RND

機　能　0 以上 1 未満の乱数を発生します。

書　式

```
RND〔（x）〕
```

x …数式。

省略形　RN.

解　説　発生した擬似乱数が、この関数の値になります。この乱数は、0 以上 1 未満の乱数になっています。

サンプルプログラム　乱数を10個発生させてみましょう。

```
LIST
10 FOR I=1 TO 10
20 PRINT I;RND(1)
30 NEXT
OK
RUN
 1   .6475668
 2   .58308077
 3   .17951131
 4   .14283574
 5   .13336551
 6   .60755396
 7   .48844886
 8   .68064535
 9   .71988237
 10   .13614893
OK
```

3.2 文字関数

3.2.1 ASC

機　能　文字をキャラクタコードに対応する数値に変換します。(⇐⇒ 3.2.2 CHR$)

書　式

```
ASC（文字式）
```

文字式……数値に変換する文字が左端にある文字列。

省略形　なし。

解　説　文字式の最初の1文字のキャラクタコードが、この関数の値になります。キャラクタコードの値は0～255の整数値です。(巻末「キャラクタコード表」参照) また、文字式がヌルストリングのときに、この関数の値は0になります。

サンプルプログラム　"A","B","C" をＡＢＣ のキャラクタコードに変換します。"ＡＢＣ" のときは最初の "A" のみコードに変換します。

```
LIST
10 A$="A":B$="B":C$="C":D$="ABC"
20 X=ASC(A$):Y=ASC(B$):Z=ASC(C$):Q=ASC(D$)
30 PRINT X;Y;Z;Q
Ok
RUN
 65  66  67  65
Ok
```

3.2.2 CHR$

機　能　数値をキャラクタコードとみなして、対応する文字に変換します。(⇐⇒ 3.2.1 ASC)

書　式

```
CHR$（x1〔, x2, x3,……〕）
```

x_1, x_2, x_3,……文字に変換する数式。
各0～255

省略形　CHR.

解　説　数値をキャラクタコードとする文字が与えられます。数値1つにつき1文字が得られますが、数値を1つずつ ,(カンマ) で区切って入れると、その数値の並びに対応する文字列が得られます。

サンプルプログラム　キャラクタコードについては巻末の「A.4.1キャラクタコード表」を参照してください。

キャラクタコードを文字に変換します。

65→A
60→B
70→L，79→O，86→V，69→E，76→L，89→Y，33→！
と各コードに対応する文字が得られます。

```
LIST
10 X$=CHR$(65)
20 Y$=CHR$(66)
30 Z$=CHR$(76,79,86,69,76,89,33)
40 PRINT X$,Y$,Z$
Ok
RUN
A           B           LOVELY!
Ok
```

3.2.3 VAL

機　能　数字を表わす文字式を数値に変換します。（⇔3.2.4 STR$）

書　式

```
VAL（文字式）
```

文字式……数字を表わす文字式。

省略形　VA.

解　説　文字式の中の文字としての数字を数値に変換します。もし文字式の最初の文字が＋，－，＆または数字でないときには、この関数の値は0になります。

サンプルプログラム

```
LIST
10 A=VAL("2356")
20 B=VAL("&H"+"EFC0")
30 C=VAL("&O"+"177666")
40 D=VAL("&B"+"1010000011011")
50 A$=STR$(A)
60 B$=HEX$(B)
70 C$=OCT$(C)
80 D$=BIN$(D)
90 PRINT A,A$
100 PRINT B,B$
110 PRINT C,C$
120 PRINT D,D$
OK
RUN
 2356         2356
-4160        EFC0
-74          177666
 2587        1010000011011
OK
```

STR$, HEX$, OCT$, BIN$についてはその関数の項を参照してください。

3.2.4 STR$

機　能　数式の値を文字の数字に変換します。（⇔3.2.3 VAL）

書　式

```
STR$（x）
```

x……数式。

省略形　なし。

解　説　数式および数値を表す文字式が、この関数の値になります。
正の数の場合、先頭にスペースが一文字はいります。

サンプルプログラム

```
LIST
10 PRINT STR$(0)
20 PRINT STR$(100)
30 PRINT STR$(255)
OK
RUN
 0
 100
 255
OK
```

3.2.5 HEX$

機　能　数式を16進数の文字列に変換します。

書　式

```
HEX$（x）
```

x……数式。 -65535～65535。（ただし、-65535～-1は1～65535と同じです。）

省略形　HE.

解　説　数式を16進数の文字列に変換したものが、この関数の値になります。

サンプルプログラム

```
LIST
10 PRINT HEX$(-32768!)
20 PRINT HEX$(0)
30 PRINT HEX$(235)
40 PRINT HEX$(65535!)
OK
RUN
8000
0
EB
FFFF
OK
```

3.2.6 OCT$

機　能　数式を8進数の文字列に変換します。

書　式

```
OCT$（x）
```

x……数式。 -65535～65535。（ただし、-65535～-1は1～65535と同じです。）

省略形　OC.

解　説　数式を8進数の文字列に変換したものが、この関数の値になります。

サンプルプログラム

```
LIST
10 PRINT OCT$(-32768!)
20 PRINT OCT$(0)
30 PRINT OCT$(235)
40 PRINT OCT$(65535!)
OK
RUN
100000
0
353
177777
OK
```

3.2.7 BIN$

機　能

数式を2進数の文字列に変換します。

書　式

```
BIN$(x)
```

x……数式。-65535～65535。(ただし、-65535～-1は、1～65535と同じです。)

省略形

BI.

解　説

数式および数値を2進数の文字列（0，1からなる）に変換したものが、この関数の値になります。

サンプルプログラム

```
LIST
10 PRINT BIN$(-32768!)
20 PRINT BIN$(0)
30 PRINT BIN$(235)
40 PRINT BIN$(65535!)
Ok
RUN
1000000000000000
0
11101011
1111111111111111
Ok
```

〈HEX$, OCT$, BIN$および定数について〉



10進数XをHEX$，OCT$，BIN$をもちいて変換すると次の表のようになります。

X(10進)	HEX$(X)	OCT$(X)	BIN$(X)
-65535	1	1	1
\	\	\	\
-32768	8000	100000	1000000000000000 (16ケタ)
\	\	\	\
-1	FFFF	177777	1111111111111111 (16ケタ)
0	0	0	0
1	1	1	1
\	\	\	\
32768	8000	100000	1000000000000000 (16ケタ)
\	\	\	\
65535	FFFF	77777	1111111111111111 (16ケタ)

16進数、8進数、2進数をVAL関数をもちいて10進数に変換すると次の表のようになります。

16進(&H)	8進(&O)	2進(&B)	10進
-&HFFFF	-&O177777	-&B1111111111111111 (16ケタ)	1
\	\	\	\
-&H8001	-&O100001	-&B1000000000000001 (16ケタ)	32767
-&H8000	-&O100000	-&B1000000000000000 (16ケタ)	-32768
\	\	\	\
-&H1	-&O1	-&B1	-1
&H0	&O0	&B0	0
&H1	&O1	&B1	1
\	\	\	\
&H7FFF	&O77777	&B111111111111111 (15ケタ)	32767
&H8000	&O100000	&B1000000000000000 (16ケタ)	-32768
\	\	\	\
&HFFFF	&O177777	&B1111111111111111 (16ケタ)	-1

3.2.8 LEFT$

機　能　文字式の左側から、指定した数だけ文字を取り出します。(⇔3.2.9 RIGHT$)

書　式

```
LEFT$(文字式,n)
```

n…文字式から取り出す文字数。0～255

省略形　LEF.

解　説　文字式の左側から取り出されたn個の文字列が、この関数の値になります。nの値が文字式の文字数より大きいときには、すべての文字が、n＝0のときには、ヌルストリングがそれぞれ関数の値になります。

サンプルプログラム

```
LIST
10 A$="ABCDEFGHI"
20 PRINT LEFT$(A$,10)
30 PRINT LEFT$(A$,4)
40 PRINT LEFT$(A$,0)
OK
RUN
ABCDEFGHI
ABCD

OK
```

3.2.9 RIGHT$

機　能　文字式の右側から、指定した数だけ文字を取り出します。(⇔3.2.8 LEFT$)

書　式

```
RIGHT$(文字式,n)
```

n…文字式からとりだす文字数。0～255

省略形　RI.

解　説　文字式の右側から取り出されたn個の文字列が、この関数の値になります。nの値が文字式の文字数より大きいときには、すべての文字が、n＝0のときには、ヌルストリングがそれぞれ関数の値になります。

サンプルプログラム

```
LIST
10 A$="ABCDEFGHI"
20 PRINT RIGHT$(A$,10)
30 PRINT RIGHT$(A$,0)
40 PRINT RIGHT$(A$,4)
OK
RUN
ABCDEFGHI

FGHI
OK
```

3.2.10 MID$

機　能　文字式の中から、指定した数だけ一連の文字を取り出します。(⇒3.2.8 LEFT$、3.2.9 RIGHT$)

書　式

MID$（文字列，開始位置[，n]）

開始位置…文字式の最初の文字の位置を1とする文字の位置。
n……文字式から取り出す文字の数。0～255

省略形　MI.

解　説　文字式の中の開始位置から取り出されたn個の文字列が、この関数の値になります。nを省略したとき、および文字式の文字数よりnが大きいときには、開始位置から右端までの文字列が、関数の値になります。開始位置が文字式の文字数より大きいときには、ヌルストリングが関数の値になります。
（この命令は2．8．8　MID$とは異なります。）

サンプルプログラム

```
LIST
10 A$="ABCDEFGHI"
20 PRINT MID$(A$,3,5)
30 PRINT MID$(A$,8)
40 PRINT MID$(A$,3,20)
OK
RUN
CDEFG
HI
CDEFGHI
OK
```

3.2.11 STRING$

機　能　任意の文字式を任意の数だけ用意します。

書　式

STRING$（n，{文字式 / 数　式}）

n……文字式の数。0～255
数式……0～255のアスキーコードとみなされます。

省略形　STRIN.

解　説　文字式または数式で指定した文字列をn個並べて用意します。

サンプルプログラム

```
LIST
10 A$="ABCDE"
20 PRINT STRING$(3,A$)
OK
RUN
ABCDEABCDEABCDE
OK
```

STRING$（3，A$）は、A$＋A$＋A$と同じです。

3.2.12 SPACE$

機　能　スペースからなる文字列を与えます。

書　式

```
SPACE$(n)
```

n…スペースの数。0～255

省略形　SPA.

解　説　n個のスペース(空白)からなる文字列が、この関数の値になります。

サンプルプログラム

```
LIST
10 PRINT "A";SPACE$(10);"A"
OK
RUN
A          A                    ← AとAの間に10個スペースを表示しています。
OK
```

3.2.13 INSTR

機　能　文字式の中に含まれる特定の文字式の位置を検出します。

書　式

```
INSTR([開始位置,]文字式1,文字式2)
```

開始位置…探し始める文字の位置。1～255
文字式1の最初の文字の位置を1とします。

省略形　INS.

解　説　文字式1の中を、開始位置から順に右へ、文字式2があるかどうか探して行きます。もしみつかれば、その先頭の文字の位置を関数の値とし、みつからなければ、関数の値を0にします。開始位置を省略すると、文字式1の最初から探し出します。

サンプルプログラム

```
LIST
10 A$="SHARP HuBASIC"
20 PRINT INSTR(A$,"Hu")          ← SHARP HuBASICという文字中でHuのある
OK                                  開始位置を表示させています。
RUN
 7
OK
```

3.2.14 LEN

機　能　文字式の文字数を与えます。

書　式

```
LEN(文字式)
```

省略形　なし

解　説　文字式に含まれるすべての文字の数が、この関数の値になります。文字の数は0～255で、文字式がヌルストリングのときは0になります。なお、空白やコントロールコードなどの画面表示されない文字も1文字として数えます。

サンプルプログラム

```
LIST
10 A$="ABCDEFGHIJKLMNOPQRSTUVWXYZ"
20 PRINT LEN(A$)
OK
RUN
 26
OK
```

アルファベットの文字列の長さは26文字です。

3.2.15 HEXCHR$

機　能　16進数を表わす文字式をキャラクタコードとみなして対応する文字列に交換します。

書　式

HEXCHR$(文字式)

文字式……16進数を表わす文字式

省略形　HEXC.

解　説　16進数を表わす文字式を最初から2文字(2バイト)ずつ区切って、その2文字に対応するキャラクタに変換して文字列を作ります。この文字列が、この関数の値になります。16進数の文字式の中にスペース(空白)があるときは、区切れとみなし、無視して次へ進みます。

サンプルプログラム

```
LIST
10 A$="1234567890ABCD"
20 PRINT HEXCHR$(A$)
OK
RUN
4Vx−オヘ
OK
```

"3 4 5 6 7 8 9 0 A B C D"を、
"34　56　78　90　AB　CD"と2文字づつ区切ってそれに対応する文字列

␣　␣　␣　␣　␣　␣
↓　↙　↙　↙　↙　↙
"4　V　X　−　オ　ヘ"　　を作ります。(1␣2に対応する文字は、表示されません。)

3.2.16 MIRROR$

機　能　文字式のバイナリーコードを転置した文字列を与えます。

書　式

MIRROR$(文字式)

省略形　MIR.

解　説　文字式のバイナリーコード(2進コード)を左右転置したコードに対応する文字列が、この関数の値になります。

サンプルプログラム

```
LIST
10 A$=CHR$(&H83)
20 PRINT BIN$(ASC(A$)),BIN$(ASC(MIRROR$(A$)))
OK
RUN
 10000011   11000001
OK
```

ユーザー定義のキャラクタなどのパターンを左右にひっくり返すときなど便利です。
(2.5.10 DEF　CHR$参照)

3.2.17 MKI$・MKS$・MKD$

機　能

数式をそのまま内部表現(バイナリィコード)に対応した文字列に変換します。
(⇔3.2.18 CVI、CVS、CVD)

書　式

MKI＄（整数値）

MKS＄（単精度値）

MKD$（倍精度値）

省略形

MK.
MKS.
MKD.

解　説

MKI$は整数値を2バイトの文字列に、
MKS$は単精度値を5バイトの文字列に、
MKD$は倍精度値を8バイトの文字列に、
それぞれ変換します。変換の仕方は、バイナリーコードをキャラクタコード(アスキーコード)とみなして対応する文字に変換します。
ファイルに対して数字を記録する場合に用いると、ファイルのスペースを節約できます。
(アスキーコードについては、付録A.4キャラクタコード表を参照)

サンプルプログラム

```
LIST
10 A%=-32:B!=3.141592:C#=1.2345678901234 56#
20 A$=MKI$(A%)
30 B$=MKS$(B!)      ← 各数値のバイナリーコードを文字列に変換。
40 C$=MKD$(C#)
50 X%=CVI(A$)
60 Y!=CVS(B$)       ← 各文字列をバイナリーコードに変換。
70 Z#=CVD(C$)
80 PRINT A%;B!;C#
90 PRINT#0,A$:PRINT#0,B$:PRINT#0,C$
100 PRINT X%;Y!;Z#
OK
RUN
-32   3.141592   1.2345678901234 56
●〒
￣I ^Oラ◆
_↑ ^FR ^Tbマチ
-32   3.141592   1.2345678901234 56
Ok
```

サンプルプログラムは、整数−32のバイナリーコード&B1111111111100000（＝&HFFE0）ですのでそれに対応したアスキーコードの文字●(&HE0),〒(&HFF)に変更されます。ただし、メモリー上では、上位8ビットと下位8ビットの格納順が逆になっていますので、文字も逆の順番で表示されます。

単精度実数、3.141592のバイナリーコードは、16進数で表わすと82　49　0F　D7　E4の5バイトですので、文字￣I ^Oラ◆に変換されます。

倍数度実数、1.2345678901234 56のバイナリーコードは、16進数で表わすと81　1E　06　52　14　62　CF　C1の8バイトですので、文字_↑ ^FR ^Tbマチに変換されます。

3.2.18 CVI・CVS・CVD

機　能　文字式を数値に変換します。(⟺ 3.2.17 MKI$、MKS$、MKD$)

書　式

CVI（2文字の文字式）

CVS（5文字の文字式）

CVD（8文字の文字式）

省略形　CVI、CVDにはなし。
CVSの省略形はCV.です。

解　説　CVIは2文字(2バイト)の文字式を整数値に、
CVSは5文字(5バイト)の文字式を単精度実数値に、
CVDは8文字(8バイト)の文字式を倍精度実数値に、
それぞれ変換します。
ファイルから読み込んだ文字データを数値に変換する場合に、これらの関数を使うと便利です。

サンプルプログラム

```
LIST
10 A$="AB":B$="ABCDE":C$="ABCDEFGH"
20 A%=CVI(A$)
30 B!=CVS(B$)
40 C#=CVD(C$)
50 X$=MKI$(A%)
60 Y$=MKS$(B!)
70 Z$=MKD$(C#)
80 PRINT A$:PRINT B$:PRINT C$
90 PRINT A%;B!;C#
100 PRINT X$:PRINTY$:PRINTZ$
OK
RUN
AB
ABCDE
ABCDEFGH
 16961  8.2273479E-20  8.22734792741845 3D-20
AB
ABCDE
ABCDEFGH
OK
```

3.3 特殊関数

3.3.1 INP

機　能　I／Oポートから1バイトのデータを読み込みます。　(⇒3.3.3 PEEK@)

書　式

```
INP(ポートアドレス)
```

ポートアドレス……読み込むポートアドレス。0～&HFFFF

省略形　なし

解　説　ポートアドレスで指定したI／Oポートから読み込んだ1バイトのデータが、この関数の値になります。値は0～255です。

サンプルプログラム　画面のホーム位置(左上隅)にカーソルをセットして、「AB」と打ち込んでから、次のプログラムを実行してみてください。

```
AB
LIST
10 X=INP(&H3000)
20 Y=INP(&H3001)
30 PRINT X,Y
40 PRINT CHR$(X);SPC(9);CHR$(Y)
OK
RUN
 65          66
A           B
OK
```

画面(テキスト画面)は、I／Oポートの&H3000～&H37FFに割りつけられています。

3.3.2 PEEK

機　能　指定したメモリーアドレスからデータを取り出します。(⇔ 2.2.32 POKE)

書　式

PEEK（アドレス）

アドレス…メインメモリー内のアドレス。0～&HFFFF

省略形　PE.

解　説　メインメモリー内の指定したアドレスから取り出した1バイトのデータが、この関数の値になります。

サンプルプログラム　PEEK関数を使ってメモリーダンププログラムを作ってみましょう。
INPUT文で、読み込み開始アドレスと終了アドレスを指定すると、その領域内からデータを読み込んでメモリーのダンプ(dump)をします。
桁をそろえて表示するために、RIGHT$関数を使用しています。

```
10 INPUT"Start Address=&H",SA$
20 INPUT"End   Address=&H",EA$
30 SA=VAL("&H"+SA$)
40 EA=VAL("&H"+EA$)
100 PRINT RIGHT$("000"+HEX$(SA),4);"   ";
110 FOR I=SA TO SA+7
120  DT=PEEK(I)
130  PRINT RIGHT$("0"+HEX$(DT),2);" ";
140 NEXT
150 SA=SA+8
160 PRINT
170 IF SA<=EA THEN 100
```

3.3.3 PEEK@

機　能　VRAM内の指定されたアドレスの内容を読み出します。(⇒3.3.1 INP)

書　式

```
PEEK@（アドレス）
```

アドレス…VRAM内のアドレスで&H2000～&HFFFF。

省略形　PE. @

解　説　VRAM(Video Random Access Memory)の指定されたアドレスの内容1バイトが、この関数の値になります。値は、0～255の整数です。

サンプルプログラム　カーソルをホーム位置(画面左上隅)にもっていって、次のようにキーボードから入力してください。ホーム位置のアドレスは&H3000、その属性のアドレスは&H2000番地なので、?のキャラクタコードと属性のコードが得られます。(付録「A.1テキスト画面とその属性ポートへのアクセス方法」参照)

```
? HEX$(PEEK@(&H3000))
3F
Ok
? PEEK@(&H2000)
 7
Ok
```

3.3.4 POS

機　能　画面のカーソルの水平位置を与える。

書　式

```
POS（0）
```

省略形　なし

解　説　画面上の現在のカーソルの水平位置の値(WIDTH40のとき0～39、WIDTH80のとき0～79)が、この関数の値になります。

サンプルプログラム

```
10 CLS4:LOCATE 12,15
20 X=POS(0)
30 LOCATE 0,0
40 PRINT X
```

3.3.5 LPOS

機　能　プリンターヘッドの現在位置を与えます。

書　式

```
LPOS（x）
```

x……数式。(何を指定してもかまいません。)

省略形　LPO.

解　説　プリンターのヘッドの現在位置が、この関数の値になります。値の範囲は0～240です。この関数をうまく使うと、プリントの簡単な制御を行うことができます。
プリンターヘッドが40桁を越えると改行するプログラムを下に示します。

サンプルプログラム

```
10 FOR I=1 TO 50
20 IF LPOS(0)>40 THEN LPRINT
30 LPRINT I;
40 NEXT
50 LPRINT
```



LPRINTは、バッファに1行分のデータがたまると出力するようになっているので、50行目にLPRINTを入れてプリントの最終行のデータ(47～50の4つ)をバッファから吐き出すようにしています。

3.3.6 VARPTR

機　能　数値変数が格納されているメインメモリーのアドレスを与えます。

書　式

```
VARPTR（x）
```

x…数値変数

省略形　VAR.

解　説　変数の内容を記憶しているメインメモリーの先頭アドレスが、この関数の値になります。(3.6.2 STRPTR参照)

サンプルプログラム

```
A=213:AD=VARPTR(A):PRINT AD;HEX$(AD)
-24902 9EBA
Ok
```

変数Aの内容が格納されているアドレスは、&H9EBA番地であることを示しています。
Aの値が負の数となっているのは、ADが32768～65535のため補数表現になっているからです。
注) BASICのバージョン及び変数の状態、プログラムの有無によって値が違うことがあります。

3.3.7 FRE・SIZE

機　能　ユーザーメモリーの未使用領域のサイズを与えます。

書　式

```
FRE（x）
```

```
SIZE
```

x……数式。(何を指定してもかまいません。)

省略形　なし

解　説　BASICのプログラムで使っていないユーザーメモリーのバイト数が、この関数の値になります。

サンプルプログラム

```
? FRE(0)
 23536
Ok
? SIZE
 23536
Ok
```

注）BASICのバージョン，変数の状態及びプログラムの有無によって値が違ってきます。

3.3.8 POINT

機　能　画面の点(ドット)のパレットコードを与えます。

書　式

```
POINT（x, y）
```

x, y…画面上の論理座標(実数値)

省略形　POI.

解　説　画面上の1ドットのパレットコード(0～7)を与えます。パレットコードについてはPALET文を参照してください。WINDOW外の点は-1の値を与えます。

サンプルプログラム

```
IF POINT(X,Y)=2 THEN 100
```

上の文は「画面上の（X,Y）のドットがパレットコードの2だったら行番号100へジャンプせよ。」という意味です。

3.3.9 STICK

機　能　ジョイスティックまたはテンキーからの入力値を与えます。

書　式

```
STICK（x）
```

x……0，1，2
　　0…テンキー
　　1…ジョイスティック1
　　2…ジョイスティック2

省略形　STI.

解　説　ジョイスティックまたはキーボードのテンキーから入力された値が、この関数の値になります。
ジョイスティックに接続している場合には、スティックを倒す方向によって、下の図のような値が入力されます。



キーボード上のテンキーからは1～9の値のみ入力され、ほかのキーが押されたときには、0が入力されます。フルキーの中の1～9のキーでも1～9の値が入力されます。

サンプルプログラム

```
10 PRINT STICK(0);:GOTO10
```

RUNすると画面に連続的に0が表示されますから、テンキーから数字キーを入力して、入力値が正しいかどうか確めてください。
止めるときは、SHIFT ＋ BREAK キーを押すか、CTRL ＋ C キーを同時に押してください。「Break in 10」と表示して止まります。

3.3.10 STRIG

機　能　ジョイスティックのトリガーボタンまたはキーボードのスペースキーの状態を与えます。

書　式

```
STRIG（x）
```

x……0、1、2
　　0…キーボードのスペースキー
　　1…ジョイスティック1 のボタン
　　2…ジョイスティック2 〃

省略形　STR.

解　説　ジョイスティックの(トリガー)ボ タンまたはキーボード上のスペースキーが、
押されているときには真の値 -1 が、
押されていないときには偽の値0が、
この関数の値になります。

サンプルプログラム

```
IF STRIG(0) GOTO "BEAM"
```

スペースキーが押されていればラベル名の "BEAM" にジャンプさせる文です。

3.3.11 CMT

機　能　カセットの状態を与えます。

書　式

CMT〔（x）〕

x……0,1,2

省略形　CM.

解　説　（1）CMTのみで引数が省略されている場合、（2）CMT（x）の書式の場合、

カセットの状態	関数CMTの値
EJECT	0
STOP	1
READ	2
FAST	3
REW	4
WRITE	10

xの値	カ　セ　ッ　ト　の　状　態	関数CMT（x）の値
0	カセットテープの回転しているとき	-1
	カセットテープの回転が止まっているとき	0
1	カセットテープがセットされているとき	-1
	カセットテープがセットされていないとき	0
2	カセットテープの消去防止用のツメを折っていないとき	-1
	カセットテープの消去防止用のツメが折ってあるとき	0

サンプルプログラム

```
LIST
10 WIDTH 40
20 LOCATE 0,10
30 ON CMT+1 GOSUB 1000,1010,1020,1030,1040,1160,1160,
   1160,1160,1160,1050
40 LOCATE 0,12
50 ON CMT(0)+2 GOSUB 1100,1110
60 ON CMT(1)+2 GOSUB 1120,1130
70 ON CMT(2)+2 GOSUB 1140,1150
80 GOTO20
1000 PRINT"ﾌﾀｶﾞ ｱｲﾃｲﾏｽ。        ":RETURN
1010 PRINT"ｶｾｯﾄﾊ ﾄﾏｯﾃｲﾏｽ。      ":RETURN
1020 PRINT"READﾁｭｳﾃﾞｽ。          ":RETURN
1030 PRINT"FFﾁｭｳﾃﾞｽ。            ":RETURN
1040 PRINT"REWﾁｭｳﾃﾞｽ。           ":RETURN
1050 PRINT"WRITEﾁｭｳﾃﾞｽ。         ":RETURN
1100 PRINT"ﾃｰﾌﾟｶｲﾃﾝﾁｭｳﾃﾞｽ。     ":RETURN
1110 PRINT"ﾃｰﾌﾟﾊ ｶｲﾃﾝｼﾃｲﾏｾﾝ。 ":RETURN
1120 PRINT"ﾃｰﾌﾟﾊ SETｻﾚﾃｲﾏｽ。   ":RETURN
1130 PRINT"ﾃｰﾌﾟﾊ SETｻﾚﾃｲﾏｾﾝ。 ":RETURN
1140 PRINT"ﾂﾒｶﾞ ｵｯﾃｱﾘﾏｾﾝ。      ":RETURN
1150 PRINT"ﾂﾒｶﾞ ｵｯﾃｱﾘﾏｽ。       ":RETURN
1160 RETURN
OK.
```

このサンプルは、カセットの状態を調べて画面に表示するプログラムです。カセット状態の変化にともなって、刻一刻と表示が変化します。
このプログラムを実行させて、カセット・コントロール・キーをいろいろ押してみて下さい。

3.3.12 EOF

機　能　シーケンシャルファイルの終了を検出します。

書　式

```
EOF（ファイル番号）
```

ファイル番号…OPEN（OPEN文参照）で指定したファイル番号。

省略形　EO.

解　説
ファイル番号で指定したファイルからデータを読み込んだときに、
ファイルの最後のデータであれば真の値 -1 が、
ファイルの最後のデータでなければ偽の値 0 が、
この関数の値になります。

サンプルプログラム

```
IF EOF(1) THEN CLOSE#1
```

上の文は「ファイル番号1のファイルが終りのデータならばファイルを閉じる。」という意味をもっています。

3.3.13 KANJI$

機　能　漢字ROMから読み込んだパターンデータが、この関数の値になります。

書　式

```
KANJI$（x）
```

x…漢字コード。整数。

省略形　KA.

解　説
専用の漢字ROMから読み込んだパターン（縦16、横16ドットのサイズ）のコードが、この関数の値になります。
KANJI$は、PATTERN文と一緒に使います。(2.6.8 PATTERN文参照)
なお、漢字コードについては、付録A.4コード表を参照してください。

サンプルプログラム

```
LIST
10 INIT
20 POSITION 0,0
30 PATTERN -16,KANJI$(&H7F1)
OK
```

(上の例で、CZ-803Cではオプションの漢字ROMが必要です。)
このサンプルがKANJI$の基本的な使い方を示しています。
実際に実行すると画面左上隅に漢字の「漢」を表示します。(2.6.8 PATTERN文参照)
CZ-803Cで、オプションの漢字ROMボードがI/Oポートにそう入されてない場合は、16×16ドットの白い四角形が2つ続いて表示されます。

3.3.14 FN

機　能　関数名と引数を与えて、関数ルーチンのコールを行います。

書　式

FN関数名〔（x_1〔,x_2〕…）〕

x_1 ……ユーザーの定義する関数名。文字型でもよい。

x_2 ……DEFFNで定義した関数の引数。

省略形　なし

解　説　2.2.29 DEF FN参照。

サンプルプログラム

```
LIST
10 DEF FNA$(X)=RIGHT$("000"+HEX$(X),4)
20 PRINT FNA$(10)
30 PRINT FNA$(1000)
40 PRINT FNA$(10000)
OK
RUN
000A
03E8
2710
OK
```

← 引数Xを16進数で表示します。

3.3.15 USR

機　能　引数を与えて、機械語サブルーチンのコールを行います。

書　式

USR〔ユーザー関数番号〕（x）

ユーザー関数番号…DEF USRで定義した番号　0～9

x ……ユーザー関数に渡すデータ(引数)

省略形　なし

解　説　引数を機械語サブルーチン(ユーザー関数)に渡して、サブルーチンのコールを行います。引数を持たすことができるので、CALL（2.2.31 CALL文参照）とは区別します。
この関数の詳細は、「A.3.2 USR命令の使い方」を参照してください。

サンプルプログラム

```
Z=USR1(10)
```

3.4 入出力用文字関数

3.4.1 MEM$

機　能　指定したメモリーから任意の数だけ文字を取り出します。

書　式

```
MEM$（アドレス，n）
```

アドレス…メインメモリー内のアドレス。0～&HFFFF

n……文字の数。0～255

省略形　MEM.

解　説　メインメモリー内の指定したアドレスから始まる領域から、n個の文字を取り出して、関数の値にします。(メモリー内の1バイトづつのデータをキャラクタコードと見なして、それに対応する文字を与えます。)

サンプルプログラム

```
M$=MEM$(&HFE00,255)
```

3.4.2 SCRN$

機　能　テキスト画面上の文字列を与えます。

書　式

```
SCRN$（x, y, n）
```

x, y…カーソルの絶対座標。xが水平位置、yが垂直位置

n　……0～255

省略形　SCRN.

解　説　テキスト画面上の水平座標x、垂直座標yからn個の文字が、この関数の値になります。
nが0のとき、関数の値はヌルストリングになります。

サンプルプログラム　カーソルをホーム位置にして10数個、文字を打ち込んで、次のプログラムを実行してみてください。
(ただし、プログラムは画面の下のほうに作ること)注）ホーム位置は画面の左上隅です。

```
ABCDEFGHIJKLMNOPQRSTU


PRINT SCRN$(0,0,10)   ←このプログラムを入力し、実行すると下のようになります。
ABCDEFGHIJ
Ok
```

3.4.3 CHARACTER$

機　能　指定したカーソル位置の文字を与えます。　(⇒3.4.2 SCRN$)

書　式

```
CHARACTER$（x, y）
```

x, y……カーソルの座標x列y行

x…水平位置　WIDTH 40のとき　0～39
　　WIDTH 80のとき　0～79

y…垂直位置　0～24

省略形　CHAR.

解　説　x列y行のカーソル位置にある文字が、この関数の値になります。これはSCRN$（x, y, 1）と同じものと考えることができます。

サンプルプログラム

```
LIST
10 LOCATE 0,0:PRINT"*"
20 A$=CHARACTER$(0,0)
30 LOCATE 0,10:PRINTA$
Ok
```

ホーム位置に＊が表示され、(0、10)の位置にも＊が表示されます。

3.4.4 CGPAT$

機　能　キャラクタコードに対応する32バイト(32文字)の文字列を与えます。

書　式

```
CGPAT$（x）
```

x……キャラクタコード。0～255

省略形　CGP.

解　説　キャラクタコードとユーザー定義のキャラクタに対応する32バイト(32文字)の文字列を返します。左から8文字がノーマルキャラクタのビットパターンで作られた文字、次の8文字がDEFCHR$文によって、ユーザーが定義したキャラクタの青色の部分のビットパターンで作られた文字、次の8文字が赤色の部分で、最後の8文字が緑色の部分で作られた文字になります。
たとえば、キャラクタHのコードは&H48ですが、このコードにDEFCHR$によってシアン色の斜線を定義した場合を例にとります。

キャラクタH(キャラクタコード&H48)のドットパターンを下の図に示します。



2進コード	10進コード
01000010	66
01000010	66
01000010	66
01111110	126
01000010	66
01000010	66
01000010	66
00000000	0

また、キャラクタコード&H48に下の図のようなキャラクタを定義します。

青と緑の部分



2進コード	10進コード
00000001	1
00000010	2
00000100	4
00001000	8
00010000	16
00100000	32
01000000	64
10000000	128

赤の部分



2進コード	10進コード
00000000	0
00000000	0
00000000	0
00000000	0
00000000	0
00000000	0
00000000	0
00000000	0

サンプルプログラム

前記の例についてのサンプルプログラムを下に示します。

```
LIST
10 SCREEN 0,0,0:WIDTH 40
20 DIM ST$(32)
30 PRINTCHR$(&H48)'モジ"   H ノ ヒョウジ"。
40 '
50 ' キャラクタ ノ テイキ"。
60 A$=HEXCHR$("0102040810204080")
70 N$=HEXCHR$("0000000000000000")
80 DEF CHR$(&H48)=A$+N$+A$
90 '
100 'キャラクタ ノ ヒョウジ"。
110 CGEN 1
120 PRINT CHR$(&H48)
130 CGEN 0
140 '
150 'テイキ"シタ キャラクタ ニ タイスル 32モジ" ヲ エル。
160 CG$=CGPAT$(&H48)
170 PRINT#0,CG$
180 '
190 '32モジ" ノ キャラクタコート" ノ ヒョウジ"。
200 FOR I=1 TO 32
210 ST$(I)=MID$(CG$,I,1)
220 PRINT USING"####";ASC(ST$(I));
230 IF (I MOD 8)=0 THEN PRINT
240 NEXT
OK
```

170行のPRINT＃0は、CGPAT$(＆H48)に含まれているコントロールキャラクタを表示するために使っています。
参考のため、200～240行でCGPAT$(＆H48)の各キャラクタのコードを求めて表示しています。
下が実行させた結果で、3行目がCGPAT$(＆H48)の32文字の文字列です。2行目の／は、実際にはシアン色で表示されます。

```
H
/
BBB-BBB@A B D H P  @_ @@@@@@@@A B D H P  @_
  66  66  66 126  66  66  66   0
   1   2   4   8  16  32  64 128
   0   0   0   0   0   0   0   0
   1   2   4   8  16  32  64 128
Ok
```

4行目は、3行目のBBB-BBB^@の各キャラクタを10進コードに変換したもので、これの2進コードがHのドットパターンを表しています。
5行目は、3行目の^A^B^D^H^P　@　の各キャラクタをコードに変換したもので、これの2進コードが2行目の／の青の部分のドットパターンを表しています。
6行目は、^@^@^@^@^@^@^@^@の各キャラクタをコードに変換したもので、／の赤の部分のドットパターンを表しますが、例ではドットがないので、0が並んでいます。
7行目は、5行目と同じで、この2進コードが／の緑の部分のドットパターンを表しています。

3.4.5 INKEY$

機　能　キーボードからの1文字を与えます。

書　式

```
INKEY$〔(n)〕
```

n……0、1、2

省略形　INK.

解　説　キーボードから入力された1文字が、この関数の値になります。
引数の状態によって、4つの機能をもっています。

（1）引数が省略された場合
　キーが押された時、その文字が関数の値になり、
　　押されていなければ、ヌルストリングが関数の値になります。

（2）n＝0の場合
　キーが押されていれば、間断なくその文字が関数の値になり、
　　押されていなければ、ヌルストリングが関数の値になります。

（3）n＝1の場合
　カーソルをブリンクし、1文字の入力があるまで待ちます。1文字入力のINPUT$と考えることができます。

（4）n＝2の場合
　キーボードからの入力状態を示すサブCPUからの1バイトの情報がこの関数の値になります。
　入力状態を表す情報のビット構成は、次のようになっています。

7	6	5	4	3	2	1	0
T	P	R	G	L	K	S	C

ビット	入力状態	入力有	入力無
T	テンキー部分	0	1
P	キーの入力	0	1
R	リピート機能	働いている0	働いていない1
G	GRAPHキー	0	1
L	CAP LOCKキー	0	1
K	カナモードキー	0	1
S	SHIFTキー	0	1
C	CTRLキー	0	1

サンプルプログラム

```
IF INKEY$="8" GOTO "UP"
```

8のキーが押されていれば、ラベル名の"up"にジャンプさせる文です。

3.4.6 INPUT$

機　能　キーボードなどの入力装置またはファイルから文字列を読み込みます。

書　式

```
INPUT$(n[,[#]ファイル番号])
```

n……読み込む文字数　0～255

ファイル番号…OPEN(OPEN文参照)で指定したファイル番号

省略形　I.$

解　説　キーボードなどの入力装置またはテープなどのファイルから読み込んだn個の文字が、この関数の値になります。

,[#]ファイル番号を省略してINPUT$(n)とすると、キーボードからn個の文字を読み込みます。なお、BREAKコード（03）を除くすべての文字の読み込みが可能です。

サンプルプログラム　5文字入力してその文字がSHARPでない時、COMPUTERと表示します。

```
IF INPUT$(5)<>"SHARP" THEN PRINT"COMPUTER"
```

次の3つのプログラムは、まったく同一の動作をします。

```
10 A$=INKEY$(1)
20 PRINT A$
```

```
10 A$=INPUT$(1)
20 PRINT A$
```

```
10 A$=INKEY$
20 IF A$="" THEN 10
30 PRINT A$
```

3.5 タイマー変数

3.5.1 TIME$

機　能　内蔵時計の時間を与えます。

書　式

```
TIME$
```

省略形　なし

解　説　内蔵時計の値が、この変数の値になります。
hh：mm：ssの文字式で表されます。

- hh …00～23
- mm…00～59
- ss …00～59

サンプルプログラム

```
PRINT TIME$
13:08:57
Ok
```

3.5.2 DAY$

機　能　内蔵カレンダーの曜日を与えます。

書　式

```
DAY$
```

省略形　DAY.

解　説　内蔵カレンダーの曜日が、この変数の値になります。

SUN.	日曜日
MON.	月曜日
TUE.	火曜日
WED.	水曜日
THU	木曜日
FRI.	金曜日
SAT.	土曜日

サンプルプログラム

```
PRINT DAY$
SUN
Ok
```

3.5.3 DATE$

機　能　内蔵カレンダーの年月日を与えます。

書　式

```
DATE$
```

省略形　DATE.

解　説　内蔵カレンダーの年月日が、この変数の値になります。年は初期状態で？？となり、DATE$の代入により変更を受けますが、時間がたっても変化しません。

サンプルプログラム

```
PRINT DATE$
80/01/01
OK
```

3.5.4 TIME

機　能　内蔵時計によるストップウォッチの時刻を与えます。

書　式

```
TIME
```

省略形　TIM.

解　説　TIMEに代入した時刻（0～86399秒までの値）に、経過秒数を加えた値が、この変数の値になります。86399秒を越えると再び0秒に戻ります。
これにより、内蔵時計に影響を与えずに時間を計測することが可能となります。

サンプルプログラム

```
LIST
10 TIME=0
20 FOR I=0 TO 10000:NEXT
30 PRINT TIME
OK
RUN
 6
OK
```

```
LIST
10 TIME=0
20 FOR I%=0 TO 10000:NEXT
30 PRINT TIME
OK
RUN
 4
OK
```

3.6 システム変数

3.6.1 CSRLIN

機　能　カーソルの垂直位置を与えます。

書　式

```
CSRLIN
```

省略形　CSR.

解　説　現在のカーソルの垂直位置が、この変数の値になります。値の範囲は0～24です。

サンプルプログラム

```
LOCATE 5,10:PRINT POS(0);CSRLIN
```

カーソル位置を(5、10)に指定し、カーソルの位置の水平位置および垂直位置を(5、10)から表示します。

```
 5  10
OK
```

3.6.2 STRPTR

機　能　VARPTRとともに用いて、文字変数が格納されているメインメモリーのアドレスを求めることができます。

書　式

```
STRPTR
```

省略形　STRP.

解　説　VARPTR (3.3.6 VARPTR文参照)では、整数型、単精度型、倍精度型の数値データについては、そのデータの格納されているメモリーの先頭アドレスが与えられますが、文字データについては、その文字列の長さ(1バイト)と、文字データの格納されている領域の先頭アドレスを0とする相対アドレス（2バイト）の計3バイトを格納してあるアドレスが与えられます。よって、文字データの格納されている実際のアドレスを知るためには、文字データの格納されている領域の先頭アドレスの絶対的な値を知る必要があります。文字データが格納されている先頭アドレスがSTRPTRの値になるので、この変数STRPTRの値とVARPTRの相対アドレスの値を加えることによって、先頭アドレスを知ることができます。

VARPTR(文字変数) ＝ [文字長 | 相対アドレス] (3バイト)

STRPTR ＝ [絶対アドレス] (2バイト)



サンプルプログラム

```
AD=STRPTR:PRINT AD;HEX$(AD)
-24889  9EC7
OK
```

3.6.3 DTL

機　能　現在読み込み中のDATA文の行番号を与えます。

書　式

```
DTL
```

省略形　DT.

解　説　DATA文のデータを読み込んでいるとき、その行番号が、この変数の値になります。

サンプルプログラム

```
IF DTL=1000 THEN RESTORE 1500
```

3.6.4 ERL

機　能　エラー発生時に、その行番号を与えます。

書　式

```
ERL
```

省略形　なし

解　説　プログラムの実行中、エラーが発生した場合、そのエラーが発生した文の行番号が、この変数の値になります。なお、ダイレクト実行でのエラー発生では影響を受けません。
2.4.1 ON ERROR GOTO文のサンプルプログラム参照。

サンプルプログラム

```
LIST
10 ON ERROR GOTO 1000
20 PRINT 10/0
30 PRINT 1E+38*1E+38
40 END
1000 IF ERL=20 THEN PRINT"Division by zero":RESUME NEXT
1010 IF ERL=30 THEN PRINT"Over flow":RESUME NEXT
1020 ON ERROR GOTO 0
Ok
RUN
Division by zero
Over flow
Ok
```

3.6.5 ERR

機　能　エラー発生時に、そのエラーコードを与えます。

書　式

```
ERR
```

省略形　なし

解　説　プログラムの実行中、エラーが発生した場合、そのエラーのエラーコードが、この変数の値になります。エラーコードについては、「エラーコード表」を参照してください。なお、ダイレクト実行でのエラー発生では影響を受けません。

2.4.1 ON ERROR GOTO文のサンプルプログラム参照。

サンプルプログラム

```
LIST
 10 A=10/0
Ok
RUN
Division by zero in 10
Ok
?ERR
 11
Ok
?ERL
 10
Ok
ERROR ERR
Division by zero
Ok
```

第4章　ファイル操作

4.1 ファイルとは

みなさんは、日常、いろいろなもののファイルをつくっていると思います。例えば、書類やレポート、新聞の切りぬきや広告類、パンフレットや小冊子などをそれぞれバインダーにとじたり、ノートにはりつけたりしてまとめていることでしょう。また、各ファイルには、見たい時にすぐ取り出せるように、表紙などに見出しや題目を書いて、書棚にしまっておきますね。パソコンで扱う"ファイル"もこれとまったく同じです。どういう内容のファイルかを示すために"ファイル名"という見出しをつけますし、どの書棚にしまっておくかを"ファイルディスクリプタ"というもので指定します。この2つを指定して、ファイルをつくったり、読み出したりしているのです。

4.2 デバイスとは

ファイルをしまっておく書棚には、どういうものがあるのでしょう。X-1では、次の種類の書棚にファイルをしまうことができます。

カセットテープ



グラフィックメモリー

外部メモリー



画　面



キーボード



プリンタ



これらの書棚を、デバイスと呼んでいます。すなわち、デバイスとは、コンピュータとファイルのやりとりをする入出力装置のことです。



書棚"カセットテープ"



書棚"グラフィックメモリー"



書棚"外部メモリー"

4.3 ファイルディスクリプタとは

あるファイルをどの書棚にしまうのか、どの書棚からとり出してくるのかの指定を、コンピュータでは "ファイルディスクリプタ" で指定します。これは、次のような各デバイスの略称のようなものです。

～デバイス～	～ファイルディスクリプタ～
カセットテープ	CAS：
グラフィックメモリー	MEM：
外部メモリー	EMM0： ～ EMM9：
画　面	CRT： or SCR：
キーボード	KEY：
プリンタ	LPT：

ここで、

- 外部メモリーというのは、メインメモリー以外にあとから増設したメモリーのことで、0から9まで指定できます。
- グラフィックメモリーというのは、通常グラフィック用に使用しているメモリーなのですが、次の命令を実行すると、特別に外部メモリーとして使えます。

```
OPTION SCREEN 2:INIT"MEM:"
Are you sure ? (y or n)Y
Ok.
```

「OPTION SCREEN2：INIT"MEM"[↵]」をダイレクトに入力しますと、「Are you sure？(y or n)」と聞いてきますので「Y」とキー入力してください。

- 画面に対してファイルを格納する、ということは、画面上にファイルの内容を表示するということです。その表示の仕方により、"CRT：" と "SCR：" の2通りがあります。(詳しくは「4.4.2.1 データファイルの作り方」のサンプルプログラムの説明を参照してください。) ただし、画面からファイルを読みこむことはできません。
- キーボードからファイルを取り出す、ということは、押されたキーを読みこむということです。ただし、キーボードにファイルを格納することはできません。
- プリンタにファイルを格納する、ということは、ファイルの内容をプリンタに印字させるということです。ただし、プリンタからファイルを読み出すことはできません。

これらファイルディスクリプタのうち、1つを指定するということは、メインメモリーと、指定されたデバイスとをパイプでつないで、ファイルの出し入れを可能にすることを意味します。下の図を見てください。コンピュータのパイプとデバイスのパイプを、スイッチによって切り換えています。このスイッチを切り換えるのが、ファイルディスクリプタの役目なのです。たとえば、ファイルディスクリプタとして "CAS：" を指定しますと、下の図のように、カセットテープとパイプがつながります。(ファイルディスクリプタのあとのコロン(：)をつけ忘れないでください)



これで、マニュアル本文の書式のところで、たびたび登場してきた
コマンド "ファイルディスクリプタ：ファイル名"
という指定の意味がおわかりいただけたと思います。

4.4 ファイルの使い方

ファイルには、プログラムファイルとデータファイルの2種類があります。プログラムファイルというのは、データを処理するためのプログラムを記録したファイルで、一般にプログラムと呼ばれているものです。また、データファイルというのは、そのプログラムで処理されるデータの集まりです。

もう少しわかりやすく説明しますと、みなさんは、勉強する時に教科書や参考書を見ながら、それに沿って、必要事項をノートに書き出したり、暗記したりして勉強を進めていくでしょう。コンピュータのファイル処理もこれと同じなのです。コンピュータは、プログラムファイルに沿って、データの入出力や処理を行ないます。そのデータの集まりがデータファイルなのです。ですから、プログラムファイルは教科書、データファイルはノートのようなものだと思ってください。

4.4.1 プログラムファイルの使い方

プログラムファイルは、パイプをつないで(すなわち、ファイルディスクリプタを指定して)、ファイル名をくっつけてやれば、簡単にパイプを通ってデバイスに記録されます。逆に、デバイスからプログラムファイルを読みこむには、ファイルディスクリプタとファイル名を指定してやれば、コンピュータは自動的に探し出して、そのプログラムファイルをメインメモリー中に転送します。

サンプルプログラム

```
SAVE"MEM:TEST"          ← ファイル名 "TEST" というプログラムファイルをグラフィックメモリーに格納する命令です。
LOAD"EMM0:SAMPLE"       ← 外部メモリー0から、ファイル名 "SAMPLE" というプログラムファイルを読み出す命令です。
```

4.4.2 データファイルの使い方

データファイルとは、プログラムによって処理されるデータの集まりで、次に述べるように、プログラムファイルとは違った操作が必要です。下に操作の概略を図にしてあげています。「4.4.2.1 データファイルの作り方」及び「4.4.2.2 データファイルの読み出し方」の説明と並行してこの図を参照してください。

●OPEN "{O / I}",〔#〕ファイル番号,〔"〔ファイルディスクリプタ：〕ファイル名"〕

指定した書棚から(ファイルディスクリプタ)、指定したファイル(ファイル名)をとり出して、指定した机の上(ファイル番号)に開きます。



サンプルプログラム

```
MAXFILES 3
OPEN"I",#1,"CAS:TEST"
```

●PRINT＃，WRITE＃　→ ファイルにデータを書きこみます。
INPUT＃，LINPUT＃，INPUT$　→ ファイルからデータを読みこみます。

PRINT＃，WRITE＃



INPUT＃，LINPUT＃，INPUT$

●CLOSE＃ファイル番号
ファイルをもとの書棚にもどします。

サンプルプログラム

```
CLOSE#1
```

4.4.2.1　データファイルの作り方

新しくファイルを作る時、みなさんはどうしますか？まず、書棚へ行って、空のファイルを探しますね。そして、それを取り出して机の上におきます。それから、ファイルに見出しをつけ、表紙を開けて書きこむ準備をします。それと同じことをコンピュータも行ないます。それがOPEN命令です。

OPEN "O" ,〔＃〕ファイル番号,〔"〔ファイルディスクリプタ：〕ファイル名"〕
（ただし〔　〕内は省略可）

ファイルの読み書きの最初には、その準備のために必ずOPEN命令を実行してください。OPENの次の"O"という指定は次のことを意味しています。すなわち、みなさんはファイルを取り出す時、書きこむのか、読むのか、どちらかの目的をもっているでしょう。コンピュータにも、ファイルを取り出す目的を知らせてやる必要があります。そのために、次のいずれかを、OPENの次に指定してやるのです。

"O" ……ファイルにデータを書きこみます。(OUTPUT："出力"の意味)
"I" ……ファイルからデータを読み出します。(INPUT："入力"の意味)

これは、ファイルディスクリプタによって、以下の指定が可能です。

CAS：　……O，I（入出力とも指定可）
MEM：　……O，I（入出力とも指定可）
EMM0：〜EMM9：　……O，I（入出力とも指定可）
CRT：，SCR：　……O　（出力のみ可）
KEY：　……I　（入力のみ可）
LPT：　……O　（出力のみ可）

次のファイル番号というのは、どの番号の机の上でファイルを扱うのかを指定するもので、1つの机には1つのファイルしか置けません。通常は、番号1の机しかありませんが、MAXFILES命令（ファイル番号の最大値を設定する命令）によって、最大15までの机を増設することができます。以後のファイル処理は、ファイル名ではなくてファイル番号で行ないます。

さて、このOPEN命令でファイルを机の上に開いた後、いよいよファイル中にデータを書きこむことになるわけですが、それに対応するものが、PRINT＃命令、及びWRITE＃命令です。

PRINT＃ファイル番号〔，x_1，x_2，……〕
WRITE＃ファイル番号〔，x_1，x_2，……〕

ファイル番号は、OPEN命令で指定した番号を使ってください。$x_1, x_2, \cdots\cdots$というのは書きこむデータのはいった変数で、数値型、文字型どちらでも使用できますが、読み出す時には、書き込んだデータの型と読み出す変数の型が一致していなければなりません。また、これらは、データを書き込み終えるまでくり返して使用することができます。また、ファイル番号と最初の変数の間にカンマ(,)を忘れないでください。

次に、PRINT＃命令とWRITE＃命令の違いを説明します。みなさんは、ノートなどにデータを書きこむ場合、どのような書式で書きこみますか？データだけ連続して書き並べる人もいれば、データとデータの間にカンマ(,)などをつけて区切りながら書いていく人もいるでしょう。コンピュータにも、この2通りの書き方ができます。すなわち、前者がPRINT＃命令、後者がWRITE＃命令に対応します。PRINT＃命令は、データの区切りがないので、データとデータの間を少しあけてファイルに書きこまねばなりませんが、WRITE＃命令は、データを詰めて書きこめるので、その分ファイルのスペースが少なくてすみます。WRITE＃命令は、データ間にカンマを書きこんだり、文字データの両端をダブルクォテーション(")で囲んで出力します。

サンプルプログラム

サンプルプログラムは、同じデータ(10000％,200000！,123456789.123＃,ABC)をPRINT＃とWRITE＃命令を使って画面に表示させ、両者の違いを比較するプログラムです。

```
LIST
10 OPEN "O",#1,"CRT:TEST"
20 A1%=10000:A2!=200000.5:A3#=123456789.123#:A4$="ABC"
30 PRINT#1,A1%;A2!;A3#;A4$
40 WRITE#1,A1%;A2!;A3#;A4$
50 CLOSE#1
60 END
OK
RUN
 10000  200000.5  123456789.123 ABC
10000,200000.5,123456789.123,"ABC"
OK
```

← ファイルディスクリプタとして"CRT:"(画面)を指定して、ファイルを開いています。
← PRINT＃命令で、データを表示します。
← WRITE＃命令で、データを表示します。
← ファイルを閉じます。
← PRINT＃のデータ書式
← WRITE＃のデータ書式

さて、ファイルにデータを書きこみ終えたら、みなさんはそのファイルをどうしますか？もちろん、ファイルを閉じて、もとあった書棚へもどしておくでしょう。そうしないと、そこの机はもう使えなくなりますし、誰かが誤まって何か書き加えたり削除しないとも限りません。コンピュータでも同じことを行ないます。それがCLOSE命令です。

CLOSE〔〔＃〕ファイル番号,〔＃〕ファイル番号,……〕

ファイル番号を指定すると、その番号のファイル(机の上にのっているファイル)だけを閉じます(もとの書棚にもどします)。ファイル番号を省略しますと、OPEN命令で開いたファイル(机の上にのっているファイル)は、すべて閉じられます(もとの書棚にもどされます)。

これで、データファイルの作成は完了しました。このように、OPEN, PRINT＃(WRITE＃), CLOSEのたった3つの命令で、簡単にファイルを作成することができます。

サンプルプログラム

サンプル(1)

```
LIST
10 CLS
20 OPEN "O",#1,"CAS:TEST1"
30 FOR I=0 TO 255
40 PRINT#1,CHR$(I);
50 NEXT I
60 CLOSE#1
70 END
OK.
```

← 画面クリア
← 外部デバイスとしてカセットテープを指定し、ファイル名"TEST1"を開きます。用途はファイル作成。ファイル番号は1
← アスキーコード0から255(&HFF)までのすべてのキャラクタ(文字データ)をファイル中に書きこみます。
← ファイル番号1のファイルを閉じます。

サンプル(1)は、ファイル名"TEST1"というデータファイルに、アスキーコードの文字を順に書き込んでいき、それをカセットテープに格納するプログラムです。動作確認後、行番号20を次のように打ち直してください。

```
20 OPEN "O",#1,"SCR:TEST1"
```

今度は、外部デバイスとしてSCR(画面)を指定しました。画面にファイルを作成するなんて、と不思議に思う方もいらっしゃると思いますが、これは結局、サンプル(2)のような普通の画面表示命令と表面的には何ら変わりありません。

サンプル
プログラム

サンプル(2)

```
LIST
10 CLS
20 FOR I=0 TO 255
30 PRINT CHR$(I);
40 NEXT I
Ok
```

しかし、応用面では全然異なっています。サンプル(1)が、行番号20のファイルディスクリプタを変えるだけで、別の外部デバイスにデータファイルを作成することができるのに対し、サンプル(2)の方は、ただ画面に表示することしかできません。では次に、サンプル(1)の行番号20を次のように打ち直してください。

```
20 OPEN "O",#1,"CRT:TEST1"
```

これを実行するとわかりますが、同じ画面へのデータ書き込みでも、"SCR:" と CRT:" では、前者が実行例(1)のようにコントロールコード(アスキーコード&H0～&H1F)を実行するのに対して、後者は実行例(2)のようにコントロールコードを実行せずにコントロールキャラクタを表示します。この違いは、サンプル(2)において、行番号30を次のように変更したプログラムと表面的にはまったく同じです。

```
30 PRINT#0,CHR$(I);
```

ファイルディスクリプタ "SCR:" と "CRT:" の違い

実行例(1)……ファイルディスクリプタとして "SCR:" を指定



実行例(2)……ファイルディスクリプタとして "CRT:" を指定



以上のように、サンプル(1)は、ファイルディスクリプタを変えるだけで、異なったタイプのデバイスに対してファイルの作成を行ないますので、非常に便利です。たとえば、"LPT:" を指定するだけで、プリンタにデータが印字されますし、"MEM:" を指定すれば、グラフィックメモリーにデータファイルが作成されます。

ただし、"MEM:" を実行する前に、次の行番号を打ち込んで下さい。

```
15  OPTION  SCREEN 2:INIT"MEM:"
```

このように、ファイルディスクリプタを指定するメリットは、ファイルの入出力命令を各デバイスに共通に使用でき、デバイスによってファイル操作が異なるというわずらわしさをなくしていることです。

4.4.2.2 データファイルの読み出し方

次に、前に作成したファイルが必要になった時、どのようにしてファイル中のデータを読み出すかをお話ししましょう。まず、目的のファイルがしまってある書棚へ行って、見出しを見てファイルを探します。ファイルが見つかったら、それを取り出して机の上に置いて中を開けます。それらを行なうのが、OPEN命令です。

OPEN "I" ,〔#〕ファイル番号,〔"〔ファイルディスクリプタ：〕ファイル名"〕
（ただし〔 〕内は省略可）

ここで、"I" というのは、4.4.2.1 で説明したように、データ閲覧のために開いたファイルだということを示しています。よって、指定した外部デバイスにそのファイル名のファイルがないとエラーが出ます。また、ファイル番号は、ファイル作成時に指定したものとは関係ありません。

目的のファイルを開いたあとは、いよいよデータの読みこみです。これには、INPUT＃, LINPUT＃, INPUT＄の3種類の命令があります。

INPUT＃ファイル番号，x_1〔，x_2,……〕
LINPUT＃ファイル番号，文字型変数
文字型変数＝INPUT＄（n〔，〔#〕ファイル番号〕）

INPUT＃命令では、読みこんだデータは、変数x_1,x_2,……にはいりますが、データと同じ型の変数を使うようにしてください。また、LINPUT＃命令は、INPUT＃命令が数値と文字のデータを読みこめるのに対して、文字データしか読みこめません。しかも、255文字分を一度に読みこんで文字型変数に入れます。INPUT＄関数も、文字データしか読みこめませんが、読みこむ文字数nをこちらで指定してやることができます。また、INPUT＄関数のファイル番号を省略すると、キーボードからの入力待ちになります。

データの入力が終了したあとは、ファイルを閉じ、再びもとの書棚へもどして、次に取り出されるファイルのために机をあけます。それを行なっているのが、CLOSE命令です。

CLOSE〔〔#〕ファイル番号,〔#〕ファイル番号,……〕

以上で、データの読みこみは完了しました。

サンプルプログラム　サンプル(3)

```
LIST
10 CLS                                          ←画面クリア
20 OPEN "I",#1,"CAS:TEST1"                      ←サンプル(1)で作成したファイルを開きます。
30 FOR I=0 TO 255
40 X$=INPUT$(1,#1)                              ←文字データを1文字読みこみます。
50 IF X$>="A" AND X$<="Z" THEN PRINT X$;        ←読みこんだ文字データが、必要なデータかどうか調べ、
60 NEXT I                                         もし必要なデータなら表示させます。
70 CLOSE#1                                      ←ファイルを閉じます。
80 END
OK
```

サンプル(3)は、サンプル(1)で作成したデータファイルの中から、アルファベットのAからZまでを読み出して、画面に表示させるプログラムです。実行する前に、カセットをサンプル(1)のファイルよりも前に巻きもどしておいてください。

このプログラムを実行すると、まず "TEST1" という名のファイルを探し出します。そして、ファイルの最初から1文字ずつ読みこんでいって、必要なデータかどうか比較します。そうして、256個目のデータを比較し終えた時、実行を完了します。画面最上段にAからZのアルファベットが表示されたことと思います。

では、次に行番号20のファイルディスクリプタを "KEY：" に直してみてください。

```
20 OPEN "I",#1,"KEY:TEST1"
```

これを実行するとどうなるでしょう。デバイスとしてキーボードを指定していますから、行番号40では、キーボードからの入力を要求してきます。INPUT＄（1，＃1）関数は、1文字づつキーを入力します。入力したキーがアルファベットの大文字ならば、行番号50の条件分岐によって画面に表示されますが、それ以外ならば表示されません。これを256文字分くり返して、プログラムは終了します。

4.4.3 ファイル関連命令

さらに、次の命令をあわせて利用すると、容易でスムーズなファイル処理が可能になります。

●DEVICE "ファイルディスクリプタ:"

FILES, LFILES, LOAD, LOADM, SAVE, SAVEM, LOAD?, VERIFY, CHAIN, MERGE, OPEN, INIT, LIST, RUN, KILL, DEVI$, DEVO$の各命令において、ある決まった書棚のファイルばかりを使うような時、いちいちその書棚名(ファイルディスクリプタ)を指定してやるのはめんどうなので、最初に、使用する書棚名を設定してやることができます。それがDEVICE命令です。DEVICE命令でファイルディスクリプタを設定してやると、上記の命令群において、ファイルディスクリプタの指定なしでも、目的の書棚からファイルの出し入れが行なえます。DEVICE命令実行後でも、ファイルディスクリプタを指定すれば、指定した書棚のファイルを使うこともできます。BASIC起動時はDEVICE "CAS:" が実行されており、ファイルディスクリプタを指定しないと、カセットテープに対する動作となります。

ただし、上記の命令群のうち、次の各命令でファイルディスクリプタを省略すると、ダブルクォテーション(")の有無によって動作が異なってきます。

INIT" ……DEVICE命令で設定したデバイスを初期化します。
INIT ……画面を初期化します。
LIST" ……DEVICE命令で設定したデバイスに対して、プログラムリストを出力します。
LIST ……画面に対して、プログラムリストを出力します。
RUN" ……DEVICE命令で設定したデバイス内の最初に見つけたプログラムをロードし実行します。
RUN ……メインメモリー中にあるプログラムを実行します。
KILL" ……DEVICE命令で設定したデバイス内の最初に見つけたファイルを消去します。
KILL ……エラーとなります。
DEVI$""[,レコード番号,文字変数,文字変数……DEVICE命令で設定したデバイスから、1レコードのデータを読みこみます。
DEVO$""[,レコード番号,文字式,文字式……DEVICE命令で設定したデバイスに、1レコード分のデータを書きこみます。

サンプルプログラム

```
DEVICE"LPT:" ↵
Ok.
LIST" ↵
```

DEVICE命令で"LPT:"(プリンタ)を設定した後、ファイルディスクリプタを省略して、LIST" を実行すると、プリンタにプログラムリストが印字されます。すなわち、LIST"LPT:" を実行したこととまったく同じになります。

●FILES〔"ファイルディスクリプタ:"〕
LFILES

ファイルディスクリプタとして指定できるのは、次のものだけです。

CAS:, MEM:, EMM0:～EMM9:

FILES命令は、指定した書棚(ファイルディスクリプタ)の前に立って、そこに保管されているファイルの見出し(ファイル名)をながめわたす命令です。

FILES命令は画面に、LFILES命令はプリンタに、ファイル名の一覧表を表示します。

●MAXFILES n (0≦n≦15)

書棚から取り出したファイルを置く机の数を設定します。その机には、ファイル番号がついていて、1つの机には1つのファイルしか置けません。無指定時には、ファイル番号1の机しかありませんが、MAXFILES命令で指定してやることにより、最大15個まで机を増やすことができます。

MAXFILES命令は、すなわち、同時に取り出すことのできるファイル数を設定したものだと思ってください。MAXFILES 0を実行すると、ファイルを取り出すことができなくなります。

サンプルプログラム

```
LIST
10 MAXFILES 3                    ← 同時に3個のファイルを開くことができます。
20 OPEN "I",#1,"CAS:TEST1"
30 OPEN "O",#2,"SCR:TEST1"
40 OPEN "O",#3,"LPT:TEST1"
 :    :
200 CLOSE
Ok
```

●INIT〔"ファイルディスクリプタ:"〕

指定したデバイスの初期化を行ないます。初期化の内容は、デバイスにより異なります。ファイルディスクリプタを省略すると、画面のモードを初期化します(ただし、ダブルクォテーション(")をつけた場合は、DEVICE命令で設定したデバイスを初期化します)。

～ ファイルディスクリプタ ～

CAS:……カセットを先頭まで巻き戻し、すべてのファイルを消去します。
必要なファイルが記録されている時は、絶対に実行しないで下さい。

MEM:
EMM0:
〜
EMM9:
……フォーマットの管理テーブルを初期化します。これを実行すると、メモリー内のファイルは使用不能になりますので、必要なファイルがはいっている時は、絶対に実行しないでください。逆に、使用前に必ず初期化してから、ファイルを格納するようにしてください。

CRT:
SCR:
……画面のモードが初期化されます。処理内容は次の通りです。
COLOR 7,0:CGEN:CFLASH:CREV:CSIZE:PALET:PRW:WINDOW:CONSOLE:SCREEN 0,0,0

KEY:
LPT:
……エラーがでますので、これらのファイルディスクリプタを指定しないでください。

ダイレクトモードでINIT命令を実行すると、「Are you sure ? (y or n)」と聞いてきますのでYキーを押してください。プログラムモードでは聞いてこずに直ちに初期化を実行します。

●KILL "〔ファイルディスクリプタ:〕ファイル名"

ファイルディスクリプタで指定したデバイス内の、指定したファイルを消去します。ファイルディスクリプタとして、次のものを指定することができます。

MEM:, EMM0:〜EMM9:

カセットテープに対しては動作しませんので注意してください。

デバイス内のすべてのファイルを消去したい時は、上記INIT命令を使用してください。

●EOF(ファイル番号)

データファイルの終了を検出し、ファイルの最後のデータならば、−1が、最後のデータでなければ0がEOFの値になります。

サンプルプログラム

```
LIST
10 CLS
20 OPEN "I",#1,"CAS:TEST1"
30 X$=INPUT$(1,#1)
40 IF X$>="A" AND X$<="Z" THEN PRINT X$;
50 IF EOF(1) THEN CLOSE#1 ELSE GOTO 30
OK
```

4.4.2.2 のサンプル(3)を、EOF関数を用いて書き直しました。行番号30の実行の結果、EOFに−1か0がはいります。それを、行番号50で調べています。IF文では、EOF(1)が−1ならば真、0ならば偽とみなして、それぞれ、THEN, ELSE以下の命令を実行します。

4.5 ファイルの種類

今までは、一口にファイル、ファイルと言ってきましたが、ファイルには次のような種類があります。



●中間コードのプログラムファイル

BASICプログラムを中間コードに変換して記録されたファイルです。
このプログラムは、次の命令によってアクセス(書きこみ・読み出し)されます。

ファイルの書きこみ……SAVE "ファイルディスクリプタ：ファイル名"
ファイルの読み出し……LOAD "ファイルディスクリプタ：ファイル名"
CHAIN "ファイルディスクリプタ：ファイル名"

このファイル形式は、ごく一般に使われているプログラムの記録方法で、最も手軽に使用されています。

●アスキーコードのプログラムファイル

BASICプログラムを各文字ごとにアスキーコードに変換して、ファイルに記録します。
このプログラムファイルは、次の命令によってアクセスされます。

ファイルの書きこみ……SAVE "ファイルディスクリプタ：ファイル名" ,A
ファイルの読み出し……LOAD "ファイルディスクリプタ：ファイル名"
MERGE "ファイルディスクリプタ：ファイル名"

アスキーコードのファイルは、中間コードに比べて、ファイルに必要な領域やアクセス時間がかかるかわりに、メインメモリー中のプログラムとデバイスに格納されているプログラムとのドッキングが可能です(2.1.19 MERGE命令)。

●バイナリーコードのプログラムファイル

機械語プログラムを2進数のままファイルに記録します。
このプログラムファイルは、次の命令によってアクセスされます。

ファイルの書きこみ……SAVEM "ファイルディスクリプタ：ファイル名",セーブ開始番地,セーブ終了番地-
〔,実行番地〕
ファイルの読みこみ……LOADM "ファイルディスクリプタ：ファイル名"〔,ロード開始番地〔,R〕〕

機械語プログラムは、BASICプログラムと違って、プログラムが格納されているメインメモリー内のアドレスを指定してやらなければなりません。

●シーケンシャルファイル

ファイル中のデータを読み出す時に、必要不必要にかかわらず、ファイルの最初のデータから順に読み出していくファイル形式です。カセットテープ上には、データファイルとしてシーケンシャルファイルしか作成できません。なぜならば、テープの性質上、ファイル中の任意のデータのアクセスが非常に困難だからです。データを1つづつFFやREWして探し出すなんてことはとてもできません。ですから、前章で説明してきたPRINT#命令やINPUT#命令などのファイルアクセス命令は、すべてこのシーケンシャルファイルに対する命令です。
シーケンシャルファイルは、次の命令によってアクセスされます。

ファイルの書きこみ

OPEN "O",〔#〕ファイル番号,〔"〔ファイルディスクリプタ：〕ファイル名"〕

↓

PRINT#ファイル番号〔,x_1,x_2,……〕

WRITE#ファイル番号〔,x_1,x_2,……〕

↓

CLOSE〔〔#〕ファイル番号,〔#〕ファイル番号,……〕

ファイルの読み出し

OPEN "I",〔#〕ファイル番号,〔"〔ファイルディスクリプタ：〕ファイル名"〕

↓

INPUT#ファイル番号,x_1〔,x_2,……〕

LINPUT#ファイル番号,文字型変数

文字型変数＝INPUT$(n〔,〔#〕ファイル番号〕)

↓

CLOSE〔〔#〕ファイル番号,〔#〕ファイル番号,……〕

シーケンシャルファイルは、データが連続して記録されるので、ファイルの使用効率がよく、1度の処理でファイルの全データを使う場合には能率よく行なえます。しかし、ファイル中の任意のデータの検索や、データの追加・削除には向きません。

● ランダムファイル

ランダムファイルというのは、シーケンシャルファイルの欠点を補なったもので、デバイスの任意の位置に、データの入出力が行なえるファイル形式です。よって、カセットテープ上では作成することができません。これは、おもにフロッピーディスク上で用いますので、このBASICにはランダムファイルのアクセス命令はありません。DISK BASICマニュアルを参照してください。

4.6 シーケンシャルファイルの利用例

このサンプルプログラムは、自動的にシーケンシャルファイルの読み書きができる"AUTO FILE"プログラムです。まず、リストを見てください。このプログラムは、3つの部分から成り立っています。メインは行番号10から90までで、シーケンシャルファイルの読み書きを選択させるルーチンです。もし、ファイルの作成を行ないたければ、1キーを押します。すると、「自由に書いてください」と2秒間表示された後、ファイル作成モードになりますので、自由に画面上にデータを書きこんでください。これは、行番号1000から1150までのサブルーチンで、"KEY:", "SCR:", "CAS:"という3つのファイルディスクリプタを指定しています。キーボードとスクリーン画面でファイルを作成し、完成後、カセットテープに書きこんだ後、また選択モードにもどります。また、ファイルの読み出しを行ないたければ、2キーを押します。これは行番号2000から2120までのサブルーチンで、"SCR:"と"CAS:"というファイルディスクリプタに、ファイルを同時にオープンさせて、カセットテープのファイル中のデータをスクリーン画面上に表示させています。これら2つのモードを利用すれば、シーケンシャルファイルの読み書きは、自由に行なうことができます。ただし、ファイル作成時の入力文字数は2001文字以下で、それ以上になると自動的にカセットにデータが記録されます。↵キーを押しても同じ結果になります。その際、ファイルネームを聞いてきますので、13文字以下で指定してやってください。

```
LIST
10 INIT"SCR:":CLS4:CLEAR                                  ← 画面、変数クリア
15 MAXFILES 2:DIM A$(2000)                                ← ファイル番号の最大値を2に設定。
20 LOCATE 10,6:PRINT"ファイル サクセイ プログラム"          A$に2001文字分のディメンジョンを宣言。
30 LOCATE 12,9:PRINT"1...ファイル サクセイ"
40 LOCATE 12,11:PRINT"2...ファイル リード"                ← 選択
50 LOCATE 12,13:PRINT"3...END"
60 LOCATE 10,15:INPUT"トレニシマスカ? ",K
70 IF K<1 AND K>3 THEN 60
80 ON K GOSUB 1000,2000,90                                ← 1を押したら行番号1000を、2を押したら2000を、3を
85 GOTO10                                                   押したら90を呼び出す。
90 CLS4:END                                               ← 選択モードにジャンプ。
                                                          ← END
1000 ' FILE WRITE
1010 CLS4:LOCATE 10,10:PRINT"シュウニ カイテ クダサイ":PAUSE 20:CLS
1020 OPEN"I",#1,"KEY:"                                    ← キーボードから入力したファイルデータを、スクリーン
1030 OPEN"O",#2,"SCR:"                                      上に表示するために両方のファイルを開いておく。
1040 I=0                                                  ← 入力文字数のカウンタリセット。
1050 A$(I)=INPUT$(1,#1)                                   ← キーボードから1文字入力。
1060 PRINT #2,A$(I);                                      ← スクリーン上に表示。
1070 IF A$(I)=CHR$(13) OR I>=2000 THEN 1100               ← 入力文字が↵キーか、2001文字以上なら、行番号1100へジャンプ
1080 I=I+1                                                ← 入力文字数をカウント。
1090 GOTO 1050                                            ← キー入力のくり返し
1100 CLOSE                                                ← ファイルをすべて閉じる。
1110 CLS4
1115 LOCATE 0,10:INPUT"カセットニ カキコミマス。 ファイル ネーム?  ",K$ ← カセットテープに作成するファイルの名前を入力。
1120 OPEN"O",#1,"CAS:"+K$                                 ← ファイルオープン。
1130 FOR J=0 TOI:PRINT#1,A$(J);:NEXT                      ← データ書きこみ
1140 CLOSE                                                ← ファイルクローズ。
1150 RETURN                                               ← 選択モードへもどります。
2000 ' FILE READ
2010 CLS4:LOCATE 0,10:INPUT"カセットカラ ヨミコミマス。 ファイル ネーム?  ",K$ ← カセットテープから読みこむファイルの名前を入力。
2020 OPEN"I",#1,"CAS:"+K$                                 ← カセットテープから読みこんだファイルデータをスクリーン
2030 OPEN"O",#2,"SCR:"                                      上に表示するために、両方のファイルを開いておく。
2040 FOR I=0 TO 2000
2050   A$(I)=INPUT$(1,#1)                                 ← カセットテープからデータを1文字読みこむ。
2060   IF EOF(1) THEN 2080                                ← そのデータがファイル最後のデータならば、行番地2080へジャンプ。
2070 NEXT                                                 ← データ読みこみをくり返す。
2080 CLS4:FOR J=0 TO I-1:PRINT#2,A$(J);:NEXT
2090 CLOSE                                                ← スクリーン上に読みこんだデータを表示します。
2100 LOCATE 0,24:PRINT"*** スキナ キーヲ オシテクダサイ ***";← 任意のキー入力で選択モードへもどります。
2110 F$=INKEY$(1)
2120 RETURN
OK
```

選択モード

ファイル作成モード

ファイル読みこみモード

付　　録

A.1　テキスト画面とその属性ポートへのアクセス方法

テキスト画面とテキスト画面の属性は共に入出力ポートに割りつけられていて、それぞれ2Kバイト（2048バイト）の容量をもっており、そのアドレスは

テキスト画面　&H3000～&H37FF

テキスト属性　&H2000～&H27FF

となっています。(図1参照)

テキスト画面とテキストの属性とはちょうど1対1に対応していて、OUTコマンドやPOKE@ステートメントを使うことにより、1バイト単位でテキストの属性を指定することができます。

PRINT文等の画面表示命令を実行すると、テキスト画面の座標に対応するI/O アドレスにキャラクタコードを送り、テキスト属性の対応するI/O アドレスに、そのキャラクタの色、反転、点滅、キャラクタジェネレータの切り換え、サイズを示すデータを送ります。

テキストの属性8ビットのビット構成は下に示す通りです。

7	6	5	4	3	2	1	0
HS	VS	CG	CF	CR	G	R	B

(1) B・R・G…カラー指定(⇒COLOR)

テキスト画面のカラーはビットB(青指定)、R(赤指定)、G(緑指定)のビットパターンの組み合わせにより、下の表のように決定されます。

ビット 2 G	1 R	0 B	
0	0	0	黒
0	0	1	青
0	1	0	赤
0	1	1	マゼンタ
1	0	0	緑
1	0	1	シアン
1	1	0	黄
1	1	1	白

(2) CR …キャラクタの反転指定(⇒CREV)

0のときテキストのキャラクタは標準モード(normal mode)

1のときテキストのキャラクタが反転モード(reverse mode)となります。

(3) CF …キャラクタの点滅指定(⇒CFLASH)

0のときテキストのキャラクタは標準モード(normal mode)

1のときテキストのキャラクタは点滅モード(flashing mode)となります。

(4) CG…ROM/ RAMの切り換え指定(⇒CGEN)

本機はユーザ定義のキャラクタジェネレータRAM(RAMCG)とあらかじめテキストのキャラクタが書き込まれているROM(ROMCG)を持っています。

ROMCG , RAMCGをアクセスするコードはそれぞれ&H00～&HFFまでの256が割り当てられています。(⇒キャラクタコード表)

テキストのキャラクタ表示コードがROMCGをアクセスするのか、RAMCGをアクセスするのかを決定するのがビットCGです。

0のときROMCGを

1のときRAMCGをアクセスします。

(5) VS・HS…キャラクタのサイズ指定（⇒CSIZE）

ビットVS, HSを下の表のようにセットすることにより、キャラクタのサイズを変えることができます。

ただしVSをセットする場合は、一行分セットしてください。

ビット	7 HS	6 VS	
	0	0	標準サイズ
	0	1	垂直2倍
	1	0	水平2倍
	1	1	垂直水平2倍



（a）40文字モード（2画面）時のI/Oアドレス



（b）80文字モード（1画面）時のI/Oアドレス

図1．テキスト画面とI/Oアドレス（カッコ内はテキスト属性のI/Oアドレス）

例えば、LOCATE 0 , 0 : COLOR 2 : CREV 1 : PRINT "A" ↵ を実行しますと、I/Oアドレス&H3000に&H41（"A" のキャラクタコード）が送られ、I/Oアドレス&H2000に&H 0 A（＝&B00001010）が送られます。

サンプル（1）

では一度、今までに説明した動作を確認してみましょう。次の命令をダイレクトに入力してください。(40カラムモードで実行してください。)

```
LOCATE  10,10 : COLOR 6 : CFLASH 1 : PRINT "1" ↵
Ok
A=PEEK@(&H319A) : B=PEEK@(&H219A) : PRINT CHR$(A), BIN$(B) ↵
1               10110
Ok
```

まず、座標(10,10)にキャラクタ"1"を、黄色、点滅モードで書き込みます。

次に、座標に対応するI/Oアドレス、&H319A（テキスト画面）、&H219A（テキスト属性）にストアされているデータを、PEEK@関数を使って、それぞれ変数A，Bに読みこんだ後、キャラクタ及び2進データに変換して表示させています。それを見ると、属性のデータ&B00010110の下位3ビット(110)で黄色が指定され（赤＋緑）、点滅(CF)のビットが"1"になっていることがわかります。

サンプル(2)

次に、PRINT命令を使わずに、POKE@命令によって直接、I/Oアドレスにデータを送り、画面に好きな文字を好きなモードで表示させてみます。以下の命令をダイレクトに打ちこんでください。("X-1"を表示させる命令群です。40カラムモードで実行してください。)

POKE@&H319A, &H58, &H2D, &H31：POKE@&H219A, &B00010010, &B00000101, &B00011100

すると、座標(10,10)から横に"X-1"の文字が、"X"は赤の点滅、"-"はシアン、"1"は緑の反転・点滅で表示されたと思います。1番目のPOKE@文の16進数データで任意のキャラクタを、2番目のPOKE@文の2進数データで任意の状態で、また、それぞれのアドレスによって、任意の座標に表示させることができます。

参考

数値データの内部表現と外部表現

SHARP HuBASICで扱う数値は、内部表現では2進浮動小数点形式で記述され、演算や比較もこの形式のまま実行されます。2進数表現の場合、数値は必ずしも正確ではないため、実数演算、比較などの処理では、誤差の発生、あるいは、外部表現(PRINT文などによって表示される表現)との食い違いの発生に注意する必要があります。

たとえば、ある演算の結果の値が、数学的には整数となる場合でも、演算の途中で整数以外の値を扱っている場合には、内部表現では必ずしも整数が得られるとは限りません。

数値データの内部表現の外部表現に与える影響について例をあげて説明してみましょう。たとえば、変数Aの初期値を0として0.1ずつ加え続けるという演算を行うと、0.1は内部表現では正確な小数の0.1でないため、加え続けているうちに誤差を蓄積して、外部表現に影響を与えます。

プログラム1

```
10 A=0
15 FOR I=0 TO 1500
20 A=A+.1
30 PRINT A;
40 NEXT I
```

注) .1=0.1

上のプログラムを実行すると、

```
RUN
 .1  .2  .3  .4  .5  .6  .7  .8  .9  1  1.1  1.2  1.3  1.4  1.5
1.6  1.7  1.8  1.9  2  2.1  2.2  2.3  2.4  2.5  2.6  2.7  2.8
2.9  3  3.1  3.2  3.3  3.4  3.5
                    ⋮
 44.7  44.8  44.9  45  45.1  45.2  45.3  45.4  45.5  45.6  45.7
45.8  45.899999  45.999999  46.099999  46.199999  46.299999
46.399999  46.499999  46.599999  46.699999  46.799999  46.899999
46.999999  47.099999  47.199999  47.299999  47.3
                    ⋮
```

というように初めは.1, .2, .3, …と増えて行きますが、45.8を過ぎると、誤差が現れます。

これは、内部表現の誤差がたまって、外部表現に影響を与えたために起こる現象です。

この不合理を取り除くための対策としては、まず第1に整数の1を加えるという整数演算を行って、その結果を10で割って小数に戻すという方法があります。

プログラム2

```
10 A=0
15 FOR I=0 TO 1500
20 A=A+1
30 PRINT A/10;
40 NEXT I
```

どうしても0.1ずつ加えたい場合には、10倍して整数型に変換してから10で割った値を出力するという方法や、いったん文字型に変換してから数値型に戻した値を出力するという方法があります。下にプログラムを示しますので各自試してみてください。

プログラム3（10倍した値をCINTを使って整数型に変換してから10で割る）

```
LIST
10 A=0
15 FOR I=0 TO 1500
20 A=A+.1
25 A=CINT(A*10)/10
30 PRINT A;
40 NEXT
OK
```

プログラム4（STR$を使って文字型に変換してからVALを使って数値型に戻す）

```
LIST
10 A=0
15 FOR I=0 TO 1500
20 A=A+.1
25 A=VAL(STR$(A))
30 PRINT A;
40 NEXT
OK
```

また、FOR～NEXT文において、プログラム5のように、初期値0から終了値1まで増加0.1で回すと、誤差のため、最後の1になるまでループしませんので、これを防ぐためにプログラム6のように、増分を整数の1として、初期値0から終了値10まで回すようにする必要があります。

プログラム5

```
LIST
10 FOR I=0 TO 1 STEP .1
20  PRINT I;
30 NEXT
OK
RUN
 0  .1  .2  .3  .4  .5  .6  .7  .8  .9
OK
```

プログラム6

```
LIST
10 FOR I=0 TO 10
20  PRINT I/10;
30 NEXT
OK
RUN
 0  .1  .2  .3  .4  .5  .6  .7  .8  .9  1
OK
```

そのほか、0.1を3で割って、次に3倍するという演算を行うと、内部表現が正確な0.1とならないためIF文の比較で演算結果と0.1が一致しないということが起こります。

プログラム7

```
LIST
10 A=.1/3*3
20 B=.1
30 PRINT"A=";A:PRINT"B=";B
40 PRINT USING"###.##############";A;B
50 IF A=B THEN PRINT"ヨイ!":END ELSE PRINT"フルイ!":END
OK

RUN
A= .1
B= .1
   0.100000000035   0.100000000006
フルイ!
OK
```

IF文の比較は内部表現によっているため、このような不合理が生じるもので、これを解消するために、プログラム8のように比較の部分を変えて、10^{-8}の精度で比較するようにするとよいでしょう。

プログラム8

```
LIST
10 A=.1/3*3
20 B=.1
30 PRINT"A=";A:PRINT"B=";B
40 PRINT USING"###.##############";A;B
50 IF ABS(A-B)<.00000001 THEN PRINT"ヨイ!":END ELSE PRINT"フルイ!":END
OK

RUN
A= .1
B= .1
   0.100000000035   0.100000000006
ヨイ!
OK
```

A.2　コマンド・ステートメント・関数の省略形一覧表

実際プログラミングをするとき、省略形を知っていると、たいへん便利なので、下にその一覧表を示します。

（注意）　たとえば、LISTの省略形はL.ですが、実際にはそれより長いLI.やLIS.も省略形として使うことができます。そこで、下に示す省略形はいずれも省略形の中で最短のものとします。また、省略形の最後には必ずピリオド.をつけます。

省略形入力後、再び表示すると、もとの形で表示されます。

もとの形	省略形	もとの形	省略形
ABS	AB.	CSTOP	CST.
APSS	AP.	CURSOR	CU.
ASC	なし	CVD	なし
ASK	AS.	CVI	なし
ATN	AT.	CVS	CV.
AUTO	A.	DATA	DA.
BEEP	BE.	DATE$	DATE.
BIN$	BI.	DAY$	DAY.
BOOT	BO.	DEFCHR$	DEFCHR.
CALL	CA.	DEFDBL	DEFD.
CANVAS	CAN.	DEFFN	なし
CDBL	CD.	DEFINT	DEFI.
CFLASH	CF.	DEFKEY	DEFK.
CGEN	CG.	DEFSNG	DEFS.
CGPAT$	CGP.	DEFSTR	DEFST.
CHAIN	CH.	DEFUSR	なし
CHNNEL	CHAN.	DELETE	D.
CHARACTER$	CHAR.	DEVICE	DEV.
CHR$	CHR.	DEVI$	なし
CINT	CIN.	DEVO$	DEVO.
CIRCLE	CI.	DIM	DI.
CLEAR	CLE.	DTL	DT.
CLICKON	CLI.O.	EDIT	E.
CLICKOFF	CLI.OF.	EJECT	EJ.
CLOSE	CLO.	END	EN.
CLR	なし	EOF	EO.
CLS	CL.	ERL	なし
CMT	CM.	ERR	なし
COLOR	COL.	ERROR	ERR.
CONSOLE	CONS.	EXP	EX.
CONT	C.	FAC	なし
COS	なし	FAST	FA.
CREV	CR.	FILES	FIL.
CRT	なし	FIX	なし
CSIZE	CS.	FN	なし
CSNG	CSN.	FOR	F.
CSRLIN	CSR.	FRAC	FR.

もとの形	省略形
FRE	なし
GET@	なし
GOSUB	GOS.
GOTO	G.
GRAPH	GR.
HCOPY	H.
HEXCHR$	HEXC.
HEX$	HE.
IF/THEN/ELSE	IF/TH./EL.
INKEY$	INK.
INP	なし
INPUT	I.
INPUT#	I.#
INPUT$	I.$
INSTR	INS.
INT	なし
KANJI$	KA.
KEY	K.
KEY0	K.0
KEYLIST	K.L.
KEYOFF	K.OF.
KEYON	K.O.
KEYSTOP	K.S.
KILL	KI.
KLIST	KL.
LABEL	LA.
LAYER	LAY.
LEFT$	LEF.
LEN	なし
LET	(完全省略)
LFILES	LF.
LIMIT	LIM.
LINE	なし
LINE INPUT#	LINEI.#
LINPUT	LIN.
LIST	L.
LLIST	LL.
LOAD	LO.
LOADM	LO.M
LOAD?	なし
LOG	なし
LOCATE	LOC.
LPOS	LPO.
LPRINT	LP.またはL?
MAXFILES	MA.
MEM$	MEM.
MERGE	M.

もとの形	省略形
MID$	MI.
MIRROR	MIR.
MKD$	MKD.
MKI$	MK.
MKS$	MKS.
MON	MO.
MUSIC	MU.
NEW	なし
NEWON	なし
NEXT	N.
OCT$	OC.
ON	O.
ON ERROR GOTO	O. ERR. G.
ON KEY GOSUB	O. K. GOS.
OPEN	OPE.
OPTION BASE	OP.B.
OPTION SCREEN	OP.SC.
OUT	OU.
PAI	なし
PAINT	PAI.
PALET	PAL.
PATTERN	PAT.
PAUSE	PA.
PEEK	PE.
PEEK@	PE.@
PLAY	PL.
POINT	POI.
POKE	PO.
POKE@	PO.@
POLY	POL.
POS	なし
POSITION	POS.
PRESET	PRE.
PRINT	P. または?
PRINT#	P.#または?#
PRINT USING	なし
PRW	なし
PSET	PS.
PUT@	なし
RAD	なし
READ	REA.
REM	'(アポストロフィ)
RENUM	REN.
REPEAT	REP.
REPEAT OFF	REP.OF.
REPEAT ON	REP.O.
RESTORE	RES.

もとの形	省略形
RESUME	RESU.
RETURN	RE.
REW	なし
RIGHT $	RI.
RND	RN.
RUN	R.
SAVE	SA.
SAVEM	SA.M
SCREEN	SC.
SCRN $	SCRN.
SCROLL	SCRO.
SEARCH	SE.
SGN	SG.
SIN	なし
SIZE	なし
SOUND	SO.
SPACE $	SPA.
SPC	SP.
SQR	SQ.
STICK	STI.
STOP	S.
STRIG	STR.
STR $	なし
STRING$	STRIN.
STRPTR	STRP.

〈論理演算子の省略〉

NOT＝NO.
AND＝AN.
XOR＝X.
IMP ＝IM.
EQV＝EQ.

もとの形	省略形
SUM	なし
SWAP	SW.
TAB	TA.
TAN	なし
TEMPO	TE.
TIME	TIM.
TIME$	なし
TROFF	TROF.
TRON	T.
TVPW OFF	TV.OF.
TVPW ON	TV.O.
UNTIL	U.
USR	なし
VAL	VA.
VARPTR	VAR.
VERIFY	VE.
VOL	VO.
WAIT	WA.
WEND	WE.
WHILE	W.
WIDTH	WI.
WINDOW	WIN.
WRITE	WR.
WRITE#	WR.#

A.3 機械語とのリンク方法

A.3.1 モニターのコマンド

本機のBASICは機械語の入力を容易にする為にMON部を設けています。
スタック・ワークエリアとしてはFF00～FFFF（16進）を使用しています。
このモニターはBASICのMON命令によって使用可能な状態になり、BASIC同様、スクリーンエディターを内蔵していて、次に示すエディット書式により、メインメモリー64Kのどのアドレスでも書きかえ可能です。

エディット書式

:アドレス＝データ␣データ␣データ……　　注）データ間の␣は、1文字分のスペース(空白)を表わしています。

：（コロン）…エディット可能な行である事を示すマーク
（BASICの文番号とMONのアドレスを区別する為）

アドレス……16進数4文字以内で指定し、先頭のデータがはいるメインメモリーのアドレス（0～FFFF）

＝（等号）…アドレス部とデータ部を区別する為のセパレーター

データ……16進数2文字以内あるいは；（セミコロン）＋文字で指定し、8ビットのデータあるいは指定文字のキャラクターコードを指定のメモリーアドレスに書き込みます。データ間は原則として、スペースで区切りますが、データが2文字の場合は必要ありません。

＊MFE００とキーインして例の様に実行してみて下さい。Aの文字を表示します。

(例)

：FE００＝３E␣;A␣CD␣13␣００␣C９ ↵ キーを押す。

＊ SHIFT ＋ BREAK キーを押す。

＊GFE００

＊P　プリンタースイッチ

D及びFコマンドの出力をプリンターにしたり画面にしたりします。
モニターにはいった時は画面モードとなっておりこのコマンドを実行するたびにモードが反転します。
プリンターがつながっていない場合はしばらくしてから、ERR？と出てコマンド待ちになりますから、プリンターをチェックするか、Pコマンドを実行して画面モードにもどして下さい。

＊D　ダンプ

＊D〔先頭アドレス[␣最終アドレス]〕

メモリーの内容を表示します。最終アドレスを省略すると先頭アドレスより、128バイトを表示します。先頭アドレスも省略すると、次のアドレスから128バイトを表示します。

次のフォーマットでダンプされます。

:HHHH＝HH␣HH␣HH␣HH␣HH␣HH␣HH␣HH␣/ABCDE.G.

(エデット可能マーク)(行先頭アドレス)(セパレーター)(8バイト16進データ)(セパレーター)(8バイト文字)

注）最後に出力される8バイト文字は、データのディスプレイコードでコントロールコードは.(ピリオド）で表示されます。

BREAK キーを押すと表示ストップします。その後、スペースキー等を押すと表示が再開されます。 SHIFT ＋ BREAK キーを押すとコマンド待ちへもどります。

＊M　メモリーセット

＊M〔先頭アドレス〕

メモリーの内容をかきかえます。先頭アドレスを省略すると、現在のポインターからかきかえます。このモードからぬけるには、SHIFT ＋ BREAK キーを押して下さい。

アドレスとデータが表示されてカーソルはデータの上にきますので、エディットの書式でデータを指定して  キーを押してください。指定のデータ数、アドレスが加算されて次の行へすすみます。

＊F　ファインド

＊F　先頭アドレス␣最終アドレス␣データ␣データ␣…

先頭アドレスより最終アドレスまで、データで指定された個数の連続したデータをさがし、見つかったらそのアドレスとデータをダンプの様式で出力します。SHIFT ＋ BREAK キーを押すと止まります。

＊G　ゴーサブ

＊G　コールアドレス

指定したコールアドレスをサブルーチンコールします。スタックポインターは、FFFE（16進）にあります。

＊T　トランスファ

＊T　先頭アドレス␣最終アドレス␣転送先頭アドレス

指定したアドレス間のデータを転送先頭アドレス以後に転送します。

＊S　セーブ

＊S　先頭アドレス␣最終アドレス␣実行先頭アドレス：ファイル名

指定したアドレス間のデータをカセットテープに記録します。

実行先頭アドレスとはLOADした時の実行先頭アドレスです。

ファイル名は、BASICと同様次の様式で ： （コロン） の後に指定して下さい。年月日時分は自動的にタイマーより読み込まれ記録されます。

ファイル名の様式

：NAME.EXT

＊L　ロード

＊L　［ロード先頭アドレス］[:ファイルネーム]

指定のファイルをカセットよりLOADします。先頭アドレスを指定すると、そのアドレスよりLOADし、指定しないと、SAVEした情報の通りLOADします。ファイルネームは指定しないと、最初に見つけたファイルをLOADします。途中でBREAKしたりチェックサムエラーが出た時は、ERR？と出てコマンド待ちにもどります。

＊V　ベリファイ

＊V〔：ファイルネーム〕

指定のファイルをカセットより読み、メインメモリーの内容と比較します。正しくSAVEされたかどうかを調べる時に使い正しくない場合はERR？と出て止まります。

＊R　リターン

モニターをコールしたシステムへリターンします。

BASICのMON命令からはいった場合は、このコマンドによりBASICへもどれます。SP及びHLは保存していますのでMONの次の命令を実行します。

スタックポインター（SP）がFF00～FFFF（16進）にあるようなシステムからコールされた場合や、スタックにリターンアドレスがはいってない場合は、この命令ではリターンできません。

Gコマンドより、そのシステムのホットスタートをCALLしてください。

A.3.2 USR命令の使い方

BASICより機械語サブルーチンを呼び出す命令には、CALL命令とUSR命令があります。

この中でUSR命令はデータの受け渡しができる他、機械語サブルーチンからのエラー処理ができる等の機能をもっていますので次にその機能を説明します。

まずUSR命令を使う為には、そのアドレスを定義してやる必要があります。これにはDEFUSRを使います。

DEFUSR〔n〕＝コールアドレス

nは0～9の番号でいくつものUSR命令を使用する時、最大10個まで定義できます。省略すると0です。

コールアドレスはユーザーの機械語サブルーチンがはいっているアドレスであり、メインメモリー64Kのどこにでも指定できます。

指定せずに使用する事はしないで下さい。定義を解除する命令はありませんので、前の定義がのこったままとなり、暴走する事があります。

DEFUSRで定義したら、次のようにして使います。

USR〔n〕(引数)　　引数…数式または文字式

例　A＝USR(B)

A$＝USR1(C$)

D＝USR(65)

数式あるいは文字式で指定されたデータをもってユーザー定義n番のサブルーチンを呼び出し、リターンした時のデータをもつ関数となります。サブルーチン中で引数データを書きかえるとその書きかえたデータをもつ関数となります。ただし、引数データのタイプは見る事はできても変える事はできません。USR関数を実行すると、各レジスタには以下のような内容が格納されて、機械語サブルーチンへジャンプします。

サブルーチンへのJump時に設定される各レジスター内容。

IXレジスタ……エラー処理のアドレスを示しています。

Aレジスターにエラー番号をセットしてJP (IX)を実行する事によりエラー処理ができます。

A レジスタ……データのタイプを示しています。

2……整数型

5……単精度型

8……倍精度型

3……文字型

HLレジスタ……データのはいっているメインメモリー上のアドレスを示しています。文字型のみデータではなく、ストリングディスクリプタと呼ばれるエリアを示していますので、このデータでは、文字のデータはわかりません。DEレジスタとBレジスタで見てください。

データは、HLのアドレスから、次のようにはいっています。

整数型(2バイト)……

Low	High

単精度型(5バイト)……

指数	MSB　仮数4バイト　LSB

倍精度型(8バイト)……

指数	MSB　仮数7バイト　LSB

文字型……

文字長	オフセット値

Low ：下位8ビット
High ：上位8ビット
LSB：最下位ビット
MSB：最上位ビット

ここで、

指数……2の何乗であるかを表わしており、常に&H80を加算した値になっています。

例えば、0乗の時&H80であり、1乗の時&H81、－1乗の時&H7Fとなっています。仮数を2進数であらわした時に得られる"1"，"0"の並びの中で、最も上位の"1"のビットを小数点第1位に設定した時、もとの2進数との桁数の差が指数の階乗になっています。

仮数……指数のところで説明したように、2進数表現で最上位の"1"のビットをMSBにもってくるので、MSBは常に"1"になるはずです。ということは、MSBのビットは、別の目的に使用してやっても何らさしつかえありません。そこで、MSBのビットを、正･負を表わす符号ビットとして使用しています(正："0"，負："1")。これによって、扱える数値の範囲を増やし、精度を上げています。

オフセット値…USR関数の引数が文字型の時、その文字データが格納されているメインメモリー内の位置を示していますが、その位置というのは、ストリングデータエリア（文字型のデータばかりが格納されている領域）の先頭アドレスをゼロにした時の相対アドレスになっています(3.6.2 STRPTR参照)。そのデータが格納されている絶対アドレスではありません。絶対アドレスは、DEレジスタに格納されています。

DEレジスタ及びBレジスタ……Aレジスタが3、つまりUSR関数の引数が文字型の時のみ意味をもち、その文字データが格納されているアドレス(絶対アドレス)がDEレジスタに、文字データの長さがBレジスタにはいっています。Bレジスタが0の時はDEレジスタは意味をもちません。メモリーに格納されている文字データを直接書き直す時は、Bレジスタの長さ以上には入れないでください。また、引数が文字変数のみの場合は、引数とUSR関数がもってかえるデータとが、同じアドレスに格納されてしまいますので、引数の値が変わってしまいます。よって、文字式を渡す場合は、USR(A$+"")というように、文字式の加算の形で渡してください。(""はヌルストリング)

サンプルプログラム

```
LIST
5 CLEAR &HD000
10 POKE &HD000,&HDD,&H22,&H0,&HE0,&H32,&H2,&HE0,&H6,&H0,
   &H4F,&H11,&H3,&HE0,&HED, &HB0,&HC9
20 DEF USR1=&HD000
30 B!=257
40 A=USR1(B!)
50 IX2=PEEK(&HE000)
60 IX1=PEEK(&HE001)
70 AR=PEEK(&HE002)
80 FOR I=1 TO AR:HL(I)=PEEK(&HE002+I):NEXT
90 PRINT "IX=&H";HEX$(IX1);HEX$(IX2),"Acc=";AR:PRINT"(HL)=";
100 FOR I=1 TO AR:PRINT" ";HEX$(HL(I));:NEXT
110 END
Ok
RUN
IX=&H2076 Acc= 5
(HL)= 89 0 80 0 0
Ok
```

サンプルプログラムは、数値型の引数をもつUSR関数を実行した時に、機械語サブルーチンへ渡される、各レジスタの内容を表示するプログラムです。
機械語サブルーチンは、&HE000にIXレジスタの内容を、&HE002にAccの内容を、&HE003にUSR関数の引数のデータ(内部表現に変換されている)をそれぞれ格納するプログラムです。

(参考)

機械語サブルーチンのアセンブルリスト

```
&HD000  DD 22 00 E0   LD (&HE000), IX
        32 02 E0      LD (&HE002), A
        06 00         LD B, 0H
        4F            LD C, A
        11 03 E0      LD DE, &HE003
        ED 30         LDIR
        C9            RET
```

A.4 コード表

A.4.1 キャラクタコード表(ASCIIコード表)

上位4ビット→

本機で使用するキャラクタ（文字）のコード表を上に示します。
キャラクタコードは16進数の００からＦＦ（&H00～&HFF）まであります。
上の表は横の数字が16進数の上位桁、縦の数字が下位桁を表しています。
たとえば、キャラクタＳのコードは&H53となります。

&H00～&H1Fは、コントロールキャラクタで、プログラム中でこれらを使うにはCHR$関数を用います。
たとえば＆H1C（CTRL ＋ ¥）はカーソルを右に移動し、＆H07（CTRL ＋ G）はポッとブザーを鳴らします。
単なる文字として表示させるには、PRINT＃０文を利用してください。
&HA0はスペースと同じように１文字の空白ですが、スペースとは区別されます。
それを打ち込むにはカナモードにしてから、SHIFT ＋ 口キー（カタカナのロ）を打ってください。

A.4.2 JIS第１水準漢字一覧表

■文字コードの読み方

(例) 亜 のコードは、1601と読みます。実際の使用にはKANJI＄をつけてKANJI＄(1601)とします。
安 のコードは、1634と読みます。実際の使用にはKANJI＄をつけてKANJI＄(1634)とします。

ア	0	1	2	3	4	5	6	7	8	9
160×		亜	唖	娃	阿	哀	愛	挨	姶	逢
161×	葵	茜	穐	悪	握	渥	旭	葦	芦	鯵
162×	梓	圧	斡	扱	宛	姐	虻	飴	絢	綾
163×	鮎	或	粟	袷	安	庵	按	暗	案	闇
164×	鞍	杏								

イ	0	1	2	3	4	5	6	7	8	9
164×			以	伊	位	依	偉	囲	夷	委
165×	威	尉	惟	意	慰	易	椅	為	畏	異
166×	移	維	緯	胃	萎	衣	謂	違	遺	医
167×	井	亥	域	育	郁	磯	一	壱	溢	逸
168×	稲	茨	芋	鰯	允	印	咽	員	因	姻
169×	引	飲	淫	胤	蔭					
170×		院	陰	隠	韻	吋				

ウ	0	1	2	3	4	5	6	7	8	9
170×							右	宇	烏	羽
171×	迂	雨	卯	鵜	窺	丑	碓	臼	渦	嘘
172×	唄	欝	蔚	鰻	姥	厩	浦	瓜	閏	噂
173×	云	運	雲							

エ	0	1	2	3	4	5	6	7	8	9
173×				荏	餌	叡	営	嬰	影	映
174×	曳	栄	永	泳	洩	瑛	盈	穎	頴	英
175×	衛	詠	鋭	液	疫	益	駅	悦	謁	越
176×	閲	榎	厭	円	園	堰	奄	宴	延	怨
177×	掩	援	沿	演	炎	焔	煙	燕	猿	縁
178×	艶	苑	薗	遠	鉛	鴛	塩			

オ	0	1	2	3	4	5	6	7	8	9
178×								於	汚	甥
179×	凹	央	奥	往	応					
180×		押	旺	横	欧	殴	王	翁	襖	鴬
181×	鷗	黄	岡	沖	荻	億	屋	憶	臆	桶
182×	牡	乙	俺	卸	恩	温	穏	音		

カ	0	1	2	3	4	5	6	7	8	9
182×									下	化
183×	仮	何	伽	価	佳	加	可	嘉	夏	嫁
184×	家	寡	科	暇	果	架	歌	河	火	珂
185×	禍	禾	稼	箇	花	苛	茄	荷	華	菓
186×	蝦	課	嘩	貨	迦	過	霞	蚊	俄	峨
187×	我	牙	画	臥	芽	蛾	賀	雅	餓	駕
188×	介	会	解	回	塊	壊	廻	快	怪	悔
189×	恢	懐	戒	拐	改					
190×		魁	晦	械	海	灰	界	皆	絵	芥
191×	蟹	開	階	貝	凱	劾	外	咳	害	崖
192×	慨	概	涯	碍	蓋	街	該	鎧	骸	浬
193×	馨	蛙	垣	柿	蠣	鈎	劃	嚇	各	廓
194×	拡	攪	格	核	殻	獲	確	穫	覚	角
195×	赫	較	郭	閣	隔	革	学	岳	楽	額
196×	顎	掛	笠	樫	橿	梶	鰍	潟	割	喝

カ	0	1	2	3	4	5	6	7	8	9
197×	恰	括	活	渇	滑	葛	褐	轄	且	鰹
198×	叶	椛	樺	鞄	株	兜	竈	蒲	釜	鎌
199×	噛	鴨	栢	茅	萱					
200×		粥	刈	苅	瓦	乾	侃	冠	寒	刊
201×	勘	勧	巻	喚	堪	姦	完	官	寛	干
202×	幹	患	感	慣	憾	換	敢	柑	桓	棺
203×	款	歓	汗	漢	澗	灌	環	甘	監	看
204×	竿	管	簡	緩	缶	翰	肝	艦	莞	観
205×	諌	貫	還	鑑	間	閑	関	陥	韓	館
206×	舘	丸	含	岸	巌	玩	癌	眼	岩	翫
207×	贋	雁	頑	顔	願					

キ	0	1	2	3	4	5	6	7	8	9
207×						企	伎	危	喜	器
208×	基	奇	嬉	寄	岐	希	幾	忌	揮	机
209×	旗	既	期	棋	棄					
210×		機	帰	毅	気	汽	畿	祈	季	稀
211×	紀	徽	規	記	貴	起	軌	輝	飢	騎
212×	鬼	亀	偽	儀	妓	宜	戯	技	擬	欺
213×	犠	疑	祇	義	蟻	誼	議	掬	菊	鞠
214×	吉	吃	喫	桔	橘	詰	砧	杵	黍	却
215×	客	脚	虐	逆	丘	久	仇	休	及	吸
216×	宮	弓	急	救	朽	求	汲	泣	灸	球
217×	究	窮	笈	級	糾	給	旧	牛	去	居
218×	巨	拒	拠	挙	渠	虚	許	距	鋸	漁
219×	禦	魚	亨	享	京					
220×		供	侠	僑	兇	競	共	凶	協	匡
221×	卿	叫	喬	境	峡	強	彊	怯	恐	恭
222×	挟	教	橋	況	狂	狭	矯	胸	脅	興
223×	蒐	郷	鏡	響	饗	驚	仰	凝	尭	暁
224×	業	局	曲	極	玉	桐	粁	僅	勤	均
225×	巾	錦	斤	欣	欽	琴	禁	禽	筋	緊
226×	芹	菌	衿	襟	謹	近	金	吟	銀	

ク	0	1	2	3	4	5	6	7	8	9
226×										九
227×	倶	句	区	狗	玖	矩	苦	躯	駆	駈
228×	駒	具	愚	虞	喰	空	偶	寓	遇	隅
229×	串	櫛	釧	屑	屈					
230×		掘	窟	沓	靴	轡	窪	熊	隈	粂
231×	栗	繰	桑	鍬	勲	君	薫	訓	群	軍
232×	郡									

ケ	0	1	2	3	4	5	6	7	8	9
232×		卦	袈	祁	係	傾	刑	兄	啓	圭
233×	珪	型	契	形	径	恵	慶	慧	憩	掲
234×	携	敬	景	桂	渓	畦	稽	系	経	継
235×	繋	罫	茎	荊	蛍	計	詣	警	軽	頚
236×	鶏	芸	迎	鯨	劇	戟	撃	激	隙	桁
237×	傑	欠	決	潔	穴	結	血	訣	月	件

ケ	0	1	2	3	4	5	6	7	8	9
238×	倹	倦	健	兼	券	剣	喧	圏	堅	嫌
239×	建	憲	懸	拳	捲					
240×		検	権	牽	犬	献	研	硯	絹	県
241×	肩	見	謙	賢	軒	遣	鍵	険	顕	験
242×	鹸	元	原	厳	幻	弦	減	源	玄	現
243×	絃	舷	言	諺	限					

コ	0	1	2	3	4	5	6	7	8	9
243×						乎	個	古	呼	固
244×	姑	孤	己	庫	弧	戸	故	枯	湖	狐
245×	糊	袴	股	胡	菰	虎	誇	跨	鈷	雇
246×	顧	鼓	五	互	伍	午	呉	吾	娯	後
247×	御	悟	梧	檎	瑚	碁	語	誤	護	醐
248×	乞	鯉	交	佼	侯	候	倖	光	公	功
249×	効	勾	厚	口	向					
250×		后	喉	坑	垢	好	孔	孝	宏	工
251×	巧	巷	幸	広	庚	康	弘	恒	慌	抗
252×	拘	控	攻	昂	晃	更	杭	校	梗	構
253×	江	洪	浩	港	溝	甲	皇	硬	稿	糠
254×	紅	紘	絞	綱	耕	考	肯	肱	腔	膏
255×	航	荒	行	衡	講	貢	購	郊	酵	鉱
256×	砿	鋼	閤	降	項	香	高	鴻	剛	劫
257×	号	合	壕	拷	濠	豪	轟	麹	克	刻
258×	告	国	穀	酷	鵠	黒	獄	漉	腰	甑
259×	忽	惚	骨	狛	込					
260×		此	頃	今	困	坤	墾	婚	恨	懇
261×	昏	昆	根	梱	混	痕	紺	艮	魂	

サ	0	1	2	3	4	5	6	7	8	9
261×										些
262×	佐	叉	唆	嵯	左	差	査	沙	瑳	砂
263×	詐	鎖	裟	坐	座	挫	債	催	再	最
264×	哉	塞	妻	宰	彩	才	採	栽	歳	済
265×	災	采	犀	砕	砦	祭	斎	細	菜	裁
266×	載	際	剤	在	材	罪	財	冴	坂	阪
267×	堺	榊	肴	咲	崎	埼	碕	鷺	作	削
268×	咋	搾	昨	朔	柵	窄	策	索	錯	桜
269×	鮭	笹	匙	冊	刷					
270×		察	拶	撮	擦	札	殺	薩	雑	皐
271×	鯖	捌	錆	鮫	皿	晒	三	傘	参	山
272×	惨	撒	散	桟	燦	珊	産	算	纂	蚕
273×	讃	賛	酸	餐	斬	暫	残			

シ	0	1	2	3	4	5	6	7	8	9
273×								仕	仔	伺
274×	使	刺	司	史	嗣	四	士	始	姉	姿
275×	子	屍	市	師	志	思	指	支	孜	斯
276×	施	旨	枝	止	死	氏	獅	祉	私	糸
277×	紙	紫	肢	脂	至	視	詞	詩	試	誌
278×	諮	資	賜	雌	飼	歯	事	似	侍	児
279×	字	寺	慈	持	時					
280×		次	滋	治	爾	璽	痔	磁	示	而
281×	耳	自	蒔	辞	汐	鹿	式	識	鴫	竺
282×	軸	宍	雫	七	叱	執	失	嫉	室	悉
283×	湿	漆	疾	質	実	蔀	篠	偲	柴	芝
284×	屢	蕊	縞	舎	写	射	捨	赦	斜	煮
285×	社	紗	者	謝	車	遮	蛇	邪	借	勺

シ	0	1	2	3	4	5	6	7	8	9
286×	尺	杓	灼	爵	酌	釈	錫	若	寂	弱
287×	惹	主	取	守	手	朱	殊	狩	珠	種
288×	腫	趣	酒	首	儒	受	呪	寿	授	樹
289×	綬	需	囚	収	周					
290×		宗	就	州	修	愁	拾	洲	秀	秋
291×	終	繍	習	臭	舟	蒐	衆	襲	讐	蹴
292×	輯	週	酋	酬	集	醜	什	住	充	十
293×	従	戎	柔	汁	渋	獣	縦	重	銃	叔
294×	夙	宿	淑	祝	縮	粛	塾	熟	出	術
295×	述	俊	峻	春	瞬	竣	舜	駿	准	循
296×	旬	楯	殉	淳	準	潤	盾	純	巡	遵
297×	醇	順	処	初	所	暑	曙	渚	庶	緒
298×	署	書	薯	藷	諸	助	叙	女	序	徐
299×	恕	鋤	除	傷	償					
300×		勝	匠	升	召	哨	商	唱	嘗	奨
301×	妾	娼	宵	将	小	少	尚	庄	床	廠
302×	彰	承	抄	招	掌	捷	昇	昌	昭	晶
303×	松	梢	樟	樵	沼	消	渉	湘	焼	焦
304×	照	症	省	硝	礁	祥	称	章	笑	粧
305×	紹	肖	菖	蒋	蕉	衝	裳	訟	証	詔
306×	詳	象	賞	醤	鉦	鍾	鐘	障	鞘	上
307×	丈	丞	乗	冗	剰	城	場	壌	嬢	常
308×	情	擾	条	杖	浄	状	畳	穣	蒸	譲
309×	醸	錠	嘱	埴	飾					
310×		拭	植	殖	燭	織	職	色	触	食
311×	蝕	辱	尻	伸	信	侵	唇	娠	寝	審
312×	心	慎	振	新	晋	森	榛	浸	深	申
313×	疹	真	神	秦	紳	臣	芯	薪	親	診
314×	身	辛	進	針	震	人	仁	刃	塵	壬
315×	尋	甚	尽	腎	訊	迅	陣	靱		

ス	0	1	2	3	4	5	6	7	8	9
315×									笥	諏
316×	須	酢	図	厨	逗	吹	垂	帥	推	水
317×	炊	睡	粋	翠	衰	遂	酔	錐	錘	随
318×	瑞	髄	崇	嵩	数	枢	趨	雛	据	杉
319×	椙	菅	頗	雀	裾					
320×		澄	摺	寸						

セ	0	1	2	3	4	5	6	7	8	9
320×					世	瀬	畝	是	凄	制
321×	勢	姓	征	性	成	政	整	星	晴	棲
322×	栖	正	清	牲	生	盛	精	聖	声	製
323×	西	誠	誓	請	逝	醒	青	静	斉	税
324×	脆	隻	席	惜	戚	斥	昔	析	石	積
325×	籍	績	脊	責	赤	跡	蹟	碩	切	拙
326×	接	摂	折	設	窃	節	説	雪	絶	舌
327×	蝉	仙	先	千	占	宣	専	尖	川	戦
328×	扇	撰	栓	栴	泉	浅	洗	染	潜	煎
329×	煽	旋	穿	箭	線					
330×		繊	羨	腺	舛	船	薦	詮	賤	践
331×	選	遷	銭	銑	閃	鮮	前	善	漸	然
332×	全	禅	繕	膳	糎					

ソ	0	1	2	3	4	5	6	7	8	9
332×						噌	塑	岨	措	曾
333×	曽	楚	狙	疏	疎	礎	祖	租	粗	素

ソ	0	1	2	3	4	5	6	7	8	9
334×	組	蘇	訴	阻	遡	鼠	僧	創	双	叢
335×	倉	喪	壮	奏	爽	宋	層	匝	惣	想
336×	捜	掃	挿	掻	操	早	曹	巣	槍	槽
337×	漕	燥	争	痩	相	窓	糟	総	綜	聡
338×	草	荘	葬	蒼	藻	装	走	送	遭	鎗
339×	霜	騒	像	増	憎					
340×		臓	蔵	贈	造	促	側	則	即	息
341×	捉	束	測	足	速	俗	属	賊	族	続
342×	卒	袖	其	揃	存	孫	尊	損	村	遜

タ	0	1	2	3	4	5	6	7	8	9
343×	他	多	太	汰	詫	唾	堕	妥	惰	打
344×	柁	舵	楕	陀	駄	騨	体	堆	対	耐
345×	岱	帯	待	怠	態	戴	替	泰	滞	胎
346×	腿	苔	袋	貸	退	逮	隊	黛	鯛	代
347×	台	大	第	醍	題	鷹	滝	瀧	卓	啄
348×	宅	托	択	拓	沢	濯	琢	託	鐸	濁
349×	諾	茸	凧	蛸	只					
350×		叩	但	達	辰	奪	脱	巽	竪	辿
351×	棚	谷	狸	鱈	樽	誰	丹	単	嘆	坦
352×	担	探	旦	歎	淡	湛	炭	短	端	箪
353×	綻	耽	胆	蛋	誕	鍛	団	壇	弾	断
354×	暖	檀	段	男	談					

チ	0	1	2	3	4	5	6	7	8	9
354×						値	知	地	弛	恥
355×	智	池	痴	稚	置	致	蜘	遅	馳	築
356×	畜	竹	筑	蓄	逐	秩	窒	茶	嫡	着
357×	中	仲	宙	忠	抽	昼	柱	注	虫	衷
358×	註	酎	鋳	駐	樗	瀦	猪	苧	著	貯
359×	丁	兆	凋	喋	寵					
360×		帖	帳	庁	弔	張	彫	徴	懲	挑
361×	暢	朝	潮	牒	町	眺	聴	脹	腸	蝶
362×	調	諜	超	跳	銚	長	頂	鳥	勅	捗
363×	直	朕	沈	珍	賃	鎮	陳			

ツ	0	1	2	3	4	5	6	7	8	9
363×								津	墜	椎
364×	槌	追	鎚	痛	通	塚	栂	掴	槻	佃
365×	漬	柘	辻	蔦	綴	鍔	椿	潰	坪	壺
366×	嬬	紬	爪	吊	釣	鶴				

テ	0	1	2	3	4	5	6	7	8	9
366×								亭	低	停	偵
367×	剃	貞	呈	堤	定	帝	底	庭	廷	弟
368×	悌	抵	挺	提	梯	汀	碇	禎	程	締
369×	艇	訂	諦	蹄	逓					
370×		邸	鄭	釘	鼎	泥	摘	擢	敵	滴
371×	的	笛	適	鏑	溺	哲	徹	撤	轍	迭
372×	鉄	典	塡	天	展	店	添	纏	甜	貼
373×	転	顚	点	伝	殿	澱	田	電		

ト	0	1	2	3	4	5	6	7	8	9
373×									兎	吐
374×	堵	塗	妬	屠	徒	斗	杜	渡	登	菟
375×	賭	途	都	鍍	砥	礪	努	度	土	奴
376×	怒	倒	党	冬	凍	刀	唐	塔	塘	套
377×	宕	島	嶋	悼	投	搭	東	桃	梼	棟
378×	盗	淘	湯	涛	灯	燈	当	痘	禱	等

ト	0	1	2	3	4	5	6	7	8	9
379×	答	筒	糖	統	到					
380×		董	蕩	藤	討	謄	豆	踏	逃	透
381×	鐙	陶	頭	騰	闘	働	動	同	堂	導
382×	憧	撞	洞	瞳	童	胴	萄	道	銅	峠
383×	鴇	匿	得	徳	瀆	特	督	禿	篤	毒
384×	独	読	栃	橡	凸	突	椴	届	鳶	苫
385×	寅	酉	瀞	噸	屯	惇	敦	沌	豚	遁
386×	頓	呑	曇	鈍						

ナ	0	1	2	3	4	5	6	7	8	9
386×					奈	那	内	乍	凪	薙
387×	謎	灘	捺	鍋	楢	馴	縄	畷	南	楠
388×	軟	難	汝							

ニ	0	1	2	3	4	5	6	7	8	9
388×				二	尼	弐	邇	匂	賑	肉
389×	虹	廿	日	乳	入					
390×		如	尿	韮	任	妊	忍	認		

ヌ	0	1	2	3	4	5	6	7	8	9
390×									濡	

ネ	0	1	2	3	4	5	6	7	8	9
390×										禰
391×	祢	寧	葱	猫	熱	年	念	捻	撚	燃
392×	粘									

ノ	0	1	2	3	4	5	6	7	8	9
392×		乃	廼	之	埜	嚢	悩	濃	納	能
393×	脳	膿	農	覗	蚤					

ハ	0	1	2	3	4	5	6	7	8	9
393×						巴	把	播	覇	杷
394×	波	派	琶	破	婆	罵	芭	馬	俳	廃
395×	拝	排	敗	杯	盃	牌	背	肺	輩	配
396×	倍	培	媒	梅	楳	煤	狽	買	売	賠
397×	陪	這	蝿	秤	矧	萩	伯	剝	博	拍
398×	柏	泊	白	箔	粕	舶	薄	迫	曝	漠
399×	爆	縛	莫	駁	麦					
400×		函	箱	硲	箸	肇	筈	櫨	幡	肌
401×	畑	畠	八	鉢	溌	発	醗	髪	伐	罰
402×	抜	筏	閥	鳩	噺	塙	蛤	隼	伴	判
403×	半	反	叛	帆	搬	斑	板	氾	汎	版
404×	犯	班	畔	繁	般	藩	販	範	釆	煩
405×	頒	飯	挽	晩	番	盤	磐	蕃	蛮	

ヒ	0	1	2	3	4	5	6	7	8	9
405×										匪
406×	卑	否	妃	庇	彼	悲	扉	批	披	斐
407×	比	泌	疲	皮	碑	秘	緋	罷	肥	被
408×	誹	費	避	非	飛	樋	簸	備	尾	微
409×	枇	毘	琵	眉	美					
410×		鼻	柊	稗	匹	疋	髭	彦	膝	菱
411×	肘	弼	必	畢	筆	逼	檜	姫	媛	紐
412×	百	謬	俵	彪	標	氷	漂	瓢	票	表
413×	評	豹	廟	描	病	秒	苗	錨	鋲	蒜
414×	蛭	鰭	品	彬	斌	浜	瀕	貧	賓	頻
415×	敏	瓶								

フ	0	1	2	3	4	5	6	7	8	9
415×			不	付	埠	夫	婦	富	冨	布
416×	府	怖	扶	敷	斧	普	浮	父	符	腐

フ	0	1	2	3	4	5	6	7	8	9
417×	膚	芙	譜	負	賦	赴	阜	附	侮	撫
418×	武	舞	葡	蕪	部	封	楓	風	葺	蕗
419×	伏	副	復	幅	服					
420×		福	腹	複	覆	淵	弗	払	沸	仏
421×	物	鮒	分	吻	噴	墳	憤	扮	焚	奮
422×	粉	糞	紛	雰	文	聞				

ヘ	0	1	2	3	4	5	6	7	8	9
422×							丙	併	兵	塀
423×	幣	平	弊	柄	並	蔽	閉	陛	米	頁
424×	僻	壁	癖	碧	別	瞥	蔑	箆	偏	変
425×	片	篇	編	辺	返	遍	便	勉	娩	弁
426×	鞭									

ホ	0	1	2	3	4	5	6	7	8	9
426×		保	舗	鋪	圃	捕	歩	甫	補	輔
427×	穂	募	墓	慕	戊	暮	母	簿	菩	倣
428×	俸	包	呆	報	奉	宝	峰	峯	崩	庖
429×	抱	捧	放	方	朋					
430×		法	泡	烹	砲	縫	胞	芳	萌	蓬
431×	蜂	褒	訪	豊	邦	鋒	飽	鳳	鵬	乏
432×	亡	傍	剖	坊	妨	帽	忘	忙	房	暴
433×	望	某	棒	冒	紡	肪	膨	謀	貌	貿
434×	鉾	防	吠	頬	北	僕	卜	墨	撲	朴
435×	牧	睦	穆	釦	勃	没	殆	堀	幌	奔
436×	本	翻	凡	盆						

マ	0	1	2	3	4	5	6	7	8	9
436×					摩	磨	魔	麻	埋	妹
437×	味	枚	毎	哩	槙	幕	膜	枕	鮪	柾
438×	鱒	桝	亦	俣	又	抹	末	沫	迄	儘
439×	繭	麿	万	慢	満					
440×		漫	蔓							

ミ	0	1	2	3	4	5	6	7	8	9
440×				味	未	魅	巳	箕	岬	密
441×	蜜	湊	蓑	稔	脈	妙	粍	民	眠	

ム	0	1	2	3	4	5	6	7	8	9
441×										務
442×	夢	無	牟	矛	霧	鵡	椋	婿	娘	

メ	0	1	2	3	4	5	6	7	8	9
442×										冥
443×	名	命	明	盟	迷	銘	鳴	姪	牝	滅
444×	免	棉	綿	緬	面	麺				

モ	0	1	2	3	4	5	6	7	8	9
444×							摸	模	茂	妄
445×	孟	毛	猛	盲	網	耗	蒙	儲	木	黙
446×	目	杢	勿	餅	尤	戻	籾	貰	問	悶
447×	紋	門	匁							

ヤ	0	1	2	3	4	5	6	7	8	9
447×				也	冶	夜	爺	耶	野	弥
448×	矢	厄	役	約	薬	訳	躍	靖	柳	薮
449×	鑓									

ユ	0	1	2	3	4	5	6	7	8	9
449×		愉	愈	油	癒					
450×		諭	輸	唯	佑	優	勇	友	宥	幽
451×	悠	憂	揖	有	柚	湧	涌	猶	猷	由
452×	祐	裕	誘	遊	邑	郵	雄	融	夕	

ヨ	0	1	2	3	4	5	6	7	8	9
452×										予
453×	余	与	誉	輿	預	傭	幼	妖	容	庸
454×	揚	揺	擁	曜	楊	様	洋	溶	熔	用
455×	窯	羊	耀	葉	蓉	要	謡	踊	遙	陽
456×	養	慾	抑	欲	沃	浴	翌	翼	淀	

ラ	0	1	2	3	4	5	6	7	8	9
456×										羅
457×	螺	裸	来	莱	頼	雷	洛	絡	落	酪
458×	乱	卵	嵐	欄	濫	藍	蘭	覧		

リ	0	1	2	3	4	5	6	7	8	9
458×									利	吏
459×	履	李	梨	理	璃					
460×		痢	裏	裡	里	離	陸	律	率	立
461×	葎	掠	略	劉	流	溜	琉	留	硫	粒
462×	隆	竜	龍	侶	慮	旅	虜	了	亮	僚
463×	両	凌	寮	料	梁	涼	猟	療	瞭	稜
464×	糧	良	諒	遼	量	陵	領	力	緑	倫
465×	厘	林	淋	燐	琳	臨	輪	隣	鱗	麟

ル	0	1	2	3	4	5	6	7	8	9
466×	瑠	塁	涙	累	類					

レ	0	1	2	3	4	5	6	7	8	9
466×						令	伶	例	冷	励
467×	嶺	怜	玲	礼	苓	鈴	隷	零	霊	麗
468×	齢	暦	歴	列	劣	烈	裂	廉	恋	憐
469×	漣	煉	簾	練	聯					
470×		蓮	連	錬						

ロ	0	1	2	3	4	5	6	7	8	9
470×					呂	魯	櫓	炉	賂	路
471×	露	労	婁	廊	弄	朗	楼	榔	浪	漏
472×	牢	狼	籠	老	聾	蠟	郎	六	麓	禄
473×	肋	録	論							

ワ	0	1	2	3	4	5	6	7	8	9
473×				倭	和	話	歪	賄	脇	惑
474×	枠	鷲	亙	亘	鰐	詫	藁	蕨	椀	湾
475×	碗	腕								

◆上記JIS第1水準以外の漢字も収納していますのでその内容は、付表の"図形文字用符号一覧表"の太枠内をご参照ください。

A.4.3 非漢字一覧表

■文字コードの読み方

(例) ? のコードは、0109と読みます。実際の使用にはKANJ$をつけてKANJ$(0109)とします。

A のコードは、0333と読みます。実際の使用にはKANJ$をつけてKANJ$(0333)とします。

記号

	0	1	2	3	4	5	6	7	8	9
010×				、	。	,	.	·	:	;	?
011×	!	″	°	′	‵	¨	^	¯	_	ヽ
012×	゛	゜	ゝ	ゞ	仝	々	〆	〇	ー	―
013×	‐	／	＼	～	∥	｜	…	‥	‘	’
014×	“	”	(	)	〔	〕	[	]	{	}
015×	〈	〉	《	》	「	」	『	』	【	】
016×	＋	－	±	×	÷	＝	≠	＜	＞	≦
017×	≧	∞	∴	♂	♀	°	′	″	℃	￥
018×	$	¢	£	%	#	&	*	@	§	☆
019×	★	○	●	◎	◇					
020×		◆	□	■	△	▲	▽	▼	※	〒
021×	→	←	↑	↓	＝					

英・数字

	0	1	2	3	4	5	6	7	8	9
031×							0	1	2	3
032×	4	5	6	7	8	9				
033×				A	B	C	D	E	F	G
034×	H	I	J	K	L	M	N	O	P	Q
035×	R	S	T	U	V	W	X	Y	Z	
036×						a	b	c	d	e
037×	f	g	h	i	j	k	l	m	n	o
038×	p	q	r	s	t	u	v	w	x	y
039×	z									

ひらがな

	0	1	2	3	4	5	6	7	8	9
040×		ぁ	あ	ぃ	い	ぅ	う	ぇ	え	ぉ
041×	お	か	が	き	ぎ	く	ぐ	け	げ	こ
042×	ご	さ	ざ	し	じ	す	ず	せ	ぜ	そ
043×	ぞ	た	だ	ち	ぢ	っ	つ	づ	て	で
044×	と	ど	な	に	ぬ	ね	の	は	ば	ぱ
045×	ひ	び	ぴ	ふ	ぶ	ぷ	へ	べ	ぺ	ほ
046×	ぼ	ぽ	ま	み	む	め	も	ゃ	や	ゅ
047×	ゆ	ょ	よ	ら	り	る	れ	ろ	ゎ	わ
048×	ゐ	ゑ	を	ん						

カタカナ

	0	1	2	3	4	5	6	7	8	9
050×		ァ	ア	ィ	イ	ゥ	ウ	ェ	エ	ォ
051×	オ	カ	ガ	キ	ギ	ク	グ	ケ	ゲ	コ
052×	ゴ	サ	ザ	シ	ジ	ス	ズ	セ	ゼ	ソ
053×	ゾ	タ	ダ	チ	ヂ	ッ	ツ	ヅ	テ	デ
054×	ト	ド	ナ	ニ	ヌ	ネ	ノ	ハ	バ	パ
055×	ヒ	ビ	ピ	フ	ブ	プ	ヘ	ベ	ペ	ホ
056×	ボ	ポ	マ	ミ	ム	メ	モ	ャ	ヤ	ュ
057×	ユ	ョ	ヨ	ラ	リ	ル	レ	ロ	ヮ	ワ
058×	ヰ	ヱ	ヲ	ン	ヴ	ヵ	ヶ			

ギリシャ文字

	0	1	2	3	4	5	6	7	8	9
060×		A	B	Γ	Δ	E	Z	H	Θ	I
061×	K	Λ	M	N	Ξ	O	Π	P	Σ	T
062×	ϒ	Φ	X	Ψ	Ω					
063×				α	β	γ	δ	ε	ζ	η
064×	θ	ι	κ	λ	μ	ν	ξ	ο	π	ρ
065×	σ	τ	υ	φ	χ	ψ	ω			

ロシア文字

	0	1	2	3	4	5	6	7	8	9
070×		А	Б	В	Г	Д	Е	Ё	Ж	З
071×	И	Й	К	Л	М	Н	О	П	Р	С
072×	Т	У	Ф	Х	Ц	Ч	Ш	Щ	Ъ	Ы
073×	Ь	Э	Ю	Я						
074×										а
075×	б	в	г	д	е	ё	ж	з	и	й
076×	к	л	м	н	о	п	р	с	т	у
077×	ф	х	ц	ч	ш	щ	ъ	ы	ь	э
078×	ю	я								

◆上記以外の非漢字も収納していますので、その内容は、付表の〝図形文字用符号一覧表〟の太枠内をご参照ください。

A.4.4　図形文字用符号コード一覧表

この表では、当漢字ROMに収納されている全ての文字内容をまとめております。

■文字コードの読み方

（例）㈱ のコード：0100と読み実際の使用にはKANJ＄(0100)とします。
　　㎏ のコード：0588と読み実際の使用にはKANJ＄(0588)とします。
　　澤 のコード：1795と読み実際の使用にはKANJ＄(1795)とします。

左記(例)のごとく、実際の使用に於いては区と点欄の０～９までの見出し数字には、各々、頭に０をつけて使用します。
（０＝00、１＝01……９＝09として使用します。）

太枠内はJIS第１水準以外の漢字および記号・符号です。

A.5エラーメッセージー覧表

エラーコード	エラーメッセージ	説　　明
1	NEXT without FOR	FORがないのにNEXTがある。
2	Syntax error	文法がまちがっている。
3	RETURN without GOSUB	GOSUBがないのにRETURNがある。
4	Out of DATA	READで読むべきデータがDATA文に用意されていない。
5	Illegal function call	規定外の数値やデータが指定されている。
6	Overflow	演算結果が許容範囲を越えた。
7	Out of memory	プログラムが大きすぎる。配列などの変数を多くとりすぎている。
8	Undefined label	GOTO, GOSUB, IFなどで指定した分岐先の行番号がない。
9	Subscript out of range	配列変数の添字が規定外である。
10	Duplicate Definition	配列が2重に定義されている。
11	Division by zero	0で割った。
12	Illegal direct	ダイレクト実行できないステートメントを実行しようとした。
13	Type mismatch	変数の型が一致しない。
14	———	———
15	String too long	文字が255文字を越えている。
16	Too complex	式が複雑すぎる。たとえば、(　)が異常に多い場合。
17	Can't continue	CONTによってプログラムの実行を再開できない。
18	Undefined function	DEFで定義されていない関数を呼んだ。
19	No RESUME	RESUMEによってプログラムの実行を再開できない。
20	RESUME without error	エラーがないのにRESUMEを実行しようとした。
21	Illegal format	エラーメッセージの定義されていないエラーが起こった。
22	Missing operand	パラメータの必要な命令に指定がない。
23	Line buffer overflow	1行の入力文字が多すぎる。
24	———	———
25	Bad screen mode	グラフィックメモリーを外部記憶として使おうとした。(WIDTH80でASKを実行した。)
26	UNTIL without REPEAT	REPEATがないのにUNTILを実行しようとした。
27	Out of tape	カセットテープがセットされていない。
28	———	———
29	Tape read ERROR	カセットテープからデータが正しく読めない。
30	Bad file mode	異ったモードのファイルを参照しようとした。(ファイルにはバイナリィ、Basicテキスト、アスキータイプの3つのモードがある。)
31	Out of stack	POPを実行しようとしたがスタックに何も入っていない。
32	WHILE without WEND	WHILEループにWENDがない。
33	WEND without WHILE	WHILEがないのにWENDがある。
34	Reserved feature	ディスクBASIC用のコマンドを実行しようとした。
35	FOR without NEXT	FORループにNEXTがない。
36	Format over	PRINT USINGで指定したフォーマットが長すぎて出力できない。
37	REPEAT without UNTIL	REPEATループにUNTILがない。
50	FIELD overflow	FIELD文でランダムファイル内のレコード長が256以上になっている。
51	Device in use	外部装置の使用中である。
52	Bad file number	オープンされていないファイルや、起動時に指定しないファイルを参照しようとした。
53	File not found	LOAD,KILL,OPENでディスクにないファイルを参照しようとした。
54	Already open	すでにOPENしているファイルを再びOPENしようとした。
55	———	———

エラーコード	エラーメッセージ	説　　明
56	Device I/O ERROR	入出力装置において入出力エラーが生じた致命的エラー。
57	File already exists	NAMEで変更しようとしたファイル名がすでに登録されている。
58	———	———
59	———	———
60	Device full	データが入出力装置の許容容量を越えた。
61	Input past end	end of file のファイルを読もうとしたか、空ファイルを読もうとした。
62	———	———
63	———	———
64	Bad allocation table	ディスケット中のFATが壊れている。
65	Bad file descripter	ディスクリプタが違う。
66	Bad record	レコード番号が規定外。
67	No password	パスワードがあわない。
68	———	———
69	———	———
70		
71	File not open	OPENされていないファイルを使用しようとした。
72	Write protected	指定されたカセットテープの消去防止用のツメがとれている。
73	Device offline	入出力装置がつながっていないのに使用しようとした。

注）ここに登録されていない上記以外のエラーメッセージは、Unprintable error と表示されます。

A.6 ファイルディスクリプタ表

		使用できるモード	
		入力	出力
SCR:	画面(コントロールコードを実行する)	×	○
CRT:	画面(コントロールコードを表示する)	×	○
KEY:	キーボード	○	×
LPT:	プリンター	×	○
CAS:	カセットファイル	○	○
MEM:	グラフィックメモリー	○	○
EMM0:	外部メモリー0	○	○
～	～		
EMM9:	外部メモリー9	○	○

SHARP HuBASICで使用するファイルディスクリプタ表を上に示します。
ファイルディスクリプタは、外部デバイスを指定する文字定数で、最後に：(コロン)をつけるのを忘れないでください。

A.7 メモリーマップ


0000
←&H0000
IOCS
モニター
BASIC
インタープリタ
←NEW ONで指定したテキストアドレス
テキスト
→テキストエンドコード"00 00"
変数エリア
←VARPTR
→変数エンドコード"00"
ファイル用ストリングバッファ
#0, #1, …
STRPTR
MAXFILESnでとられるエリア
(n+1)×(16+256)バイト
ストリングデータ
バッファ
テンポラリーストリング
エリア
フリーエリア
STACK
←(CLEAR,LIMIT)−&H100
FAC
←CLEAR,LIMITで指定したアドレス
ユーザー用機械語
フリーエリア
FF00
←&HFF00
FFFF
IPL用ワークエリア
←&HFFFF

A.8 SHARP HuBASICプログラマのための ワンポイントアドバイス

1．INPUT文などでデータを入力した後や、プログラムをエディットした後には必ず [↵] キーを押すようにしましょう。

2．LIST, RUN, LOAD, SAVEなどを実行する前には必ず [SHIFT] ＋ [CLR HOME] キーを押して画面をクリアしてから行うように習慣づけてください。

3．カーソルを画面の下方向に動かすときには必ず [↓] キーを使うようにし、[↵] キーはデータやプログラムを入力するときにだけ使うようにしましょう。

4．画面が複雑になったり、カーソルが見えなくなったりしてどうしようもなくなったときには、次の順番で操作を行ってください。

（1）ディスプレイテレビの電源は大丈夫か確認する。
（2）ディスプレイテレビとコンピュータとの接続部分は大丈夫か確認する。
（3）ディスプレイテレビがコンピュータモードになっているか確認する。
（4）キーボード上の [SHIFT] ＋ [BREAK] キーを押す。
（5）[CTRL] ＋ [D] キーを押す。
（6）[SHIFT] ＋ [CLR HOME] キーを押す。
（7）[S] [C] [.] [↵] と押す。
（8）本機の背面のリセットスイッチを押す。
（9）本機の電源をいったん切って、5秒後に再び入れる。

5．プログラムを能率的に作成するには、次のキー操作に精通するとよいでしょう。

（1）[CTRL] ＋ [A] …INS（インサート）モードの設定、解除を行う。
(INSモード…文字列を間に挿入する。)
（2）[CTRL] ＋ [E] …カーソルから右を行の終わりまで消す。
（3）[CTRL] ＋ [Z] …カーソル以降のテキスト画面をすべてクリアする。
（4）[↑]、[↓]、[←]、[→] を使ってカーソルの移動を行う。
（5）[SHIFT] ＋ [CLR HOME] …テキスト画面を全てクリアする。

6．CLEARはサブルーチン内で実行できないので注意してください。メインプログラムの最初に実行するのが常識です。

7．プログラムで小数を使わない場合は、最初にDEFINT A－Zで変数の整数型宣言をしておくと、プログラムの実行速度が上がります。

8．FOR～NEXT文でNEXTの後の変数を省略するとループ速度が速くなります。

9．TRONは画面が乱れるだけでデバッグ効果はあまり期待できません。むしろ、プログラムの随所にSTOPを入れてデバッグするほうがよいでしょう。

10．プログラムの先頭にINIT文を入れるようにしましょう。

11．ON KEY GOSUB文などでファンクションキーが効かなくなったときは [0] [↵] と押せば動くようになります。

12．TIME$は1度設定したらさわらないで、時間の計測にはTIMEを使いましょう。

13．プログラムの実行中に [BREAK] キーのみ押すと一時停止します。はなすと実行を再開します。

14．APSSの動作を確実にするため、カセットテープは15分テープを使って下さい。
また、すでに記録してあるテープを再使用する場合には、必ず消去してから使用して下さい。

15．プログラム実行中にエラーが発生したときには、[E] [.] [.] [↵] と押すと、便利にエディットできます。

16．カセットが作動している間、キーからの入力を受けつけにくくなる場合がありますが、カセットの動作が終わると、一気に入力されます。

17．プログラムはなるべく1本のカセットテープに1つの割合でSAVEすると、プログラムをLOADするとき速くて便利です。

18．プログラムは同じものを2回続けてカセットテープにSAVEするように心がけてください。1つが壊れても、もう1つが使えます。カーソルをSAVE文のところにもっていって [↵] キーを押せば簡単です。

19．カセットデッキのフタを閉じるときは静かにフタの左側を押すようにしてください。

20．ファイル名として，(カンマ)：(コロン)；(セミコロン) は使用できません。

A.9 サンプルプログラム集

A.9.1 キャラクタ定義 1

```
 10 REM SAMPLE PROGRAM
20 SCREEN0,0,0:OPTION SCREEN1:CGEN:WIDTH40
:COLOR7,0:WINDOW:PALET:CSIZE:CREV:CFLASH:PRW
30 FOR I=0 TO 255
40 DEFCHR$(I)=STRING$(3,LEFT$(CGPAT$(I),8))
50 NEXT
60 CGEN 1:CSIZE 0
70 FOR I=0 TO 255
80 PRINT #0,CHR$(I);
90 NEXT
100 FOR I=0 TO 255
110 DEFCHR$(I)=LEFT$(STRING$(3,LEFT$(CGPAT$(I),8)),23)+CHR$(85)
120 NEXT
130 CSIZE
```

A.9.2 キャラクタ定義 2

```
 10 REM SAMPLE PROGRAM
20 SCREEN0,0,0:WIDTH 40:CREV:CFLASH:COLOR 7,0:CLS4:CGEN:CSIZE
30 CGEN 1:CSIZE 0
40 FOR I=0 TO 255
50 PRINT #0,CHR$(I);
60 NEXT I
70 FOR I=0 TO 255
80 DEFCHR$(I)=LEFT$(STRING$(3,LEFT$(CGPAT$(I),8)),23)+CHR$(255)
90 NEXT I
```

ユーザー定義文字に、ノーマル文字に、下線を引いたパターンを定義します。

A.9.3 チェック模様と斜線

```
10 REM SAMPLE PROGRAM
20 SCREEN0,0,0:WIDTH 80:COLOR7,0:WINDOW:
PALET:CSIZE:CREV:CFLASH:PRW:CLS4
30 COLOR2:A$=HEXCHR$("55AA")
40 FOR I=0 TO 99
50 POSITION0,I*2
60 FOR J=0 TO 79
70 PATTERN-2,A$
80 NEXT J
90 NEXT I
100 COLOR5:A$=HEXCHR$("0100040010004000")
110 FOR I=0 TO 24
120 POSITION0,I*8
130 FOR J=0 TO 79
140 PATTERN-8,A$
150 NEXT J
160 NEXT I
170 END
```

一回目はチェック模様が表示され、二回目は斜線が表示されます。

A.9.4 線と図

```
10 REM SAMPLE PROGRAM
20 SCREEN0,0,0:WIDTH40:COLOR7,0:WINDOW:
PALET:CSIZE:CREV:CFLASH:PRW:CLS4
30 X0=INT(RND(1)*320)
40 Y0=INT(RND(1)*200)
50 X1=INT(RND(1)*320)
60 Y1=INT(RND(1)*200)
70 C=INT(RND(1)*7)+1
80 LINE(X0,Y0)-(X1,Y1),PSET,C,BF
90 X0=X1:Y0=Y1
100 GOTO 50
```

画面につながった線を引きます。
また、80行をとりかえると中間色で四角形を書きます。

```
80 LINE(X0,Y0)-(X1,Y1),PSET,BF,HEXCHR$("AAFFEE55FFBB")
```

A.9.5 ペイントデモプログラム

```
10 REM SAMPLE PROGRAM
20 SCREEN 0,0,0:WIDTH80:COLOR7,0:WINDOW:
PALET:CSIZE:CREV:CFLASH:PRW:CLS4
30 FORI=0TO7:CIRCLE(I*80+40,50),38,I,1,0,360:
LINE(I*80,100)-(I*80+79,199),PSET,7,B:
PAINT(I*80+40,50),I,I:NEXT
40 PAINT(81,101),HEXCHR$("AA0000550000"),7:
PAINT(161,101),HEXCHR$("00AA00005500"),7:
PAINT(241,101),HEXCHR$("AAAA00555500"),7
50 PAINT(321,101),HEXCHR$("0000AA000055"),7:
PAINT(401,101),HEXCHR$("AA00AA550055"),7:
PAINT(481,101),HEXCHR$("00AAAA005555"),7:
PAINT(561,101),HEXCHR$("AAAAAA555555"),7
60 END
```

円を色つきで表示し、その内部をぬりつぶし、長方形を書いてその内部をすこしうすい色でぬりつぶします。

A.9.6 ペイントタイルパターンデモプログラム

```
10 REM SAMPLE PROGRAM
20 SCREEN 0,0,0:OPTION SCREEN1:CGEN:WIDTH80:COLOR7,0:
WINDOW:PALET:CSIZE:CREV:CFLASH:PRW
30 CLS4:LOCATE 2,10:PRINT"AUTO or MANUAL(A/M)?";
40 A$=INKEY$(1):IF A$="A"THEN FL=1 ELSE FL=0
50 FOR I=0 TO 7
60 LINE(I*160+40,20)-(I*160+119,59),PSET,7,B
70 NEXT
80 RESTORE 250
90 FOR I=0 TO 3
100 IF FL=1 THEN KEY0,STR$(I+1)+CHR$(13)
110 LOCATE I*20+2,11:INPUT"ﾀﾝｽｳ",A
120 PAUSE 250:PAUSE 250
130 IF A<1 OR A>8 GOTO 110
140 A$=""
150 FOR J=1 TO A
160 IF FL=1 THEN READB$:KEY0,B$+CHR$(13)
170 LOCATE I*20+2,11+J:INPUT "ﾊﾟﾀｰﾝ",B$:B$=HEXCHR$(LEFT$(B$,6))
180 PAUSE 250:PAUSE 250
190 IF LEN(B$)<>3 GOTO 170
200 A$=A$+B$
210 NEXT
220 PAINT(I*160+41,21),A$,7
230 NEXT I
240 GOTO 30
250 DATA AAFFFF,AAFF55,55FFAA,222200,555500,888800,
010001,020002,040004,080008
```

ペイントのタイルパターンを見るプログラムです。AUTO or MANUALとたずねてくるので、大文字でAかMを押してください。Aを押すと自動で段数、パターンが入力されます。Mを押すと段数の入力です。1から8まで入力してください。次にパターンです。16進数6ケタの入力で、段数分入力がくりかえされます。

A.9.7 星

```
10 REM SAMPLE PROGRAM
20 SCREEN0,0,0:WIDTH 40:COLOR7,0:WINDOW:
PALET:CSIZE:CREV:CFLASH:PRW:CLS4
30 FOR X=26 TO 319 STEP52
40 FOR Y=20 TO 199 STEP40
50 C=INT(RND(1)*7)+1
60 POLY(X,Y),28,C,144,90,810
70 NEXT Y
80 NEXT X
```

画面に星型を色つきで表示します。

A.9.8 三原色

```
10 REM SAMPLE PROGRAM
20 SCREEN0,0,0:WIDTH 40:COLOR 7,0:WINDOW:PALET:CSIZE:
CREV:CFLASH:PRW:CLS4
30 CIRCLE(160,70),50,7,1,0,360:CIRCLE(134,115),50,7,1,0,
360:CIRCLE(186,115),50,7,1,0,360
40 PAINT(160,50),1,7:PAINT(117,125),2,7:PAINT(203,125),4,
7:PAINT(117,75),3,7:PAI
NT(160,150),6,7:PAINT(203,75),5,7:PAINT(160,100),7,7
50 PAINT(117,125),2,7:PAINT(203,125),4,7:PAINT(117,75),3,
7:PAINT(160,150),6,7:PAINT(203,75),5,7:PAINT(160,100),7,7
60 CIRCLE(160,70),50,1,1,0,180:CIRCLE(134,115),50,2,1,
120,300:CIRCLE(186,115),50,4,1,-120,60:CIRCLE(160,70),50,
3,1,181,240:CIRCLE(134,115),50,6,1,300,360:CIRCLE(186,115)
,50,5,1,60,120
70 CIRCLE(160,70),50,5,1,300,360:CIRCLE(134,115),50,3,1,
60,120:CIRCLE(186,115),50,6,1,180,240
80 LOCATE 15,11:PRINT"■■■\__/■■■":LOCATE 15,12:PRINT"■■
oo  ■■":LOCATE15,13:PRINT"■■■■/  \■■■■":CANVAS1,2,4
90 LAYER1,2,3,4:PAUSE 10:LAYER1,2,4,3:PAUSE 10:LAYER1,3,2
,4:PAUSE 10:LAYER1,3,4,2:PAUSE10:LAYER1,4,2,3:PAUSE10:LAY
ER1,4,3,2:PAUSE10
100 LAYER2,1,3,4:PAUSE 10:LAYER2,1,4,3:PAUSE 10:LAYER2,3,
1,4:PAUSE 10:LAYER2,3,4,1:PAUSE10:LAYER2,4,1,3:PAUSE10:
LAYER2,4,3,1:PAUSE10
110 LAYER3,1,2,4:PAUSE 10:LAYER3,1,4,2:PAUSE 10:LAYER3,2,
1,4:PAUSE 10:LAYER3,2,4,1:PAUSE10:LAYER3,4,1,2:PAUSE10:L
AYER3,4,2,1:PAUSE10:GOTO 90
```

画面に色のちがう丸を表示し、その中央にキャラクタパターンを入れます。そして、優先順位をいろいろ変えています。

A.9.9 カエルの歌

```
10 REM SAMPLE PROGRAM
20 REM FROG FROG FROG
30 SCREEN0,0,0:OPTION SCREEN1:CGEN:WIDTH40:COLOR7,
0:WINDOW:PALET:CSIZE:CREV:CFLASH:PRW
40 TEMPO 120
50 MUSIC"O4V10:O5V8:O3V12"
60 MUSIC"C5DEFEDCR:R5RRRRRRR:R5RRRRRRR"
70 MUSIC"CDEFEDCR:EFGAGFER:RRRRRRRR"
80 MUSIC"CDEFEDCR:EFGAGFER:CRCRCRCR"
90 MUSIC"CDEFEDCR:EFGAGFER:CRCRCRCR"
100 MUSIC"CDEFEDCR:RRRRRRRR:CRCRCRCR"
110 MUSIC"CDEFEDCR:EFGAGFER:RRRRRRRR"
120 MUSIC"CDEFEDCR:RRRRRRRR:RRRRRRRR"
130 MUSIC"R:R:R"
140 GOTO 60
```

A.9.10 TEL.サウンド

```
10 REM SAMPLE PROGRAM
20 REM TELEPHON
30 SCREEN0,0,0:OPTION SCREEN1:CGEN:WIDTH40:COLOR7,0:
WINDOW:PALET:CSIZE:CREV:CFLASH:PRW
40 TEMPO 300
50 MUSIC"E0A0EAEAEAEAEAEAEAEAEAEAEAEAEAEAEAEAEAEAEAEAEAE
AEAR9R9R9"
60 GOTO 50
```

A.9.11　SOUND命令による効果音

はじめは波の音です。

```
10 ' Wave Sound
20 SOUND 7,&B11110111          'A ch Noise Select
30 SOUND 6,43                  'Noise Freq. Set
40 SOUND 8,16                  'A ch Env. mode Set
50 SOUND 11,0:SOUND 12,89      'Env. Freq. Set
60 SOUND 13,14                 'Env. Pat. Set
70 END
```

ノイズの平均周波数を約3 KHz、エンベロープ周期を約3秒に設定します。エンベロープパターンはNo. 14ですから、一度指定すると、後は連続して音が出ます。音を止めるには、CTRL ＋D キーを押します。
次は、蒸気機関車のような音です。

```
10 ' SL Sound
20 SOUND 0,16:SOUND 1,2        'A ch Freq. Set
30 SOUND 6,31                  'Noise Freq. Set
40 SOUND 7,&B11101110          'A ch Tone & B ch Noise Select
50 SOUND 8,16:SOUND 9,16       'A,B ch Env. mode Set
60 SOUND 11,60:SOUND 12,5      'Env. Freq. Set
70 SOUND 13,14                 'Env. Pat. Set
80 END
```

Aチャンネルから約240Hzの音と、Bチャンネルから約4 KHzのノイズを、エンベロープ周期約0.2秒で出力します。
次のプログラムを実行してください。競馬の感じが出ていませんか。

```
10 ' HORSE RACING Sound
20 SOUND 0,0:SOUND 1,8         'A ch Freq. Set
30 SOUND 7,&B11111110          'A ch Tone Select
40 SOUND 8,16                  'A ch Env. mode Set
50 SOUND 11,0:SOUND 12,1       'Env. Freq. Set
60 SOUND 13,0                  'Env. Pat. Set
70 PAUSE 2
80 SOUND 13,0                  'Env. Pat. Set
90 PAUSE 1
100 SOUND 13,0                 'Env. Pat. Set
110 PAUSE 1
120 GOTO 60
```

Aチャンネルに約60Hz、エンベロープ周期に約0.03秒を設定し、エンベロープパターン0番を指定します。これをPAUSE 2～1の間隔でくり返し出力します。

更に、ジェット機の離陸時のような音を発生させるプログラムです。

```
10 ' JET Sound
20 SOUND 7,&B110101              'A ch Noise & B ch Tone Select
30 SOUND 8,15:SOUND 9,15         'A,B ch Vol. Set
40 SOUND 6,5                     'Noise Freq. Set
50 FOR T=255 TO 35 STEP -1
60 T1=T MOD 256                  'R2 Data Calc.
70 T2=T ¥ 256                    'R3 Data Calc.
80 SOUND 2,T1:SOUND 3,T2         'B ch Freq. Set
90 SOUND 6,T/7.8                 'Noise Freq. Set
100 PAUSE 1
110 NEXT
120 PAUSE 50
130 FOR V=15 TO 0 STEP -1
140 PAUSE 2
150 SOUND 8,V:SOUND 9,V          'A,B ch Vol. Set
160 NEXT
170 END
```

Aチャンネルからノイズ、Bチャンネルからトーンを出しますが、はじめ、ノイズ周波数を約25KHzにします。その後、トーン周波数を約490～3570Hzに、又これに対して、ノイズを約3.8～28KHzまで順次変化させていきます。そして、フェードアウトするために音量を15～0へとしぼります。

最後にSOUND命令とPLAY命令を使った2重奏のサンプルプログラムを示しますが20行、30行でエンベロープの周期やパターンを変えると違った感じになります。

```
10 'Music
20 SOUND 11,20:SOUND 12,60
30 SOUND 13,9
40 SOUND7,&B111100
50 RESTORE120
60 PLAY120
70 READ A$
80 IF A$="END" THEN END
90 SOUND 8,16:SOUND9,16
100 PLAY A$
110 GOTO70
120 DATA"O5+D5G3A3B3+C3+D5G5G5:O4B7A5B8"
130 DATA"+E5+C3+D3+E3+#F3+G5G5G5:+C8B8"
140 DATA"+C5+D3+C3B3A3B5+C3B3A3G3:A8G8"
150 DATA"#F5G3A3B3G3B5A7:+D5B5G5+D7+C5"
160 DATA"O5+D5G3A3B3+C3+D5G5G5:O4B7A5B8"
170 DATA"+E5+C3+D3+E3+#F3+G5G5G5:+C8B8"
180 DATA"+C5+D3+C3B3A3B5+C3B3A3G3:A8G8"
190 DATA"A5B3A3G3#F3G8:+C5+D5D5G8"
200 DATA END
```

A.9.12 サイコロ 1

```
10 REM SAMPLE PROGRAM
20 SCREEN0,0,0:WIDTH 40:COLOR 7,0:WINDOW:PALET:CSIZE:CREV:
CFLASH:PRW:CLS4
40 CLS 0
50 WINDOW (80,50)-(239,149),(0,0)-(319,199)
60 GOSUB 120
70 FOR I=0 TO 3:FOR J=0 TO 3
80 IF (I MOD 3)<>0 AND(J MOD 3)<>0 GOTO 100
90 WINDOW(I*80,J*50)-(I*80+79,J*50+49),(0,0)-(319,199):
GOSUB 120
100 NEXT J:NEXT I
110 END
120 POLY(160,100),90,7,60,30,390:POLY(160,100),90,7,0,30:
POLY(160,100),90,7,0,150:POLY(160,100),90,7,0,270
130 PAINT(160,55),HEXCHR$("FFAA55FF55AA"),7:PAINT(121,122),
HEXCHR$("AAFF5555FFAA"),7:PAINT(198,122),
HEXCHR$("55AAFFAA55FF"),7:RETURN
```

中央に、ウィンドウを使ってすこし小さくしたハコを書きます。
また、ウィンドウを使ってもっと小さいハコをそのまわりに書きます。

A.9.13 サイコロ 2

```
10 REM SAMPLE PROGRAM
20 SCREEN0,0,0:WIDTH 40:COLOR 7,0:WINDOW:PALET:CSIZE:
CREV:CFLASH:PRW:CLS4
40 CLS 0 :PALET 7,0
50 WINDOW (80,50)-(239,149),(0,0)-(319,199)
60 GOSUB 120
70 FOR I=0 TO 3:FOR J=0 TO 3
80 IF (I MOD 3)<>0 AND(J MOD 3)<>0 GOTO 100
90 WINDOW(I*80,J*50)-(I*80+79,J*50+49),(0,0)-(319,199):
GOSUB 120
100 NEXT J:NEXT I
110 FOR I%=0 TO 7:PALET I%,I%:PALET(I%-1)-(I%=0)*8,0:
PAUSE 50:NEXT :GOTO 110
120 POLY(160,100),90,7,60,30,390:POLY(160,100),90,7,0,30:
POLY(160,100),90,7,0,150:POLY(160,100),90,7,0,270
130 PAINT(160,55),HEXCHR$("FFAA55FF55AA"),7:PAINT(121,122)
,HEXCHR$("AAFF5555FFAA"),7:PAINT(198,122),
HEXCHR$("55AAFFAA55FF"),7:RETURN
```

画面にハコを書き、パレットを使い色をかえ、まるでストロボの光があたっているように見えます。

A.9.14 サイコロ 3

```
10 REM SAMPLE PROGRAM
20 SCREEN 0,0,0:WIDTH 40:COLOR 7,0:WINDOW:
PALET:CSIZE:CREV:CFLASH:PRW:CLS4
30 CLS 0:PALET 7,0
40 WINDOW (80,50)-(239,149),(0,0)-(319,199)
50 GOSUB 140
60 FOR I=0 TO 3:FOR J=0 TO 3
70 IF (I MOD 3)<>0 AND(J MOD 3)<>0 GOTO 90
80 WINDOW(I*80,J*50)-(I*80+79,J*50+49),(0,0)-(319,199):GOSUB 140
90 NEXT J:NEXT I
100 FOR I=0 TO 7:PALET I,0:NEXT
105 WINDOW
110 Y=1:Z=1:PRW193
120 LINE(0,Y+Z)-(39,Y+Z),"■",BF:PAUSE10:LINE(0,Y-Z)-(39,Y-Z)
," ",BF:Y=Y+Z:IF Y=1 OR Y=23 THEN Z=-Z
130 GOTO 120
140 POLY(160,100),90,7,60,30,390:POLY(160,100),90,7,0,30:
POLY(160,100),90,7,0,150:POLY(160,100),90,7,0,270
150 PAINT(160,55),HEXCHR$("FFAA55FF55AA"),7:PAINT(121,122),
HEXCHR$("AAFF5555FFAA"),7:PAINT(198,122),HEXCHR$
("55AAFFAA55FF"),7:RETURN
```

いくつかのハコを書いて、そこにサーチライトキャラクタライン(3行分)を上下に動かし、ハコの部分を写し出します。

A.9.15 漢字を任意の位置に表示

```
5 PALET
10 INIT:WIDTH 80
20 CONSOLE 23,2
30 XD=0:YD=0
40 POSITION XD*16,YD*16
50 GOSUB 250
60 KCO=800
70 INPUT "ｶﾝｼﾞ ﾉ ｺｰﾄﾞ ｦ ｲﾚﾃｸﾀﾞｻｲ";KCO
80 IF KCO=0 THEN 120
90 PATTERN -16,KANJI$(KCO)
100 IF XD<38 THEN XD=XD+1
110 GOTO 40
120 GOSUB 250:PRINT"ｶｰｿﾙ ｦ ｲﾄﾞｳ ｼﾃｸﾀﾞｻｲ";:CU$=INKEY$(1)
130 CU=ASC(CU$)-27
140 ON CU GOTO 170,190,230,210
150 IF CU$=CHR$(13) THEN 40
160 GOTO 120
170 IF XD<38 THEN XD=XD+1
180 GOTO 120
190 IF XD<>0 THEN XD=XD-1
200 GOTO 120
210 IF YD<10 THEN YD=YD+1
220 GOTO 120
230 IF YD<>0 THEN YD=YD-1
240 GOTO 120
250 CONSOLE 0,22:CLS
260 CURSOR XD*2,YD*2+1
270 PRINT"__";
280 CONSOLE 23,2:CLS
290 RETURN
```

● このプログラムは、ディスプレイ画面に縦11文字、横39文字の範囲内で任意の位置に漢字を表示するプログラムです。このプログラムを実行させると画面左端に "＝" のキャラクタが表示されます。
● この表示が漢字を表示する位置を示します。(ここでは、これをカーソルと呼ぶことにします。)このとき画面下部に"カンジ ノ コード ヲ イレテクダサイ" というメッセージが表示されますので、表示をしたい漢字のコードを入力するとそれに対応した漢字が表示され、カーソルが右に1文字分移動し、ふたたび漢字コードの入力待ちとなります。
● 表示位置を任意に指定する場合は、漢字コードの入力待ちのときに0を入力してください。すると "カーソル ヲ イドウ シテクダサイ" という表示がされますので、カーソルキーにより任意の位置へカーソルを移動させることができます。
● カーソルの移動ができましたら ↵ キーを押すと漢字コードの入力待ちとなりますので、表示をしたい漢字のコードを入力してください。

注) CZ-803Cでは、オプションの漢字ROM CZ-8KRが必要です。

BASICテープのコピー作成方法

本機に付属のカセットテープ「SHARP HuBASIC CZ8CB01」は、テープの長時間使用による摩耗、あるいは不慮の事故による破損によって使用できなくなることが考えられます。あらかじめ予備のBASICテープを作成しておくようにしてください。

BASICテープをコピーするには、コピーのためのプログラムを作る必要がありますので、その手順を次に示します。

①BASICテープをカセットデータレコーダにセットして、「SHARP HuBASIC CZ8CB01」を起動した後、テープを巻き戻したままにしておきます。

②下に示すBASICプログラムをキーボードから入力します。

```
10 '
20 '      BASIC ﾃｰﾌﾟ ｺﾋﾟｰ ﾌﾟﾛｸﾞﾗﾑ
30 '
40 CLEAR &HFE00
50 RESTORE 100
60 FOR I=0 TO 49
70 READ A$:POKE &HFE00+I,VAL("&H"+A$)
80 NEXT
90 CALL &HFE00
100 DATA 21,60,FE,01,20,00,CD,41,00,2A
110 DATA 74,FE,ED,4B,72,FE,CD,44,00,3E
120 DATA 01,CD,1B,00,FE,20,20,F7,21,60
130 DATA FE,01,20,00,CD,3B,00,2A,74,FE
140 DATA ED,4B,72,FE,CD,3E,00,C3,13,FE
```

③上のプログラムをRUNすると、自動的にCZ8CB01がロードされ、終了後カーソルが点滅し、入力待ちになります。

④カセットデータレコーダに、巻き戻してある新しい空テープをセットしてから、キーボード上のスペースキーを押すと、CZ8CB01のセーブが始まります。

⑤セーブが終了すると、カーソルが点滅し始めますが、そのままの状態にしておくと、テープが空回りし続けますので、キーボード上のカセットストップキーを押して止めてください。

⑥セーブが終了した時点で、④の操作を繰り返すことにより、何本でもコピーを取ることができます。

⑦プログラムの実行を止めるときは、本機の背面にある黒いRESETボタンを押してプログラムを消し、BASIC起動時の状態に戻してください。

⑧コピーテープを作成したら、消去防止用のツメを折って、大切に保管してください。

A.10 類似命令一覧表

●メインメモリーとのデータ・アクセス

POKE(ステートメント) ……任意のメインメモリー(&H0～&HFFFF)にデータを書きこみます。(アドレス指定)
PEEK(関数) ……任意のメインメモリーからデータを読みこみます。(アドレス指定)
MEM$(ステートメント)……任意のメインメモリーに文字を転送します。(アドレス指定)
MEM$(関数)……任意のメインメモリーから文字を読みこみます。(アドレス指定)

●I/Oポートに対するデータ・アクセス

OUT(ステートメント)……任意のI/Oポート(&H0～&HFFFF)にデータを送ります。(アドレス指定)
INP(関数) ……任意のI/Oポート(&H0～&HFFFF)からデータを読みこみます。(アドレス指定)
POKE@(ステートメント)……VRAM(I/Oポート&H2000～&HFFFF)にデータを書きこみます。(アドレス指定)
PEEK@(関数)……VRAM(I/Oポート&H2000～&HFFFF)からデータを読みこみます。(アドレス指定)
PUT@(ステートメント)……テキスト及びグラフィック画面(I/Oポート&H2000～&HFFFF)上に配列変数内のデータを表示させます。(座標指定)
GET@(ステートメント)……テキスト及びグラフィック画面(I/Oポート&H2000～&HFFFF)上のデータを配列変数内に読みこみます。(座標指定)
SCRN$(関数)……テキスト画面(I/Oポート&H2000～&H3FFF)上の文字列を読みこみます。(座標指定)

●その他

POS(関数) ……カーソルの水平方向の座標を与えます。
CSRLIN(システム変数) ……カーソルの垂直方向の座標を与えます。
VARPTR(関数)……数値変数が格納されているアドレスを与えます。
STRPTR(システム変数)……文字変数が格納されているアドレスを与えます。

●ファイルに対するデータの入出力

PRINT#, WRITE#, INPUT#, LINPUT#, INPUT$, DEVI$, DEVO$

●特に画面に対するデータの入出力

PRINT, WRITE, INPUT, LINPUT, INKEY$

●特にプリンタに対するデータの入出力

LPRINT

参　考

三角関数と双曲線関数

組み込み数値関数にない三角関数と双曲線関数およびそれらの逆関数を定義するための例を次に示します。使用する場合には各関数の定義域に注意する必要があります。

secant	FNA(X)=1/COS(X)
cosecant	FNB(X)=1/SIN(X)
cotangent	FNC(X)=1/TAN(X)
arcsine	FND(X)=ATN(X/SQR(1−X^2))
arccosine	FNE(X)=−ATN(X/SQR(1−X^2))+π/2
arcsecant	FNF(X)=ATN(SQR(X^2−1))+(SGN(X)−1)*π/2
arccosecant	FNG(X)=ATN(1/SQR(X^2−1))+(SGN(X)−1)*π/2
arccotangent	FNH(X)=−ATN(X)+π/2
hyperbolic sine	FNI(X)=(EXP(X)−EXP(−X))/2
hyperbolic cosine	FNJ(X)=(EXP(X)+EXP(−X))/2
hyperbolic tangent	FNK(X)=−EXP(−X)/(EXP(X)+EXP(−X))*2+1
hyperbolic secant	FNL(X)=2/(EXP(X)+EXP(−X))
hyperbolic cosecant	FNM(X)=2/(EXP(X)−EXP(−X))
hyperbolic cotangent	FNN(X)=EXP(−X)/(EXP(X)−EXP(−X))*2+1
arc-hyperbolic sine	FNO(X)=LOG(X+SQR(X^2+1))
arc-hyperbolic cosine	FNP(X)=LOG(X+SQR(X^2−1))
arc-hyperbolic tangent	FNQ(X)=LOG((1+X)/(1−X))/2
arc-hyperbolic secant	FNR(X)=LOG((SQR(1−X^2)+1)/X)
arc-hyperbolic cosecant	FNS(X)=LOG((SGN(X)*SQR(X^2+1)+1)/X)
arc-hyperbolic cotangent	FNT(X)=LOG((X+1)/(X−1))/2

シャープ株式会社

本　　社　〒５４５　大阪市阿倍野区長池町22番22号
電話　06（621）1221（大代表）
電子機器事業本部　〒329-21　栃木県矢板市早川町174番地
電話　02874（3）1131（大代表）

お客様へ……お買いあげ年月日、お買いあげ店名を記入されますと、修理などの依頼のときに便利です。

お買いあげ年月日	年　月　日
お買いあげ店名	
	電話番号
もよりの お客様ご相談窓口	
	電話番号

TMAN-3559CEZZ
4G T0446-A